
第1章 はじめに

1 .1  関数型JavaScr ipt

関数型プログラミングの⼿法は、かねてよりJavaScriptでも⽤いられてきました。

UnderscoreJSなどのライブラリは、 map や filter 、 reduce といったよく知られた関数を活
⽤して、⼩さいプログラムを組み合わせて⼤きなプログラムを作れるようにします。

    var sumOfPrimes = 

        _.chain(_.range(1000)) 

         .filter(isPrime) 

         .reduce(function(x, y) { 

             return x + y; 

         }) 

         .value(); 

NodeJSにおける⾮同期プログラミングでは、第⼀級の値としての関数をコールバックを定義す
るために多⽤しています。

    require('fs').readFile(sourceFile, function (error, data) { 

      if (!error) { 

https://leanpub.com/purescript/read
http://underscorejs.org/


        require('fs').writeFile(destFile, data, function (error) { 

          if (!error) { 

            console.log("File copied"); 

          } 

        }); 

      } 

    }); 

Reactやvirtual-domなどのライブラリは、アプリケーションの状態についての純粋な関数として
その外観をモデル化しています。

関数は単純な抽象化を可能にし、優れた⽣産性をもたらしてくれます。しかし、JavaScriptで
の関数型プログラミングには⽋点があります。JavaScriptは冗⻑で、型付けされず、強⼒な抽象
化を⽋いているのです。また、無秩序に書かれたJavaScriptコードでは、式の理解がとても困難
です。

PureScriptはこのような問題を解決すべく作られたプログラミング⾔語です。PureScriptは、
とても表現⼒豊かでありながらわかりやすく読みやすいコードを書けるようにする、軽量な構⽂
を備えています。強⼒な抽象化を提供する豊かな型システムも採⽤しています。また、
JavaScriptやJavaScriptへとコンパイルされる他の⾔語と相互運⽤するときに重要な、⾼速で理解
しやすいコードを⽣成します。PureScriptをひとことで⾔えば、純粋関数型プログラミングの理
論的な強⼒さと、JavaScriptのお⼿軽で緩いプログラミングスタイルとの、とても現実的なバラ
ンスを狙った⾔語だということを理解して頂けたらと思います。

1 .2  型と型推論

動的型付けの⾔語と静的型付けの⾔語をめぐる議論についてはよく知られています。
PureScriptは静的型付けの⾔語、つまり正しいプログラムはコンパイラによってその動作を⽰す
ような型を与えられる⾔語です。逆にいえば、型を  与えることができないプログラムは誤ったプ
ログラムであり、コンパイラによって拒否されます。動的型付けの⾔語とは異なり、PureScript
では型はコンパイル時のみに存在し、実⾏時には型の表現はありません。

PureScriptの型は、これまでJavaやC#のような他の⾔語で⾒たような型とは、いろいろな意味
で異なっていることにも注意することが⼤切です。おおまかに⾔えばPureScriptの型はJavaやC#
と同じ⽬的を持っているものの、PureScriptの型はMLとHaskellのような⾔語に影響を受けてい
ます。開発者がプログラムについての強い主張を表明できるので、PureScriptの型は表現⼒豊か
なのです。最も重要なのは、PureScriptの型システムは型推論(type inference)をサポートしてい
ることです。型推論があれば明⽰的な型注釈は必要最低限となり、型システムを厄介者ではなく

http://facebook.github.io/react/
https://github.com/Matt-Esch/virtual-dom


道具にしてくれます。簡単な例を⽰すと、次のコードは数を定義していますが、それが Number
型だという注釈はコードのどこにもありません。

iAmANumber = 

  let square x = x * x 

  in square 42.0 

次のもっと複雑な例では、コンパイラにとって未知の型が存在しているときでさえも、型注釈
なしで型の正しさを確かめることができるということが⽰されています。

iterate f 0 x = x 

iterate f n x = iterate f (n - 1) (f x) 

ここで x の型は不明ですが、 x がどんな型を持っているかにかかわらず、 iterate が型シ
ステムの規則に従っていることをコンパイラは検証することができます。

静的型はプログラムの正しさについての確信を得るためだけではなく、その正しさによって開
発を助ける、ということをあなたに納得させる(もしくは、あなたの理解を確認する)ことをこの
本では試みます。最も単純な抽象化を使わないかぎりJavaScriptでコードの⼤規模なリファクタ
リングすることは難しいですが、型検証器のある表現⼒豊かな型システムは、リファクタリング
さえ楽しく対話的な体験にしてくれます。

加えて、型システムによって提供されたこのセーフティネットは、より⾼度な抽象化をも可能
にします。実際に、関数型プログラミング⾔語Haskellによって知られるようになった、型主導
の強⼒な抽象化である『型クラス』をPureScriptは備えています。

1 .3  多⾔語Webプログラミング

関数型プログラミングはすでに多くの成功を収めています。特に成功している応⽤例をいくつ
か挙げると、データ解析、構⽂解析、コンパイラの実装、ジェネリックプログラミング、並列処
理などがあります。

PureScriptのような関数型⾔語は、アプリケーション開発の最初から最後までを 実施すること
が可能です。値や関数の型を提供することで既存のJavaScriptコードをインポートし、通常の
PureScriptコードからこれらの関数を使⽤する機能をPureScriptは提供しています。この⼿法に
ついては本書の後半で⾒ていくことになります。



しかしながら、PureScriptの強みのひとつは、JavaScriptを対象とする他  の⾔語との相互運⽤
性にあります。アプリケーションの開発の⼀部にだけPureScriptを使⽤し、JavaScriptの残りの部
分を記述するのに他の⾔語を使⽤するという⽅法もあります。

いくつかの例を⽰します。

中核となる処理はPureScriptで記述し、ユーザーインターフェイスはJavaScriptで記述する
JavaScriptや、他のJavaScriptにコンパイルされる⾔語でアプリケーションを書き、PureScriptで
そのテストを書く
既存のアプリケーションのユーザインタフェースのテストを⾃動化するためにPureScriptを使⽤
する

この本では⼩規模な課題をPureScriptで解決することに焦点を当てますが、ここで学ぶ⼿法は
⼤規模なアプリケーションに組み込むこともできます。JavaScriptからPureScriptコードを呼び出
す⽅法、およびその逆についても⾒ていきます。

1 .4  ソフトウェア要件

この本でのソフトウェア要件は最⼩限です。第１章では開発環境の構築を⼀から案内します。
これから使⽤するツールは、ほとんどの現代のオペレーティングシステムの標準リポジトリで使
⽤できるものです。

PureScriptコンパイラ⾃体は、コンパイル済みバイナリ形式でダウンロードすることもできま
すし、最新のHaskellコンパイラが稼働しているシステム上でソースからビルドすることもでき
ます。次の章ではこの⼿順を説明していきます。

本書のこのバージョンのコードは、 0.11.* バージョンのPureScriptコンパイラと互換性があ
ります。

1 .5  読者について

読者はJavaScriptの基本をすでに理解しているものと仮定します。すでにNPMやBowerのよう
なJavaScriptのエコシステムでの経験があれば、⾃⾝の好みに応じて標準設定をカスタマイズし
たい場合などに役に⽴ちますが、そのような知識は必要ではありません。



関数型プログラミングの予備知識は必要ありませんが、あっても害にはならないでしょう。実
例には新しいアイデアがつきものですから、これから使う関数型プログラミングからこうした概
念に対する直感的な理解を得ることができるはずです。

PureScriptはプログラミング⾔語Haskellに強く影響を受けているため、Haskellに通じている
読者はこの本の中で提⽰された概念や構⽂の多くに⾒覚えがあるでしょう。しかしながら、読者
はPureScriptとHaskellの間にはいくつか重要な違いがあることも理解しておかなければなりませ
ん。ここで紹介する概念の多くはHaskellでも同じように解釈できるとはいえ、どちらかの⾔語
での考え⽅を他⽅の⾔語でそのまま応⽤しようとすることは、必ずしも適切ではありません。

1 .6  本書の読み進めかた

本書の各章は、概ね章ごとに完結しています。しかしながら、多少の関数型プログラミングの
経験がある初⼼者でも、まずは各章を順番に進めていくことをおすすめします。最初の数章で
は、本書の後半の内容を理解するために必要な基礎知識を養います。関数型プログラミングの考
え⽅に⼗分通じた読者(特にMLやHaskellのよう強く型付けされた⾔語での経験を持つ読者)な
ら、本書の前半の章を読まなくても、後半の章のコードの⼤まかな理解を得ることがおそらく可
能でしょう。

各章ではそれぞれひとつの実⽤的な例に焦点をあて、新しい考え⽅を導⼊するための動機付け
として⽤います。各章のコードは本書のGitHubのリポジトリから⼊⼿できます。各章にはソー
スコードから抜粋したコード⽚が掲載されていますが、完全に理解するためには本書に掲載され
たコードと平⾏してリポジトリのソースコードを読む必要があります。対話式環境 PSCi で実⾏
し理解を確かめられるように、⻑めの節には短いコード⽚が掲載されていることがあります。

コード例は次のように等幅フォントで⽰されています。

module Example where 

 

import Control.Monad.Eff.Console (log) 

 

main = log "Hello, World!" 

先頭にドル記号がついた⾏は、コマンドラインに⼊⼒されたコマンドです。

$ pulp build 

https://github.com/paf31/purescript-book


通常、これらのコマンドはLinuxやMac OSの利⽤者ならそのまま適⽤できますが、Windows
の利⽤者はファイル区切り⽂字を変更する、シェルの組み込み機能をWindowsの相当するもの
に置き換えるなどの⼩さな変更を加える必要があるかもしれません。

pulp repl 対話式プロンプトに⼊⼒するコマンドは、⾏の先頭に⼭括弧が付けられていま
す。

> 1 + 2 

3 

各章には演習が付いており、それぞれ難易度も⽰されています。各章の内容を完全に理解する
ために、演習に取り組むことを強くお勧めします。

この本は初⼼者にPureScriptへの導⼊を提供することを⽬的としており、問題についてのお決
まりの解決策の⼀覧を提供するような種類の本ではありません。初⼼者にとってこの本を読むの
は楽しい挑戦になるはずですし、本書の内容を読み演習に挑戦すればだいたいの利益を得られる
でしょうが、なにより重要なのは、あなたが⾃分⾃⾝のコードを書いてみることです。

1 .7  困ったときには

もしどこかでつまずいたときには、PureScriptを学べるオンラインで利⽤可能な資料がたくさ
んあります。

PureScript IRCチャン  ネルはあなたが抱える問題についてチャットするのに最適な場所です。
IRCクライアントでirc.freenode.netをポイントし、#purescriptチャンネルに接続してください。
PureScriptのウェブサイトにはPureScriptの開発者によって書かれたブログ記事や、初⼼者向け
の動画、その他のリソースへのリンクがあります。
PureScriptコンパイラのドキュメントは、⾔語の主要な機能についての簡単  なコード例がありま
す。
Try PureScript!ではユーザーがWebブラウザでPureScriptコードをコンパイルすることができま
す。また、ウェブサイトにはコードの簡単な例がいくつか含まれています。
Pursuitは、PureScriptの型や関数を検索することができるデータベースです。

もしあなたが例を読んで学ぶことを好むなら、GitHubの purescript 組織、
purescript-node 組織および purescript-contrib 組織にはPureScriptコードの例がたくさん

あります。

http://purescript.org/
https://github.com/purescript/documentation
http://try.purescript.org/
http://pursuit.purescript.org/


1 .8  著者について

私はPureScriptコンパイラの最初の開発者です。私はカリフォルニア州ロサンゼルスを拠点に
しており、8ビットパーソナルコンピュータ、Amstrad CPC上のBASICでまだ幼い時にプログラ
ミングを始めました。それ以来、私はいくつものプログラミング⾔語(JavaやScala、C#、F#、
HaskellそしてPureScript)で業務に携わってきました。

プロとしての経歴が始まって間もなく、私は関数型プログラミングと数学の関係を理解するよ
うになり、そしてプログラミング⾔語Haskellとの恋に落ちました。

JavaScriptでの経験をもとに、私はPureScriptコンパイラの開発を始めることにしました。私
は⾃分がHaskellのような⾔語から取り上げた関数型プログラミングの⼿法を使っていることに
気が付きましたが、それを応⽤するためのもっと理にかなった環境を求めていました。そのとき
検討した案のなかには、Haskellをその意味論を維持しながらJavaScriptへとコンパイルするいろ
いろな試み(Fay、Haste、GHCJS)もありましたが、私が興味を持っていたのは、この問題への
別の切り⼝からのアプローチ、すなわちHaskellのような⾔語の構⽂と型システムを楽しみなが
らJavaScriptの意味論も維持するということが、どのようにすれば可能になるのかでした。

私はウェブサイトを運営しており、Twitterで連絡をとることもできます。

1 .9  謝辞

現在の状態に到達するまでPureScriptを⼿伝ってくれた多くの協⼒者に感謝したいと思いま
す。コンパイラやツール、ライブラリ、ドキュメント、テストでの組織的で弛まぬ努⼒がなかっ
たら、プロジェクトは間違いなく失敗していたことでしょう。

この本の表紙に表⽰されたPureScriptのロゴはGareth Hughesによって作成されたもので、
Creative Commons Attribution 4.0 licenseの条件の下で再利⽤させて頂いています 。

最後に、この本の内容に関する反応や訂正をくださったすべての⽅に、⼼より感謝したいと思
います。

http://blog.functorial.com/
http://twitter.com/paf31
https://creativecommons.org/licenses/by/4.0/


第2章 開発環境の準備

2 .1  この章の⽬標

この章の⽬標は、作業⽤のPureScript開発環境を準備し、最初のPureScriptプログラムを書く
ことです。

これから書く最初のコードはごく単純なPureScriptライブラリで、直⾓三⾓形の対⾓線の⻑さ
を計算する関数ひとつだけを提供します。

2 .2  導⼊

PureScript開発環境を準備するために、次のツールを使います。

purs  - PureScriptコンパイラ本体
npm  - 残りの開発ツールをインストールできるようにする、Nodeパッケージマネージャ
Pulp    - さまざまな作業をパッケージマネージャと連動して⾃動化するコマンドラインツール

この章ではこれらのツールのインストール⽅法と設定を説明します。

2 .3  PureScr iptのインストール

PureScriptコンパイラをインストールするときにお勧めなのは、PureScriptのウェブサイトか
らバイナリ配布物としてダウンロードする⽅法です。PureScriptコンパイラおよび関連する実⾏
ファイルが、パス上で利⽤できるかどうか確認をしてください。試しに、コマンドラインで
PureScriptコンパイラを実⾏してみましょう。

$ purs 

PureScriptコンパイラをインストールする他の選択肢としては、次のようなものがあります。

NPMを使⽤する。 npm install -g purescript

http://purescript.org/
http://npmjs.org/
https://github.com/purescript-contrib/pulp
http://purescript.org/


ソースコードからコンパイルを⾏う。この⽅法については、PureScriptのWebサイトが参考にな
ります。

2 .4  各ツールのインストール

もしNodeJSがインストールされていないなら、NodeJSをインストールする必要があります。
そうするとシステムに npm パッケージマネージャもインストールされるはずです。 npm がイン
ストールされ、パス上で利⽤可能であることを確認してください。

npm がインストールされたら、 pulp と bower もインストールする必要があります。プロジ
ェクトがどこで作業しているかにかかわらずこれらのコマンドラインツールが利⽤可能であるよ
うにするため、通常はグローバルにインストールしておくのがいいでしょう。

$ npm install -g pulp bower 

これで、最初のPureScriptプロジェクトを作成するために必要なすべてのツールの⽤意ができ
たことになります。

2 .5  Hel lo ,  PureScr ipt !

まずはシンプルに始めましょう。PureScriptコンパイラ pulp を直接使⽤して、基本的なHello
World! プログラムをコンパイルします。 最初に空のディレクトリ my-project を作成し、そこ
で pulp init を実⾏します。

$ mkdir my-project 

$ cd my-project 

$ pulp init 

 

* Generating project skeleton in ~/my-project 

 

$ ls 

 

bower.json src  test 

Pulpは src と test という2つのディレクトリと設定ファイル bower.json を作成してくれま
す。 src ディレクトリにはソースコードファイルを保存し、 test ディレクトリにはテストコー

http://nodejs.org/


ドファイルを保存します。 test ディレクトリはこの本の後半で使います。

src/Main.purs という名前のファイルに、以下のコードを貼り付けてください。

module Main where 

 

import Control.Monad.Eff.Console 

 

main = log "Hello, World!" 

これは⼩さなサンプルコードですが、  いくつかの重要な概念を⽰しています。

すべてのソースファイルはモジュールヘッダから始まります。モジュール名は、ドットで区切ら
れた⼤⽂字で始まる1つ以上の単語から構成されています。ここではモジュール名としてひとつ
の単語だけが使⽤されていますが、 My.First.Module というようなモジュール名も有効です。
モジュールは、モジュール名の各部分を区切るためのドットを含めた、完全な名前を使⽤してイ
ンポートされます。ここでは log 関数を提供する Control.Monad.Eff.Console モジュールを
インポートしています。
この main プログラムの定義本体は、関数適⽤の式になっています。PureScriptでは、関数適⽤
は関数名のあとに引数を空⽩で区切って書くことで表します。

それではこのコードをビルドして実⾏してみましょう。次のコマンドを実⾏します。

$ pulp run 

 

* Building project in ~/my-project 

* Build successful. 

Hello, World! 

おめでとうございます! はじめてPureScriptで作成されたプログラムのコンパイルと実⾏が
できました。

2 .6  ブラウザ向けのコンパイル

Pulpは pulp browserify を実⾏して、PureScriptコードをブラウザで使うことに適した
JavaScriptに変換することができます。

$ pulp browserify 

 

* Browserifying project in ~/my-project 



* Building project in ~/my-project 

* Build successful. 

* Browserifying... 

これに続いて、⼤量のJavaScriptコードがコンソールに表⽰されます。 これはBrowserifyの出
⼒で、Preludeと呼ばれる標準のPureScriptライブラリに加え、 src ディレクトリのコードにも
適⽤されます。このJavaScriptコードをファイルに保存し、HTML⽂書に含めることもできま
す。これを試しに実⾏してみると、ブラウザのコンソールに"Hello、World！"という⽂章が出⼒
されます。

2 .7  使⽤されていないコードを取り除く

Pulpは代替コマンド pulp build を提供しています。 -O オプションで未使⽤コードの削除を
適⽤すると、不要なJavaScriptを出⼒から取り除くことができます。

$ pulp build -O --to output.js 

 

* Building project in ~/my-project 

* Build successful. 

* Bundling Javascript... 

* Bundled. 

この場合も、⽣成されたコードはHTML⽂書で使⽤できます。 output.js を開くと、次のよ
うなコンパイルされたモジュールがいくつか表⽰されます。

(function(exports) { 

  "use strict"; 

 

  var Control_Monad_Eff_Console = PS["Control.Monad.Eff.Console"]; 

 

  var main = Control_Monad_Eff_Console.log("Hello, World!"); 

  exports["main"] = main; 

})(PS["Main"] = PS["Main"] || {}); 

ここでPureScriptコンパイラがJavaScriptコードを⽣成する⽅法の要点が⽰されています。

すべてのモジュールはオブジェクトに変換され、そのオブジェクトにはそのモジュールのエクス
ポートされたメンバが含まれています。モジュールは即時関数パターンによってスコープが限定
されたコードで初期化されています。
PureScriptは可能な限り変数の名前をそのまま使おうとします。

http://browserify.org/


PureScriptにおける関数適⽤は、そのままJavaScriptの関数適⽤に変換されます。
引数のない単純な呼び出しとしてメインメソッド呼び出しが⽣成され、すべてのモジュールが定
義された後に実⾏されます。
PureScriptコードはどんな実⾏時ライブラリにも依存しません。コンパイラによって⽣成される
すべてのコードは、あなたのコードが依存するいずれかのPureScriptモジュールをもとに出⼒さ
れているものです。

PureScriptはシンプルで理解しやすいコードを⽣成すること重視しているので、これらの点は
⼤切です。実際に、ほとんどのコード⽣成処理はごく軽い変換です。PureScriptについての理解
が⽐較的浅くても、ある⼊⼒からどのようなJavaScriptコードが⽣成されるかを予測することは
難しくありません。

2 .8  CommonJSモジュールのコンパイル

pulpは、PureScriptコードからCommonJSモジュールを⽣成するためにも使⽤できます。 これ
は、NodeJSを使⽤する場合やCommonJSモジュールを使⽤してコードを⼩さなコンポーネント
に分割する⼤きなプロジェクトを開発する場合に便利です。

CommonJSモジュールをビルドするには、（ -O オプションなしで） pulp build コマンド
を使います。

$ pulp build 

 

* Building project in ~/my-project 

* Build successful. 

⽣成されたモジュールはデフォルトで output ディレクトリに置かれます。 各PureScriptモジ
ュールは、それ⾃⾝のサブディレクトリにある独⾃のCommonJSモジュールにコンパイルされ
ます。

2 .9  Bowerによる依存関係の追跡

この章の⽬的となっている diagonal 関数を書くためには、平⽅根を計算できるようにする
必要があります。 purescript-math パッケージにはJavaScriptの Math オブジェクトのプロパテ
ィとして定義されている関数の型定義が含まれていますので、 purescript-math パッケージを



インストールしてみましょう。 npm の依存関係でやったのと同じように、次のようにコマンド
ラインに⼊⼒すると直接このパッケージをダウンロードできます。

$ bower install purescript-math --save 

--save オプションは依存関係を bower.json 設定ファイルに追加させます。

purescript-math ライブラリは、依存するライブラリと⼀緒に bower_components サブディ
レクトリにインストールされます。

2 .10 対⾓線の⻑さの計算

それでは外部ライブラリの関数を使⽤する例として diagonal 関数を書いてみましょう。

まず、 src/Main.purs ファイルの先頭に次の⾏を追加し、 Math モジュールをインポートし
ます。

import Math (sqrt) 

また、数値の加算や乗算のようなごく基本的な演算を定義する Prelude モジュールをインポ
ートすることも必要です。

import Prelude 

そして、次のように diagonal 関数を定義します。

diagonal w h = sqrt (w * w + h * h) 

この関数の型を定義する必要はないことに注意してください。 diagonal は2つの数を取り数
を返す関数である、とコンパイラは推論することができます。しかし、ドキュメントとしても役
⽴つので、通常は型注釈を提供しておくことをお勧めします。

それでは、新しい diagonal 関数を使うように main 関数も変更してみましょう。

main = logShow (diagonal 3.0 4.0) 



pulp run を使⽤して、モジュールを再コンパイルします。

$ pulp run 

 

* Building project in ~/my-project 

* Build successful. 

5.0 

2 .11 対話式処理系を使⽤したコードのテスト

PureScriptコンパイラには PSCi と呼ばれる対話式のREPL(Read-eval-print loop)が付属して
います。 PSCi はコードをテストなど思いついたことを試すのにとても便利です。それでは、
psci を使って diagonal 関数をテストしてみましょう。

pulp repl コマンドを使ってソースモジュールを⾃動的に PSCi にロードすることができま
す。

$ pulp repl 

> 

コマンドの⼀覧を⾒るには、 :? と⼊⼒します。

> :? 

The following commands are available: 

 

    :?                        Show this help menu 

    :quit                     Quit PSCi 

    :reset                    Reset 

    :browse           Browse  

    :type               Show the type of  

    :kind               Show the kind of  

    :show        import       Show imported modules 

    :show        loaded       Show loaded modules 

    :paste       paste        Enter multiple lines, terminated by ^D 

Tabキーを押すと、⾃分のコードで利⽤可能なすべての関数、及びBowerの依存関係とプレリ
ュードモジュールのリストをすべて⾒ることができるはずです。

Prelude モジュールを読み込んでください。



> import Prelude 

幾つか数式を評価してみてください。 PSCi で評価を⾏うには、1⾏以上の式を⼊⼒し、Ctrl+
Dで⼊⼒を終了します。

> 1 + 2 

3 

 

> "Hello, " <> "World!" 

"Hello, World!" 

それでは PSCi で diagonal 関数を試してみましょう。

> import Main 

> diagonal 5.0 12.0 

 

13.0 

また、 PSCi で関数を定義することもできます。

> double x = x * 2 

 

> double 10 

20 

コード例の構⽂がまだよくわからなくても⼼配はいりません。 この本を読み進めるうちにわ
かるようになっていきます。

最後に、 :type コマンドを使うと式の型を確認することができます。

> :type true 

Boolean 

 

> :type [1, 2, 3] 

Array Int 

PSCi で試してみてください。もしどこかでつまずいた場合は、メモリ内にあるコンパイル済
みのすべてのモジュールをアンロードするリセットコマンド ：reset を使⽤してみてくださ
い。



演 習

1. (簡単) Math モジュールで定義されている pi 定数を使⽤し、指定された半径の円
の⾯積を計算する関数 circleArea を書いてみましょう。また、 PSCi を使⽤して
その関数をテストしてください。 (ヒント： import math ⽂を修正して、 pi を
インポートすることを忘れないようにしましょう)

2. (やや難しい) purescript-globals パッケージを依存関係としてインストールする
には、 bower install を使います。PSCiでその機能を試してみてください。 (ヒン
ト： PSCiの ：browse コマンドを使うと、モジュールの内容を閲覧することがで
きます)

まとめ

この章では、Pulpツールを使⽤して簡単なPureScriptプロジェクトを設定しました。

また、最初のPureScript関数を書き、コンパイルし、NodeJSを使⽤して実⾏することができ
ました。

以降の章では、コードをコンパイルやデバッグ、テストするためにこの開発設定を使⽤します
ので、これらのツールや使⽤⼿順に⼗分習熟しておくとよいでしょう。



第3章 関数とレコード

3 .1  この章の⽬標

この章では、関数およびレコードというPureScriptプログラムのふたつの構成要素を導⼊しま
す。さらに、どのようにPureScriptプログラムを構造化するのか、どのように型をプログラム開
発に役⽴てるかを⾒ていきます。

連絡先のリストを管理する簡単  な住所録アプリケーションを作成していきます。このコード例
により、PureScriptの構⽂からいくつかの新しい概念を導⼊します。

このアプリケーションのフロントエンドは対話式処理系 PSCi を使うようにしていますが、
JavaScriptでフロントエンドを書くこともできるでしょう。実際に後の章で、フォームの検証と
保存および復元の機能追加について詳しく説明します。

3 .2  プロジェクトの準備

この章のソースコードは src/Data/AddressBook.purs というファイルに含まれています。こ
のファイルは次のようなモジュール宣⾔とインポート⼀覧から始まります。

module Data.AddressBook where 

 

import Prelude 

 

import Control.Plus (empty) 

import Data.List (List(..), filter, head) 

import Data.Maybe (Maybe) 

ここでは、いくつかのモジュールをインポートします。

Control.Plus モジュールには後ほど使う empty 値が定義されています。
purescript-lists パッケージで提供されている Data.List モジュールをインポートしていま

す。 purescript-lists パッケージはbowerを使⽤してインストールすることができ、連結リス
トを使うために必要ないくつかの関数が含まれています。
Data.Maybe モジュールは、値が存在したりしなかったりするような、オプショナルな値を扱う

ためのデータ型と関数を定義しています。



(訳者注・ダブルドット(..)を使⽤すると、指定された型コンストラクタのすべてのデータコンス
トラクタをインポートできます。)

このモジュールのインポート内容が括弧内で明⽰的に列挙されていることに注⽬してくださ
い。明⽰的な列挙はインポート内容の衝突を避けるのに役に⽴つので、⼀般に良い習慣です。

ソースコードリポジトリを複製したと仮定すると、この章のプロジェクトは次のコマンドを使
⽤してPulpを使⽤して構築できます。

$ cd chapter3 

$ bower update 

$ pulp build 

3 .3  単純な型

JavaScriptのプリミティブ型に対応する組み込みデータ型として、PureScriptでは数値型と⽂
字列型、真偽型の３つが定義されており、それぞれ Number 、 String 、 Boolean と呼ばれて
います。これらの型はすべてのモジュールに暗黙にインポートされる Prim モジュールで定義さ
れています。 pulp repl の :type コマンドを使⽤すると、簡単な値の型を確認できます。

$ pulp repl 

 

> :type 1.0 

Number 

 

> :type "test" 

String 

 

> :type true 

Boolean 

PureScriptには他にも、配列とレコード、関数などの組み込み型が定義されています。

整数は、⼩数点以下を省くことによって、型 Number の浮動⼩数点数の値と区別されます。

> :type 1 

Int 

⼆重引⽤符を使⽤する⽂字列リテラルとは異なり、⽂字リテラルは⼀重引⽤符で囲みます。



> :type 'a' 

Char 

配列はJavaScriptの配列に対応していますが、JavaScriptの配列とは異なり、PureScriptの配列
のすべての要素は同じ型を持つ必要があります。

> :type [1, 2, 3] 

Array Int 

 

> :type [true, false] 

Array Boolean 

 

> :type [1, false] 

Could not match type Int with Boolean. 

最後の例で起きているエラーは型検証器によって報告されたもので、配列の2つの要素の型を
単⼀化(Unification)しようとして失敗したこと⽰しています。

レコードはJavaScriptのオブジェクトに対応しており、レコードリテラルはJavaScriptのオブジ
ェクトリテラルと同じ構⽂になっています。

> author = { name: "Phil", interests: ["Functional Programming", "JavaScript"] } 

 

> :type author 

{ name :: String 

, interests :: Array String 

} 

この型が⽰しているのは、オブジェクト author は、

String 型のフィールド name
Array String つまり String の配列の型のフィールド interests

という２つのフィールド(field)を持っているということです。

レコードのフィールドは、ドットに続けて参照したいフィールドのラベルを書くと参照するこ
とができます。

> author.name 

"Phil" 

 

> author.interests 

["Functional Programming","JavaScript"] 



PureScriptの関数はJavaScriptの関数に対応しています。PureScriptの標準ライブラリは多くの
関数の例を提供しており、この章ではそれらをもう少し詳しく⾒ていきます。

> import Prelude 

> :type flip 

forall a b c. (a -> b -> c) -> b -> a -> c 

 

> :type const 

forall a b. a -> b -> a 

ファイルのトップレベルでは、等号の直前に引数を指定することで関数を定義することができ
ます。

add :: Int -> Int -> Int 

add x y = x + y 

バックスラッシュに続けて空⽩⽂字で区切られた引数名のリストを書くことで、関数をインラ
インで定義することもできます。PSCiで複数⾏の宣⾔を⼊⼒するには、 ：paste コマンドを使
⽤して"paste mode"に⼊ります。このモードでは、Control-Dキーシーケンスを使⽤して宣⾔を
終了します。

> :paste 

… add :: Int -> Int -> Int 

… add = \x y -> x + y 

… ^D 

PSCi でこの関数が定義されていると、次のように関数の隣に２つの引数を空⽩で区切って書
くことで、関数をこれらの引数に適⽤(apply)することができます。

> add 10 20 

30 

3 .4  量化された型

前の節ではPreludeで定義された関数の型をいくつか⾒てきました。たとえば flip 関数は次
のような型を持っていました。



> :type flip 

forall a b c. (a -> b -> c) -> b -> a -> c 

この forall キーワードは、 flip が全称量化された型(universally quantified type)を持って
いることを⽰しています。これは、 a や b 、 c をどの型に置き換えても、 flip はその型で
うまく動作するという意味です。

例えば、 a を Int 、 b を String 、 c を String というように選んでみたとします。この
場合、 flip の型を次のように特殊化(specialize)することができます。

(Int -> String -> String) -> String -> Int -> String 

量化された型を特殊化したいということをコードで⽰す必要はありません。特殊化は⾃動的に
⾏われます。たとえば、すでにその型の flip を持っていたかのように、次のように単に flip
を使⽤することができます。

> flip (\n s -> show n <> s) "Ten" 10 

 

"10Ten" 

a 、 b 、 c の型はどんな型でも選ぶことができるといっても、型の不整合は⽣じないよう
にしなければなりません。 flip に渡す関数の型は、他の引数の型と整合性がなくてはなりませ
ん。第２引数として⽂字列 "Ten" 、第３引数として数 10 を渡したのはそれが理由です。もし
引数が逆になっているとうまくいかないでしょう。

> flip (\n s -> show n <> s) 10 "Ten" 

 

Could not match type Int with type String 

3 .5  字下げについての注意

JavaScriptとは異なり、PureScriptのコードは字下げの⼤きさに影響されます(indentation-
sensitive)。これはHaskellと同じようになっています。コード内の空⽩の多寡は無意味ではな
く、Cのような⾔語で中括弧によってコードのまとまりを⽰しているように、PureScriptでは空
⽩がコードのまとまりを⽰すのに使われているということです。



宣⾔が複数⾏にわたる場合は、２つめの⾏は最初の⾏の字下げより深く字下げしなければなり
ません。

したがって、次は正しいPureScriptコードです。

add x y z = x + 

  y + z 

しかし、次は正しいコードではありません。

add x y z = x + 

y + z 

後者では、PureScriptコンパイラはそれぞれの⾏ごとにひとつ、つまり2つの宣⾔であると構
⽂解析します。

⼀般に、同じブロック内で定義された宣⾔は同じ深さで字下げする必要があります。例えば
PSCi でlet⽂の宣⾔は同じ深さで字下げしなければなりません。次は正しいコードです。

> :paste 

… x = 1 

… y = 2 

… ^D 

しかし、これは正しくありません。

> :paste 

… x = 1 

…  y = 2 

… ^D 

PureScriptのいくつかの予約語（例えば where や of 、 let ）は新たなコードのまとまりを
導⼊しますが、そのコードのまとまり内の宣⾔はそれより深く字下げされている必要がありま
す。

example x y z = foo + bar 

  where 

    foo = x * y 

    bar = y * z 



ここで foo や bar の宣⾔は example の宣⾔より深く字下げされていることに注意してくだ
さい。

ただし、ソースファイルの先頭、最初の module 宣⾔における予約語 where だけは、この規
則の唯⼀の例外になっています。

3 .6  独⾃の型の定義

PureScriptで新たな問題に取り組むときは、まずはこれから扱おうとする値の型の定義を書く
ことから始めるのがよいでしょう。最初に、住所録に含まれるレコードの型を定義してみます。

type Entry = { firstName :: String, lastName :: String, address :: Address } 

これは Entry という型同義語(type synonym、型シノニム)を定義しています。 型 Entry は
等号の右辺と同じ型ということです。レコードの型はいずれも⽂字列である firstName 、
lastName 、 phone という３つのフィールドからなります。前者の２つのフィールドは型
String を持ち、 address は以下のように定義された型 Address を持っています。

type Address = { street :: String, city :: String, state :: String } 

それでは、２つめの型同義語も定義してみましょう。住所録のデータ構造としては、単に項⽬
の連結リストとして格納することにします。

type AddressBook = List Entry 

List Entry は Array Entry とは同じではないということに注意してください。
Array Entry は住所録の項⽬の配列を意味しています。

3 .7  型構築⼦と種

List は型構築⼦(type constructor、型コンストラクタ)の⼀例になっています。 List そのも
のは型ではなく、何らかの型 a があるとき List a が型になっています。つまり、 List は型
引数(type argument) a をとり、新たな型 List a を構築するのです。



ちょうど関数適⽤と同じように、型構築⼦は他の型に並べることで適⽤されることに注意して
ください。型 List Entry は実は型構築⼦ List が型 Entry に適⽤されたものです。これは住
所録項⽬のリストを表しています。

(型注釈演算⼦ :: を使って)もし型 List の値を間違って定義しようとすると、今まで⾒たこ
とのないような種類のエラーが表⽰されるでしょう。

> import Data.List 

> Nil :: List 

In a type-annotated expression x :: t, the type t must have kind Type 

これは種エラー(kind error)です。値がその型で区別されるのと同じように、型はその種(kind)
によって区別され、間違った型の値が型エラーになるように、間違った種の型は種エラーを引き
起こします。

Number や String のような、値を持つすべての型の種を表す Type と呼ばれる特別な種があ
ります。

型構築⼦にも種があります。たとえば、種 Type -> Type はちょうど List のような型から型
への関数を表しています。ここでエラーが発⽣したのは、値が種 Type であるような型を持つと
期待されていたのに、 List は種 Type -> Type を持っているためです。

PSCi で型の種を調べるには、 :kind 命令を使⽤します。例えば次のようになります。

> :kind Number 

Type 

 

> import Data.List 

> :k List 

Type -> Type 

 

> :kind List String 

Type 

PureScriptの種システムは他にも⾯⽩い種に対応していますが、それらについては本書の他の
部分で⾒ていくことになるでしょう。

3 .8  住所録の項⽬の表⽰



それでは最初に、⽂字列で住所録の項⽬を表現するような関数を書いてみましょう。まずは関
数に型を与えることから始めます。型の定義は省略することも可能ですが、ドキュメントとして
も役⽴つので型を書いておくようにすると良いでしょう。型宣⾔は関数の名前とその型を :: 記
号で区切るようにして書きます。

showEntry :: Entry -> String 

showEntry は引数として Entry を取り string を返す関数であるということを、この型シグ
ネチャは⾔っています。 showEntry の定義は次のとおりです。

showEntry entry = entry.lastName <> ", " <> 

                  entry.firstName <> ": " <>  

                  showAddress entry.address 

この関数は Entry レコードの３つのフィールドを連結し、単⼀の⽂字列にします。ここで使
⽤される showAddress は address フィールドを連接し、単⼀の⽂字列にする関数です。
showAddress の定義は次のとおりです。

showAddress :: Address -> String 

showAddress addr = addr.street <> ", " <> 

                   addr.city <> ", " <> 

                   addr.state 

関数定義は関数の名前で始まり、引数名のリストが続きます。関数の結果は等号の後ろに定義
します。フィールドはドットに続けてフィールド名を書くことで参照することができます。
PureScriptでは、⽂字列連結はJavaScriptのような単⼀のプラス記号ではなく、ダイアモンド演
算⼦（ <> ）を使⽤します。

3 .9  はやめにテスト、たびたびテスト

PSCi 対話式処理系では反応を即座に得られるので、試⾏錯誤を繰り返したいときに向いてい
ます。それではこの最初の関数が正しく動作するかを PSCi を使⽤して確認してみましょう。

まず、これまで書かれたコードをビルドします。

$ pulp build 



次に、 PSCi を起動し、この新しいモジュールをインポートするために import 命令を使いま
す。

$ pulp repl 

 

> import Data.AddressBook 

レコードリテラルを使うと、住所録の項⽬を作成することができます。レコードリテラルは
JavaScriptの無名オブジェクトと同じような構⽂で名前に束縛します。

> address = { street: "123 Fake St.", city: "Faketown", state: "CA" } 

 それでは、この例に関数を適⽤してみてください。

> showAddress address 

 

"123 Fake St., Faketown, CA" 

そして、例で作成した address を含む住所録の entry レコードを作成し showEntry に適⽤
させましょう。

> entry = { firstName: "John", lastName: "Smith", address: address } 

> showEntry entry 

 

"Smith, John: 123 Fake St., Faketown, CA" 

3 .10 住所録の作成

今度は住所録の操作を⽀援する関数をいくつか書いてみましょう。空の住所録を表す値とし
て、空のリストを使います。

emptyBook :: AddressBook 

emptyBook = empty 

既存の住所録に値を挿⼊する関数も必要でしょう。この関数を insertEntry と呼ぶことにし
ます。関数の型を与えることから始めましょう。



insertEntry :: Entry -> AddressBook -> AddressBook 

insertEntry は、最初の引数として Entry 、第⼆引数として AddressBook を取り、新しい
AddressBook を返すということを、この型シグネチャは⾔っています。

既存の AddressBook を直接変更することはしません。その代わりに、同じデータが含まれて
いる新しい AddressBook を返すようにします。このように、 AddressBook は永続データ構造
(persistent data structure)の⼀例となっています。これはPureScriptにおける重要な考え⽅です。
変更はコードの副作⽤であり、コードの振る舞いについての判断するのを難しくします。そのた
め、我々は可能な限り純粋な関数や不変のデータを好むのです。

Data.List の Cons 関数を使⽤すると insertEntry を実装できます。 PSCi を起動し
:type コマンドを使って、この関数の型を⾒てみましょう。

$ pulp repl 

 

> import Data.List 

> :type Cons 

 

forall a. a -> List a -> List a 

Cons は、なんらかの型 a の値と、型 a を要素に持つリストを引数にとり、同じ型の要素を
持つ新しいリストを返すということを、この型シグネチャは⾔っています。 a を Entry 型とし
て特殊化してみましょう。

Entry -> List Entry -> List Entry 

しかし、 List Entry はまさに AddressBook ですから、次と同じになります。

Entry -> AddressBook -> AddressBook 

今回の場合、すでに適切な⼊⼒があります。 Entry と AddressBook に Cons を適⽤すると、
新しい AddressBook を得ることができます。これこそまさに私たちが求めていた関数です！

insertEntry の実装は次のようになります。

insertEntry entry book = Cons entry book 



等号の左側にある２つの引数 entry と book がスコープに導⼊されますから、これらに
Cons 関数を適⽤して結果の値を作成しています。

3 .11 カリー化された関数

PureScriptでは、関数は常にひとつの引数だけを取ります。 insertEntry 関数は２つの引数
を取るように⾒えますが、これは実際にはカリー化された関数(curried function)の⼀例となって
います。

insertEntry の型に含まれる -> は右結合の演算⼦であり、つまりこの型はコンパイラによ
って次のように解釈されます。

Entry -> (AddressBook -> AddressBook) 

すなわち、 insertEntry は関数を返す関数である、ということです！この関数は単⼀の引数
Entry を取り、それから単⼀の引数 AddressBook を取り新しい AddressBook を返す新しい関

数を返すのです。

これは例えば、最初の引数だけを与えると insertEntry を部分適⽤(partial application)でき
ることを意味します。 PSCi でこの結果の型を⾒てみましょう。

> :type insertEntry example 

 

AddressBook -> AddressBook 

期待したとおり、戻り値の型は関数になっていました。この結果の関数に、ふたつめの引数を
適⽤することもできます。

> :type (insertEntry example) emptyBook 

AddressBook 

ここで括弧は不要であることにも注意してください。次の式は同等です。

> :type insertEntry example emptyBook 

AddressBook 



これは関数適⽤が左結合であるためで、なぜ単に空⽩で区切るだけで関数に引数を与えること
ができるのかも説明にもなっています。

本書では今後、「２引数の関数」というように表現することがあることに注意してください。
これはあくまで、最初の引数を取り別の関数を返す、カリー化された関数を意味していると考え
てください。

今度は insertEntry の定義について考えてみます。

insertEntry :: Entry -> AddressBook -> AddressBook 

insertEntry entry book = Cons entry book 

もし式の右辺に明⽰的に括弧をつけるなら、 （Cons entry）book となります。
insertEntry entry はその引数が単に関数 (Cons entry) に渡されるような関数だということ

です。この2つの関数はどんな⼊⼒についても同じ結果を返しますから、つまりこれらは同じ関
数です！よって、両辺から引数 book を削除できます。

insertEntry :: Entry -> AddressBook -> AddressBook 

insertEntry entry = Cons entry 

そして、同様の理由で両辺から entry も削除することができます。

insertEntry :: Entry -> AddressBook -> AddressBook 

insertEntry = Cons 

この処理はイータ変換(eta conversion)と呼ばれ、引数を参照することなく関数を定義するポ
イントフリー形式(point-free form)へと関数を書き換えるのに使うことができます。

insertEntry の場合には、イータ変換によって「 insertEntry は単にリストに対する cons
だ」と関数の定義はとても明確になりました。しかしながら、常にポイントフリー形式のほうが
いいのかどうかには議論の余地があります。

3 .12 あなたの住所録は？

最⼩限の住所録アプリケーションの実装で必要になる最後の関数は、名前で⼈を検索し適切な
Entry を返すものです。これは⼩さな関数を組み合わせることでプログラムを構築するとい

う、関数型プログラミングで鍵となる考え⽅のよい応⽤例になるでしょう。



まずは住所録をフィルタリングし、該当する姓名を持つ項⽬だけを保持するようにするのがい
いでしょう。それから、結果のリストの先頭の(head)要素を返すだけです。

この⼤まかな仕様に従って、この関数の型を計算することができます。まず PSCi を起動し、
filter 関数と head 関数の型を⾒てみましょう。

$ pulp repl 

 

> import Data.List 

> :type filter 

 

forall a. (a -> Boolean) -> List a -> List a 

 

> :type head 

 

forall a. List a -> Maybe a 

型の意味を理解するために、これらの2つの型の⼀部を取り出してみましょう。

filter はカリー化された２引数の関数です。最初の引数は、リストの要素を取り Boolean
値を結果として返す関数です。第２引数は要素のリストで、返り値は別のリストです。

head は引数としてリストをとり、 Maybe a という今まで⾒たことがないような型を返しま
す。 Maybe a は型 a のオプショナルな値、つまり a の値を持つか持たないかのどちらかの値
を⽰しており、JavaScriptのような⾔語で値がないことを⽰すために使われる null の型安全な
代替⼿段を提供します。これについては後の章で詳しく扱います。

filter と head の全称量化された型は、PureScriptコンパイラによって次のように特殊化
(specialized)されます。

filter :: (Entry -> Boolean) -> AddressBook -> AddressBook 

 

head :: AddressBook -> Maybe Entry 

検索する関数の引数として姓と名前を渡す必要があるのもわかっています。

filter に渡す関数も必要になることもわかります。この関数を filterEntry と呼ぶことに
しましょう。 filterEntry は Entry -> Boolean という型を持っています。
filter filterEntry という関数適⽤の式は、 AddressBook -> AddressBook という型を持つ

でしょう。もしこの関数の結果を head 関数に渡すと、型 Maybe Entry の結果を得ることにな
ります。



これまでのことをまとめると、この findEntry 関数の妥当な型シグネチャは次のようになり
ます。

findEntry :: String -> String -> AddressBook -> Maybe Entry 

findEntry は、姓と名前の2つの⽂字列、および AddressBook を引数にとり、 Maybe Entry
という型の値を結果として返すということを、この型シグネチャは⾔っています。結果の
Maybe Entry という型は、名前が住所録で発⾒された場合にのみ Entry の値を持ちます。

そして、 findEntry の定義は次のようになります。

findEntry firstName lastName book = head $ filter filterEntry book 

  where 

    filterEntry :: Entry -> Boolean 

    filterEntry entry = entry.firstName == firstName && entry.lastName == lastName 

⼀歩づつこのコードの動きを調べてみましょう。

findEntry は、どちらも⽂字列型である firstName と lastName 、 AddressBook 型の
book という3つの名前をスコープに導⼊します

定義の右辺では filter 関数と head 関数が組み合わされています。まず項⽬のリストをフィ
ルタリングし、その結果に head 関数を適⽤しています。

真偽型を返す関数 filterEntry は where 節の内部で補助的な関数として定義されています。
このため、 filterEntry 関数はこの定義の内部では使⽤できますが、外部では使⽤することが
できません。また、 filterEntry はそれを包む関数の引数に依存することができ、
filterEntry は指定された Entry をフィルタリングするために引数 firstName と lastName

を使⽤しているので、 filterEntry が findEntry の内部にあることは必須になっています。

最上位での宣⾔と同じように、必ずしも filterEntry の型シグネチャを指定しなくてもよい
ことに注意してください。ただし、ドキュメントとしても役に⽴つので型シグネチャを書くこと
は推奨されています。

3 .13 中置の関数適⽤

上でみた findEntry のコードでは、少し異なる形式の関数適⽤が使⽤されています。 head
関数は中置の $ 演算⼦を使って式 filter filterEntry book に適⽤されています。



これは head (filter filterEntry book) という通常の関数適⽤と同じ意味です。

($) はPreludeで定義されている apply 関数の別名で、次のように定義されています。

apply :: forall a b. (a -> b) -> a -> b 

apply f x = f x 

 

infixr 0 apply as $ 

ここで、 apply は関数と値をとり、その値にその関数を適⽤します。 infixr キーワードは
($) を apply の別名として定義します。

しかし、なぜ通常の関数適⽤の代わりに $ を使ったのでしょうか？ その理由は $ は右結合
で優先順位の低い演算⼦だということにあります。これは、深い⼊れ⼦になった関数適⽤のため
の括弧を、 $ を使うと取り除くことができることを意味します。

たとえば、ある従業員の上司の住所がある道路を⾒つける、次の⼊れ⼦になった関数適⽤を考
えてみましょう。

street (address (boss employee)) 

これは $ を使⽤して表現すればずっと簡単になります。

street $ address $ boss employee 

3 .14 関数合成

イータ変換を使うと insertEntry 関数を簡略化できたのと同じように、引数をよく考察する
と findEntry の定義を簡略化することができます。

引数 book が関数 filter filterEntry に渡され、この適⽤の結果が head に渡されること
に注⽬してください。これは⾔いかたを変えれば、 filter filterEntry と head の合成
(composition) に book は渡されるということです。

PureScriptの関数合成演算⼦は <<< と >>> です。前者は「逆⽅向の合成」であり、後者は
「順⽅向の合成」です。



いずれかの演算⼦を使⽤して findEntry の右辺を書き換えることができます。逆順の合成を
使⽤すると、右辺は次のようになります。

(head <<< filter filterEntry) book 

この形式なら最初の定義にイータ変換の技を適⽤することができ、 findEntry は最終的に次
のような形式に到達します。

findEntry firstName lastName = head <<< filter filterEntry 

  where 

    ... 

右辺を次のようにしても同じです。

filter filterEntry >>> head 

どちらにしても、これは「 findEntry はフィルタリング関数と head 関数の合成である」と
いう findEntry 関数のわかりやすい定義を与えます。

どちらの定義のほうがわかりやすいかの判断はお任せしますが、このように関数を部品として
捉え、関数はひとつの役⽬だけをこなし、機能を関数合成で組み⽴てるというように考えると有
⽤なことがよくあります。

3 .15 テスト、テスト、テスト……

これでこのアプリケーションの中核部分が完成しましたので、 PSCi を使って試してみましょ
う。

$ pulp repl 

 

> import Data.AddressBook 

まずは空の住所録から項⽬を検索してみましょう（これは明らかに空の結果が返ってくること
が期待されます）。

> findEntry "John" "Smith" emptyBook 

 



No type class instance was found for 

 

    Data.Show.Show { firstName :: String 

                   , lastName :: String 

                   , address :: { street :: String 

                                , city :: String 

                                , state :: String 

                                } 

                   } 

エラーです！でも⼼配しないでください。これは単に 型 Entry の値を⽂字列として出⼒する
⽅法を PSCi が知らないという意味のエラーです。

findEntry の返り値の型は Maybe Entry ですが、これは⼿作業で⽂字列に変換することがで
きます。

showEntry 関数は Entry 型の引数を期待していますが、今あるのは Maybe Entry 型の値で
す。この関数は Entry 型のオプショナルな値を返すことを忘れないでください。⾏う必要があ
るのは、オプショナルな値の中に項⽬の値が存在すれば showEntry 関数を適⽤し、そうでなけ
れば存在しないという値をそのまま伝播することです。

幸いなことに、Preludeモジュールはこれを⾏う⽅法を提供しています。 map 演算⼦は
Maybe のような適切な型構築⼦まで関数を「持ち上げる」ことができます（この本の後半で関

⼿について説明するときに、この関数やそれに類似する他のものについて詳しく⾒ていきま
す）。

> import Prelude 

> map showEntry (findEntry "John" "Smith" emptyBook) 

 

Nothing 

今度はうまくいきました。この返り値 Nothing は、オプショナルな返り値に値が含まれてい
ないことを⽰しています。期待していたとおりです。

もっと使いやすくするために、 Entry を⽂字列として出⼒するような関数を定義し、毎回
showEntry を使わなくてもいいようにすることもできます。

printEntry firstName lastName book 

  = map showEntry (findEntry firstName lastName book) 

それでは空でない住所録を作成してもう⼀度試してみましょう。先ほどの項⽬の例を再利⽤し
ます。



> book1 = insertEntry entry emptyBook 

 

> printEntry "John" "Smith" book1 

 

Just ("Smith, John: 123 Fake St., Faketown, CA") 

今度は結果が正しい値を含んでいました。 book1 に別の名前で項⽬を挿⼊して、ふたつの名
前がある住所録 book2 を定義し、それぞれの項⽬を名前で検索してみてください。

演 習

1. （簡単） findEntry 関数の定義の主な部分式の型を書き下し、 findEntry 関数
についてよく理解しているか試してみましょう。たとえば、 findEntry の定義の
なかにある head 関数の型は AddressBook -> Maybe Entry と特殊化されていま
す。

2. （簡単） findEntry の既存のコードを再利⽤し、与えられた電話番号から
Entry を検索する関数を書いてみましょう。また、 PSCi で実装した関数をテス

トしてみましょう。

3. （やや難しい） 指定された名前が AddressBook に存在するかどうかを調べて真
偽値で返す関数を書いてみましょう。 (ヒント：リストが空かどうかを調べる
Data.List.null 関数の型を psci で調べてみてみましょう)

4. （難しい） 姓名が重複している項⽬を住所録から削除する関数
removeDuplicates を書いてみましょう。 (ヒント：値どうしの等価性を定義する

述語関数に基づいてリストから重複要素を削除する関数 Data.List.nubBy の型
を、 psci を使⽤して調べてみましょう)

まとめ

この章では、関数型プログラミングの新しい概念をいくつか導⼊しました。



対話的モード PSCi を使⽤して関数を調べるなど思いついたことを試す⽅法
検証や実装の道具としての型の役割
多引数関数を表現する、カリー化された関数の使⽤
関数合成で⼩さな部品を組み合わせてのプログラムの構築
where 節を利⽤したコードの構造化
Maybe 型を使⽤してnull値を回避する⽅法

イータ変換や関数合成のような⼿法を利⽤した、よりわかりやすいコードへの再構成

次の章からは、これらの考えかたに基づいて進めていきます。



第4章 再帰、マップ、畳み込み

4 .1  この章の⽬標

この章では、アルゴリズムを構造化するときに再帰関数をどのように使うかについて⾒ていき
ましょう。再帰は関数型プログラミングの基本的な⼿法であり、この本の全体に渡って使われま
す。

また、PureScriptの標準ライブラリから標準的な関数をいくつか取り扱います。 map や fold

のようなよく知られた関数だけでなく、 filter や concatMap といった珍しいけれど便利なも
のについても⾒ていきます。

この章では、仮想的なファイルシステムを操作する関数のライブラリを動機付けに⽤います。
この章で学ぶ⼿法を応⽤して、擬似的なファイルシステムによって表されるファイルのプロパテ
ィを計算する関数を記述します。

4 .2  プロジェクトの準備

この章のソースコードには、 src/Data/Path.purs と src/FileOperations.purs という２つ
のファイルが含まれています。

Data.Path モジュールには、仮想ファイルシステムが含まれています。このモジュールの内
容を変更する必要はありません。

FileOperations モジュールには、 Data.Path APIを使⽤する関数が含まれています。演習へ
の回答はこのファイルだけで完了することができます。

このプロジェクトには以下のBower依存関係があります。

purescript-maybe : Maybe 型構築⼦が定義されています
purescript-arrays : 配列を扱うための関数が定義されています
purescript-strings : JavaScriptの⽂字列を扱うための関数が定義されています
purescript-foldable-traversable : 配列の畳み込みやその他のデータ構造に関する関数が定

義されています
purescript-console : コンソールへの出⼒を扱うための関数が定義されています



4 .3  はじめに

再帰は⼀般のプログラミングでも重要な⼿法ですが、特に純粋関数型プログラミングでは当た
り前のように⽤いられます。この章で⾒ていくように、再帰はプログラムの変更可能な状態を減
らすために役⽴つからです。

再帰は分割統治(divide and conquer)戦略と密接な関係があります。分割統治とはすなわち、
いろいろな⼊⼒に対する問題を解決するために、⼊⼒を⼩さな部分に分割し、それぞれの部分に
ついて問題を解いて、部分ごとの答えから最終的な答えを組み⽴てるということです。

それでは、PureScriptにおける再帰の簡単な例をいくつか⾒てみましょう。

次は階乗関数(factorial function)のよくある例です。

fact :: Int -> Int 

fact 0 = 1 

fact n = n * fact (n - 1) 

部分問題へ問題を分割することによって階乗関数がどのように計算されるかがわかります。よ
り⼩さい数へと階乗を計算していくということです。ゼロに到達すると、答えは直ちに求まりま
す。

次は、フィボナッチ関数(Fibonacci function)を計算するという、これまたよくある例です。

fib :: Int -> Int 

fib 0 = 1 

fib 1 = 1 

fib n = fib (n - 1) + fib (n - 2) 

やはり、部分問題の解決策を考えることで全体を解決していることがわかります。このと
き、 fib (n - 1) と fib (n - 2) という式に対応した、２つの部分問題があります。これらの
２つの部分問題が解決されていれば、この部分的な答えを加算することで、全体の答えを組み⽴
てることができます。

4 .4  配列上での再帰

再帰関数の定義は、 Int 型だけに限定されるものではありません！本書の後半でパターン照
合(pattern matching)を扱うときに、いろいろなデータ型の上での再帰関数について⾒ていきま



すが、ここでは数と配列に限っておきます。

⼊⼒がゼロでないかどうかについて分岐するのと同じように、配列の場合も、配列が空でない
かどうかについて分岐していきます。再帰を使⽤して配列の⻑さを計算する次の関数を考えてみ
ます。

import Prelude 

 

import Data.Array (null) 

import Data.Array.Partial (tail) 

import Partial.Unsafe (unsafePartial) 

 

length :: forall a. Array a -> Int 

length arr = 

  if null arr 

    then 0 

    else 1 + length (unsafePartial tail arr) 

この関数では配列が空かどうかで分岐するために if ... then ... else 式を使っています。
この null 関数は配列が空のときに true を返します。空の配列の⻑さはゼロであり、空でない
配列の⻑さは配列の先頭を取り除いた残りの部分の⻑さより１⼤きいというわけです。

JavaScriptで配列の⻑さを調べるのには、この例はどうみても実⽤的な⽅法とはいえません
が、次の演習を完了するための⼿がかりとしては充分でしょう。

演 習

1. (簡単) ⼊⼒が偶数であるとき、かつそのときに限り true に返すような再帰関数
を書いてみましょう。

2. (少し難しい) 配列内の偶数の数を数える再帰関数を書いてみましょう。ヒン
ト： Data.Array.Partial モジュールの unsafePartial head 関数を使うと、空で
ない配列の最初の要素を⾒つけることができます。



4 .5  マップ

map 関数は配列に対する再帰関数のひとつです。この関数を使うと、配列の各要素に順番に
関数を適⽤することで、配列の要素を変換することができます。そのため、配列の内容は変更さ
れますが、その形状(ここでは「⻑さ」)は保存されます。

本書で後ほど型クラス(type class)を扱うとき、形状を保存しながら型構築⼦のクラスを変換
する関⼿(functor)と呼ばれる関数を紹介しますが、その時に map 関数は関⼿の具体例であるこ
とがわかるでしょう。

それでは、 PSCi で map 関数を試してみましょう。

$ pulp repl 

 

import Prelude 

map (\n -> n + 1) [1, 2, 3, 4, 5] 

 

[2, 3, 4, 5, 6] 

map がどのように使われているかに注⽬してください。最初の引数には配列がどのように対
応付けられるかを⽰す関数、第２引数には配列そのものを渡します。

4 .6  中置演算⼦

バッククォート(`)で関数名を囲むと、対応関係を表す関数と配列のあいだに、 map 関数を書
くことができます。

(\n -> n + 1) `map` [1, 2, 3, 4, 5] 

 

[2, 3, 4, 5, 6] 

この構⽂は中置関数適⽤と呼ばれ、どんな関数でもこのように中置することができます。普通
は2引数の関数に対して使うのが最も適切でしょう。

配列を扱うときは、 map 関数と等価な <$> という演算⼦が存在します。この演算⼦は他の⼆
項演算⼦と同じように中置で使⽤することができます。

(\n -> n + 1) <$> [1, 2, 3, 4, 5] 

 



[2, 3, 4, 5, 6] 

それでは map の型を⾒てみましょう。

:type map 

forall a b f. Functor f => (a -> b) -> f a -> f b 

実は map の型は、この章で必要とされているものよりも⼀般的な型になっています。今回の
⽬的では、 map は次のようなもっと具体的な型であるかのように考えるとよいでしょう。

forall a b. (a -> b) -> Array a -> Array b 

この型では、 map 関数に適⽤するときには a と b という２つの型を⾃由に選ぶことができ
る、ということも⽰されています。 a は元の配列の要素の型で、 b は⽬的の配列の要素の型で
す。もっと⾔えば、 map が配列要素の型を変化させても構わないということです。たとえ
ば、 map を使⽤すると数値を⽂字列に変換することができます。

show <$> [1, 2, 3, 4, 5] 

 

["1","2","3","4","5"] 

中置演算⼦ <$> は特別な構⽂のように⾒えるかもしれませんが、実はPureScriptの普通の関数
です。中置構⽂を使⽤した単なる適⽤にすぎません。実際、括弧でその名前を囲むと、この関数
を通常の関数のように使⽤することができます。これは、 map 代わりに、括弧で囲まれ
た (<$>) という名前を使って配列に関数を適⽤できるということです。

(<$>) show [1, 2, 3, 4, 5] 

["1","2","3","4","5"] 

新しい中置演算⼦を定義するには、関数と同じ記法を使います。演算⼦名を括弧で囲み、あと
は普通の関数のようにその中置演算⼦を定義します。たとえば、 Data.Array モジュールでは次
のように range 関数と同じ振る舞いの中置演算⼦ (..) を定義しています。

infix 8 range as .. 

この演算⼦は次のように使うことができます。



import Data.Array 

1 .. 5 

[1, 2, 3, 4, 5] 

 

show <$> (1 .. 5) 

["1","2","3","4","5"] 

注意： 独⾃の中置演算⼦は、⾃然な構⽂を持った領域特化⾔語を定義するのに優れた⼿段に
なりえます。ただし、使⽤には充分注意してください。初⼼者が読めないコードになることがあ
りますから、新たな演算⼦の定義には慎重になるのが賢明です。

上記の例では、 1 .. 5 という式は括弧で囲まれていましたが、実際にはこれは必要ありませ
ん。なぜなら、 Data.Array モジュールは、 <$> に割り当てられた優先順位より⾼い優先順位
を .. 演算⼦に割り当てているからです。上の例では、 .. の優先順位は、予約語 infix のあと
に書かれた数の 8  と定義されていました。ここでは <$> の優先順位よりも⾼い優先順位を ..

に割り当てており、このため括弧を付け加える必要がないということです。

show <$> 1 .. 5 

   

["1","2","3","4","5"] 

中置演算⼦に左結合性または右結合性を与えたい場合は、代わりに予約語 infixl と infixr

を使います。

4 .7  配列のフィルタリング

Data.Array モジュールでは他にも、 map と同様によく使われる関数 filter も提供していま
す。この関数は、述語関数に適合する要素のみを残し、既存の配列から新しい配列を作成する機
能を提供します。

たとえば、1から10までの数で、偶数であるような数の配列を計算したいとします。これは次
のように⾏うことができます。

import Data.Array 

 

filter (\n -> n `mod` 2 == 0) (1 .. 10) 

[2,4,6,8,10] 



演 習

1. (簡単) map 関数や <$> 関数を使⽤して、 配列に格納された数のそれぞれの平⽅
を計算する関数を書いてみましょう。

2. (簡単) filter 関数を使⽤して、数の配列から負の数を取り除く関数を書いてみ
ましょう。

3. (やや難しい) filter 関数と同じ意味の中置演算⼦ <$?> を定義してみましょ
う。先ほどの演習の回答を、この新しい演算⼦を使⽤して書き換えてください。ま
た、 PSCi でこの演算⼦の優先順位と結合性を試してみてください。

4 .8  配列の平坦化

配列に関する標準的な関数として Data.Array で定義されているものには、 concat 関数もあ
ります。 concat は配列の配列をひとつの配列へと平坦化します。

import Data.Array 

:type concat  

forall a. Array (Array a) -> Array a 

 

concat [[1, 2, 3], [4, 5], [6]] 

[1, 2, 3, 4, 5, 6] 

関連する関数として、 concat と map を組み合わせたような concatMap と呼ばれる関数もあ
ります。 map は(相異なる型も可能な)値からの値への関数を引数に取りますが、それに対し
て concatMap は値から値の配列の関数を取ります。

実際に動かして⾒てみましょう。

import Data.Array 

 

:type concatMap 

forall a b. (a -> Array b) -> Array a -> Array b 

 

concatMap (\n -> [n, n * n]) (1 .. 5) 

[1,1,2,4,3,9,4,16,5,25] 



ここでは、数をその数とその数の平⽅の２つの要素からなる配列に写す関
数 \n -> [n, n * n] を引数に concatMap を呼び出しています。結果は、1から5の数と、その
それぞれの数の平⽅からなる、10個の数になります。

concatMap がどのように結果を連結しているのかに注⽬してください。渡された関数を元の
配列のそれぞれの要素について⼀度づつ呼び出し、その関数はそれぞれ配列を⽣成します。最後
にそれらの配列を単⼀の配列に押し潰し、それが結果となります。

map と filter 、 concatMap は、「配列内包表記」(array comprehensions)と呼ばれる、配列
に関するあらゆる関数の基盤を形成しています。

4 .9  配列内包表記

数 n のふたつの因数を⾒つけたいとしましょう。これを⾏うための簡単  な⽅法としては、総
当りで調べる⽅法があります。つまり、 1 から n の数のすべての組み合わせを⽣成し、それを
乗算してみるわけです。もしその積が n なら、 n の因数の組み合わせを⾒つけたということに
なります。

配列内包表記を使⽤すると、この計算を実⾏することができます。 PSCi を対話式の開発環境
として使⽤し、ひとつづつこの⼿順を進めていきましょう。

n 以下の数の組み合わせの配列を⽣成する最初の⼿順は、 concatMap を使えば⾏うことがで
きます。

1 .. n のそれぞれの数を配列 1 .. n へとマッピングすることから始めましょう。

pairs n = concatMap (\i -> 1 .. n) (1 .. n) 

この関数をテストしてみましょう。

pairs 3 

[1,2,3,1,2,3,1,2,3] 

これは求めているものとはぜんぜん違います。単にそれぞれの組み合わせの２つ⽬の要素を返
すのではなく、ペア全体を保持することができるように、内側の 1 .. n の複製について関数を
マッピングする必要があります。



:paste 

… pairs' n = 

…   concatMap (\i -> 

…     map (\j -> [i, j]) (1 .. n) 

…   ) (1 .. n) 

… ^D 

 

pairs' 3 

[[1,1],[1,2],[1,3],[2,1],[2,2],[2,3],[3,1],[3,2],[3,3]] 

いい感じになってきました。しかし、 [1, 2] と [2, 1] の両⽅があるように、重複した組み
合わせが⽣成されています。 j を i から n の範囲に限定することで、２つ⽬の場合を取り除く
ことができます。

:paste 

… pairs'' n = 

…   concatMap (\i -> 

…     map (\j -> [i, j]) (i .. n) 

…   ) (1 .. n) 

… ^D 

 

pairs'' 3 

[[1,1],[1,2],[1,3],[2,2],[2,3],[3,3]] 

すばらしいです！因数の候補のすべての組み合わせを⼿に⼊れたので、 filter を使えば、そ
の積が n であるような組み合わせを選び出すことができます。

import Data.Foldable 

 

factors n = filter (\pair -> product pair == n) (pairs'' n) 

 

factors 10 

[[1,10],[2,5]] 

このコードでは、 purescript-foldable-traversable ライブラリの Data.Foldable モジュー
ルにある product 関数を使っています。

うまくいきました！重複のなく、因数の組み合わせの正しい集合を⾒つけることができまし
た。

4 .10 do記法



機能は実現できましたが、このコードの可読性は⼤幅に向上することができます。 map

や concatMap は基本的な関数であり、do記法(do notation)と呼ばれる特別な構⽂の基礎になっ
ています(もっと厳密にいえば、それらの⼀般化である map と bind が基礎をなしています)。

注意： map と concatMap が配列内包表記を書けるようにしているように、もっと⼀般的な演
算⼦である map と bind はモナド内包表記(monad comprehensions)と呼ばれているものを書け
るようにします。本書の後半ではモナド(monad)の例をたっぷり⾒ていくことになります。

do記法を使うと、先ほどの factors 関数を次のように書き直すことができます。

factors :: Int -> Array (Array Int) 

factors n = filter (\xs -> product xs == n) $ do 

  i <- 1 .. n 

  j <- i .. n 

  pure [i, j] 

予約語 do はdo記法を使うコードのブロックを導⼊します。このブロックは幾つかの型の式で
構成されています。

配列の要素を名前に束縛する式。これは後ろ向きの⽮印 <- で ⽰されていて、その左側は名前、
右側は配列の型を持つ式です。
名前に配列の要素を束縛しない式。最後の⾏の pure [i, j] が、この種類の式の⼀例です。
- let キーワードを使⽤し、式に名前を与える式(ここでは使われていません)。

この新しい記法を使うと、アルゴリズムの構造がわかりやすくなることがあります。⼼のなか
で <- を「選ぶ」という単語に置き換えるとすると、「1からnの間の要素 i を選び、それからi
からnの間の要素 j を選び、 [i, j] を返す」というように読むことができるかもしれません。

最後の⾏では、 pure 関数を使っています。この関数は PSCi で評価することができますが、
型を明⽰する必要があります。

pure [1, 2] :: Array (Array Int) 

[[1, 2]] 

配列の場合、 pure は単に１要素の配列を作成します。実際に、 factors 関数を変更し
て、 pure の代わりにこの形式を使うようにすることもできます。

factors :: Int -> Array (Array Int) 

factors n = filter (\xs -> product xs == n) $ do 

  i <- 1 .. n 



  j <- i .. n 

  [[i, j]] 

そして、結果は同じになります。

4 .11 ガード

factors 関数を更に改良する⽅法としては、このフィルタを配列内包表記の内側に移動する
というものがあります。これは purescript-control ライブラリにある Control.MonadZero モ
ジュールの guard 関数を使⽤することで可能になります。

import Control.MonadZero (guard) 

 

factors :: Int -> Array (Array Int) 

factors n = do 

  i <- 1 .. n 

  j <- i .. n 

  guard $ i * j == n 

  pure [i, j] 

pure と同じように、 guard 関数がどのように動作するかを理解するために、 PSCi で guard

関数を適⽤して調べてみましょう。 guard 関数の型は、ここで必要とされるものよりもっと⼀
般的な型になっています。

import Control.MonadZero 

 

:type guard 

forall m. MonadZero m => Boolean -> m Unit 

今回の場合は、 PSCi は次の型を報告するものと考えてください。

Boolean -> Array Unit 

次の計算の結果から、配列における guard 関数について今知りたいことはすべてわかりま
す。

import Data.Array 

 

length $ guard true 



1 

 

length $ guard false 

0 

つまり、 guard が true に評価される式を渡された場合、単⼀の要素を持つ配列を返すので
す。もし式が false と評価された場合は、その結果は空です。

ガードが失敗した場合、配列内包表記の現在の分岐は、結果なしで早めに終了されることを意
味します。これは、 guard の呼び出しが、途中の配列に対して filter を使⽤するのと同じだと
いうことです。これらが同じ結果になることを確認するために、 factors の⼆つの定義を試し
てみてください。

演 習

1. (簡単) factors 関数を使⽤して、整数の引数が素数であるかどうかを調べる関
数 isPrime を定義してみましょう。

2. (やや難しい) 2つの配列の直積集合を⾒つけるための関数を書いてみましょ
う。直積集合とは、要素 a 、 b のすべての組み合わせの集合のことです。ここ
で a は最初の配列の要素、 b は２つ⽬の配列の要素です。

3. (やや難しい) ピタゴラスの三つ組数とは、 a² + b² = c² を満たすような３つの
数の配列 [a, b, c] のことです。配列内包表記の中で guard 関数を使⽤して、
数 n を引数に取り、どの要素も n より⼩さいようなピタゴラスの三つ組数すべて
を求める関数を書いてみましょう。その関数は Int -> Array (Array Int) という
型を持っていなければなりません。

4. (難しい) factors 関数を使⽤して、数 n のすべての因数分解を求める関
数 factorizations を定義してみましょう。数 n の因数分解とは、それらの積
が n であるような整数の配列のことです。ヒント：１は因数ではないと考えてくだ
さい。また、無限再帰に陥らないように注意しましょう。



4 .12 畳み込み

再帰を利⽤して実装される興味深い関数としては、配列に対する左畳み込み(left fold)と右畳
み込み(right fold)があります。

PSCi を使って、 Data.Foldable モジュールをインポートし、 foldl と foldr 関数の型を調
べることから始めましょう。

import Data.Foldable 

 

:type foldl 

forall a b f. Foldable f => (b -> a -> b) -> b -> f a -> b 

 

:type foldr 

forall a b f. Foldable f => (a -> b -> b) -> b -> f a -> b 

これらの型は、現在興味があるものよりも⼀般的です。この章の⽬的では、 PSCi は以下の
(より具体的な)答えを与えていたと考えておきましょう。

:type foldl 

forall a b. (b -> a -> b) -> b -> Array a -> b 

 

:type foldr 

forall a b. (a -> b -> b) -> b -> Array a -> b 

どちらの型でも、 a は配列の要素の型に対応しています。型 b は、配列を⾛査(traverse)した
ときの結果を累積する「累積器」(accumulator)の型だと考えることができます。

foldl 関数と foldr 関数の違いは⾛査の⽅向です。 foldr が「右から」配列を畳み込むのに
対して、 foldl は「左から」配列を畳み込みます。

これらの関数の動きを⾒てみましょう。 foldl を使⽤して数の配列の和を求めてみます。
型 a は Number になり、結果の型 b も Number として選択することができます。ここでは、次の
要素を累積器に加算する Number -> Number -> Number という型の関数、 Number 型の累積器の
初期値、和を求めたい Number の配列という、3つの引数を提供する必要があります。最初の引
数としては、加算演算⼦を使⽤することができますし、累積器の初期値はゼロになります。

foldl (+) 0 (1 .. 5) 

15 



この場合では、引数が逆になっていても (+) 関数は同じ結果を返すので、 foldl と foldr の
どちらでも問題ありません。

foldr (+) 0 (1 .. 5) 

15 

foldl と foldr の違いを説明するために、畳み込み関数の選択が影響する例も書いてみまし
ょう。加算関数の代わりに、⽂字列連結を使⽤して⽂字列を作ってみます。

foldl (\acc n -> acc <> show n) "" [1,2,3,4,5] 

"12345" 

 

foldr (\n acc -> acc <> show n) "" [1,2,3,4,5] 

"54321" 

これは、2つの関数の  違いを⽰しています。左畳み込み式は、以下の関数適⽤と同等です。

((((("" <> show 1) <> show 2) <> show 3) <> show 4) <> show 5) 

それに対し、右畳み込みは以下に相当します。

((((("" <> show 5) <> show 4) <> show 3) <> show 2) <> show 1) 

4 .13 末尾再帰

再帰はアルゴリズムを定義するための強⼒な⼿法ですが、問題も抱えています。JavaScriptで
再帰関数を評価するとき、⼊⼒が⼤きすぎるとスタックオーバーフローでエラーを起こす可能性
があるのです。

PSCi で次のコードを⼊⼒すると、この問題を簡単に検証できます。

f 0 = 0 

f n = 1 + f (n - 1) 

 

f 10 

10 

   



f 10000 

RangeError: Maximum call stack size exceeded 

これは問題です。関数型プログラミングの基本的な⼿法として再帰を採⽤しようとするなら、
無限かもしれない再帰でも扱える⽅法が必要です。

PureScriptは末尾再帰最適化(tail recursion optimization)の形でこの問題に対する部分的な解
決策を提供しています。

注意：この問題へのより完全な解決策としては、いわゆるトランポリン(trampolining)を使⽤
したライブラリで実装する⽅法がありますが、それはこの章で扱う範囲を超えています。この内
容に興味のある読者は purescript-free や purescript-tailrec パッケージのドキュメントを
参照してみてください。

末尾再帰の最適化が可能かどうかには条件があります。末尾位置(tail position)にある関数の再
帰的な呼び出しは、スタックフレームが確保されないジャンプに置き換えることができます。呼
び出しは、関数が戻るより前の最後の呼び出しであるとき、末尾位置にあるといいます。なぜこ
の例でスタックオーバーフローを観察したのかはこれが理由です。この f の再帰呼び出しは、
末尾位置ではないからです。

実際には、PureScriptコンパイラは再帰呼び出しをジャンプに置き換えるのではなく、再帰的
な関数全体をwhileループに置き換えます。

以下はすべての再帰呼び出しが末尾位置にある再帰関数の例です。

fact :: Int -> Int -> Int 

fact 0 acc = acc 

fact n acc = fact (n - 1) (acc * n) 

fact への再帰呼び出しは、この関数の中で起こる最後のものである、つまり末尾位置にある
ことに注意してください。

4 .14 累積器

末尾再帰ではない関数を末尾再帰関数に変える⼀般的な⽅法としては、累積器引数
(accumulator parameter)を使⽤する⽅法があります。結果を累積するために返り値を使うと末尾
再帰を妨げることがありますが、それとは対照的に累積器引数は返り値を累積する関数へ追加さ
れる付加的な引数です。



たとえば、⼊⼒配列を逆順にする、この配列の再帰を考えてみましょう。

reverse :: forall a. Array a -> Array a 

reverse [] = [] 

reverse xs = snoc (reverse (unsafePartial tail xs)) 

                  (unsafePartial head xs) 

この実装は末尾再帰ではないので、⼤きな⼊⼒配列に対して実⾏されると、⽣成された
JavaScriptはスタックオーバーフローを発⽣させるでしょう。しかし、代わりに、結果を蓄積す
るための２つ⽬の引数を関数に導⼊することで、これを末尾再帰に変えることができます。

reverse :: forall a. Array a -> Array a 

reverse = reverse' [] 

  where 

    reverse' acc [] = acc 

    reverse' acc xs = reverse' (unsafePartial head xs : acc) 

                               (unsafePartial tail xs) 

ここでは、配列を逆転させる作業を補助関数 reverse' に委譲しています。関数`reverse'が末
尾再帰的であることに注⽬してください。 その唯⼀の再帰呼び出しは、最後の場合の末尾位置
にあります。これは、⽣成されたコードがwhileループとなり、⼤きな⼊⼒でもスタックが溢れ
ないことを意味します。

reverse のふたつめの実装を理解するためには、部分的に構築された結果を状態として扱う
ために、補助関数 reverse' で累積器引数の使⽤することが必須であることに注意してくださ
い。結果は空の配列で始まりますが、⼊⼒配列の要素ひとつごとに、ひとつづつ⼤きくなってい
きます。後の要素は配列の先頭に追加されるので、結果は元の配列の逆になります！

累積器を「状態」と考えることもできますが、直接に変更がされているわけではないことにも
注意してください。この累積器は不変の配列であり、計算に沿って状態受け渡すために、単に関
数の引数を使います。

4 .15 明⽰的な再帰より畳み込みを選ぶ

末尾再帰を使⽤して再帰関数を記述することができれば末尾再帰最適化の恩恵を受けることが
できるので、すべての関数をこの形で書こうとする誘惑にかられます。しかし、多くの関数は配
列やそれに似たデータ構造に対する折り畳みとして直接書くことができることを忘れがちで
す。 map や fold のようなコンビネータを使って直接アルゴリズムを書くことには、コードの単



純さという利点があります。これらのコンビネータはよく知られており、アルゴリズムの意図を
はっきりとさせるのです。

例えば、 先ほどの reverse の例は、畳み込みとして少なくとも2つの⽅法で書くことができ
ます。 foldr を使⽤すると次のようになります。

import Data.Foldable 

:paste 

… reverse :: forall a. Array a -> Array a 

… reverse = foldr (\x xs -> xs <> [x]) [] 

… ^D 

reverse [1, 2, 3] 

 

[3,2,1] 

foldl を使って reverse を書くことは、読者への課題として残しておきます。

演 習

1. (簡単) foldl を使って、真偽値の配列の要素すべてが真かどうか調べてみてくだ
さい。

2. (やや難しい) 関数 foldl (==) false xs が真を返すような配列 xs とはどのよ
うなものか説明してください。

3. (やや難しい) 累積器引数を使⽤して、次の関数を末尾再帰形に書きなおしてくだ
さい。

import Prelude 

import Data.Array.Partial (head, tail) 

count :: forall a. (a -> Boolean) -> Array a -> Int 

count _ [] = 0 

count p xs = if p (unsafePartial head xs) 

               then count p (unsafePartial tail xs) + 1 

               else count p (unsafePartial tail xs) 

4. (やや難しい) foldl を使って reverse を書いてみましょう。



4 .16 仮想ファイルシステム

この節では、これまで学んだことを応⽤して、模擬的なファイルシステムで動作する関数を書
いていきます。事前に定義されたAPIで動作するように、マップ、畳み込み、およびフィルタを
使⽤します。

Data.Path モジュールでは、次のように仮想ファイルシステムのAPIが定義されています。

ファイルシステム内のパスを表す型 Path があります。
ルートディレクトリを表すパス root があります。
ls 関数はディレクトリ内のファイルを列挙します。
filename 関数は Path のファイル名を返します。
size 関数は Path が⽰すファイルの⼤きさを返します。
isDirectory 関数はファイルかディレクトリかを調べます。

型については、型定義は次のようになっています。

root :: Path 

 

ls :: Path -> Array Path 

 

filename :: Path -> String 

 

size :: Path -> Maybe Number 

 

isDirectory :: Path -> Boolean 

PSCi でこのAPIを試してみましょう。

$ pulp repl 

import Data.Path 

 

root 

/ 

 

isDirectory root 

true 

 

ls root 

[/bin/,/etc/,/home/] 

FileOperations モジュールでは、 Data.Path APIを操作するための関数を定義されていま
す。 Data.Path モジュールを変更したり実装を理解したりする必要はありません。すべ



て FileOperations モジュールだけで作業を⾏います。

4 .17 すべてのファイルの⼀覧

それでは、内側のディレクトリまで、すべてのファイルを列挙する関数を書いてみましょう。
この関数は以下のような型を持つでしょう。

allFiles :: Path -> Array Path 

再帰を使うとこの関数を定義することができます。まずは ls を使⽤してディレクトリの直接
の⼦を列挙します。それぞれの⼦について再帰的に allFiles を適⽤すると、それぞれパスの配
列が返ってくるでしょう。 concatMap を適⽤すると、この結果を同時に平坦化することができ
ます。

最後に、 : 演算⼦を使って現在のファイルも含めます。

allFiles file = file : concatMap allFiles (ls file) 

注意：cons演算⼦ ： は、実際には不変な配列に対してパフォーマンスが悪いので、⼀般的に
は推奨されません。 リンクリストやシーケンスなどの他のデータ構造を使⽤すると、パフォー
マンスを向上させることができます。

それでは PSCi でこの関数を試してみましょう。

import FileOperations 

import Data.Path 

 

allFiles root 

   

[/,/bin/,/bin/cp,/bin/ls,/bin/mv,/etc/,/etc/hosts, ...] 

すばらしい！do記法で配列内包表記を使ってもこの関数を書くことができるので⾒ていきま
しょう。

逆向きの⽮印は配列から要素を選択するのに相当することを思い出してください。最初の⼿順
は、引数の直接の⼦から要素を選択することです。それから、単にそのファイルに対してこの再
帰関数を呼びします。do記法を使⽤しているので、再帰的な結果をすべて連結する concatMap

が暗黙に呼び出されています。



新しいコードは次のようになります。

allFiles' :: Path -> Array Path 

allFiles' file = file : do 

  child <- ls file 

  allFiles' child 

PSCi で新しいコードを試してみてください。同じ結果が返ってくるはずです。どちらのほう
がわかりやすいかの選択はお任せします。

演 習

1. (簡単) ディレクトリのすべてのサブディレクトリの中まで、ディレクトリを除く
すべてのファイルを返すような関数 onlyFiles を書いてみてください。

2. (やや難しい) このファイルシステムで最⼤と最⼩のファイルを決定するような畳
み込みを書いてください。

3. (難しい) ファイルを名前で検索する関数 whereIs を書いてください。この関数
は型 Maybe Path の値を返すものとします。この値が存在するなら、そのファイル
がそのディレクトリに含まれているということを表します。この関数は次のように
振る舞う必要があります。

> whereIs "/bin/ls" 

Just (/bin/) 

 

> whereIs "/bin/cat" 

Nothing 

ヒント：do記法で配列内包表記を使⽤して、この関数を記述してみてください。

まとめ



この章では、アルゴリズムを簡潔に表現する⼿段として、PureScriptでの再帰の基本を説明し
ました。また、独⾃の中置演算⼦や、マップ、フィルタリングや畳み込みなどの配列に対する標
準関数、およびこれらの概念を組み合わせた配列内包表記を導⼊しました。最後に、スタックオ
ーバーフローエラーを回避するために末尾再帰を使⽤することの重要性、累積器引数を使⽤して
末尾再帰形に関数を変換する⽅法を⽰しました。



第5章 パターン照合

5 .1  この章の⽬標

この章では、代数的データ型とパターン照合という、ふたつの新しい概念を導⼊します。ま
た、⾏多相というPureScriptの型システムの興味深い機能についても簡単に取り扱います。

パターン照合(Pattern matching)は関数  型プログラミングにおける⼀般的な⼿法で、複数の場
合に実装を分解することにより、開発者は潜在的に複雑な動作の関数を簡潔に書くことができま
す。

代数的データ型はPureScriptの型システムの機能で、パターン照合とも密接に関連していま
す。

この章の⽬的は、代数的データ型やパターン照合を使⽤して、単純なベクターグラフィックス
を描画し操作するためのライブラリを書くことです。

5 .2  プロジェクトの準備

この章のソースコードはファイル src/Data/Picture.purs で定義されています。

このプロジェクトでは、これまで⾒てきたBowerパッケージを引き続き使⽤しますが、それに
加えて次の新しい依存関係が追加されます。

purescript-globals : ⼀般的なJavaScriptの値や関数の取り扱いを可能にします。
purescript-math : JavaScriptの Math オブジェクトの関数群を利⽤可能にします。

Data.Picture モジュールは、簡単な図形を表すデータ型 Shape や、図形の集合である型
Picture 、及びこれらの型を扱うための関数を定義しています。

このモジュールでは、データ構造の畳込みを⾏う関数を提供する Data.Foldable モジュール
もインポートします。

module Data.Picture where 

 

import Prelude 

import Data.Foldable (foldl) 



Data.Picture モジュールでは、 Global と Math モジュールもインポートするため as キー
ワードを使⽤します。

import Global as Global 

import Math as Math 

これは型や関数をモジュール内で使⽤できるようにしますが、 Global.infinity

や Math.max といった修飾名でのみ使⽤にできるようにします。これは重複したインポートをさ
け、使⽤するモジュールを明確にするのに有効な⽅法です。

注意：同じモジュール名を修飾名に使⽤する場合には不要な作業です。⼀般的に
は import Math as M などの短い名前がよく使われています。

5 .3  単純なパターン照合

それではコード例を⾒ることから始めましょう。パターン照合を使⽤して2つの整数の最⼤公
約数を計算する関数は、次のようになります。

gcd :: Int -> Int -> Int 

gcd n 0 = n 

gcd 0 m = m 

gcd n m = if n > m 

            then gcd (n - m) m 

            else gcd n (m - n) 

このアルゴリズムはユークリッドの互除法と呼ばれています。その定義をオンラインで検索す
ると、おそらく上記のコードによく似た数学の⽅程式が⾒つかるでしょう。パターン照合の利点
のひとつは、上記のようにコードを場合分けして定義することができ、数学関数の定義と似たよ
うな簡潔で宣⾔型なコードを書くことができることです。

パターン照合を使⽤して書かれた関数は、条件と結果の組み合わせによって動作します。この
定義の各⾏は選択肢(alternative)や場合(case)と呼ばれています。等号の左辺の式はパターンと
呼ばれており、それぞれの場合は空⽩で区切られた1つ以上のパターンで構成されています。等
号の右側の式が評価され値が返される前に引数が満たさなければならない条件について、これら
の場合は説明しています。それぞれの場合は上からこの順番に試されていき、最初に⼊⼒に適合
した場合が返り値を決定します。



たとえば、 gcd 関数は次の⼿順で評価されます。

まず最初の場合が試されます。第２引数がゼロの場合、関数は n （最初の引数）を返します。
そうでなければ、2番⽬の場合が試されます。最初の引数がゼロの場合、関数は m （第２引数）
を返します。
それ以外の場合、関数は最後の⾏の式を評価して返します。

パターンは値を名前に束縛することができることに注意してください。この例の各⾏では n
という名前と m という名前の両⽅、またはどちらか⼀⽅に、⼊⼒された値を束縛しています。
これより、⼊⼒の引数から名前を選ぶためのさまざまな⽅法に対応した、さまざまな種類のパタ
ーンを⾒ていくことになります。

5 .4  単純なパターン

上記のコード例では、2種類のパターンを⽰しました。

Int 型の値が正確に⼀致する場合にのみ適合する、数値リテラルパターン
引数を名前に束縛する、変数パターン

単純なパターンには他にも種類があります。

⽂字列リテラルと真偽リテラル
どんな引数とも適合するが名前に束縛はしない、アンダースコア（ _ ）で表されるワイルドカ
ードパターン

ここではこれらの単純なパターンを使⽤した、さらに2つの例を⽰します。

fromString :: String -> Boolean 

fromString "true" = true 

fromString _      = false 

 

toString :: Boolean -> String 

toString true  = "true" 

toString false = "false" 

PSCi でこれらの関数を試してみてください。

5 .5  ガード



ユークリッドの互除法の例では、 m > n のときと m <= n のときの２つに分岐するために
if .. then .. else 式を使っていました。こういうときには他にガード(guard)を使うという

選択肢もあります。

ガードは真偽値の式で、パターンによる制約に加えてそのガードが満たされたときに、その場
合の結果になります。ガードを使⽤してユークリッドの互除法を書き直すと、次のようになりま
す。

gcd :: Int -> Int -> Int 

gcd n 0 = n 

gcd 0 n = n 

gcd n m | n > m = gcd (n - m) m 

        | otherwise = gcd n (m - n) 

３⾏⽬ではガードを使⽤して、最初の引数が第２引数よりも厳密に⼤きいという条件を付け加
えています。

この例が⽰すように、ガードは等号の左側に現れ、パイプ⽂字（ | ）でパターンのリストと
区切られています。

演 習

1. （簡単）パターン照合を使⽤して、階乗関数を書いてみましょう。ヒント:⼊⼒
がゼロのときとゼロでないときの、ふたつの場合を考えてみてください。

2. （やや難しい）⼆項係数を計算するためのパスカルの公式(Pascal's Rule、パスカ
ルの三⾓形を参照のこと)について調べてみてください。パスカルの公式を利⽤
し、パターン照合を使って⼆項係数を計算する関数を記述してください。

5 .6  配列リテラルパターン

配列リテラルパターン(array literal patterns)は、固定⻑の配列に対して照合を⾏う⽅法を提供
します。たとえば、空の配列であることを特定する関数 isEmpty を書きたいとします。最初の



選択肢に空の配列パターン（ [] ）を⽤いるとこれを実現できます。

isEmpty :: forall a. Array a -> Boolean 

isEmpty [] = true 

isEmpty _ = false 

次の関数では、⻑さ5の配列と適合し、配列の5つの要素をそれぞれ異なった⽅法で束縛して
います。

takeFive :: Array Int -> Int 

takeFive [0, 1, a, b, _] = a * b 

takeFive _ = 0 

最初のパターンは、第１要素と第２要素がそれぞれ0と1であるような、5要素の配列にのみ適
合します。その場合、関数は第3要素と第4要素の積を返します。それ以外の場合は、関数は0を
返します。 PSCi で試してみると、たとえば次のようになります。

> :paste 

… takeFive [0, 1, a, b, _] = a * b 

… takeFive _ = 0 

… ^D 

 

> takeFive [0, 1, 2, 3, 4] 

6 

 

> takeFive [1, 2, 3, 4, 5] 

0 

 

> takeFive [] 

0 

配列のリテラルパターンでは、固定⻑の配列と⼀致させることはできますが、不特定の⻑さの
配列を照合させる⼿段を提供していません。PureScriptでは、そのような⽅法で不変な配列を分
解すると、実⾏速度が低下する可能性があるためです。不特定の⻑さの配列に対して照合を⾏う
ことができるデータ構造が必要な場合は、 Data.List を使うことをお勧めします。そのほかの
操作について、より優れた漸近性能を提供するデータ構造も存在します。

5 .7  レコードパターンと⾏多相



レコードパターン(Record patterns)は(ご想像のとおり)レコードに照合します。

レコードパターンはレコードリテラルに⾒た⽬が似ていますが、レコードリテラルでラベルと
式をコロンで区切るのとは異なり、レコードパターンではラベルとパターンを等号で区切りま
す。

たとえば、次のパターンは first と last と呼ばれるフィールドが含まれた任意のレコード
にマッチし、これらのフィールドの値はそれぞれ x と y という名前に束縛されます。

showPerson :: { first :: String, last :: String } -> String 

showPerson { first: x, last: y } = y <> ", " <> x 

レコードパターンはPureScriptの型システムの興味深い機能である⾏多相(row polymorphism)
の良い例となっています。もし上の showPerson を型シグネチャなしで定義していたとすると、
この型はどのように推論されるのでしょうか？⾯⽩いことに、推論される型は上で与えた型とは
同じではありません。

> showPerson { first: x, last: y } = y <> ", " <> x 

 

> :type showPerson 

forall r. { first :: String, last :: String | r } -> String 

この型変数 r とは何でしょうか？ PSCi で showPerson を使ってみると、⾯⽩いことがわか
ります。

> showPerson { first: "Phil", last: "Freeman" } 

"Freeman, Phil" 

 

> showPerson { first: "Phil", last: "Freeman", location: "Los Angeles" } 

"Freeman, Phil" 

レコードにそれ以外のフィールドが追加されていても、 showPerson 関数はそのまま動作す
るのです。型が String であるようなフィールド first と last がレコードに少なくとも含ま
れていれば、関数適⽤は正しく型付けされます。しかし、フィールドが不⾜していると、
showPerson の呼び出しは不正となります。

> showPerson { first: "Phil" } 

 

Type of expression lacks required label "last" 



showPerson の推論された型シグネチャは、 String であるような first と last というフ
ィールドと、それ以外の任意のフィールドを持った任意のレコードを引数に取り、 String を返
す、というように読むことができます。

この関数はレコードフィールドの⾏ r について多相的なので、⾏多相と呼ばれるわけです。

次のように書くことができることにも注意してください。

> showPerson p = p.last <> ", " <> p.first 

この場合も、 PSCi は先ほどと同じ型を推論するでしょう。

後ほど拡張可能作⽤(Extensible effects)について議論するときに、再び⾏多相について⾒てい
くことになります。

5 .8  ⼊れ⼦になったパターン

配列パターンとレコードパターンはどちらも⼩さなパターンを組み合わせることで⼤きなパタ
ーンを構成しています。これまでの例では配列パターンとレコードパターンの内部に単純なパタ
ーンを使⽤していましたが、パターンが⾃由に⼊れ⼦にすることができることも知っておくのが
⼤切です。⼊れ⼦になったパターンを使うと、潜在的に複雑なデータ型に対して関数が条件分岐
できるようになります。

たとえば、次のコードでは、レコードパターンと配列パターンを組み合わせて、レコードの配
列と照合させています。

type Address = { street :: String, city :: String } 

 

type Person = { name :: String, address :: Address } 

 

livesInLA :: Person -> Boolean 

livesInLA { address: { city: "Los Angeles" } } = true 

livesInLA _ = false 

5 .9  名前付きパターン



パターンには名前を付けることができ、⼊れ⼦になったパターンを使うときにスコープに追加
の名前を導⼊することができます。任意のパターンに名前を付けるには、 @ 記号を使います。

たとえば、次のコードは1つ以上の要素を持つ任意の配列と適合しますが、配列の先頭を x と
いう名前、配列全体を arr という名前に束縛します。

sortPair :: Array Int -> Array Int 

sortPair arr@[x, y] 

  | x <= y = arr 

  | otherwise = [y, x] 

sortPair arr = arr 

その結果、ペアがすでにソートされている場合は、新しい配列を複製する必要がありません。

演 習

1. （簡単）レコードパターンを使って、2つの Person レコードが同じ都市にいる
か探す関数 sameCity を定義してみましょう。

2. （やや難しい）⾏多相を考慮すると、 sameCity 関数の最も⼀般的な型は何でし
ょうか？先ほど定義した livesInLA 関数についてはどうでしょうか？

3. （やや難しい）配列リテラルパターンを使って、１要素の配列の唯⼀のメンバー
を抽出する関数 fromSingleton を書いてみましょう。１要素だけを持つ配列でな
い場合、関数は指定されたデフォルト値を返さなければなりません。この関数は
forall a. a -> Array a -> a .という型を持っていなければなりません。

5 .10 Case式

パターンはソースコードの最上位にある関数だけに現れるわけではありません。 case 式を使
⽤すると計算の途中の値に対してパターン照合を使うことができます。case式には無名関数に似
た種類の便利さがあります。関数に名前を与えることがいつも望ましいわけではありません。パ
ターン照合を使いたいためだけで関数に名前をつけるようなことを避けられるようになります。



例を⽰しましょう。次の関数は、配列の"longest zero suffix"(和がゼロであるような、最も⻑
い配列の末尾)を計算します。

import Data.Array.Partial (tail) 

import Partial.Unsafe (unsafePartial) 

 

lzs :: Array Int -> Array Int 

lzs [] = [] 

lzs xs = case sum xs of 

           0 -> xs 

           _ -> lzs (unsafePartial tail xs) 

例えば次のようになります。

> lzs [1, 2, 3, 4] 

[] 

 

> lzs [1, -1, -2, 3] 

[-1, -2, 3] 

この関数は場合ごとの分析によって動作します。もし配列が空なら、唯⼀の選択肢は空の配列
を返すことです。配列が空でない場合は、さらに2つの場合に分けるためにまず case 式を使⽤
します。配列の合計がゼロであれば、配列全体を返します。そうでなければ、配列の残りに対し
て再帰します。

5 .11 パターン照合の失敗

case式のパターンを順番に照合していって、もし選択肢のいずれの場合も⼊⼒が適合しなかっ
た時は何が起こるのでしょうか？この場合、パターン照合失敗によって、case式は実⾏時に失敗
します。

簡単な例でこの動作を⾒てみましょう。

import Partial.Unsafe (unsafePartial) 

 

partialFunction :: Boolean -> Boolean 

partialFunction = unsafePartial \true -> true 



この関数はゼロの⼊⼒に対してのみ適合する単⼀の場合を含みます。このファイルをコンパイ
ルして PSCi でそれ以外の値を与えてテストすると、実⾏時エラーが発⽣します。

> partialFunction false 

 

Failed pattern match 

どんな⼊⼒の組み合わせに対しても値を返すような関数は全関数(total function)と呼ばれ、そ
うでない関数は部分関数(partial function)と呼ばれています。

⼀般的には、可能な限り全関数として定義したほうが良いと考えられています。もしその関数
が正しい⼊⼒に対して値を返さないことがあるとわかっているなら、⼤抵は a に対して型
Maybe a の返り値にし、失敗を⽰すときには Nothing を使うようにしたほうがよいでしょう。

この⽅法なら、型安全な⽅法で値の有無を⽰すことができます。

PureScriptコンパイラは、パターンマッチが不完全で関数が全関数ではないことを検出すると
エラーを⽣成します。部分関数が安全である場合、 unsafePartial 関数を使ってこれらのエラ
ーを抑制することができます(その部分関数が安全だとあなたが⾔い切れるなら！)。もし上記の
unsafePartial 関数の呼び出しを取り除くと、コンパイラは次のエラーを⽣成します。

A case expression could not be determined to cover all inputs. 

The following additional cases are required to cover all inputs: 

 

  false 

これは値 false が、定義されたどのパターンとも⼀致しないことを⽰しています。これらの
警告には、複数の不⼀致のケースが含まれることがあります。

上記の型シグネチャも省略した場合は、次のようになります。

partialFunction true = true 

このとき、PSCiは興味深い型を推論します｡

:type partialFunction 

 

Partial => Boolean -> Boolean 

本書ではのちに => 記号を含むいろいろな型を⾒ることができます（これらは型クラスに関連
しています）。しかし、今のところは、PureScriptは型システムを使って部分関数を追跡してい



ること、開発者は型検証器にコードが安全であることを明⽰する必要があることを確認すれば⼗
分です。

コンパイラは、定義されたパターンが冗⻑であることを検出した場合（すでに定義されたパタ
ーンに⼀致するケースのみ）でも警告を⽣成します。

redundantCase :: Boolean -> Boolean 

redundantCase true = true 

redundantCase false = false 

redundantCase false = false 

このとき、最後のケースは冗⻑であると正しく検出されます。

Redundant cases have been detected. 

The definition has the following redundant cases: 

 

  false 

注意：PSCiは警告を表⽰しないので、この例を再現するには、この関数をファイルとして保
存し、 pulp build を使ってコンパイルします。

5 .12 代数的データ型

この節では、PureScriptの型システムでパターン照合に原理的に関係している代数的データ型
(Algebraic data type, ADT）と呼ばれる機能を導⼊します。

しかしまずは、ベクターグラフィックスライブラリの実装というこの章の課題を解決する基礎
として、簡単な例を切り⼝にして考えていきましょう。

直線、矩形、円、テキストなどの単純な図形の種類を表現する型を定義したいとします。オブ
ジェクト指向⾔語では、おそらくインタフェースもしくは抽象クラス Shape を定義し、使いた
いそれぞれの図形について具体的なサブクラスを定義するでしょう。

しかしながら、この⽅針は⼤きな⽋点をひとつ抱えています。 Shape を抽象的に扱うために
は、実⾏したいと思う可能性のあるすべての操作を事前に把握し、 Shape インタフェースに定
義する必要があるのです。このため、モジュール性を壊さずに新しい操作を追加することが難し
くなります。



もし図形の種類が事前にわかっているなら、代数的データ型はこうした問題を解決する型安全
な⽅法を提供します。モジュール性のある⽅法で Shape に新たな操作を定義し、型安全なまま
保守することを可能にします。

代数的データ型として表現された Shape がどのように記述されるかを次に⽰します。

data Shape 

  = Circle Point Number 

  | Rectangle Point Number Number 

  | Line Point Point 

  | Text Point String 

次のように Point 型を代数的データ型として定義することもできます。

data Point = Point 

  { x :: Number 

  , y :: Number 

  } 

この Point データ型は、興味深い点をいくつか⽰しています。

代数的データ型の構築⼦に格納されるデータは、プリミティブ型に限定されるわけではありませ
ん。構築⼦はレコード、配列、あるいは他の代数的データ型を含めることもできます。
代数的データ型は複数の構築⼦があるデータを記述するのに便利ですが、構築⼦がひとつだけの
ときでも便利です。
代数的データ型の構築⼦は、代数的データ型⾃⾝と同じ名前の場合もあります。これはごく⼀般
的であり、 Point データ構築⼦と Point 型構築⼦を混同しないようにすることが⼤切です。こ
れらは異なる名前空間にあります。

この宣⾔ではいくつかの構築⼦の和として Shape を定義しており、各構築⼦に含まれたデー
タはそれぞれ区別されます。 Shape は、中央 Point と半径を持つ Circle か、 Rectangle 、
Line 、 Text のいずれかです。他には Shape 型の値を構築する⽅法はありません。

代数的データ型の定義は予約語 data から始まり、それに新しい型の名前と任意個の型引数が
続きます。その型のデータ構築⼦は等号の後に定義され、パイプ⽂字（ | ）で区切られます。

それではPureScriptの標準ライブラリから別の例を⾒てみましょう。オプショナルな値を定義
するのに使われる Maybe 型を本書の冒頭で扱いました。 purescript-maybe パッケージでは
Maybe を次のように定義しています。

data Maybe a = Nothing | Just a 



この例では型引数 a の使⽤⽅法を⽰しています。パイプ⽂字を「または」と読むことにする
と、この定義は「 Maybe a 型の値は、無い( Nothing )、またはただの( Just )型 a の値だ」と
英語のように読むことができます。

データ構築⼦は再帰的なデータ構造を定義するために使⽤することもできます。更に例を挙げ
ると、要素が型 a の単⽅向連結リストのデータ型を定義はこのようになります。

data List a = Nil | Cons a (List a) 

この例は purescript-lists パッケージから持ってきました。ここで Nil 構築⼦は空のリス
トを表しており、 Cons は先頭となる要素と他の配列から空でないリストを作成するために使わ
れます。 Cons の２つ⽬のフィールドでデータ型 List a を使⽤しており、再帰的なデータ型に
なっていることに注⽬してください。

5 .13 代数的データ型の使⽤

代数的データ型の構築⼦を使⽤して値を構築するのはとても簡単です。対応する構築⼦に含ま
れるデータに応じた引数を⽤意し、その構築⼦を単に関数のように適⽤するだけです。

例えば、上で定義した Line 構築⼦は２つの Point を必要としていますので、 Line 構築⼦
を使って Shape を構築するには、型 Point のふたつの引数を与えなければなりません。

exampleLine :: Shape 

exampleLine = Line p1 p2 

  where 

    p1 :: Point 

    p1 = Point { x: 0.0, y: 0.0 } 

 

    p2 :: Point 

    p2 = Point { x: 100.0, y: 50.0 } 

p1 及び p2 を構築するため、レコードを引数として Point 構築⼦を適⽤しています。

代数的データ型で値を構築することは簡単ですが、これをどうやって使ったらよいのでしょう
か？ここで代数的データ型とパターン照合との重要な接点が⾒えてきます。代数的データ型の値
がどの構築⼦から作られたかを調べたり、代数的データ型からフィールドの値を取り出す唯⼀の
⽅法は、パターン照合を使⽤することです。



例を⾒てみましょう。 Shape を String に変換したいとしましょう。 Shape を構築するのに
どの構築⼦が使⽤されたかを調べるには、パターン照合を使⽤しなければなりません。これには
次のようにします。

showPoint :: Point -> String 

showPoint (Point { x: x, y: y }) = 

  "(" <> show x <> ", " <> show y <> ")" 

 

showShape :: Shape -> String 

showShape (Circle c r)      = ... 

showShape (Rectangle c w h) = ... 

showShape (Line start end)  = ... 

showShape (Text p text) = ... 

各構築⼦はパターンとして使⽤することができ、構築⼦への引数はそのパターンで束縛するこ
とができます。 showShape の最初の場合を考えてみましょう。もし Shape が Circle 構築⼦適
合した場合、２つの変数パターン c と r を使って Circle の引数（中⼼と半径）がスコープに
導⼊されます。その他の場合も同様です。

showPoint は、パターン照合の別の例にもなっています。 showPoint はひとつの場合しかあ
りませんが、 Point 構築⼦の中に含まれたレコードのフィールドに適合する、⼊れ⼦になった
パターンが使われています。

5 .14 レコード同名利⽤

showPoint 関数は引数内のレコードと⼀致し、 x と y プロパティを同じ名前の値に束縛し
ます。 PureScriptでは、このようなパターン⼀致を次のように単純化できます。

showPoint :: Point -> String 

showPoint (Point { x, y }) = ... 

ここでは、プロパティの名前のみを指定し、名前に導⼊したい値を指定する必要はありませ
ん。 これはレコード同名利⽤(record pun)と呼ばれます。

レコード同名利⽤をレコードの構築に使⽤することもできます。例えば、スコープ内に x と
y という名前の値があれば、 Point {x、y} を使って Point を作ることができます。

origin :: Point 

origin = Point { x, y } 



  where 

    x = 0.0 

    y = 0.0 

これは、状況によってはコードの可読性を向上させるのに役⽴ちます。

演 習

1. （簡単）半径 10 で中⼼が原点にある円を表す Shape の値を構築してくださ
い。

2. （やや難しい）引数の Shape を原点を中⼼として 2.0 倍に拡⼤する、 Shape
から Shape への関数を書いてみましょう。

3. （やや難しい） Shape からテキストを抽出する関数を書いてください。この関
数は Maybe String を返さなければならず、もし⼊⼒が Text を使⽤して構築され
たのでなければ、返り値には Nothing 構築⼦を使ってください。

5 .15 newtype宣⾔

代数的データ型の特別な場合に、newtypeと呼ばれる重要なものあります。newtypeは予約語
data の代わりに予約語 newtype を使⽤して導⼊します。

newtype宣⾔では過不⾜なくひとつだけの構築⼦を定義しなければならず、その構築⼦は過不
⾜なくひとつだけの引数を取る必要があります。つまり、newtype宣⾔は既存の型に新しい名前
を与えるものなのです。実際、newtypeの値は、元の型と同じ実⾏時表現を持っています。しか
し、これらは型システムの観点から区別されます。これは型安全性の追加の層を提供するので
す。

例として、ピクセルとインチのような単位を表現するために、 Number の型レベルの別名を定
義したくなる場合があるかもしれません。



newtype Pixels = Pixels Number 

newtype Inches = Inches Number 

こうすると Inches を期待している関数に Pixels 型の値を渡すことは不可能になりますが、
実⾏時の効率に余計な負荷が加わることはありません。

newtypeは次の章で型クラスを扱う際に重要になります。newtypeは実⾏時の表現を変更する
ことなく型に異なる振る舞いを与えることを可能にするからです。

5 .16 ベクターグラフィックスライブラリ

これまで定義してきたデータ型を使って、ベクターグラフィックスを扱う簡単なライブラリを
作成していきましょう。

ただの Shape の配列であるような、 Picture という型同義語を定義しておきます。

type Picture = Array Shape 

デバッグしていると Picture を String として表⽰できるようにしたくなることもあるでし
ょう。これはパターン照合を使⽤して定義された showPicture 関数で⾏うことができます。

showPicture :: Picture -> Array String 

showPicture = map showShape 

それを試してみましょう。 モジュールを pulp build でコンパイルし、 pulp repl でPSCi
を開きます。

$ pulp build 

$ pulp repl 

 

> import Data.Picture 

 

> :paste 

… showPicture 

…   [ Line (Point { x: 0.0, y: 0.0 }) 

…          (Point { x: 1.0, y: 1.0 }) 

…   ] 

… ^D 



 

["Line [start: (0.0, 0.0), end: (1.0, 1.0)]"] 

5 .17 外接矩形の算出

このモジュールのコード例には、 Picture の最⼩外接矩形を計算する関数 bounds が含まれ
ています。

Bounds は外接矩形を定義するデータ型です。また、構築⼦をひとつだけ持つ代数的データ型
として定義されています。

data Bounds = Bounds 

  { top    :: Number 

  , left   :: Number 

  , bottom :: Number 

  , right  :: Number 

  } 

Picture 内の Shape の配列を⾛査し、最⼩の外接矩形を累積するため、 bounds は
Data.Foldable の foldl 関数を使⽤しています。

bounds :: Picture -> Bounds 

bounds = foldl combine emptyBounds 

  where 

    combine :: Bounds -> Shape -> Bounds 

    combine b shape = union (shapeBounds shape) b 

畳み込みの初期値として空の Picture の最⼩外接矩形を求める必要がありますが、
emptyBounds で定義される空の外接矩形がその条件を満たしています。

累積関数 combine は where ブロックで定義されています。 combine は foldl の再帰呼び出
しで計算された外接矩形と、配列内の次の Shape を引数にとり、ユーザ定義の演算⼦ union を
使ってふたつの外接矩形の和を計算しています。 shapeBounds 関数は、パターン照合を使⽤し
て、単⼀の図形の外接矩形を計算します。



演 習

1. （やや難しい） ベクターグラフィックライブラリを拡張し、 Shape の⾯積を計
算する新しい操作 area を追加してください。この演習では、テキストの⾯積は0
であるものとしてください。

2. （難しい） Shape を拡張し、新しいデータ構築⼦ Clipped を追加してくださ
い。 Clipped は他の Picture を矩形に切り抜き出ます。切り抜かれた Picture
の境界を計算できるよう、 shapeBounds 関数を拡張してください。これは Shape
を再帰的なデータ型にすることに注意してください。

まとめ

この章では、関数型プログラミングから基本だが強⼒なテクニックであるパターン照合を扱い
ました。複雑なデータ構造の部分と照合するために、簡単なパターンだけでなく配列パターンや
レコードパターンをどのように使⽤するかを⾒てきました。

またこの章では、パターン照合に密接に関連する代数的データ型を導⼊しました。代数的デー
タ型がデータ構造のわかりやすい記述をどのように可能にするか、新たな操作でデータ型を拡張
するためのモジュール性のある⽅法を提供することを⾒てきました。

最後に、多くの既存のJavaScript関数に型を与えるために、強⼒な抽象化である⾏多相を扱い
ました。この本の後半ではこれらの概念を再び扱います。

本書では今後も代数的データ型とパターン照合を使⽤するので、今のうちにこれらに習熟して
おくと後で役⽴つでしょう。これ以外にも独⾃の代数的データ型を作成し、パターン照合を使⽤
してそれらを使う関数を書くことを試してみてください。



第6章 型クラス

6 .1  章の⽬標

この章では、PureScriptの型システムによって可能になる強⼒な抽象化の⼿法である、型クラ
スを導⼊します。

この章ではデータ構造をハッシュするためのライブラリを題材に説明していきます。データ⾃
⾝の構造について直接考えることなく複雑な構造のデータのハッシュ値を求めるために、型クラ
スの仕組みがどのようにして働くのかを⾒ていきます。

また、PureScriptのPreludeや標準ライブラリに含まれる、標準的な型クラスも⾒ていきます。
PureScriptのコードは概念を簡潔に表現するために型クラスの強⼒さに⼤きく依存しているの
で、これらのクラスに慣れておくと役に⽴つでしょう。

6 .2  プロジェクトの準備

この章のソースコードは、ファイル src/data/Hashable.purs で定義されています。

このプロジェクトには以下のBower依存関係があります。

purescript-maybe : オプショナルな値を表す Maybe データ型が定義されています。
purescript-tuples : 値の組を表す Tuple データ型が定義されています。
purescript-either : ⾮交和を表す Either データ型が定義されています。
purescript-strings : ⽂字列を操作する関数が定義されています。
purescript-functions : PureScriptの記述⽤の補助関数が定義されています。

モジュール Data.Hashable では、これらのBowerパッケージによって提供されるモジュール
のいくつかをインポートしています。

module Data.Hashable where 

 

import Data.Maybe 

import Data.Tuple 

import Data.Either 

import Data.String 

import Data.Function 



6 .3  ⾒せてください！(Show Me!)

型クラスの最初に扱う例は、すでに何回か⾒たことがある関数と関係しています。 show は、
何らかの値を取り、それを⽂字列として表⽰する関数です。

show は Prelude モジュールの Show と呼ばれる型クラスで次のように定義されています。

class Show a where 

  show :: a -> String 

このコードでは、型変数 a でパラメータ化された、 Show という新しい型クラス(type class)
を宣⾔しています。

型クラスインスタンスには、型クラスで定義された関数の、その型に特殊化された実装が含ま
れています。

例えば、Preludeにある Boolean 値に対する Show 型クラスインスタンスの定義は次のとおり
です。

instance showBoolean :: Show Boolean where 

  show true = "true" 

  show false = "false" 

このコードは showBool  ean という名前の型クラスのインスタンスを宣⾔します。

PureScriptでは、⽣成されたJavaScriptの可読性を良くするために、型クラスインスタンスに
名前をつけます。このとき、 Boolean 型は Show 型クラスに属しているといいます。

PSCi で、いろいろな型の値を Show 型クラスを利⽤して表⽰してみましょう。

> import Prelude 

> show true 

"true" 

 

> show 1.0 

"1.0" 

 

> show "Hello World" 

"\"Hello World\"" 



この例ではさまざまなプリミティブ型の値を show しましたが、もっと複雑な型を持つ値を
show することもできます。

> import Data.Tuple 

 

> show (Tuple 1 true) 

"(Tuple 1 true)" 

 

> import Data.Maybe 

 

> show (Just "testing") 

"(Just \"testing\")" 

型 Data.Either の値を表⽰しようとすると、興味深いエラーメッセージが表⽰されます。

> import Data.Either 

> show (Left 10) 

 

The inferred type 

 

    forall a. Show a => String 

 

has type variables which are not mentioned in the body of the type. 

Consider adding a type annotation. 

ここでの問題は show しようとしている型に対する Show インスタンスが存在しないというこ
とではなく、 PSCi がこの型を推論できなかったということです。このエラーメッセージで未知
の型 a と表⽰されているのがそれです。

:: 演算⼦を使って式に対して型注釈を加えると、 PSCi が正しい型クラスインスタンスを選
ぶことができるようになります。

> show (Left 10 :: Either Int String) 

"(Left 10)" 

Show インスタンスをまったく持っていない型もあります。関数の型 -> がその⼀例です。
Int から Int への関数を show しようとすると、型検証器によってその通りのエラーメッセー

ジが表⽰されます。

> import Prelude 

> show $ \n -> n + 1 

 



No type class instance was found for 

 

  Data.Show.Show (Int -> Int) 

演 習

1. (簡単)前章の showShape 関数を使って、 Shape 型に対しての Show インスタンス
を定義してみましょう。

6 .4  標準的な型クラス

この節では、Preludeや標準ライブラリで定義されている標準的な型クラスをいくつか⾒てい
きましょう。これらの型クラスはPureScript特有の抽象化の基礎としてあちこちで使われている
ので、これらの関数の基本についてよく理解しておくことを強くお勧めします。

Eq型クラス

Eq 型クラスは、2つの値が等しいかどうかを調べる eq 関数を定義しています。等値演算⼦
( == )は eq の別名にすぎません。

class Eq a where 

  eq :: a -> a -> Boolean 

異なる型の2つの値を⽐較しても意味がありませんから、いずれの演算⼦も2つの引数が同じ
型を持つ必要があることに注意してください。

PSCi で Eq 型クラスを試してみましょう。

> 1 == 2 

false 

 

> "Test" == "Test" 

true 



Ord型クラス

Ord 型クラスは順序付け可能な型に対して2つの値を⽐較する compare 関数を定義します。
compare 関数が定義されていると、⽐較演算⼦ < 、 > と、その仲間 <= 、 >= も定義されま

す。

data Ordering = LT | EQ | GT 

 

class Eq a <= Ord a where 

  compare :: a -> a -> Ordering 

compare 関数は2つの値を⽐較して Ordering の3つの値のうちいずれかを返します。

LT - 最初の引数が2番⽬の値より⼩さいとき
EQ - 最初の引数が2番⽬の値と等しい(または⽐較できない)とき
GT - 最初の引数が2番⽬の値より⼤きいとき

compare 関数についても PSCi で試してみましょう。

> compare 1 2 

LT 

 

> compare "A" "Z" 

LT 

Num型クラス

Field 型クラスは加算、減算、乗算、除算などの数値演算⼦を使⽤可能な型を⽰します。必
要に応じて再利⽤できるように、これらの演算⼦を抽象化するわけです。

注意: 関数呼び出しが型クラスの実装に基いて呼び出されるのとは対照的に、型クラス Eq や
Ord のクラスと同様に、 Field 型のクラスはPureScriptでは特別に扱われ、 1 + 2 * 3 のよう

な単純な式は単純なJavaScriptへと変換されます。

class EuclideanRing a <= Field a 

Field 型クラスは、いくつかのより抽象的な上位クラス(Super Class)が組み合わさってでき
ています。これは、その型は Field 型クラスの操作をすべてを提供しているわけではないが、
その⼀部を提供する、というように抽象的に説明することができます。この型クラスは抽象的な
すべてではないいくつかの数値演算⼦をサポートしています。例えば、⾃然数の型は加算および
乗算については閉じていますが、減算については閉じていないため、この型は Semiring クラス



(これは Num の上位クラスです)のインスタンスですが、 Ring や Field のインスタンスではあ
りません。

上位クラスについては、この章の後半で詳しく説明します。しかし、すべての数値型クラスの
階層について述べるのはこの章の⽬的から外れているため、この内容に興味のある読者
は purescript-prelude 内の Field に関するドキュメントを参照してください。

半群とモノイド

Semigroup (半群)型クラスは、連結演算⼦ append を提供する型を⽰します。

class Semigroup a where 

  append :: a -> a -> a 

普通の⽂字列連結について⽂字列は半群をなし、同様に配列も半群をなします。その他の標準
的なインスタンスの幾つかは、 purescript-monoid パッケージで提供されています。

以前に⾒た <> 連結演算⼦は、 append の別名として提供されています。

purescript-monoid パッケージで提供されている Monoid 型クラスは、 mempty と呼ばれる
空の値の概念で Semigroup 型クラスを拡張します。

class Semigroup m <= Monoid m where 

  mempty :: m 

⽂字列や配列はモノイドの簡単な例になっています。

Monoid 型クラスインスタンスでは、「空」の値から始めて新たな値を合成していき、その型
で累積した結果を返すにはどうするかを記述する型クラスです。例えば、畳み込みを使っていく
つかのモノイドの値の配列を連結する関数を書くことができます。 PSCi で試すと次のようにな
ります。

> import Data.Monoid 

> import Data.Foldable 

 

> foldl append mempty ["Hello", " ", "World"]   

"Hello World" 

 

> foldl append mempty [[1, 2, 3], [4, 5], [6]] 

[1,2,3,4,5,6] 



purescript-monoid パッケージにはモノイドと半群の多くの例を提供しており、これらを本
書で扱っていきます。

Foldable型クラス

Monoid 型クラスは畳み込みの結果になるような型を⽰しますが、 Foldable 型クラスは、畳
み込みの元のデータとして使えるような型構築⼦を⽰しています。

また、 Foldable 型クラスは、配列や Maybe などのいくつかの標準的なコンテナのインスタ
ンスを含む purescript-foldable-traversable パッケージで提供されています。

Foldable クラスに属する関数の型シグネチャは、これまで⾒てきたものよりも少し複雑で
す。

class Foldable f where 

  foldr :: forall a b. (a -> b -> b) -> b -> f a -> b 

  foldl :: forall a b. (b -> a -> b) -> b -> f a -> b 

  foldMap :: forall a m. Monoid m => (a -> m) -> f a -> m 

この定義は f を配列の型構築⼦だと特殊化して考えてみるとわかりやすくなります。この場
合、すべての a について f a を Array a に置き換える事ができますが、 foldl と foldr の
型が、最初に⾒た配列に対する畳み込みの型になるとわかります。

foldMap についてはどうでしょうか？これは
forall a m. Monoid m => (a -> m) -> Array a -> m になります。この型シグネチャは、型
m が Monoid 型クラスのインスタンスであればどんな型でも返り値の型として選ぶことができ

ると⾔っています。配列の要素をそのモノイドの値へと変換する関数を提供すれば、そのモノイ
ドの構造を利⽤して配列を畳み込み、ひとつの値にして返すことができます。

それでは PSCi で foldMap を試してみましょう。

> import Data.Foldable 

 

> foldMap show [1, 2, 3, 4, 5] 

"12345" 

ここではモノイドとして⽂字列を選び、 Int を⽂字列として表⽰する show 関数を使いまし
た。それから、数の配列を渡し、それぞれの数を show してひとつの⽂字列へと連結した結果出
⼒されました。



畳み込み可能な型は配列だけではありません。 purescript-foldable-traversable では
Maybe や Tuple のような型の Foldable インスタンスが定義されており、 purescript-lists

のような他のライブラリでは、そのライブラリのそれぞれのデータ型に対して Foldable イン
スタンスが定義されています。 Foldable は順序付きコンテナ(ordered container)の概念を抽象
化するのです。

関⼿と型クラス則

PureScriptで副作⽤を伴う関数型プログラミングのスタイルを可能にするための Functor と
Applicative 、 Monad といった型クラスがPreludeでは定義されています。これらの抽象につ

いては本書で後ほど扱いますが、まずは「持ち上げ演算⼦」 map の形ですでに⾒てきた
Functor 型クラスの定義を⾒てみましょう。

class Functor f where 

  map :: forall a b. (a -> b) -> f a -> f b 

演算⼦ map 関数（別名 <$> ）は関数をそのデータ構造まで「持ち上げる」(lift)ことができま
す。ここで「持ち上げ」という⾔葉の具体的な定義は問題のデータ構造に依りますが、すでにい
くつかの単純な型についてその動作を⾒てきました。

> import Prelude 

 

> map (\n -> n < 3) [1, 2, 3, 4, 5] 

[true, true, false, false, false] 

 

> import Data.Maybe 

> import Data.String (length) 

 

> map length (Just "testing") 

(Just 7) 

map 演算⼦は様々な構造の上でそれぞれ異なる振る舞いをしますが、 map 演算⼦の意味はど
のように理解すればいいのでしょうか。

直感的には、 map 演算⼦はコンテナのそれぞれの要素へ関数を適⽤し、その結果から元のデ
ータと同じ形状を持った新しいコンテナを構築するのだというように理解することができます。
しかし、この概念を厳密にするにはどうしたらいいでしょうか？

Functor の型クラスのインスタンスは、関⼿則(functor laws)と呼ばれる法則を順守するもの
と期待されています。



map id xs = xs

map g (map f xs) = map (g <<< f) xs

最初の法則は恒等射律(identity law)です。これは、恒等関数をその構造まで持ち上げると、元
の構造をそのまま返す恒等射になるということと⾔っています。恒等関数は⼊⼒を変更しません
から、これは理にかなっています。

第⼆の法則は合成律(composition law)です。構造をひとつの関数で写してから２つめの関数で
写すのは、２つの関数の合成で構造を写すのと同じだ、と⾔っています。

「持ち上げ」の⼀般的な意味が何であれ、データ構造に対する持ち上げ関数の正しい定義はこ
れらの法則に従っていなければなりません。

標準の型クラスの多くには、このような法則が付随しています。⼀般に、型クラスに与えられ
た法則は、型クラスの関数に構造を与え、インスタンスについて調べられるようにします。興味
のある読者は、すでに⾒てきた標準の型クラスに属する法則について調べてみてもよいでしょ
う。

演 習

1. (簡単)次のnewtypeは複素数を表します。

newtype Complex = Complex  

    { real :: Number 

    , imaginary :: Number  

    } 

Complex について、 Show と Eq のインスタンスを定義してください。

6 .5  型注釈

型クラスを使うと、関数の型に制約を加えることができます。例を⽰しましょう。 Eq 型クラ
スのインスタンスで定義された等値性を使って、３つの値が等しいかどうかを調べる関数を書き



たいとします。

threeAreEqual :: forall a. Eq a => a -> a -> a -> Boolean 

threeAreEqual a1 a2 a3 = a1 == a2 && a2 == a3 

この型宣⾔は forall を使って定義された通常の多相型のようにも⾒えます。しかし、太い⽮
印 => で型の残りの部分から区切られた、型クラス制約(type class constraint) Eq a がありま
す。

インポートされたモジュールのどれかに a に対する Eq インスタンスが存在するなら、どん
な型 a を選んでも threeAsEqual を呼び出すことができる、とこの型は⾔っています。

制約された型には複数の型クラスインスタンスを含めることができますし、インスタンスの型
は単純な型変数に限定されません。 Ord と Show のインスタンスを使って2つの値を⽐較する例
を次に⽰します。

showCompare :: forall a. Ord a => Show a => a -> a -> String 

showCompare a1 a2 | a1 < a2 = 

  show a1 <> " is less than " <> show a2 

showCompare a1 a2 | a1 > a2 = 

  show a1 <> " is greater than " <> show a2 

showCompare a1 a2 = 

  show a1 <> " is equal to " <> show a2 

=> シンボルを複数回使って複数の制約を指定できることに注意してください。複数の引数の
カリー化された関数を定義するのと同様です。しかし、2つの記号を混同しないように注意して
ください。

a -> b は型 a から型 b への関数の型を表します。
a => b は制約 a を型 b に適⽤します。

PureScriptコンパイラは、型の注釈が提供されていない場合、制約付き型を推測しようとしま
す。これは、関数に対して可能な最も⼀般的な型を使⽤したい場合に便利です。

PSCi で Semiring のような標準の型クラスのいずれかを使って、このことを試してみましょ
う。

> import Prelude 

 

> :type \x -> x + x 

forall a. Semiring a => a -> a 



ここで、この関数には Int -> Int または Number -> Number と注釈を付けることが考えられ
ますが、最も⼀般的な型が Semiring で動作するため、PSCiでは Int と Number の両⽅で関数
を実⾏させることができます。

6 .6  インスタンスの重複

PureScriptには型クラスのインスタンスに関する重複インスタンス規則(Overlapping instances
rule)という規則があります。型クラスのインスタンスが関数呼び出しのところで必要とされると
きはいつでも、PureScriptは正しいインスタンスを選択するために型検証器によって推論された
情報を使⽤します。そのとき、その型の適切なインスタンスがちょうどひとつだけ存在しなけれ
ばなりません。

これを実証するために、適当な型に対して2つの異なる型クラスのインスタンスを作成してみ
ましょう。次のコードでは、型 T の2つの重複する Show インスタンスを作成しています。

module Overlapped where 

import Prelude 

data T = T 

 

instance showT1 :: Show T where 

  show _ = "Instance 1" 

   

instance showT2 :: Show T where 

  show _ = "Instance 2" 

このモジュールはエラーなくコンパイルされます。 PSCi を起動し、型 T の Show インスタ
ンスを⾒つけようとすると、重複インスタンス規則が適⽤され、エラーになります。

Overlapping instances found for Prelude.Show T 

重複インスタンスルールが適⽤されるのは、型クラスのインスタンスの⾃動選択が予測可能な
処理であるようにするためです。もし型に対してふたつの型クラスインスタンスを許し、モジュ
ールインポートの順序に従ってどちらかを選ぶようにすると、実⾏時のプログラムの振る舞いが
予測できなくなってしまい好ましくありません。

適切な法則を満たすふたつ妥当な型クラスインスタンスが存在しうるなら、既存の型を包む
newtypeを定義するのが⼀般的な⽅法です。重複インスタンスのルールの下でも、異なる
newtypeなら異なる型クラスインスタンスを持つことが許されるので、問題はなくなります。こ



の⼿法はPureScriptの標準ライブラリでも使われており、例えば purescript-monoid では、
Maybe a 型は Monoid 型クラスの妥当なインスタンスを複数持っています。

6 .7  インスタンスの依存関係

制約された型を使うと関数の実装が型クラスインスタンスに依存できるように、型クラスイン
スタンスの実装は他の型クラスインスタンスに依存することができます。これにより、型を使っ
てプログラムの実装を推論するという、プログラム推論の強⼒な形式を提供します。

Show 型クラスを例に考えてみましょう。要素を show する⽅法があるとき、その要素の配列
を show する型クラスインスタンスを書くことができます。

instance showArray :: Show a => Show (Array a) where 

  ... 

型クラスインスタンスが複数の他のインスタンスに依存する場合、括弧で囲んでそれらのイン
スタンスをコンマで区切り、それを => シンボルの左側に置く必要があります。

instance showEither :: (Show a, Show b) => Show (Either a b) where 

  ... 

これらの2つの型クラスインスタンスは purescript-prelude ライブラリにあります。

プログラムがコンパイルされると、 Show の正しい型クラスのインスタンスは show の引数の
推論された型に基づいて選ばれますが、このあたりの複雑さに開発者が関与することはありませ
ん。

演 習

1. (簡単)次は型 a の要素の空でない配列の型を定義しています。

data NonEmpty a = NonEmpty a (Array a) 



`Eq a`と `Eq (Array a)`のインスタンスを再利用して、型 `NonEmpty a`に対する `Eq`インスタンスを書

. (やや難しい) Semigroup インスタンスを Array に再利⽤して NonEmpty a の Semigroup イン
スタンスを作成しましょう。

. (やや難しい) NonEmpty の Functor インスタンスを書いてみましょう。

. (やや難しい) Ord のインスタンスを持つ型 a があれば、他の値より⼤きい新しい 無限の値を追
加することができます。

```haskell 

data Extended a = Finite a | Infinite 

``` 

`a`の `Ord`インスタンスを再利用して、 `Extended a`の `Ord`インスタンスを書いてみましょう。 

. (難しい) NonEmpty の Foldable インスタンスを書いてみましょう。ヒント：配列の Foldable
インスタンスを再利⽤してみましょう。

. (難しい) 順序付きコンテナを定義する(そして Foldable のインスタンスを持っている)よう
な型構築⼦ f が与えられたとき、追加の要素を先頭に含めるような新たなコンテナ型を作るこ
とができます。

data OneMore f a = OneMore a (f a) 

このコンテナ OneMore f もまた順序を持っています。ここで、新しい要素は任意の f の要素
よりも前にきます。この OneMore f の Foldable インスタンスを書いてみましょう。

instance foldableOneMore :: Foldable f => Foldable (OneMore f) where 

6 .8  多変数型クラス

型クラスは必ずしもひとつの型だけを型変数としてとるわけではありません。型変数がひとつ
だけなのが最も⼀般的ですが、実際には型クラスはゼロ個以上の型変数を持つことができます。

それでは2つの型引数を持つ型クラスの例を⾒てみましょう。



module Stream where 

 

import Data.Array as Array 

import Data.Maybe (Maybe) 

import Data.String as String 

 

class Stream stream element where 

  uncons :: stream -> Maybe { head :: element, tail :: stream } 

 

instance streamArray :: Stream (Array a) a where 

  uncons = Array.uncons 

 

instance streamString :: Stream String Char where 

  uncons = String.uncons 

この Stream モジュールでは、 uncons 関数を使ってストリームの先頭から要素を取り出すこ
とができる、要素のストリームのような型を⽰すクラス Stream が定義されています。

Stream 型クラスは、ストリーム⾃⾝の型だけでなくその要素の型も型変数として持っている
ことに注意してください。これによって、ストリームの型が同じでも要素の型について異なる型
クラスインスタンスを定義することができます。

このモジュールでは、 uncons がパターン照合で配列の先頭の要素を取り除くような配列のイ
ンスタンスと、⽂字列から最初の⽂字を取り除くような⽂字列のインスタンスという、２つの型
クラスインスタンスが定義されています。

任意のストリーム上で動作する関数を記述することができます。例えば、ストリームの要素に
基づいて Monoid に結果を累積する関数は次のようになります。

import Prelude 

import Data.Maybe (Maybe(..)) 

import Data.Monoid (class Monoid, mempty) 

 

foldStream :: forall l e m. Stream l e => Monoid m => (e -> m) -> l -> m 

foldStream f list = 

  case uncons list of 

    Nothing -> mempty 

    Just cons -> f cons.head <> foldStream f cons.tail 

PSCi で使って、異なる Stream の型や異なる Monoid の型について foldStream を呼び出し
てみましょう。



6 .9  関数従属性

多変数型クラスは⾮常に便利ですが、混乱しやすい型や型推論の問題にもつながります。簡単
な例として、上記の Stream クラスを使って genericTail 関数をストリームに書くことを考え
てみましょう。

genericTail xs = map _.tail (uncons xs) 

これはやや複雑なエラーメッセージを出⼒します。

The inferred type 

 

  forall stream a. Stream stream a => stream -> Maybe stream 

 

has type variables which are not mentioned in the body of the type. 

Consider adding a type annotation. 

エラーは、 genericTail 関数が Stream 型クラスの定義で⾔及された element 型を使⽤し
ないので、その型は未解決のままであることを指しています。

さらに、特定の型のストリームに genericTail を適⽤することができません。

> map _.tail (uncons "testing") 

 

The inferred type 

 

  forall a. Stream String a => Maybe String 

 

has type variables which are not mentioned in the body of the type. 

Consider adding a type annotation. 

ここでは、コンパイラが streamString インスタンスを選択することを期待しています。結局
のところ、 String は Char のストリームであり、他の型のストリームであってはなりません。

コンパイラは⾃動的にその排除を⾏うことはできず、 streamString インスタンスに引き渡す
ことはできません。しかし、型クラス定義にヒントを追加すると、コンパイラを助けることがで
きます。

class Stream stream element | stream -> element where 

  uncons :: stream -> Maybe { head :: element, tail :: stream } 



ここで、 stream -> element は関数従属性(functional dependency)と呼ばれます。関数従属
性は、多変数型クラスの型引数間の関数関係を宣⾔します。この関数の依存関係は、ストリーム
型から（⼀意の）要素型への関数があることをコンパイラに伝えるので、コンパイラがストリー
ム型を知っていれば要素型へ適⽤できます。

このヒントは、コンパイラが上記の genericTail 関数の正しい型を推論するのに⼗分です。

> :type genericTail 

forall stream element. Stream stream element => stream -> Maybe stream 

 

> genericTail "testing" 

(Just "esting") 

多種の型のクラスを使⽤して特定のAPIを設計する場合、関数従属性は⾮常に有⽤です。

6 .10 型変数のない型クラス

ゼロ個の型変数を持つ型クラスを定義することもできます！これらは関数に対するコンパイル
時のアサーションに対応しており、型システム内のコードの⼤域的な性質を追跡することができ
ます。

たとえば、型システムを使って部分関数の使⽤を追跡したいとしましょう。すでに
Data.Array.Partial で定義されている head と tail の部分関数を確認します。

head :: forall a. Partial => Array a -> a 

 

tail :: forall a. Partial => Array a -> Array a 

Partial モジュールの Partial 型クラスのインスタンスを定義していないことに注意してく
ださい。こうすると⽬的を達成できます。このままの定義では head 関数を使⽤しようとすると
型エラーになるのです。

> head [1, 2, 3] 

 

No type class instance was found for 

 

  Prim.Partial 



代わりに、これらの部分関数を利⽤するすべての関数で Partial 制約を再発⾏する⽅法がで
きます。

secondElement :: forall a. Partial => Array a -> a 

secondElement xs = head (tail xs) 

前章で⾒た unsafePartial 関数を使⽤し、部分関数を通常の関数（unsafely）として扱うこ
とができます。 この関数は Partial.Unsafe モジュールで定義されています。

unsafePartial :: forall a. (Partial => a) -> a 

Partial 制約は関数の⽮印の左側の括弧の中に現れますが、外側の forall では現れませ
ん。 つまり、 unsafePartial は部分的な値から通常の値への関数です。

6 .11 上位クラス

インスタンスを別のインスタンスに依存させることによって型クラスのインスタンス間の関係
を表現することができるように、いわゆる上位クラス(superclass)を使って型クラス間の関係を
表現することができます。

あるクラスのどんなインスタンスも、その他のあるクラスのインスタンスで必要とされている
とき、前者の型クラスは後者の型クラスの上位クラスであるといい、クラス定義で逆向きの太い
⽮印(<=��)を使い上位クラス関係を⽰します。

すでに上位クラスの関係の⼀例について⾒ています。 Eq クラスは Ord の上位クラスです。
Ord クラスのすべての型クラスインスタンスについて、その同じ型に対応する Eq インスタン

スが存在しなければなりません。 compare 関数が2つの値が⽐較できないと報告した時は、そ
れらが実は同値であるかどうかを決定するために Eq クラスを使いたくなることが多いでしょう
から、これは理にかなっています。

⼀般に、下位クラスの法則が上位クラスのメンバに⾔及しているとき、上位クラス関係を定義
するのは理にかなっています。例えば、 Ord と Eq のインスタンスのどんな組についても、も
しふたつの値が Eq インスタンスのもとで同値であるなら、 compare 関数は EQ を返すはずだ
とみなすのは妥当です。⾔い換えれば、 a == b ならば compare a b == EQ です。法則の階層
上のこの関係は、 Eq と Ord の間の上位クラス関係を説明します。



この場合に上位クラス関係を定義する別の考え⽅としては、この２つのクラスの間には明らか
に"is-a"の関係があることです。下位クラスのすべてのメンバは、上位クラスのメンバでもある
ということです。

演習

. (やや難しい) 整数の空でない配列の最⼤値を求める部分関数を定義します。あなたの関数の型は
Partial => Array Int - > Int でなければなりません。 unsafePartial を使ってPSCiであな

たの関数をテストしてください。 ヒント： Data.Foldable の maximum 関数を使います。

. (やや難しい) 次の Action クラスは、ある型の動作(action)を定義する、多変数型クラスで
す。

class Monoid m <= Action m a where 

  act :: m -> a -> a 

actはモノイドがどうやって他の型の値を変更するのに使われるのかを説明する関数です。こ
の動作が モノイドの連結演算⼦に従っていると期待しましょう。

act mempty a = a

act (m1 <> m2) a = act m1 (act m2 a)

この動作は Monoid クラスによって定義された操作を順守します。

たとえば、乗算を持つ⾃然数のモノイドを形成します。

newtype Multiply = Multiply Int 

 

instance semigroupMultiply :: Semigroup Multiply where 

  append (Multiply n) (Multiply m) = Multiply (n * m) 

 

instance monoidMultiply :: Monoid Multiply where 

  mempty = Multiply 1 

このモノイドは、⽂字列の何度かの繰り返しとして⽂字列に対して動作します。このアクショ
ンを実装するインスタンスを作成します。

instance repeatAction :: Action Multiply String 

このインスタンスが上記の法則を満たしているか確かめましょう。



. (やや難しい) インスタンス Action m a => Action m（Array a） を書いてみましょう。ここ
で、 配列上の動作は要素の順序で実⾏されるように定義されるものとします。

. (難しい) 以下のnewtypeが与えられたとき、 Action m (Self m) のインスタンスを書いてみ
ましょう。ここで、モノイド m は連結によって⾃⾝に作⽤するものとします。

newtype Self m = Self m 

1.（難しい）多変数型のクラス Action の引数は、いくつかの関数従属性によって関連づけら
れるべきですか。それはなぜでしょうか。

6 .12 ハッシュの型クラス

この最後の節では、章の残りを費やしてデータ構造をハッシュするライブラリを作ります。

このライブラリの⽬的は説明だけであり、堅牢なハッシングの仕組みの提供を⽬的としていな
いことに注意してください。

ハッシュ関数に期待される性質とはどのようなものでしょうか？

ハッシュ関数は決定的でなくてはなりません。つまり、同じ値には同じハッシュ値を対応させな
ければなりません
ハッシュ関数はいろいろなハッシュ値の集合で結果が⼀様に分布しなければなりません。

最初の性質はまさに型クラスの法則のように⾒える⼀⽅で、２番⽬の性質はもっとぼんやりと
した規約に従っていて、PureScriptの型システムによって確実に強制できるようなものではなさ
そうです。しかし、これは型クラスについて次のような直感的理解を与えるはずです。

newtype HashCode = HashCode Int 

 

hashCode :: Int -> HashCode 

hashCode h = HashCode (h `mod` 65535) 

 

class Eq a <= Hashable a where 

  hash :: a -> HashCode 

これに、 a == b ならば hash a == hash b という関係性の法則が付随しています。



この節の残りの部分を費やして、 Hashable 型クラスに関連付けられているインスタンスと
関数のライブラリを構築していきます。

決定的な⽅法でハッシュ値を結合する⽅法が必要になります。

combineHashes :: HashCode -> HashCode -> HashCode 

combineHashes (HashCode h1) (HashCode h2) = hashCode (73 * h1 + 51 * h2) 

combineHashes 関数は、２つのハッシュ値を混ぜて結果を0-65535の間に分布します。

それでは、⼊⼒の種類を制限する Hashable 制約を使う関数を書いてみましょう。ハッシュ
関数を必要とするよくある⽬的としては、2つの値が同じハッシュ値にハッシュされるかどうか
を決定することです。 hashEqual 関係はそのような機能を提供します。

hashEqual :: forall a. Hashable a => a -> a -> Boolean 

hashEqual = eq `on` hash 

この関数はハッシュ同値性を定義するために Data.Function の on 関数を使っていますが、
このハッシュ同値性の定義は『それぞれの値が hash 関数に渡されたあとで２つの値が等しいな
ら、それらの値は「ハッシュ同値」である』というように宣⾔的に読めるはずです。

プリミティブ型の Hashable インスタンスをいくつか書いてみましょう。まずは整数のイン
スタンスです。 HashCode は実際には単なるラップされた整数なので、これは簡単で
す。 hashCode ヘルパー関数を使うことができます。

instance hashInt :: Hashable Int where 

  hash = hashCode 

パターン照合を使うと、 Boolean 値の単純なインスタンスを定義することもできます。

instance hashBoolean :: Hashable Boolean where 

  hash false = hashCode 0 

  hash true  = hashCode 1 

整数のインスタンスでは、 Data.Char の toCharCode 関数を使うと Char をハッシュするイ
ンスタンスを作成できます。

instance hashChar :: Hashable Char where 

  hash = hash <<< toCharCode 



（要素型が Hashable のインスタンスでもあるならば）配列の要素に hash 関数を map して
から、 combineHashes 関数の結果を使ってハッシュを左側に畳み込むことで、配列のインスタ
ンスを定義します。

instance hashArray :: Hashable a => Hashable (Array a) where 

  hash = foldl combineHashes (hashCode 0) <<< map hash 

すでに書いたより単純なインスタンスを使⽤して新たなインスタンスを構築する⽅法に注⽬し
てください。 String を Char の配列に変換し、この新たな Array インスタンスを使っ
て String インスタンスを定義しましょう。

instance hashString :: Hashable String where 

  hash = hash <<< toCharArray 

これらの Hashable インスタンスが先ほどの型クラスの法則を満たしていることを証明する
にはどうしたらいいでしょうか。同じ値が等しいハッシュ値を持っていることを確認する必要が
あります。 Int 、 Char 、 String 、 Boolean の場合は、 Eq の意味では同じ値でも厳密には
同じではない、というような型の値は存在しないので簡単です。

もっと⾯⽩い型についてはどうでしょうか。この場合、配列の⻑さに関する帰納を使うと、型
クラスの法則を証明することができます。⻑さゼロの唯⼀の配列は [] です。配列の Eq の定義
により、任意の⼆つの空でない配列は、それらの先頭の要素が同じで配列の残りの部分が等しい
とき、その時に限り等しくなります。この帰納的な仮定により、配列の残りの部分は同じハッシ
ュ値を持ちますし、もし Hashable a インスタンスがこの法則を満たすなら、先頭の要素も同
じハッシュ値をもつことがわかります。したがって、２つの配列は同じハッシュ値を持ち、
Hashable（Array a） も同様に型クラス法則を満たしています。

この章のソースコードには、 Maybe と Tuple 型のインスタンスなど、他にも Hashable イン
スタンスの例が含まれています。

演習

. (簡単) PSCi を使って、各インスタンスのハッシュ関数をテストしてください。

. (やや難しい) 同値性の近似として hashEqual 関数のハッシュ同値性を使い、配列が重複する
要素を持っているかどうかを調べる関数を書いてください。ハッシュ値が⼀致したペアが⾒つか
った場合は、 == を使って値の同値性を厳密に検証することを忘れないようにしてください。
ヒント： Data.Array の nubBy 関数を使⽤してみてください。



. (やや難しい) 型クラスの法則を満たす、次のnewtypeの Hashable インスタンスを書いてくだ
さい。

newtype Hour = Hour Int 

 

instance eqHour :: Eq Hour where 

  eq (Hour n) (Hour m) = mod n 12 == mod m 12 

newtypeの Hour とその Eq インスタンスは、12進数である型を表します。したがって、1と
13は等しいと⾒なされます。そのインスタンスが型クラスの法則を満たしていることを証明して
ください。

. (難しい) Maybe 、 Either 、 Tuple の Hashable インスタンスが型クラスの法則を満たして
いることを証明してください。

まとめ

この章では、型に基づく抽象化で、コードの再利⽤のための強⼒な形式化を可能にする型クラ
スを導⼊しました。PureScriptの標準ライブラリから標準の型クラスを幾つか⾒てきました。ま
た、ハッシュ値を計算する型クラスに基づく独⾃のライブラリを定義しました。

この章では型クラス法則の考え⽅を導⼊するとともに、抽象化のための型クラスを使うコード
について、その性質を証明する⼿法を導⼊しました。型クラス法則は等式推論(equational
reasoning)と呼ばれる⼤きな分野の⼀部であり、プログラミング⾔語の性質と型システムはプロ
グラムについて論理的な推論をできるようにするために使われています。これは重要な考え⽅
で、本書では今後あらゆる箇所で⽴ち返る話題となるでしょう。



第7章 Appl icat iveによる検証

7 .1  この章の⽬標

この章では、 Applicative 型クラスによって表現されるApplicative関⼿(applicative functor)
という重要な抽象化と新たに出会うことになります。名前が難しそうに思えても⼼配しないでく
ださい。フォームデータの検証という実⽤的な例を使ってこの概念を説明していきます。
Applicative関⼿を使うと、⼤量の決まり⽂句を伴うような⼊⼒項⽬の内容を検証するためのコー
ドを、簡潔で宣⾔的な記述へと変えることができるようになります。

また、Traversable関⼿(traversable functor)を表現する Traversable という別の型クラスに
も出会います。現実の問題への解決策からこの概念が⾃然に⽣じるということがわかるでしょ
う。

この章では第3章に引き続き住所録を例として扱います。今回は住所録のデータ型を拡張し、
これらの型の値を検証する関数を書きます。これらの関数は、例えばデータ⼊⼒フォームの⼀部
で、使⽤者へエラーを表⽰するウェブユーザインタフェースで使われると考えてください。

7 .2  プロジェクトの準備

この章のソース·コードは、次のふたつのファイルで定義されています。

src/Data/AddressBook.purs

src/Data/AddressBook/Validation.purs

このプロジェクトは多くのBower依存関係を持っていますが、その⼤半はすでに⾒てきたもの
です。新しい依存関係は２つです。

purescript-control  - Applicative のような型クラスを使⽤して制御フローを抽象化する関
数が定義されています
purescript-validation  - この章の主題である Applicative による検証 のための関⼿が定義

されています。

Data.AddressBook モジュールには、このプロジェクトのデータ型とそれらの型に対す
る Show インスタンスが定義されており、 Data.AddressBook.Validation モジュールにはそれ
らの型の検証規則含まれています。



7 .3  関数適⽤の⼀般化

Applicative関⼿の概念を理解するために、まずは以前扱った型構築⼦ Maybe について考えて
みましょう。

このモジュールのソースコードでは、次のような型を持つ address 関数が定義されていま
す。

address :: String -> String -> String -> Address 

この関数は、通りの名前、市、州という３つの⽂字列から型 Address の値を構築するために
使います。

この関数は簡単に適⽤できますので、 PSCi でどうなるか⾒てみましょう。

> import Data.AddressBook 

 

> address "123 Fake St." "Faketown" "CA" 

Address { street: "123 Fake St.", city: "Faketown", state: "CA" } 

しかし、通り、市、州の三つすべてが必ずしも⼊⼒されないものとすると、三つの場合がそれ
ぞれ省略可能であることを⽰すために Maybe 型を使⽤したくなります。

考えられる場合としては、市が省略されている場合があるでしょう。もし address 関数を直
接適⽤しようとすると、型検証器からエラーが表⽰されます。

> import Data.Maybe 

> address (Just "123 Fake St.") Nothing (Just "CA") 

 

Could not match type Maybe String with type String 

address は Maybe String 型ではなく⽂字列型の引数を取るので、もちろんこれは型エラー
になります。

しかし、もし address 関数を「持ち上げる」ことができれば、 Maybe 型で⽰される省略可能
な値を扱うことができるはずだと期待することは理にかなっています。実際
に、 Control.Apply で提供されている関数 lift3 が、まさに求めているものです。

> import Control.Apply 

> lift3 address (Just "123 Fake St.") Nothing (Just "CA") 



 

Nothing 

このとき、引数のひとつ(市)が⽋落していたので、結果は Nothing になります。もし3つの引
数すべてが Just 構築⼦を使って与えられれば、結果は値を含むことになります。

> lift3 address (Just "123 Fake St.") (Just "Faketown") (Just "CA") 

   

Just (Address { street: "123 Fake St.", city: "Faketown", state: "CA" }) 

lift3 という関数の名前は、3引数の関数を持ち上げるために使⽤できることを⽰していま
す。関数を持ち上げる同様の関数で、引数の数が異なるものが、 Control.Apply で定義されて
います。

7 .4  任意個の引数を持つ関数の持ち上げ

これで、 lift2 や lift3 のような関数を使えば、引数が2個や3個の関数を持ち上げることが
できるのはわかりました。でも、これを任意個の引数の関数へと⼀般化することはできるのでし
ょうか。

lift3 の型を⾒てみるとわかりやすいでしょう。

> :type lift3 

forall a b c d f. Apply f => (a -> b -> c -> d) -> f a -> f b -> f c -> f d 

上の Maybe の例では型構築⼦ f は Maybe ですから、 lift3 は次のように特殊化されます。

forall a b c d. (a -> b -> c -> d) -> Maybe a -> Maybe b -> Maybe c -> Maybe d 

この型が⾔っているのは、3引数の任意の関数を取り、その関数を引数と返り値が Maybe で包
まれた新しい関数へと持ち上げる、ということです。

もちろんどんな型構築⼦ f についても持ち上げができるわけではないのですが、それで
は Maybe 型を持ち上げができるようにしているものは何なのでしょうか。さて、先ほどの型の
特殊化では、 f に対する型クラス制約から Apply 型クラスを取り除いていました。 Apply は
Preludeで次のように定義されています。



class Functor f where 

  map :: forall a b. (a -> b) -> f a -> f b 

class Functor f <= Apply f where 

  apply :: forall a b. f (a -> b) -> f a -> f b 

Apply 型クラスは Functor の下位クラスであり、追加の関数 apply が定義していま
す。 Prelude モジュールでは <$> を、 map の別名として、 <*> を apply の別名として定義して
います。 map とよく似た型を持つ追加の関数 apply が定義されています。 map と apply の違い
は、 map がただの関数を引数に取るのに対し、 apply の最初の引数は型構築⼦ f で包まれてい
るという点です。これをどのように使うのかはこれからすぐに⾒ていきますが、その前にま
ず Maybe 型について Apply 型クラスをどう実装するのかを⾒ていきましょう。

instance functorMaybe :: Functor Maybe where 

  map f (Just a) = Just (f a) 

  map f Nothing  = Nothing 

 

instance applyMaybe :: Apply Maybe where 

  apply (Just f) (Just x) = Just (f x) 

  apply _        _        = Nothing 

この型クラスのインスタンスが⾔っているのは、任意のオプショナルな値にオプショナルな関
数を適⽤することができ、その両⽅が定義されている時に限り結果も定義される、ということで
す。

それでは、 map と apply を⼀緒に使ってどうやって引数が任意個の関数を持ち上げるのかを
⾒ていきましょう。

１引数の関数については、 map をそのまま使うだけです。

２引数の関数についても考えてみます。型 a -> b -> c を持つカリー化された関数 f がある
としましょう。これは型 a -> (b -> c) と同じですから、 map を f に適⽤すると
型 f a -> f (b -> c) の新たな関数を得ることになります。持ち上げられた(型 f a の)最初の
引数にその関数を部分適⽤すると、型 f (b -> c) の新たな包まれた関数が得られます。それか
ら、２番⽬の持ち上げられた(型 f b の)引数へ apply を適⽤することができ、型 f c の最終的
な値を得ます。

まとめると、 x :: f a と y :: f b があるとき、式 (f <$> x) <*> y の型は f c になりま
す(この式は apply (map f x) y と同じ意味だということを思い出しましょう)。Preludeで定義
された優先順位の規則に従うと、 f <$> x <*> y というように括弧を外すことができます。



⼀般的にいえば、最初の引数に <$> を使い、残りの引数に対しては <*> を使います。 lift3

で説明すると次のようになります。

lift3 :: forall a b c d f 

       . Apply f 

      => (a -> b -> c -> d) 

      -> f a 

      -> f b 

      -> f c 

      -> f d 

lift3 f x y z = f <$> x <*> y <*> z  

この式の型がちゃんと整合しているかの確認は、読者への演習として残しておきます。

例として、 <$> と <*> をそのまま使うと、 Maybe 上に address 関数を持ち上げることができ
ます。

> address <$> Just "123 Fake St." <*> Just "Faketown" <*> Just "CA" 

Just (Address { street: "123 Fake St.", city: "Faketown", state: "CA" }) 

 

> address <$> Just "123 Fake St." <*> Nothing <*> Just "CA" 

Nothing 

このように、引数が異なる他のいろいろな関数を Maybe 上に持ち上げてみてください。

7 .5  Appl ica t ive型クラス

これに関連する Applicative という型クラスが存在しており、次のように定義されていま
す。

class Apply f <= Applicative f where 

  pure :: forall a. a -> f a 

Applicative は Apply の下位クラスであり、 pure 関数が定義されています。 pure は値を取
り、その型の型構築⼦ f で包まれた値を返します。

Maybe についての Applicative インスタンスは次のようになります。

instance applicativeMaybe :: Applicative Maybe where 

  pure x = Just x 



Applicative関⼿は関数を持ち上げることを可能にする関⼿だと考えるとすると、 pure は引数
のない関数の持ち上げだというように考えることができます。

7 .6  Appl ica t iveに対する直感的理解

PureScriptの関数は純粋であり、副作⽤は持っていません。Applicative関⼿は、関⼿ f によっ
て表現されたある種の副作⽤を提供するような、より⼤きな「プログラミング⾔語」を扱えるよ
うにします。

たとえば、関⼿ Maybe はオプショナルな値の副作⽤を表現しています。その他の例として
は、型 err のエラーの可能性の副作⽤を表す Either err や、⼤域的な構成を読み取る副作⽤を
表すArrow関⼿(arrow functor) r -> があります。ここでは Maybe 関⼿についてだけを考えるこ
とにします。

もし関⼿ f が作⽤を持つより⼤きなプログラミング⾔語を表すとすると、 Apply

と Applicative インスタンスは⼩さなプログラミング⾔語(PureScript)から新しい⼤きな⾔語へ
と値や関数を持ち上げることを可能にします。

pure は純粋な(副作⽤がない)値をより⼤きな⾔語へと持ち上げますし、関数については上で
述べたとおり map と apply を使うことができます。

ここで新たな疑問が⽣まれます。もしPureScriptの関数と値を新たな⾔語へ埋め込むの
に Applicative が使えるなら、どうやって新たな⾔語は⼤きくなっているというのでしょう
か。この答えは関⼿ f に依存します。もしなんらかの x について pure x で表せないような
型 f a の式を⾒つけたなら、その式はそのより⼤きな⾔語だけに存在する項を表しているとい
うことです。

f が Maybe のときの式 Nothing がその例になっています。 Nothing を何らかの x につい
て pure x というように書くことはできません。したがって、PureScriptは省略可能な値を表す
新しい項 Nothing を含むように拡⼤されたと考えることができます。

7 .7  その他の作⽤について

それでは、他にも Applicative 関⼿へと関数を持ち上げる例をいろいろ⾒ていきましょう。



次は、 PSCi で定義された3つの名前を結合して完全な名前を作る簡単なコード例です。

> import Prelude 

> fullName first middle last = last <> ", " <> first <> " " <> middle 

> fullName "Phillip" "A" "Freeman" 

Freeman, Phillip A 

この関数は、クエリパラメータとして与えられた3つの引数を持つ、(とても簡単な!)ウェブサ
ービスの実装であるとしましょう。使⽤者が3つの引数すべてを与えたことを確かめたいので、
引数が存在するかどうかを表す Maybe 型をつかうことになるでしょう。 fullName を Maybe の
上へ持ち上げると、省略された引数を確認するウェブサービスを実装することができます。

> import Data.Maybe 

> fullName <$> Just "Phillip" <*> Just "A" <*> Just "Freeman" 

Just ("Freeman, Phillip A") 

> fullName <$> Just "Phillip" <*> Nothing <*> Just "Freeman" 

Nothing 

この持ち上げた関数は、引数のいずれかが Nothing なら Nothing 返すことに注意してくださ
い。

これで、もし引数が不正ならWebサービスからエラー応答を送信することができるので、なか
なかいい感じです。しかし、どのフィールドが間違っていたのかを応答で表⽰できると、もっと
良くなるでしょう。

Meybe 上へ持ち上げる代わりに Either String 上へ持ち上げるようにすると、エラーメッセ
ージを返すことができるようになります。まずは⼊⼒を Either String を使ってエラーを発信
できる計算に変換する演算⼦を書きましょう。

> :paste 

… withError Nothing err = Left err 

… withError (Just a) _  = Right a 

… ^D 

注意： Either err Applicative関⼿において、 Left 構築⼦は失敗を表しており、 Right 構築
⼦は成功を表しています。

これで Either String 上へ持ち上げることで、それぞれの引数について適切なエラーメッセ
ージを提供できるようになります。



> :paste 

… fullNameEither first middle last = 

…   fullName <$> (first  `withError` "First name was missing") 

…            <*> (middle `withError` "Middle name was missing") 

…            <*> (last   `withError` "Last name was missing") 

… ^D 

 

> :type fullNameEither 

Maybe String -> Maybe String -> Maybe String -> Either String String 

この関数は Maybe の3つの省略可能な引数を取り、 String のエラーメッセージか String の
結果のどちらかを返します。

いろいろな⼊⼒でこの関数を試してみましょう。

> fullNameEither (Just "Phillip") (Just "A") (Just "Freeman") 

(Right "Freeman, Phillip A") 

 

> fullNameEither (Just "Phillip") Nothing (Just "Freeman") 

(Left "Middle name was missing") 

 

> fullNameEither (Just "Phillip") (Just "A") Nothing 

(Left "Last name was missing") 

このとき、すべてのフィールドが与えられば成功の結果が表⽰され、そうでなければ省略され
たフィールドのうち最初のものに対応するエラーメッセージが表⽰されます。しかし、もし複数
の⼊⼒が省略されているとき、最初のエラーしか⾒ることができません。

> fullNameEither Nothing Nothing Nothing 

 

(Left "First name was missing") 

これでも⼗分なときもありますが、エラー時にすべての省略されたフィールドの⼀覧がほしい
ときは、 Either String よりも強⼒なものが必要です。この章の後半でこの解決策を⾒ていき
ます。

7 .8  作⽤の結合

抽象的にApplicative関⼿を扱う例として、Applicative関⼿ f によって表現された副作⽤を総
称的に組み合わせる関数をどのように書くのかをこの節では⽰します。



これはどういう意味でしょうか？何らかの a について型 f a の包まれた引数の配列があると
しましょう。型 List (f a) の配列があるということです。直感的には、これは f によって追跡
される副作⽤を持つ、返り値の型が a の計算の配列を表しています。これらの計算のすべてを
順番に実⾏することができれば、 List a 型の結果の配列を得るでしょう。しかし、まだ f によ
って追跡される副作⽤が残ります。つまり、元の配列の中の作⽤を「結合する」ことにより、
型 List (f a) の何かを型 List a の何かへと変換することができると考えられます。

任意の固定⻑配列の⻑さ n について、その引数を要素に持った⻑さ n の配列を構築するよう
な n 引数の関数が存在します。たとえば、もし n が 3 なら、関数
は \x y z -> x : y : z : Nil です。 この関数の型は a -> a -> a -> List a で
す。 Applicative インスタンスを使うと、この関数を f の上へ持ち上げて関数
型 f a -> f a -> f a -> f (List a) を得ることができます。しかし、いかなる n についても
これが可能なので、いかなる引数の配列についても同じように持ち上げられることが確かめられ
ます。

したがって、次のような関数を書くことができるはずです。

combineList :: forall f a. Applicative f => List (f a) -> f (List a) 

この関数は副作⽤を持つかもしれない引数の配列をとり、それぞれの副作⽤を適⽤すること
で、 f に包まれた単⼀の配列を返します。

この関数を書くためには、引数の配列の⻑さについて考えます。配列が空の場合はどんな作⽤
も実⾏する必要はありませんから、 pure を使⽤して単に空の配列を返すことができます。

combineList Nil = pure Nil 

実際のところ、これが可能な唯⼀の  定義です！

⼊⼒の配列が空でないならば、型 f a の先頭要素と、型 List (f a) の配列の残りについて考
えます。また、再帰的に配列の残りを結合すると、型 f (List a) の結果を得ることができま
す。 <$> と <*> を使うと、 cons 関数を先頭と配列の残りの上に持ち上げることができます。

combineList (Cons x xs) = Cons <$> x <*> combineList xs 

繰り返しになりますが、これは与えられた型に基づいている唯⼀の妥当な実装です。

Maybe 型構築⼦を例にとって、 PSCi でこの関数を試してみましょう。



> import Data.List 

> import Data.Maybe 

> combineList (fromFoldable [Just 1, Just 2, Just 3]) 

(Just (Cons 1 (Cons 2 (Cons 3 Nil)))) 

> combineList (fromFoldable [Just 1, Nothing, Just 2]) 

Nothing 

Meybe へ特殊化して考えると、配列のすべての要素が Just であるとき、そのときに限りこの
関数は Just を返します。そうでなければ、 Nothing を返します。オプショナルな結果を返す計
算の配列は、そのすべての計算が結果を持っていたときに全体も結果を持っているという、オプ
ショナルな値に対応したより⼤きな⾔語での振る舞いに対する直感的な理解とこれは⼀致してい
ます。

しかも、 combineArray 関数はどんな Applicative に対しても機能します！ Either err を使
ってエラーを発信するかもしれなかったり、 r -> を使って⼤域的な状態を読み取る計算を連鎖
させるときにも combineArray 関数を使うことができるのです。

combineArray 関数については、後ほど Traversable 関⼿について考えるときに再び扱いま
す。

演 習

1. (簡単) lift2 を使って、オプショナルな引数に対して働く、数に対する演算
⼦ + 、 - 、 * 、 / の持ち上げられたバージョンを書いてください。

2. (やや難しい) 上で与えられた lift3 の定義について、 <$> と <*> の型が整合し
ていることを確認して下さい。

3. (難しい) 次の型を持つ関数 combineMaybe を書いてください。

combineMaybe : forall a f. (Applicative f) => Maybe (f a) -> f (Maybe a) 

この関数は副作⽤をもつオプショナルな計算をとり、オプショナルな結果をもつ
副作⽤のある計算を返します。



7 .9  Appl ica t iveによる検証

この章のソースコードでは住所録アプリケーションで使われるいろいろなデータ型が定義され
ています。詳細はここでは割愛しますが、 Data.AddressBook モジュールからエクスポートされ
る重要な関数は次のような型を持っています。

address :: String -> String -> String -> Address 

 

phoneNumber :: PhoneType -> String -> PhoneNumber 

 

person :: String -> String -> Address -> Array PhoneNumber -> Person 

ここで、 PhoneType は次のような代数的データ型として定義されています。

data PhoneType = HomePhone | WorkPhone | CellPhone | OtherPhone 

これらの関数は住所録の項⽬を表す Person を構築するのに使います。例え
ば、 Data.AddressBook には次のような値が定義されています。

examplePerson :: Person 

examplePerson = 

  person "John" "Smith" 

         (address "123 Fake St." "FakeTown" "CA") 

                [ phoneNumber HomePhone "555-555-5555" 

                , phoneNumber CellPhone "555-555-0000" 

                ] 

PSCi でこれらの値使ってみましょう(結果は整形されています)。

> import Data.AddressBook 

> examplePerson 

Person 

  { firstName: "John", 

  , lastName: "Smith", 

  , address: Address 

    { street: "123 Fake St." 

    , city: "FakeTown" 

    , state: "CA" 

    }, 

  , phones: [ PhoneNumber 

              { type: HomePhone 

              , number: "555-555-5555" 

              } 



            , PhoneNumber 

              { type: CellPhone 

              , number: "555-555-0000" 

              } 

            ] 

  } 

前の章では型 Person のデータ構造を検証するのに Either String 関⼿をどのように使うか
を⾒ました。例えば、データ構造の２つの名前を検証する関数が与えられたとき、データ構造全
体を次のように検証することができます。

nonEmpty :: String -> Either String Unit 

nonEmpty "" = Left "Field cannot be empty" 

nonEmpty _ = Right unit 

 

validatePerson :: Person -> Either String Person 

validatePerson (Person o) = 

  person <$> (nonEmpty o.firstName *> pure o.firstName) 

         <*> (nonEmpty o.lastName *> pure o.lastName) 

         <*> pure o.address 

         <*> pure o.phones 

最初の２⾏では nonEmpty 関数を使って空⽂字列でないことを検証しています。もし⼊⼒が空
なら nonEMpty はエラーを返し( Left 構築⼦で⽰されています)、そうでなければ Right 構築⼦
を使って空の値( unit )を正常に返します。２つの検証を実⾏し、右辺の検証の結果を返すこと
を⽰す連鎖演算⼦ *> を使っています。ここで、⼊⼒を変更せずに返す検証器として右辺では単
に pure を使っています。

最後の２⾏では何の検証も実⾏せず、単に address フィールドと phones フィールドを残りの
引数として person 関数へと提供しています。

この関数は PSCi でうまく動作するように⾒えますが、以前⾒たような制限があります。

> validatePerson $ person "" "" (address "" "" "") [] 

(Left "Field cannot be empty") 

Either String Applicative関⼿は遭遇した最初のエラーだけを返します。でもこの⼊⼒で
は、名前の不⾜と姓の不⾜という２つのエラーがわかるようにしたくなるでしょう。

purescript-validation ライブラリは別のApplicative関⼿も提供されています。これは単
に V と呼ばれていて、何らかの半群(Semigroup)でエラーを返す機能があります。たとえ



ば、 V (Array String) を使うと、新しいエラーを配列の最後に連結していき、 String の配列
をエラーとして返すことができます。

Data.Validation モジュールは Data.AddressBook モジュールのデータ構造を検証するため
に V (Array String) Applicative関⼿を使っています。

Data.AddressBook.Validation モジュールにある検証の例としては次のようになります。

type Errors = Array String 

 

nonEmpty :: String -> String -> V Errors Unit 

nonEmpty field "" = invalid ["Field '" <> field <> "' cannot be empty"] 

nonEmpty _     _  = pure unit 

 

lengthIs :: String -> Number -> String -> V Errors Unit 

lengthIs field len value | S.length value /= len = 

  invalid ["Field '" <> field <> "' must have length " <> show len] 

lengthIs _ _ _ = 

  pure unit 

 

validateAddress :: Address -> V Errors Address 

validateAddress (Address o) = 

  address <$> (nonEmpty "Street" o.street  *> pure o.street) 

          <*> (nonEmpty "City"   o.city    *> pure o.city) 

          <*> (lengthIs "State"  2 o.state *> pure o.state) 

validateAddress は Address を検証します。 street と city が空でないかどうか、 state

の⽂字列の⻑さが2であるかどうかを検証します。

nonEmpty と lengthIs の２つの検証関数はいずれも、 Data.Validation モジュールで提供さ
れている invalid 関数をエラーを⽰すために使っていることに注⽬してくださ
い。 Array String 半群を扱っているので、 invalid は引数として⽂字列の配列を取ります。

PSCi でこの関数を使ってみましょう。

> import Data.AddressBook 

> import Data.AddressBook.Validation 

 

> validateAddress $ address "" "" "" 

(Invalid [ "Field 'Street' cannot be empty" 

         , "Field 'City' cannot be empty" 

         , "Field 'State' must have length 2" 

         ]) 

 

> validateAddress $ address "" "" "CA" 

(Invalid [ "Field 'Street' cannot be empty" 



         , "Field 'City' cannot be empty" 

         ]) 

これで、すべての検証エラーの配列を受け取ることができるようになりました。

7 .10 正規表現検証器

validatePhoneNumber 関数では引数の形式を検証するために正規表現を使っています。重要
なのは matches 検証関数で、この関数は Data.String.Regex モジュールのて定義されてい
る Regex を使って⼊⼒を検証しています。

matches :: String -> R.Regex -> String -> V Errors Unit 

matches _ regex value | R.test regex value = 

  pure unit 

matches field _ _ = 

  invalid ["Field '" <> field <> "' did not match the required format"] 

繰り返しになりますが、 pure は常に成功する検証を表しており、エラーの配列の伝達に
は invalid が使われています。

これまでと同じような感じで、 validatePhoneNumber は matches 関数から構築されていま
す。

validatePhoneNumber :: PhoneNumber -> V Errors PhoneNumber 

validatePhoneNumber (PhoneNumber o) = 

  phoneNumber <$> pure o."type" 

              <*> (matches "Number" phoneNumberRegex o.number *> pure o.number) 

また、 PSCi でいろいろな有効な⼊⼒や無効な⼊⼒に対して、この検証器を実⾏してみてくだ
さい。

> validatePhoneNumber $ phoneNumber HomePhone "555-555-5555" 

Valid (PhoneNumber { type: HomePhone, number: "555-555-5555" }) 

 

> validatePhoneNumber $ phoneNumber HomePhone "555.555.5555" 

Invalid (["Field 'Number' did not match the required format"]) 



演 習

1. (簡単) 正規表現の検証器を使って、 Address 型の state フィールドが2⽂字の
アルファベットであることを確かめてください。ヒント： phoneNumberRegex のソ
ースコードを参照してみましょう。

2. (やや難しい) matches 検証器を使って、⽂字列に全く空⽩が含まれないことを検
証する検証関数を  書いてください。この関数を使って、適切な場合に nonEmpty を
置き換えてください。

7 .11 Traversab le関⼿

残った検証器は、これまで⾒てきた検証器を組み合わせて Person 全体を検証す
る validatePerson です。

arrayNonEmpty :: forall a. String -> Array a -> V Errors Unit 

arrayNonEmpty field [] = 

  invalid ["Field '" <> field <> "' must contain at least one value"] 

arrayNonEmpty _ _ = 

  pure unit 

 

validatePerson :: Person -> V Errors Person 

validatePerson (Person o) = 

  person <$> (nonEmpty "First Name" o.firstName *> 

              pure o.firstName) 

         <*> (nonEmpty "Last Name" o.lastName *> 

              pure o.lastName) 

               <*> validateAddress o.address 

         <*> (arrayNonEmpty "Phone Numbers" o.phones *> 

              traverse validatePhoneNumber o.phones) 

ここに今まで⾒たことのない興味深い関数がひとつあります。最後の⾏で使われてい
る traverse です。

traverse は Data.Traversable モジュールの Traversable 型クラスで定義されています。

class (Functor t, Foldable t) <= Traversable t where 

  traverse :: forall a b f. Applicative f => (a -> f b) -> t a -> f (t b) 

  sequence :: forall a f. Applicative f => t (f a) -> f (t a) 



Traversable はTraversable関⼿の型クラスを定義します。これらの関数の型は少し難しそう
に⾒えるかもしれませんが、 validatePerson は良いきっかけとなる例です。

すべてのTraversable関⼿は Functor と Foldable のどちらでもあります(Foldable 関⼿は構造
をひとつの値へとまとめる、畳み込み操作を提供する型構築⼦であったことを思い出してくださ
い)。それ加えて、 Traversable 関⼿はその構造に依存した副作⽤のあつまりを連結する機能を
提供します。

複雑そうに聞こえるかもしれませんが、配列の場合に特殊化して簡単に考えてみましょう。配
列型構築⼦は Traversable である、つまり次のような関数が存在するということです。

traverse :: forall a b f. Applicative f => (a -> f b) -> Array a -> f (Array b) 

直感的には、Applicative関⼿ f と、型 a の値をとり型 b の値を返す( f で追跡される副作⽤
を持つ)関数が与えられたとき、型 [a] の配列の要素それぞれにこの関数を適⽤し、型 [b] の
( f で追跡される副作⽤を持つ)結果を得ることができます。

まだよくわからないでしょうか。それでは、更に f を V Errors Applicative関⼿に特殊化して
考えてみましょう。 traversable が次のような型の関数だとしましょう。

traverse :: forall a b. (a -> V Errors b) -> Array a -> V Errors (Array b) 

この型シグネチャは、型 a についての検証関数 f があれば、 traverse f は型 Array a の配
列についての検証関数であるということを⾔っています。これはまさに今必要になってい
る Person データ構造体の phones フィールドを検証する検証器そのものです！それぞれの要素
が成功するかどうかを検証する検証関数を作るために、 validatePhoneNumber を traverse へ
渡しています。

⼀般に、 traverse はデータ構造の要素をひとつづつ辿っていき、副作⽤のある計算を実⾏し
て結果を累積します。

Traversable のもう⼀つの関数、 sequence の型シグネチャには⾒覚えがあるかもしれませ
ん。

sequence :: forall a f. (Applicative m) => t (f a) -> f (t a) 



実際、先ほど書いた combineArray 関数は Traversable 型の sequence 関数が特殊化されたも
のに過ぎません。 t を配列型構築⼦として、 combineArray 関数の型をもう⼀度考えてみましょ
う。

combineList :: forall f a. Applicative f => List (f a) -> f (List a) 

Traversable 関⼿は、作⽤のある計算の集合を集めてその作⽤を連鎖させるという、データ
構造⾛査の考え⽅を把握できるようにするものです。実際、 sequence と traversable

は Traversable を定義するのにどちらも同じくらい重要です。これらはお互いが互いを利⽤し
て実装することができます。これについては興味ある読者への演習として残しておきます。

配列の Traversable インスタンスは Data.Traversable モジュールで与えられていま
す。 traverse の定義は次のようになっています。

-- traverse :: forall a b f. Applicative f => (a -> f b) -> List a -> f (List b) 

traverse _ Nil = pure Nil 

traverse f (Cons x xs) = Cons <$> f x <*> traverse f xs 

⼊⼒が空の配列のときには、単に pure を使って空の配列を返すことができます。配列が空で
ないときは、関数 f を使うと先頭の要素から型 f b の計算を作成することができます。また、
配列の残りに対して traverse を再帰的に呼び出すことができます。最後に、Applicative関⼿ f

までcons演算⼦ (:) を持ち上げて、２つの結果を組み合わせます。

Traversable関⼿の例はただの配列以外にもあります。以前に⾒た Maybe 型構築⼦
も Traversable のインスタンスを持っています。 PSCi で試してみましょう。

> import Data.Maybe 

> import Data.Traversable 

 

> traverse (nonEmpty "Example") Nothing 

(Valid Nothing) 

 

> traverse (nonEmpty "Example") (Just "") 

(Invalid ["Field 'Example' cannot be empty"]) 

 

> traverse (nonEmpty "Example") (Just "Testing") 

(Valid (Just unit)) 

これらの例では、 Nothing の値の⾛査は検証なしで Nothing の値を返し、 Just x を⾛査す
ると x を検証するのにこの検証関数が使われるということを⽰しています。つまり、 traverse

は型 a についての検証関数をとり、 Maybe a についての検証関数を返すのです。



他にも、何らかの型 a についての Tuple a や Either a や、連結リストの型構築⼦ List とい
ったTraversable関⼿があります。⼀般的に、「コンテナ」のようなデータの型構築⼦は⼤
抵 Traversable インスタンスを持っています。例として、演習では⼆分⽊の型の Traversable

インスタンスを書くようになっています。

演 習

1. (やや難しい) 左から右へと副作⽤を連鎖させる、次のような⼆分⽊データ構造に
ついての Traversable インスタンスを書いてください。

data Tree a = Leaf | Branch (Tree a) a (Tree a) 

これは⽊の⾛査の順序に対応しています。⾏きがけ順の⾛査についてはどうでし
ょうか。帰りがけ順では？

2. (やや難しい) Data.Maybe を使って Person の address フィールドを省略可能に
なるようにコードを変更してください。ヒント： traverse を使って型 Maybe a の
フィールドを検証してみましょう。

3. (難しい) traverse を使って sequence を書いてみましょう。また、 sequence

を使って traverse を書けるでしょうか？

7 .12 Appl ica t ive関⼿による並列処理

これまでの議論では、Applicative関⼿がどのように「副作⽤を結合」させるかを説明するとき
に、「結合」(combine)という単語を選びました。しかしながら、これらのすべての例におい
て、Applicative関⼿は作⽤を「連鎖」(sequence)させる、というように⾔っても同じく妥当で
す。 Traverse 関⼿はデータ構造に従って作⽤を順番に結合させる sequence 関数を提供する、
という直感的理解とこれは⼀致するでしょう。

しかし⼀般には、Applicative関⼿はこれよりももっと⼀般的です。Applicative関⼿の規則は、
その計算を実⾏する副作⽤にどんな順序付けも強制しません。実際、並列に副作⽤を実⾏するた



めのApplicative関⼿というものは妥当になりえます。

たとえば、 V 検証関⼿はエラーの配列を返しますが、その代わりに Set 半群を選んだとして
もやはり正常に動き、このときどんな順序でそれぞれの検証器を実⾏しても問題はありません。
データ構造に対して並列にこれを実⾏することさえできるのです！

別の例とし、 purescript-parallel パッケージは、並列計算をサポートする Parallel 型ク
ラスを与えます。⾮同期計算を表現する型構築⼦ parallel は、並列に結果を計算す
る Applicative インスタンスを持つことができます。

f <$> parallel computation1 

  <*> parallel computation2 

この計算は、 computation1 と computation2 を⾮同期に使って値を計算を始めるでしょう。
そして両⽅の結果の計算が終わった時に、関数 f を使ってひとつの結果へと結合するでしょ
う。

この考え⽅の詳細は、本書の後半でコールバック地獄の問題に対してApplicative関⼿を応⽤す
るときに⾒ていきます。

Applicative関⼿は並列に結合されうる副作⽤を捕捉する⾃然な⽅法です。

まとめ

この章では新しい考え⽅をたくさん扱いました。

関数適⽤の概念を副作⽤の考え⽅を表現する型構築⼦へと⼀般化する、Applicative関⼿の概念
を導⼊しました。
データ構造の検証という課題にApplicative関⼿がどのような解決策を与えるか、単⼀のエラーの
報告からデータ構造を横断するすべてのエラーの報告へ変換できるApplicative関⼿を⾒てきまし
た。
要素が副作⽤を持つ値の結合に使われることのできるコンテナであるTraversable関⼿の考え⽅
を表現する、 Traversable 型クラス導⼊しました。

Applicative関⼿は多くの問題に対して優れた解決策を与える興味深い抽象化です。本書を通じ
て何度も⾒ることになるでしょう。今回は、どうやって検証を⾏うかではなく、何を検証器が検
証すべきなのかを定義することを可能にする、宣⾔的なスタイルで書く⼿段をApplicative関⼿は
提供しました。⼀般に、Applicative関⼿は領域特化⾔語の設計のための便利な道具になります。



次の章では、これに関連するモナドという型クラスについて⾒ていきましょう。



第8章 Effモナド

8 .1  この章の⽬標

第7章では、オプショナルな型やエラーメッセージ、データの検証など、副作⽤を扱いを抽象
化するApplicative関⼿を導⼊しました。この章では、より表現⼒の⾼い⽅法で副作⽤を扱うため
の別の抽象化、モナドを導⼊します。

この章の⽬的は、なぜモナドが便利な抽象化なのか、do記法とどう関係するのかについて説
明することです。ブラウザでユーザインターフェイスを構築する副作⽤を扱うためのある種のモ
ナドを使って、前の章の住所録の例を作ることにしましょう。これから扱うEffモナドは、
PureScriptにおけるとても重要なモナドです。Effモナドはいわゆるネイティブな作⽤をカプセル
化するのに使われます。

8 .2  プロジェクトの準備

このプロジェクトのソースコードは前の章のソースコードの上に構築しますが、以前のプロジ
ェクトのモジュールは、このプロジェクトの src ディレクトリに含まれています。

このプロジェクトでは、以下のBowerの依存関係が追加されています。

purescript-eff  - Effモナドを提供します。
purescript-react  - Reactユーザインターフェイスへ接続するライブラリを提供します。

前章のモジュールに加えて、この章では Main モジュールを使⽤します。このモジュールはエ
ントリポイントであるとともに、UIの描写も⾏います。

このプロジェクトをコンパイルするには、まずReactをインストールするため npm install を
実⾏し、それから pulp browserify --to dist/Main.js でビルドを⾏います。このプロジェク
トを実⾏するには、 html/index.html ファイルをウェブブラウザで開いてください。

8 .3  モナドとdo記法



do記法は配列内包表記を扱うときに最初に導⼊されました。配列内包表記は Data.Array モ
ジュールの concatMap 関数の構⽂糖として提供されています。

次の例を考えてみましょう。２つのサイコロを振って出た⽬を数え、出た⽬の合計が n のと
きそれを得点とすることを考えます。次のような⾮決定的なアルゴリズムを使うとこれを実現す
ることができます。

最初の投擲で値 x を選択します。
２回⽬の投擲で値 y を選択します。
もし x と y の和が n なら組 {x, y} を返し、そうでなければ失敗します。

配列内包表記を使うと、この⾮決定的アルゴリズムを⾃然に書くことができます。

import Prelude 

 

import Control.Plus (empty) 

import Data.Array ((..)) 

 

countThrows :: Int -> Array (Array Int) 

countThrows n = do 

  x <- 1 .. 6 

  y <- 1 .. 6 

  if x + y == n 

    then pure [x, y] 

    else empty 

PSCi で動作を⾒てみましょう。

> countThrows 10 

[[4,6],[5,5],[6,4]] 

 

> countThrows 12   

[[6,6]] 

前の章では、オプショナルな値に対応したより⼤きなプログラミング⾔語へとPureScriptの関
数を埋め込む、 Maybe  Applicative関⼿についての直感的理解を養いました。同様に配列モナド
についても、⾮決定選択に対応したより⼤きなプログラミング⾔語へPureScriptの関数を埋め込
む、というような直感的理解を得ることができます。

⼀般に、ある型構築⼦ m のモナドは、型 m a の値を持つdo記法を使う⽅法を提供します。上
の配列内包表記では、すべての⾏に何らかの型 a についての型 Array a の計算が含まれている
ことに注⽬してください。⼀般に、do記法ブロックのすべての⾏は、何らかの型 a とモナド m
について、型 m a の計算を含んでいます。モナド m はすべての⾏で同じでなければなりません



(つまり、副作⽤の種類は固定されます)が、型 a は異なることもあります(⾔い換えると、ここ
の計算は異なる型の結果を持つことができます)。

型構築⼦ Maybe が適⽤された、do記法の別の例を⾒てみましょう。XMLノードを表す型
XML と演算⼦があるとします。

child :: XML -> String -> Maybe XML 

この演算⼦はノードの⼦の要素を探し、もしそのような要素が存在しなければ Nothing を返
します。

この場合、do記法を使うと深い⼊れ⼦になった要素を検索することができます。XML⽂書と
して符号化された利⽤者情報から、利⽤者の住んでいる市町村を読み取りたいとします。

userCity :: XML -> Maybe XML 

userCity root = do 

  prof <- child root "profile" 

  addr <- child prof "address" 

  city <- child addr "city" 

  pure city 

userCity 関数は⼦の要素である profile を探し、 profile 要素の中にある address 要
素、最後に address 要素から city 要素を探します。これらの要素のいずれかが⽋落している
場合は、返り値は Nothing になります。そうでなければ、返り値は city ノードから Just を
使って構築されています。

最後の⾏にある pure 関数は、すべての Applicative 関⼿について定義されているのでし
た。 Maybe の Applicative 関⼿の pure 関数は Just として定義されており、最後の⾏を
Just city へ変更しても同じように正しく動きます。

8 .4  モナド型クラス

Monad 型クラスは次のように定義されています。

class Apply m <= Bind m where 

  bind :: forall a b. m a -> (a -> m b) -> m b 

class (Applicative m, Bind m) <= Monad m 



ここで鍵となる関数は Bind 型クラスで定義されている演算⼦ bind で、 Functor 及び
Apply 型クラスにある <$> や <*> などの演算⼦と同じ様に Prelude では >>= として bind

の別名が定義されています。

Monad 型クラスは、すでに⾒てきた Applicative 型クラスの操作で Bind を拡張します。

Bind 型クラスの例をいくつか⾒てみるのがわかりやすいでしょう。配列についての Bind の
妥当な定義は次のようになります。

instance bindArray :: Bind Array where 

  bind xs f = concatMap f xs 

これは以前にほのめかした配列内包表記と concatMap 関数の関係を説明しています。

Maybe 型構築⼦についての Bind の実装は次のようになります。

instance bindMaybe :: Bind Maybe where 

  bind Nothing  _ = Nothing 

  bind (Just a) f = f a 

この定義はdo記法ブロックを通じて伝播された⽋落した値についての直感的理解を補強する
ものです。

Bind 型クラスとdo記法がどのように関係しているかを⾒て⾏きましょう。最初に何らかの計
算結果から値を束縛するような、簡単などdo記法ブロックについて考えてみましょう。

do value <- someComputation 

   whatToDoNext 

PureScriptコンパイラはこのようなパターンを⾒つけるたびにコードを次にように置き換えま
す。

bind someComputation \value -> whatToDoNext 

下記のように表記することもできます。

someComputation >>= \value -> whatToDoNext 

この計算 whatToDoNext は value に依存することができます。



連続した複数の束縛がある場合でも、この規則が先頭のほうから複数回適⽤されます。例え
ば、先ほど⾒た userCity の例では次のように構⽂糖が脱糖されます。

userCity :: XML -> Maybe XML 

userCity root = 

  child root "profile" >>= \prof -> 

    child prof "address" >>= \addr -> 

      child addr "city" >>= \city -> 

        pure city 

do記法を使って表現されたコードは、 >>= 演算⼦を使って書かれた同じ意味のコードよりし
ばしば読みやすくなることも特筆すべき点です。⼀⽅で、明⽰的に >>= を使って束縛が書く
と、point-free形式でコードを書く機会を増やすことになります。ただし、通常は読みやすさを
優先すべきでしょう。

8 .5  モナド則

Monad 型クラスはモナド則(monad laws)と呼ばれる3つの規則を持っています。これらは
Monad 型クラスの理にかなった実装から何を期待できるかを教えてくれます。

do記法を使⽤してこれらの規則を説明していくのが最も簡単でしょう。

Ident i ty律

右単位元則(right-identity law)が３つの規則の中で最も簡単です。この規則はdo記法ブロック
の最後の式であれば、 pure の呼び出しを排除することができると⾔っています。

do 

  x <- expr 

  pure x 

右単位元則は、この式は単なる expr と同じだと⾔っています。

左単位元則(left-identity law)は、もしそれがdo記法ブロックの最初の式であれば、 pure の呼
び出しを除去することができると述べています。

do 

  x <- pure y 

  next 



このコードの名前 x を式 y で置き換えたものと next は同じです。

最後の規則は結合則(associativity law)です。これは⼊れ⼦になったdo記法ブロックをどう扱う
のかについて教えてくれます。

c1 = do 

  y <- do 

    x <- m1 

    m2 

  m3 

上記のコード⽚は、次のコードと同じです。

c2 = do 

  x <- m1 

  y <- m2 

  m3 

これら計算にはそれぞれ、3つのモナドの式 m1 、 m2 、 m3 が含まれています。どちらの場
合でも m1 の結果は名前 x に束縛され、 m2 の結果は名前 y に束縛されます。

c1 では２つの式 m1 と m2 がそれぞれのdo記法ブロック内にグループ化されています。

c2 では m1 、 m2 、 m3 の３つすべての式が同じdo記法ブロックに現れています。

結合規則は ⼊れ⼦になったdo記法ブロックをこのように単純化しても安全であるというこ
とを⾔っています。

注意: do記法がどのように bind の呼び出しへと脱糖されるかの定義により、 c1 と c2 はい
ずれも次のコードと同じです。 `

c3 = do 

  x <- m1 

  do 

    y <- m2 

    m3 

8 .6  モナドと畳み込み



抽象的にモナドを扱う例として、この節では Monad 型クラスの何らかの型構築⼦と⼀緒に機
能するある関数を⽰していきます。これはモナドによるコードが副作⽤を伴う「より⼤きな⾔
語」でのプログラミングと対応しているという直感的理解を補強しますし、モナドによるプログ
ラミングがもたらす⼀般性も⽰しています。

これから foldM と呼ばれる関数を書いてみます。これは以前扱った foldl 関数をモナドの⽂
脈へと⼀般化します。型シグネチャは次のようになっています。

foldM :: forall m a b 

       . Monad m 

      => (a -> b -> m a) 

      -> a 

      -> List b 

      -> m a 

モナド m が現れている点を除いて、 foldl の型と同じであることに注意しましょう。

foldl :: forall a b 

       . (a -> b -> a) 

      -> a 

      -> List b 

      -> a 

直感的には、 foldM はさまざまな副作⽤の組み合わせに対応した⽂脈での配列の畳み込みを
⾏うと捉えることができます。

例として m が Maybe であるとすると、この畳み込みはそれぞれの段階で Nothing を返すこ
とで失敗することができます。それぞれの段階ではオプショナルな結果を返しますから、それゆ
え畳み込みの結果もオプショナルになります。

もし m として配列の型構築⼦ Array を選ぶとすると、畳み込みのそれぞれの段階で複数の結
果を返すことができ、畳み込みは結果それぞれに対して次の⼿順を継続します。最後に、結果の
集まりは、可能な経路すべての畳み込みから構成されることになります。これはグラフの⾛査と
対応しています！

foldM を書くには、単に⼊⼒の配列について場合分けをするだけです。

配列が空なら、型 a の結果を⽣成するための選択肢はひとつしかありません。第２引数を返
します。

foldM _ a Nil = pure a 



a をモナド m まで持ち上げるために pure を使わなくてはいけないことも忘れないようにし
てください。

配列が空でない場合はどうでしょうか？その場合、型 a の値、型 b の値、型
a -> b -> m a の関数があります。もしこの関数を適⽤すると、型 m a のモナドの結果を⼿に

⼊れることになります。この計算の結果を逆向きの⽮印 <- で束縛することができます。

あとは配列の残りに対して再帰するだけです。実装は簡単です。

foldM f a (b : bs) = do 

  a' <- f a b 

  foldM f a' bs 

do記法を除けば、この実装は配列に対する foldl の実装とほとんど同じであることにも注意
してください。

PSCi でこれを定義し、試してみましょう。除算可能かどうかを調べて、失敗を⽰すために
Maybe 型構築⼦を使う、整数の「安全な除算」関数を定義するとしましょう。

safeDivide :: Int -> Int -> Maybe Int 

safeDivide _ 0 = Nothing 

safeDivide a b = Just (a / b) 

これで、 foldM で安全な除算の繰り返しを表現することができます。

> import Data.List 

 

> foldM safeDivide 100 (fromFoldable [5, 2, 2]) 

(Just 5) 

 

> foldM safeDivide 100 (fromFoldable [2, 0, 4]) 

Nothing 

もしいずれかの時点で整数にならない除算が⾏われようとしたら、 foldM safeDivide 関数
は Nothing を返します。そうでなければ、 Just 構築⼦に包まれた除算の繰り返した累積の結
果を返します。

8 .7  モナドとAppl icat ive



クラス間に上位クラス関係があるため、 Monad 型クラスのすべてのインスタンスは
Applicative 型クラスのインスタンスでもあります。

しかしながら、どんな Monad のインスタンスについても Applicative 型クラスの実装が、
それ以上の条件なしで存在し、次のような ap が与えられます。

ap :: forall m a b. Monad m => m (a -> b) -> m a -> m b 

ap mf ma = do 

  f <- mf 

  a <- ma 

  pure (f a) 

もし m が Monad 型クラスの規則に従っているなら、 m が ap で与えられるような、妥当な
Applicative インスタンスが存在します。

興味のある読者は、これまで登場した Array 、 Maybe 、 Either e といったモナドについ
て、この ap が apply と⼀致することを確かめてみてください。

もしすべてのモナドがApplicative関⼿でもあるなら、Applicative関⼿についての直感的理解を
すべてのモナドについても適⽤することができるはずです。特に、更なる副作⽤の組み合わせで
増強された「より⼤きな⾔語」でのプログラミングとモナドがいろいろな意味で⼀致することを
当然に期待することができます。 map と apply を使って、引数が任意個の関数をこの新しい⾔
語へと持ち上げることができるはずです。

しかし、モナドはApplicative関⼿で可能な以上のことを⾏うことができ、重要な違いはdo記
法の構⽂で強調されています。利⽤者情報を符号化したXML⽂書から利⽤者の都市を検索す
る、 userCity の例についてもう⼀度考えてみましょう。

userCity :: XML -> Maybe XML 

userCity root = do 

  prof <- child root "profile" 

  addr <- child prof "address" 

  city <- child addr "city" 

  pure city 

２番⽬の計算が最初の結果 prof に依存し、３番⽬の計算が２番⽬の計算の結果 addr に依存
するというようなことをdo記法は可能にします。 Applicative 型クラスのインターフェイスだ
けを使うのでは、このような以前の値への依存は不可能です。

pure と apply だけを使って userCity を書こうとしてみれば、これが不可能であることが
わかるでしょう。Applicativeは関数の互いに独⽴した引数を持ち上げることだけを可能にします



が、モナドはもっと興味深いデータ依存関係に関わる計算を書くことを可能にします。

前の章では Applicative 型クラスは並列処理を表現できることを⾒ました。持ち上げられた
関数の引数は互いに独⽴していますから、これはまさにその通りです。 Monad 型クラスは計算
が前の計算の結果に依存できるようにしますから、同じようにはなりません。モナドはその副作
⽤を順番に組み合わせしなければいけません。

演 習

1. （簡単）  purescript-arrays パッケージの Data.Array モジュールから
head 関数と tail 関数の型を探してください。 Maybe モナドとdo記法を使い、
head と tail を組み合わせて、３要素以上の配列の3番⽬の要素を返すような関

数を作ってください。その関数は適当な Maybe 型を返さなければいけません。

2. （やや難しい） 与えられた幾つかの硬貨を組み合わせてできる可能性のあるす
べての合計を決定する関数 sum を、 foldM を使って書いてみましょう。⼊⼒の硬
貨は、硬貨の価値の配列として与えられます。この関数は次のような結果にならな
くてはいけません。

> sums [] 

[0] 

 

> sums [1, 2, 10] 

[0,1,2,3,10,11,12,13] 

ヒント： foldM を使うと１⾏でこの関数を書くことが可能です。重複する要素を
取り除いたり、結果を昇順に並び替えたりするのに、 nub 関数や sort 関数を使
いたくなるかもしれません。

3. （やや難しい） Maybe 型構築⼦について、 ap 関数と apply 演算⼦が⼀致する
ことを確認してください。

4. （やや難しい） purescript-maybe パッケージで定義されている Maybe 型につ
いての Monad インスタンスが、モナド則を満たしていることを検証してくださ
い。



5. （やや難しい） 配列上の filter の関数を⼀般化した関数 filterM を書いてく
ださい。この関数は次の型シグネチャを持つ必要があります。

filterM :: forall m a. Monad m => (a -> m Boolean) -> List a -> m (List a) 

PSCi で Maybe と Array モナドを使ってその関数を試してみてください。

6. （難しい） すべてのモナドは、次で与えられるような既定の Functor インスタ
ンスがあります。

map f a = do 

  x <- a 

  pure (f x) 

モナド則を使って、すべてのモナドが次を満たすことを証明してください。

lift2 f (pure a) (pure b) = pure (f a b) 

ここで、 Applicative インスタンスは上で定義された ap 関数を使⽤していま
す。 lift2 が次のように定義されていたことを思い出してください。

lift2 :: forall f a b c. Applicative f => (a -> b -> c) -> f a -> f b -> f c 

lift2 f a b = f <$> a <*> b 

8 .8  ネイティブな作⽤

ここではPureScriptの中核となる重要なモナド、 Eff モナドについて⾒ていきます。

Eff モナドは Control.Monad.Eff モジュール、およびPreludeで定義されています。これは
いわゆるネイティブな作⽤を扱うために使います。

ネイティブな副作⽤とは何でしょうか。ネイティブな副作⽤とは、従来のJavaScriptの式が持
つ副作⽤と、PureScript特有の式が持つ副作⽤を区別するものです。ネイティブな作⽤には次の
ようなものがあります。

コンソール⼊出⼒



乱数⽣成
例外
変更可能な状態の読み書き

また、ブラウザでは次のようなものがあります。

DOM操作
XMLHttpRequest / AJAX呼び出し
WebSocketによる相互作⽤
Local Storageの読み書き

すでに「ネイティブでない」副作⽤の例については数多く⾒てきています。

Maybe データ型で表現される省略可能な値
Either データ型で表現されるエラー

配列やリストで表現される多価関数

これらの区別はわかりにくいので注意してください。エラーメッセージは例外の形で
JavaScriptの式の副作⽤となることがあります。その意味では例外はネイティブな副作⽤を表し
ていて、 Eff を使⽤して表現することができます。しかし、 Either を使⽤して実装されたエ
ラーメッセージはJavaScriptランタイムの副作⽤ではなく、 Eff を使うスタイルでエラーメッセ
ージを実装するのは適切ではありません。そのため、ネイティブなのは作⽤⾃体というより、実
⾏時にどのように実装されているかです。

8 .9  副作⽤と純粋性

PureScriptのような⾔語が純粋であるとすると、疑問が浮かんできます。副作⽤がないなら、
どうやって役に⽴つ実際のコードを書くことができるというのでしょうか。

その答えはPureScriptの⽬的は副作⽤を排除することではないということです。これは、純粋
な計算と副作⽤のある計算とを型システムにおいて区別することができるような⽅法で、副作⽤
を表現することを⽬的としているのです。この意味で、⾔語はあくまで純粋だということです。

副作⽤のある値は、純粋な値とは異なる型を持っています。このように、例えば副作⽤のある
引数を関数に渡すことはできず、予期せず副作⽤持つようなことが起こらなくなります。

Eff モナドで管理された副作⽤を実⾏する唯⼀の⽅法は、型 Eff eff a の計算をJavaScript
から実⾏することです。



ビルドツールPulp(や他のツール)は、オプションを与えることで、アプリケーションの起動時
に main 計算を呼び出すためのJavaScriptコードを簡単に追加で⽣成できるようにしています。
main は Eff モナドでの計算であることが要求されます。

このように、 main によって使われる副作⽤が期待されることを、開発者は正確に知ることが
できます。加えて、 main がどのような種類の副作⽤を持つかを制限するのに Eff モナドを使
うことができるので、例えば、アプリケーションはコンソールと相互作⽤するが、それ以外は何
もしない、ということを確実に⾔うことができます。

8 .10 Effモナド

Eff モナドの⽬的は、副作⽤のある計算に型付けされたAPIを提供すると同時に、効率的な
JavaScriptを⽣成することにあります。これは拡張可能作⽤(extensible effects)のモナドとも呼ば
れており、これについては後述します。

例を⽰しましょう。次のコードでは乱数を⽣成するための関数が定義されている
purescript-random モジュールを使⽤しています。

module Main where 

 

import Prelude 

 

import Control.Monad.Eff.Random (random) 

import Control.Monad.Eff.Console (logShow) 

 

main = do 

  n <- random 

  logShow n 

このファイルが Main.purs という名前で保存されているなら、次のコマンドでコンパイルす
ることができます。

$ pulp run 

コンパイルされたJavaScriptを実⾏すると、コンソールに出⼒ 0 と 1 の間で無作為に選ばれ
た数が表⽰されるでしょう。

このプログラムは、乱数⽣成とコンソール⼊出⼒というJavaScriptランタイムが提供する２種
類のネイティブな作⽤を、do記法で組み合わせて使っています。



8 .11 拡張可能作⽤

PSCi でモジュールを読み込み、 main の型を調べてみましょう。

> import Main 

 

> :type main 

forall eff. Eff (console :: CONSOLE, random :: RANDOM | eff) Unit 

この型はかなり複雑そうに⾒えますが、PureScriptのレコードの⽐喩で簡単に説明することが
できます。

レコード型を使った簡単な関数を考えてみましょう。

fullName person = person.firstName <> " " <> person.lastName 

この関数は firstName と lastName というプロパティを含むレコードから完全な名前の⽂字
列を作成します。もし PSCi でこの関数の型を同様に調べたとすると、次のように表⽰されるで
しょう。

forall r. { firstName :: String, lastName :: String | r } -> String 

この型は「少なくとも fullName は firstName と lastName という2つのフィールドを持つよ
うなレコードをとり、 String を返す。」というように読みます。

渡したレコードが firstName と lastName いうプロパティさえ持っていれば、その他に余計
なフィールドを持っていたとしても fullName は気にしません。

> firstName { firstName: "Phil", lastName: "Freeman", location: "Los Angeles" } 

Phil Freeman 

同様に、上の main の型は「 main は副作⽤のある計算で、乱数⽣成とコンソール⼊出⼒、お
よびそれ以外の任意の種類の副作⽤を備えた任意の環境で実⾏することができ、型 Unit の値を
返す」というように解釈できます。

これは 「拡張可能作⽤」という名前の由来になっています。必要な副作⽤さえ備えていれ
ば、その副作⽤の集まりをいつでも拡張できるということです。



8 .12 作⽤の混在

拡張可能作⽤は Eff モナドで異なる型の副作⽤を混在(interleave)させることを可能にしま
す。

先ほど使った random 関数は次のような型を持っています。

forall eff1. Eff (random :: RANDOM | eff1) Number 

この作⽤の集まり (random :: RANDOM | eff1) は main で⾒たものと同じではありません。

しかし、作⽤が⼀致するように random の型を特殊化できます。 eff1 に
(console :: CONSOLE | eff) を選べば、これらの２つの作⽤の集合は同じになります。

同様に logShow は main の作⽤に合わせて特殊化できる型を持っています。

forall eff2. Show a => a -> Eff (console :: CONSOLE | eff2) Unit 

この場合は、 eff2 に (random :: Random | eff) を選ばなくてはなりません。

それが含む副作⽤を⽰す random と logShow の型がポイントで、より⼤きな副作⽤の集まり
を持ったより⼤きな計算を構築するために、他の副作⽤を混ぜ合わせることができるのです。

main の型注釈を与えなくてもよいことに注意してください。コンパイラは random と
logShow の多相的な型が与えられた main について、最も⼀般的な型を⾒つけることができま

す。

8 .13 Effの種

main の型は今まで⾒てきた他の型とは異なります。それを説明するためには、まず Eff の
種について考える必要があります。値がその型によって分類されるように、型がその種によって
分類されることを思い出してください。これまでは Type （型の種）と -> （型構築⼦のための
種を構築する）だけから構築された種のみを⾒てきました。

Eff の種を⾒るには、 PSCi で :kind コマンドを使います。

> import Control.Monad.Eff 

 



> :kind Eff 

# Control.Monad.Eff.Effect -> Type -> Type 

今まで⾒たことのない記号が２つあります。

Control.Monad.Eff.Effect は副作⽤の型についての型レベルのラベルを表す作⽤の種です。
これを理解するためには、上の main で⾒た2つのラベルがいずれも種
Control.Monad.Eff.Effect を持っていることに注⽬してください。

> import Control.Monad.Eff.Console 

> import Control.Monad.Eff.Random 

 

> :kind CONSOLE 

Control.Monad.Eff.Effect 

 

> :kind RANDOM 

Control.Monad.Eff.Effect 

# 種構築⼦は⾏の種を構築するのに使われます。⾏とは順序なしラベル付きの集合のことで
す。

そして、 Eff は作⽤の⾏と作⽤の返り値の型という２つの引数を持っています。つまり、
Eff の最初の引数は、作⽤の型の順序なしラベル付きの集合であり、２つめの引数は返り値の

型だということです。

これで、先ほどの main の型を読むことができるようになりました。

forall eff. Eff (console :: CONSOLE, random :: RANDOM | eff) Unit 

Eff の最初の引数は (console :: CONSOLE, random :: RANDOM | eff) です。これは
CONSOLE 作⽤と Random 作⽤を含む⾏です。パイプ記号 | は、ラベルが付けられた作⽤と、そ

れに混ぜあわせたいそれ以外の任意の作⽤を表す⾏変数(row variable) eff を区切っています。

Eff の2番⽬の引数は、計算の戻り値の型 Unit です。

8 .14 オブジェクトと⾏

拡張可能作⽤とレコードに深いつながりをもたらしている Eff の種を考えてみましょう。



上で定義した関数 fullName を考えます。

fullName :: forall r. { firstName :: String, lastName :: String | r } -> String 

fullName person = person.firstName <> " " <> person.lastName 

種 Type の型だけが値を持つので、関数の⽮印の左辺にある型の種は Type でなければなりま
せん。

中括弧は実際には構⽂糖であり、PureScriptコンパイラによって理解されている完全な型は次
のようなものです。

fullName :: forall r. Record (firstName :: String, lastName :: String | r) -> String 

中括弧がなくなっており、 Record 構築⼦が追加されていることに注意してください。
Record は Prim モジュールで定義されている組み込みの型構築⼦です。 Record の種を調べて

みると、次のようになっています。

> :kind Record 

# Type -> Type 

つまり、 Record は型の⾏をとり型を構築する型構築⼦なのです。これがレコードについての
⾏多相関数を書くことを可能にしているのです。

この型システムでは、拡張可能作⽤を扱うのに、⾏多相レコード(拡張可能レコード)を使うと
きと同じ機構が使われています。唯⼀の違いは、ラベルに現れる型の種です。レコードは型の⾏
によってパラメータ化され、 Eff は作⽤の⾏によってパラメータ化されるのです。

これと同じ型システムの機能は、型構築⼦の⾏や、⾏の⾏でパラメータ化される型を構築する
のにさえ使われることがあります！

8 .15 きめ細かな作⽤

作⽤の⾏は推論されるので、⼤抵の場合は Eff を使うときに型注釈は必須ではありません
が、計算でどの作⽤が期待されるのかをコンパイラに⽰すために型注釈が使われることがありま
す。

先ほどの例を、作⽤の閉じた⾏で注釈すると次のようになります。



main :: Eff (console :: CONSOLE, random :: RANDOM) Unit 

main = do 

  n <- random 

  logShow n 

⾏変数 eff がないことに注意してください。こうすると、異なった作⽤の型を使う計算を誤
って含めることはできません。このように、コードが持つことを許される副作⽤を制御すること
ができるのです。

8 .16 ハンドラとアクション

logShow や random のような関数はアクションと呼ばれます。アクションはそれらの関数の
右辺に Eff 型を持っており、その⽬的は新たな効果を導⼊することにあります。

これは Eff 型が関数の引数の型として現れるハンドラとは対照的です。アクションが集合へ
必要な作⽤を追加するのに対し、ハンドラは集合から作⽤を除去します。

例として、 purescript-exceptions パッケージを考えてみます。このパッケージでは
throwException と catchException という⼆つの関数が定義されています。

throwException :: forall a eff 

                . Error 

               -> Eff (exception :: EXCEPTION | eff) a 

 

catchException :: forall a eff 

                . (Error -> Eff eff a) 

               -> Eff (exception :: EXCEPTION | eff) a 

               -> Eff eff a 

throwException はアクションです。 Eff は右辺に現れていて、新しく Exception 作⽤を導
⼊します。

catchException はハンドラです。 Eff は関数の第２引数の型として出現しており、作⽤全
体としては Exception 作⽤を除去します。

特定の作⽤を必要とするコードの部分を限定するために型システムを使うことができるので、
これは便利です。作⽤のあるコードをハンドラで包むことにより、その作⽤を許さないコードブ
ロックの中に埋め込むことができます。



例えば、 Exception 作⽤を使って例外を投げるコード⽚を書き、それからそのコードを
catchException で包むことによって、例外を許さないコード⽚の中にその計算を埋め込むこと

ができるのです。

JSONドキュメントからアプリケーションの設定を読みたいとしましょう。⽂書を構⽂解析す
る過程で例外を投げることがあります。設定を読み構⽂解析するこの処理は、次のような型シグ
ネチャを持つ関数として書くことができます。

readConfig :: forall eff. Eff (exception :: EXCEPTION | eff) Config 

それから、 main 関数で catchException を使⽤して Exception 作⽤を処理することができ
ます。

main = do 

    config <- catchException printException readConfig 

    runApplication config 

  where 

    printException e = do 

      log (message e) 

      pure defaultConfig 

purescript-eff パッケージでも、副作⽤なしの計算を取り、それを純粋な値として安全に評
価する runPure ハンドラが定義されています。

type Pure a = Eff () a 

 

runPure :: forall a. Pure a -> a 

8 .17 可変状態

Preludeには ST 作⽤というまた別の作⽤も定義されています。

ST 作⽤は変更可能な状態を操作するために使われます。純粋関数プログラミングを知ってい
るなら、共有される変更可能な状態は問題を引き起こしやすいということも知っているでしょ
う。しかしながら、 ST 作⽤は型システムを使って安全で局所的な状態変化を可能にし、状態の
共有を制限するのです。



ST 作⽤は Control.Monad.ST モジュールで定義されています。これがどのように動作するか
を確認するには、そのアクションの型を⾒る必要があります。

newSTRef :: forall a h eff. a -> Eff (st :: ST h | eff) (STRef h a) 

 

readSTRef :: forall a h eff. STRef h a -> Eff (st :: ST h | eff) a 

 

writeSTRef :: forall a h eff. STRef h a -> a -> Eff (st :: ST h | eff) a 

 

modifySTRef :: forall a h eff. STRef h a -> (a -> a) -> Eff (st :: ST h | eff) a 

newSTRef は型 STRef h a の変更可能な参照領域を新しく作るのに使われます。 STRef h a
は readSTRef アクションを使って状態を読み取ったり、 writeSTRef アクションや
modifySTRef アクションで状態を変更するのに使われます。型 a は領域に格納された値の型

で、型 h は型システムのメモリ領域を表しています。

例を⽰します。⼩さな時間刻みで簡単な更新関数の実⾏を何度も繰り返すことによって、重⼒
に従って落下する粒⼦の落下の動きをシミュレートしたいとしましょう。

粒⼦の位置と速度を保持する変更可能な参照領域を作成し、領域に格納された値を更新するの
にforループ( Control.Monad.Eff の forE アクション)を使うことでこれを実現することができ
ます。

import Prelude 

 

import Control.Monad.Eff (Eff, forE) 

import Control.Monad.ST (ST, newSTRef, readSTRef, modifySTRef) 

 

simulate :: forall eff h. Number -> Number -> Int -> Eff (st :: ST h | eff) Number 

simulate x0 v0 time = do 

  ref <- newSTRef { x: x0, v: v0 } 

  forE 0 (time * 1000) \_ -> do 

    modifySTRef ref \o -> 

      { v: o.v - 9.81 * 0.001 

      , x: o.x + o.v * 0.001 

      } 

    pure unit 

  final <- readSTRef ref 

  pure final.x 

計算の最後では、参照領域の最終的な値を読み取り、粒⼦の位置を返しています。

この関数が変更可能な状態を使っていても、その参照区画 ref がプログラムの他の部分で使
われるのが許されない限り、これは純粋な関数のままであることに注意してください。 ST 作⽤



が禁⽌するものが正確には何であるのかについては後ほど⾒ます。

ST 作⽤で計算を実⾏するには、 runST 関数を使⽤する必要があります。

runST :: forall a eff. (forall h. Eff (st :: ST h | eff) a) -> Eff eff a 

ここで注⽬して欲しいのは、領域型 h が関数⽮印の左辺にある括弧の内側で量化されている
ということです。 runST に渡したどんなアクションでも、 任意の領域 h がなんであれ動作する
ということを意味しています。

しかしながら、ひとたび参照領域が newSTRef によって作成されると、その領域の型はすで
に固定されており、 runST によって限定されたコードの外側で参照領域を使おうとしても型エ
ラーになるでしょう。 runST が安全に ST 作⽤を除去できるのはこれが理由なのです！

実際に、 ST はこの例の唯⼀の作⽤なので、 runPure と runST を併⽤すると simulate を純
粋な関数に変えることができます、

simulate' :: Number -> Number -> Number -> Number 

simulate' x0 v0 time = runPure (runST (simulate x0 v0 time)) 

PSCi でこの関数を実⾏してみてください。

> import Main 

 

> simulate' 100.0 0.0 0.0 

100.00 

 

> simulate' 100.0 0.0 1.0 

95.10 

 

> simulate' 100.0 0.0 2.0 

80.39 

 

> simulate' 100.0 0.0 3.0 

55.87 

 

> simulate' 100.0 0.0 4.0 

21.54 

もし simulate の定義を runST の呼び出しのところへ埋め込むとすると、次のようになりま
す。



simulate :: Number -> Number -> Int -> Number 

simulate x0 v0 time = runPure $ runST do 

  ref <- newSTRef { x: x0, v: v0 } 

  forE 0 (time * 1000) \_ -> do 

    modifySTRef ref \o ->   

      { v: o.v - 9.81 * 0.001 

      , x: o.x + o.v * 0.001   

      } 

    pure unit   

  final <- readSTRef ref 

  pure final.x 

参照区画はそのスコープから逃れることができないことがコンパイラにわかりますし、安全に
var に変換することができます。 runST の呼び出しの本体に対して⽣成されたJavaScriptは次の

ようになります。

var ref = { x: x0, v: v0 }; 

 

Control_Monad_Eff.forE(0)(time * 1000 | 0)(function (i) { 

  return function __do() { 

    ref = (function (o) { 

      return { 

        v: o.v - 9.81 * 1.0e-3, 

        x: o.x + o.v * 1.0e-3 

      }; 

    })(ref); 

    return Prelude.unit; 

  }; 

})(); 

 

return ref.x; 

局所的な変更可能状態を扱うとき、特に Eff モナドで効率のよいループを⽣成する forE 、
foreachE 、 whileE 、 untilE のようなアクションを⼀緒に使うときには、 ST 作⽤は短い

JavaScriptを⽣成できる良い⽅法となります。

演 習

1. （やや難しい） もし分⺟で分⼦を割り切れないなら throwException を使って
例外を投げるように safeDivide 関数を書き直してください。



2. （難しい） PIを推定するには次のような簡単な⽅法があります。単位正⽅形内
にある多数の N 個の点を無作為に選び、内接する円に含まれるものの個数 n を数
えます。このとき 4n/N が円周率 pi の概算となります。 forE 関数、 Random 作
⽤、 ST 作⽤を使って、この⽅法で円周率 pi を推定する関数を書いてください。

8 .18 DOM作⽤

この章の最後の節では、 Eff モナドでの作⽤についてこれまで学んだことを、実際のDOM
操作の問題に応⽤します。

DOMを直接扱ったり、オープンソースのDOMライブラリを扱う、⾃由に利⽤可能な
PureScriptパッケージが幾つかあります。

purescript-dom  - 低レベルなJavaScript DOM APIのバインディング
purescript-jquery  - jQueryライブラリのバインディング

上記のライブラリを抽象化するPureScript向けのライブラリもあります。

purescript-thermite  - purescript-react 上で構築されるライブラリ
purescript-halogen  - 仮想DOMを抽象的する型安全なライブラリ

この章では purescript-react を使⽤し、住所簿にインターフェイスを追加しますが、興味
のあるユーザは異なるアプローチで進めることをおすすめします。

8 .19 住所録のユーザーインタフェース

purescript-react を使⽤するためにReactコンポーネントと同じ様にアプリケーションを定
義します。Reactコンポーネントは、コード内のHTML要素を純粋なデータ構造体として記述
し、効率的にDOMにレンダリングします。さらに、コンポーネントはボタンのクリックなどの
イベントに応答できます。 purescript-react ライブラリは Eff モナドを使ってこれらのイベ
ントをどのように扱うかを記述します。

Reactライブラリの完全なチュートリアルはこの章の範囲をはるかに超えていますが、読者は
必要に応じてマニュアルを参照することをお勧めします。⽬的に応じて、Reactは Eff モナドの
実⽤的な例を提供してくれます。

http://github.com/purescript-contrib/purescript-dom
http://github.com/paf31/purescript-jquery
http://jquery.org/
http://github.com/paf31/purescript-thermite
http://github.com/slamdata/purescript-halogen


まずは利⽤者が住所録に新しい項⽬を追加できるフォームを構築することにしましょう。フォ
ームには、さまざまなフィールド（姓、名前、都市、州など）を⼊⼒するテキストボックス、お
よび検証エラーが表⽰される領域が含まれます。テキストボックスに利⽤者がテキストを⼊⼒す
ると、検証エラーが更新されます。

シンプルさを保つために、フォームは固定の形状とします。電話番号は種類（⾃宅、携帯電
話、仕事、その他）ごとに別々のテキストボックスへ分けることにします。

次の⾏を除いて、HTMLファイルは基本的に空です。

<script type="text/javascript" src="../dist/Main.js"></script> 

この⾏には、Pulpによって⽣成されたJavaScriptコードが含まれています。これをファイルの
最後に配置して、コードからアクセスしようとする関連要素が上にあることを確認します。
Main.js ファイルを再構築するには、 pulp browserify コマンドを使うことができます。最初

に dist ディレクトリが存在し、ReactをNPM依存関係としてインストールしたことを確認して
ください。

$ npm install # Install React 

$ mkdir dist/ 

$ pulp browserify --to dist/Main.js 

Main モジュールは住所録コンポーネントを作成して画⾯に表⽰する main 関数を定義してい
ます。 main 関数は CONSOLE 作⽤と DOM 作⽤のみを使⽤しており、型シグニチャは次のことを
⽰します。

main :: Eff (console :: CONSOLE, dom :: DOM) Unit 

まず、 main はコンソールにステータスメッセージを記録します。

main = void do 

  log "Rendering address book component" 

その後、 main はDOM APIを使⽤してドキュメント本体への参照（ doc ）を取得します。

  doc <- window >>= document 



これは混在した作⽤の⼀例になっていることに注⽬してください。 log 関数は CONSOLE 作⽤
を使い、 window と document 関数は両⽅とも DOM 作⽤を使います。 main の型は両⽅の作⽤
を利⽤することを⽰します。

main は window アクションを使ってウィンドウオブジェクトへの参照を取得し、その結果を
document 関数に >>= を使って渡します。 document はウィンドウオブジェクトをとり、その

ドキュメントへの参照を返します。

do記法の定義により、これを次のようにも書けることに注意してください。

  w <- window 

  doc <- document w 

どちらが読みやすいかどうかは個⼈の好みの問題です。前者は名前が付けられた関数の引数が
なく、point-free形式の⼀例となっています。その⼀⽅で、後者ではウィンドウオブジェクトの
名前として w が使われています。

Main モジュールは addressBook と呼ばれる住所録コンポーネントを定義します。その定義
を理解するために、まずいくつかの概念を理解する必要があります。

Reactコンポーネントを作成するには、最初にコンポーネントのテンプレートのように動作す
るReactクラスを作成する必要があります。 purescript-react では、 createClass 関数を使っ
てクラスを作成することができます。 createClass はクラスの仕様を必要とします。この本質
は、コンポーネントのライフサイクルについて処理するために使われる Eff アクションの集合
です。開発者が注⽬するべきなのは Render アクションです。

Reactライブラリが提供するいくつかの関連する関数の型は次のとおりです。

createClass 

  :: forall props state eff 

   . ReactSpec props state eff 

  -> ReactClass props 

 

type Render props state eff 

   = ReactThis props state 

  -> Eff ( props :: ReactProps 

         , refs :: ReactRefs Disallowed 

         , state :: ReactState ReadOnly 

         | eff 

         ) ReactElement 

 

spec 

  :: forall props state eff 



   . state 

  -> Render props state eff 

  -> ReactSpec props state eff 

Render 型同義語は、いくつかの型シグネチャを単純化するために提供され、コンポーネントの
レンダリング機能を表します。
Render アクションは（ ReactThis 型の）コンポーネントへの参照をとり、 Eff モナドに
ReactElement を返します。 ReactElement はレンダリング後の意図したDOMの状態を記述す

るデータ構造体です。
すべてのReactコンポーネントは、ある型の状態を定義します。ボタンのクリックなどのイベン
トに応じて状態を変更することができます。 purescript-react では、初期状態値が spec 関数
で提供されます。
Render 型の作⽤の⾏は、いくつかの⾯⽩い作⽤を使⽤して、特定の関数からReactコンポーネ

ントの状態へのアクセスを制限します。たとえば、レンダリングのあいだ、「refs」オブジェク
トへのアクセスは Disallowed であり、コンポーネント状態へのアクセスは ReadOnly です。

Main モジュールは、住所録コンポーネントの状態の型と初期状態を定義します。

newtype AppState = AppState 

  { person :: Person 

  , errors :: Errors 

  } 

 

initialState :: AppState 

initialState = AppState 

  { person: examplePerson 

  , errors: [] 

  } 

状態には、（フォームコンポーネントを使⽤して編集可能にする） Person レコードと、既存
の検証コードを使⽤して⼊⼒されるエラーの配列が含まれています。

次に、コンポーネントの定義を⾒てみましょう。

addressBook :: forall props. ReactClass props 

すでに述べたように、 addressBook は createClass と spec を使⽤してReactクラスを作成し
ます。ここから初期状態の値と Render アクションを得ることができます。取得した Render ア
クションでいったい何ができるのでしょうか？ 例えば、 purescript-react は以下のような単
純なアクションを提供しています。



readState 

  :: forall props state access eff 

   . ReactThis props state 

  -> Eff ( state :: ReactState ( read :: Read 

                               | access 

                               ) 

         | eff 

         ) state 

 

writeState 

  :: forall props state access eff 

   . ReactThis props state 

  -> state 

  -> Eff ( state :: ReactState ( write :: Write 

                               | access 

                               ) 

         | eff 

         ) state 

readState と writeState アクションは拡張可能作⽤を伴って、ReactState作⽤を使って
Reactの状態にアクセスできるようにしますが、他の⾏の ReactState 作⽤をパラメータ化する
ことで、読み書き権限がさらに分離されることに注意してください！

これは、PureScriptの⾏ベースの作⽤に関する興味深い点を⽰しています。⾏内に現れる作⽤
は単純な１要素である必要はなく、様々な構造を持つことができ、この柔軟性によってコンパイ
ル時にいくつかの有⽤な制限が可能になります。 purescript-react ライブラリがこの制限をし
なかった場合、 Render アクションで状態を書き込もうとすると、実⾏時に例外を受け取ること
になります。適切な制限を⾏うことで、このような間違いがコンパイル時に捕捉されるようにな
りました。

これで addressBook コンポーネントの定義を読むことができるようになりました。まずは現
在のコンポーネントの状態を読むことから始めましょう。

addressBook = createClass $ spec initialState \ctx -> do 

  AppState { person: Person person@{ homeAddress: Address address } 

           , errors 

           } <- readState ctx 

次の点に注意してください。

名前 ctx は ReactThis を参照しており、必要に応じて状態を読み書きするために使⽤すること
ができます。



AppState内のレコードは、レコードパターンを使⽤して照合しています。これにはerrorsフィー
ルドのレコード同名利⽤も含まれます。便利なように、状態の構造のそれぞれの部分に明⽰的な
名前をつけています。

Render はDOMの次の状態を表す ReactElement 構造体を返さなければならないのでし
た。 Render アクションはいくつかの補助関数から定義されています。その補助関数の1つは
renderValidationErrors です。これは Errors 構造体を ReactElement の配列に変換します。

renderValidationError :: String -> ReactElement 

renderValidationError err = D.li' [ D.text err ] 

 

renderValidationErrors :: Errors -> Array ReactElement 

renderValidationErrors [] = [] 

renderValidationErrors xs = 

  [ D.div [ P.className "alert alert-danger" ] 

          [ D.ul' (map renderValidationError xs) ] 

  ] 

purescript-react では、 ReactElement は通常、単⼀のHTML要素を div のような関数を
適⽤することで作成します。これらの関数は通常、属性の配列と⼦要素の配列を引数として取り
ます。しかし、ここでは ul' のようにプライム記号(', prime character)で終わる名前は属性配列
を省略し、代わりにデフォルトの属性を使⽤します。

ここでは通常のデータ構造体を単純に操作しているので、 map のような関数を使って様々な
要素を構築することができます。

2番⽬の補助関数は formField です。これは、単⼀フォームフィールドのテキスト⼊⼒を含む
ReactElement を作成します。

formField 

  :: String 

  -> String 

  -> String 

  -> (String -> Person) 

  -> ReactElement 

formField name hint value update = 

  D.div [ P.className "form-group" ] 

        [ D.label [ P.className "col-sm-2 control-label" ] 

                  [ D.text name ] 

        , D.div [ P.className "col-sm-3" ] 

                [ D.input [ P._type "text" 

                          , P.className "form-control" 

                          , P.placeholder hint 

                          , P.value value 

                          , P.onChange (updateAppState ctx update) 



                          ] [] 

                ] 

        ] 

繰り返しますが、単純な要素から様々な要素を構成し、それぞれの要素に属性を適⽤していま
す。ここで注⽬すべき属性の1つは、 input 要素に適⽤される onChange 属性です。これはイベ
ントハンドラで、ユーザーがテキストボックス内のテキストを編集するときにコンポーネントの
状態を更新するために使⽤されます。イベントハンドラは、3番⽬の補助関数 updateAppState
を使⽤して定義されています。

updateAppState 

  :: forall props eff 

   . ReactThis props AppState 

  -> (String -> Person) 

  -> Event 

  -> Eff ( console :: CONSOLE 

         , state :: ReactState ReadWrite 

         | eff 

         ) Unit 

updateAppState は、 ReactThis 値の形式でコンポーネントへの参照、 Person レコードを
更新する関数、そして Event レコードを取ります。まず、（valueOf補助関数を使⽤して）
change イベントからテキストボックスの新しい値を抽出し、それを使って新しい Person 状態

を作成します。

  for_ (valueOf e) \s -> do 

    let newPerson = update s 

次に、検証関数を実⾏し、それに応じて（writeStateを使⽤して）コンポーネントの状態を更
新します。

    log "Running validators" 

    case validatePerson' newPerson of 

      Left errors -> 

        writeState ctx (AppState { person: newPerson 

                                 , errors: errors 

                                 }) 

      Right _ -> 

        writeState ctx (AppState { person: newPerson 

                                 , errors: [] 

                                 }) 



これは、コンポーネント実装の基本をカバーしています。しかし、コンポーネントの仕組みを
完全に理解するためには、この章に付随する情報をお読みください。

pulp browserify --to dist/Main.js を実⾏して、それからWebブラウザで
html/index.html を開き、ユーザインターフェイスを試してみてください。フォームフィール

ドにいろいろな値を⼊⼒すると、ページ上に出⼒された検証エラーを⾒ることができるでしょ
う。

このユーザインタフェースには明らかに改善すべき点がたくさんあります。演習ではアプリケ
ーションがより使いやすくなるような⽅法を追究していきます。

演 習

1. (簡単) このアプリケーションを変更し、職場の電話番号を⼊⼒できるテキストボ
ックスを追加してください。

2. (やや難しい) 検証エラーを ul 要素を使ってリストで表⽰するかわりに、それぞ
れのエラーについてひとつづつ alert スタイルで div を作成するように、コード
を変更してください。

3. (難しい、拡張) このユーザーインターフェイスの問題のひとつは、検証エラーが
その発⽣源であるフォームフィールドの隣に表⽰されていないことです。コードを
変更してこの問題を解決してください。

ヒント：検証器によって返されるエラーの型は、エラーの原因となっているフィ
ールドを⽰すために拡張する必要があります。次のようなエラー型を使⽤したくな
るかもしれません。

data Field = FirstNameField 

           | LastNameField 

           | StreetField 

           | CityField 

           | StateField 

           | PhoneField PhoneType 

 

data ValidationError = ValidationError String Field 

 

type Errors = Array ValidationError 



適切なフォーム要素を選択するように、 Field を querySelector アクションの
呼び出しに変更する関数を書く必要があるでしょう。

まとめ

この章ではPureScriptでの副作⽤の扱いについての多くの考え⽅を導⼊しました。

Monad 型クラスと、それに関連するdo記法の導⼊をしました。
モナド則を導⼊し、do記法使って書かれたコードを変換する⽅法を説明しました。
異なる副作⽤で動作するコードを書くために、モナドを抽象的に扱う⽅法を説明しました。
モナドがApplicative関⼿の⼀例であること、両者がどのように副作⽤のある計算を可能にするの
か、2つの⼿法の違いを説明しました。
ネイティブな作⽤の概念を定義し、ネイティブな副作⽤を処理するために使⽤する Eff モナド
を導⼊しました。
どのように Eff モナドが拡張可能作⽤を提供するか、複数の種類のネイティブな作⽤を同じ計
算に混在させる⽅法を説明しました。
作⽤やレコードが種システムでどのように扱われるか、拡張可能レコードと拡張可能作⽤の関連
を⾒ました。
乱数⽣成、例外、コンソール⼊出⼒、変更可能な状態、およびDOM操作といった、さまざまな
作⽤を扱うために Eff モナドを使いました。

Eff モナドは現実のPureScriptコードにおける基本的なツールです。本書ではこのあとも、
様々な場⾯で副作⽤を処理するために Eff モナドを使っていきます。



第9章 キャンバスグラフィックス

9 .1  この章の⽬標

この章のコード例では、PureScriptでHTML5のCanvas APIを使⽤して2Dグラフィックスを⽣
成する purescript-canvas パッケージに焦点をあててコードを拡張していきます。

9 .2  プロジェクトの準備

このモジュールのプロジェクトでは、以下のBowerの依存関係が新しく追加されています。

purescript-canvas - HTML5のCanvas APIのメソッドの型が定義されています。
purescript-refs - ⼤域的な変更可能領域への参照を扱うための副作⽤を提供しています。

この章のソースコードは、それぞれに main メソッドが定義されている複数のモ  ジュールへと
分割されています。この章の節の内容はそれぞれ異なるファイルで実装されており、それぞれの
節で対応するファイルの main メソッドを実⾏できるように、Pulpビルドコマンドを変更する
ことで Main モジュールが変更できるようになっています。

HTMLファイル html/index.html には、各例で使⽤される単⼀の canvas 要素、およびコン
パイルされたPureScriptコードを読み込む script 要素が含まれています。各節のコードをテス
トするには、ブラウザでこのHTMLファイルを開いてください。

9 .3  単純な図形

Example/Rectangle.purs ファイルにはキャンバスの中⼼に⻘い四⾓形をひとつ描画するとい
う簡単な例が含まれています。このモジュールは、 Control.Monad.Eff モジュールと、Canvas
APIを扱うための Eff モナドのアクションが定義されている Graphics.Canvas モジュールをイ
ンポートします。

他のモジュールでも同様ですが、 main アクションは最初に getCanvasElementById アクシ
ョンを使ってCanvasオブジェクトへの参照を取得しています。また、 getContext2D アクショ
ンを使ってキャンバスの2Dレンダリングコンテキストを参照しています。



main = void $ unsafePartial do 

  Just canvas <- getCanvasElementById "canvas" 

  ctx <- getContext2D canvas 

注意：この unsafePartial の呼び出しは必須です。これは getCanvasElementById の結果の
パターンマッチングが部分的で、 Just 値構築⼦だけと照合するためです。ここではこれで問題
ありませんが、実際の製品のコードではおそらく Nothing 値構築⼦と照合させ、適切なエラー
メッセージを提供したほうがよいでしょう。

これらのアクションの型は、 PSCi を使うかドキュメントを⾒ると確認できます。

getCanvasElementById :: forall eff. String -> 

  Eff (canvas :: Canvas | eff) (Maybe CanvasElement) 

 

getContext2D :: forall eff. CanvasElement -> 

  Eff (canvas :: Canvas | eff) Context2D 

CanvasElement と Context2D は Graphics.Canvas モジュールで定義されている型です。こ
のモジュールでは、モジュール内のすべてのアクションで使⽤されている Canvas 作⽤も定義さ
れています。

グラフィックスコンテキスト ctx は、キャンバスの状態を管理し、プリミティブな図形を描
画したり、スタイルや⾊を設定したり、座標変換を適⽤するためのメソッドを提供しています。

ctx の取得に続けて、 setFillStyle アクションを使って塗りのスタイルを⻘⼀⾊の塗りつ
ぶしに設定しています。

  setFillStyle "#0000FF" ctx 

setFillStyle アクションがグラフィックスコンテキストを引数として取っていることに注意
してください。これは Graphics.Canvas で共通のパターンです。

最後に、 fillPath アクションを使⽤して矩形を塗りつぶしています。 fillPath は次のよう
な型を持っています。

fillPath :: forall eff a. Context2D -> 

                          Eff (canvas :: Canvas | eff) a -> 

                          Eff (canvas :: Canvas | eff) a 



fillPath はグラフィックスコンテキストとレンダリングするパスを構築する別のアクション
を引数にとります。パスは rect アクションを使うと構築することができます。 rect はグラフ
ィックスコンテキストと矩形の位置及びサイズを格納するレコードを引数にとります。

  fillPath ctx $ rect ctx 

    { x: 250.0 

    , y: 250.0 

    , w: 100.0 

    , h: 100.0 

    } 

mainモジュールの名前として Example.Rectangle を指定して、この⻑⽅形のコード例をビル
ドしましょう。

$ mkdir dist/ 

$ pulp build -O --main Example.Rectangle --to dist/Main.js 

それでは html/index.html ファイルを開き、このコードによってキャンバスの中央に⻘い四
⾓形が描画されていることを確認してみましょう。

9 .4  ⾏多相を利⽤する

パスを描画する⽅法は他にもあります。 arc 関数は円弧を描画します。 moveTo 関数、
lineTo 関数、 closePath 関数は細かい線分を組み合わせることでパスを描画します。

Shapes.purs ファイルでは⻑⽅形と円弧セグメント、三⾓形の、3つの図形を描画していま
す。

rect 関数は引数としてレコードをとることを⾒てきました。実際には、⻑⽅形のプロパティ
は型同義語で定義されています。

type Rectangle = { x :: Number 

                 , y :: Number 

                 , w :: Number 

                 , h :: Number  

                 } 

x と y プロパティは左上隅の位置を表しており、 w と h のプロパティはそれぞれ幅と⾼さ
を表しています。



arc 関数に以下のような型を持つレコードを渡して呼び出すと、円弧を描画することができ
ます。

type Arc = { x     :: Number 

           , y     :: Number 

           , r     :: Number 

           , start :: Number 

           , end   :: Number 

           } 

ここで、 x と y プロパティは弧の中⼼、 r は半径、 start と end は弧の両端の⾓度を弧
度法で表しています。

たとえば、次のコードは中⼼ (300、300) 、半径 50 の円弧を塗りつぶします。

  fillPath ctx $ arc ctx 

    { x      : 300.0 

    , y      : 300.0 

    , r      : 50.0 

    , start  : Math.pi * 5.0 / 8.0 

    , end    : Math.pi * 2.0 

    } 

Number 型の x と y というプロパティが Rectangle レコード型と Arc レコード型の両⽅に
含まれていることに注意してください。どちらの場合でもこの組は点を表しています。これは、
いずれのレコード型にも適⽤できる、⾏多相な関数を書くことができることを意味します。

たとえば、 Shapes モジュールでは x と y のプロパティを変更し図形を並⾏移動する
translate 関数を定義されています。

translate :: forall r. Number -> Number -> 

              { x :: Number, y :: Number | r } -> 

              { x :: Number, y :: Number | r } 

translate dx dy shape = shape 

  { x = shape.x + dx 

  , y = shape.y + dy 

  } 

この⾏多相型に注⽬してください。これは triangle が x と y というプロパティと、それに
加えて他の任意のプロパティを持ったどんなレコードでも受け⼊れるということを⾔っていま
す。 x フィールドと y フィールドは更新されますが、残りのフィールドは変更されません。



これはレコード更新構⽂の例です。 shape { ... } という式は、 shape を元にして、括弧の
中で指定されたように値が更新されたフィールドを持つ新たなレコードを作ります。波括弧の中
の式はレコードリテラルのようなコロンではなく、等号でラベルと式を区切って書くことに注意
してください。

Shapes の例からわかるように、 translate 関数は Rectangle レコードと Arc レコード双
⽅に対して使うことができます。

Shape の例で描画される3つめの型は線分ごとのパスです。対応するコードは次のようになり
ます。

  setFillStyle "#FF0000" ctx 

 

  fillPath ctx $ do 

    moveTo ctx 300.0 260.0 

    lineTo ctx 260.0 340.0 

    lineTo ctx 340.0 340.0 

    closePath ctx 

ここでは3つの関数が使われています。

moveTo はパスの現在位置を指定された座標へ移動させます。
lineTo は現在の位置と指定された座標の間に線分を描画し、現在の位置を更新します。
closePath は開始位置と現在位置を結ぶ線分を描画し、パスを閉じます。

このコード⽚を実⾏すると、⼆等辺三⾓形を塗りつぶされます。

mainモジュールとして Example.Shapes を指定して、この例をビルドしましょう。

$ pulp build -O --main Example.Shapes --to dist/Main.js 

そしてもう⼀度 html/index.html を開き、結果を確認してください。キャンバスに３つの異
なる図形が描画されるはずです。

演 習



1. (簡単) これまでの例のそれぞれについて、 strokePath 関数や
setStrokeStyle 関数を使ってみましょう。

2. (簡単) 関数の引数の内部でdo記法ブロックを使うと、 fillPath 関数と
strokePath 関数で共通のスタイルを持つ複雑なパスを描画することができます。

同じ fillPath 呼び出しで隣り合った２つの矩形を描画するように、 Rectangle
のコード例を変更してみてください。線分と円弧を組み合わせてを、円の扇形を描
画してみてください。

3. (やや難しい) 次のような２次元の点を表すレコードが与えられたとします。

type Point = { x :: Number, y :: Number } 

多数の点からなる閉じたパスを描く関数 renderPath 書いてください。

renderPath :: forall eff. Context2D -> Array Point ->  

                                       Eff (canvas :: Canvas | eff) Unit 

次のような関数を考えます。

f :: Number -> Point 

この関数は引数として 1 から 0 の間の Number をとり、 Point を返します。
renderPath 関数を利⽤して関数 f のグラフを描くアクションを書いてくださ

い。そのアクションは有限個の点を f からサンプリングすることによって近似し
なければなりません。

関数 f を変更し、異なるパスが描画されることを確かめてください。

9 .5  無作為に円を描く

Example/Random.purs ファイルには2種類の異なる副作⽤が混在した Eff モナドを使う例が
含まれています。この例では無作為に⽣成された円をキャンバスに100個描画します。



main アクションはこれまでのようにグラフィックスコンテキストへの参照を取得し、ストロ
ークと塗りつぶしスタイルを設定します。

  setFillStyle "#FF0000" ctx 

  setStrokeStyle "#000000" ctx 

次のコードでは forE アクションを使って 0 から 100 までの整数について繰り返しをしてい
ます。

  for_ (1 .. 100) \_ -> do 

これらの数は 0 から 1 の間に無作為に分布しており、それぞれ x 座標、 y 座標、半径 r
を表しています。

    x <- random 

    y <- random 

    r <- random 

次のコードでこれらの変数に基づいて Arc を作成し、最後に現在のスタイルに従って円弧の
塗りつぶしと線描が⾏われます。

    let path = arc ctx 

         { x     : x * 600.0 

         , y     : y * 600.0 

         , r     : r * 50.0 

         , start : 0.0 

         , end   : Math.pi * 2.0 

         } 

    fillPath ctx path 

    strokePath ctx path 

forE に渡された関数が正しい型を持つようにするため、最後の⾏は必要であることに注意し
てください。

mainモジュールとして Example.Random を指定して、この例をビルドしましょう。

$ pulp build -O --main Example.Random --to dist/Main.js 

html/index.html を開いて、結果を確認してみましょう。



9 .6  座標変換

キャンバスは簡単な図形を描画するだけのものではありません。キャンバスは変換⾏列を扱う
ことができ、図形は描画の前に形状を変形してから描画されます。図形は平⾏移動、回転、拡⼤
縮⼩、および斜め変形することができます。

purescript-canvas ライブラリではこれらの変換を以下の関数で提供しています。

translate :: forall eff. TranslateTransform -> Context2D 

                                            -> Eff (canvas :: Canvas | eff) Context2D 

rotate    :: forall eff. Number             -> Context2D 

                                            -> Eff (canvas :: Canvas | eff) Context2D 

scale     :: forall eff. ScaleTransform     -> Context2D 

                                            -> Eff (canvas :: Canvas | eff) Context2D 

transform :: forall eff. Transform          -> Context2D 

                                            -> Eff (canvas :: Canvas | eff) Context2D 

translate アクションは TranslateTransform レコードのプロパティで指定した⼤きさだけ
平⾏移動を⾏います。

rotate アクションは最初の引数で指定されたラジアンの値に応じて原点を中⼼とした回転を
⾏います。

scale アクションは原点を中⼼として拡⼤縮⼩します。 ScaleTransform レコードは X 軸と
y 軸に沿った拡⼤率を指定するのに使います。

最後の transform はこの４つのうちで最も⼀般的なアクションです。このアクションは⾏列
に従ってアフィン変換を⾏います。

これらのアクションが呼び出された後に描画される図形は、⾃動的に適切な座標変換が適⽤さ
れます。

実際には、これらの関数のそれぞれの作⽤は、コンテキストの現在の変換⾏列に対して変換⾏
列を右から乗算していきます。つまり、もしある作⽤の変換をしていくと、その作⽤は実際には
逆順に適⽤されていきます。次のような座標変換のアクションを考えてみましょう。

transformations ctx = do 

  translate { translateX: 10.0, translateY: 10.0 } ctx 

  scale { scaleX: 2.0, scaleY: 2.0 } ctx 

  rotate (Math.pi / 2.0) ctx 

 

  renderScene 



このアクションの作⽤では、まずシーンが回転され、それから拡⼤縮⼩され、最後に平⾏移動
されます。

9 .7  コンテキストの保存

⼀般的な使い⽅としては、変換を適⽤してシーンの⼀部をレンダリングし、それからその変換
を元に戻します。

Canvas APIにはキャンバスの状態のスタックを操作する save と restore メソッドが備わっ
ています。 purescript-canvas ではこの機能を次のような関数でラップしています。

save    :: forall eff. Context2D -> Eff (canvas :: Canvas | eff) Context2D 

restore :: forall eff. Context2D -> Eff (canvas :: Canvas | eff) Context2D 

save アクションは現在のコンテキストの状態(現在の変換⾏列や描画スタイル)をスタックに
プッシュし、 restore アクションはスタックの⼀番上の状態をポップし、コンテキストの状態
を復元します。

これらのアクションにより、現在の状態を保存し、いろいろなスタイルや変換を適⽤し、プリ
ミティブを描画し、最後に元の変換と状態を復元することが可能になります。例えば、次の関数
はいくつかのキャンバスアクションを実⾏しますが、その前に回転を適⽤し、そのあとに変換を
復元します。

rotated ctx render = do 

  save ctx 

  rotate Math.pi ctx 

  render 

  restore ctx 

こういったよくある使いかたの⾼階関数を利⽤した抽象化として、 purescript-canvas ライ
ブラリでは元のコンテキスト状態を維持しながらいくつかのキャンバスアクションを実⾏する
withContext 関数が提供されています。

withContext :: forall eff a. Context2D ->  

                             Eff (canvas :: Canvas | eff) a -> 

                             Eff (canvas :: Canvas | eff) a           

withContext を使うと、先ほどの rotated 関数を次のように書き換えることができます。



rotated ctx render = withContext ctx do 

    rotate Math.pi ctx 

    render 

9 .8  ⼤域的な変更可能状態

この節では purescript-refs パッケージを使って Eff モナドの別の作⽤について実演して
みます。

Control.Monad.Eff.Ref モジュールでは⼤域的に変更可能な参照のための型構築⼦、および
関連する作⽤を提供します。

> import Control.Monad.Eff.Ref 

 

> :kind Ref 

Type -> Type 

 

> :kind REF 

Control.Monad.Eff.Effect 

型 RefVal a の値は型 a 値を保持する変更可能な領域への参照で、前の章で⾒た STRef h a
によく似ています。その違いは、 ST 作⽤は runST を⽤いて除去することができますが、 Ref
作⽤はハンドラを提供しないということです。 ST は安全に局所的な状態変更を追跡するために
使⽤されますが、 Ref は⼤域的な状態変更を追跡するために使⽤されます。そのため、 Ref は
慎重に使⽤する必要があります。

Example/Refs.purs ファイルには canvas 要素上のマウスクリックを追跡するのに Ref 作⽤
を使⽤する例が含まれています。

このコー  ドでは最初に newRef アクションを使って値 0 で初期化された領域への新しい参照
を作成しています。

  clickCount <- newRef 0 

クリックイベントハンドラの内部では、 modifyRef アクションを使⽤してクリック数を更新
しています。

    modifyRef clickCount (\count -> count + 1) 



readRef アクションは新しいクリック数を読み取るために使われています。

    count <- readRef clickCount 

render 関数では、クリック数に応じて変換を矩形に適⽤しています。

    withContext ctx do 

      let scaleX = Math.sin (toNumber count * Math.pi / 4.0) + 1.5 

      let scaleY = Math.sin (toNumber count * Math.pi / 6.0) + 1.5 

 

      translate { translateX: 300.0, translateY:  300.0 } ctx 

      rotate (toNumber count * Math.pi / 18.0) ctx 

      scale { scaleX: scaleX, scaleY: scaleY } ctx 

      translate { translateX: -100.0, translateY: -100.0 } ctx 

 

      fillPath ctx $ rect ctx 

        { x: 0.0 

        , y: 0.0 

        , w: 200.0 

        , h: 200.0 

        } 

このアクションでは元の変換を維持するために withContext を使⽤しており、それから続く
変換を順に適⽤しています(変換が下から上に適⽤されることを思い出してください)。

中⼼が原点に来るように、矩形を (-100, -100) 平⾏移動します。
矩形を原点を中⼼に拡⼤縮⼩します。
矩形を原点を中⼼に 10 度の倍数だけ回転します。
中⼼がキャンバスの中⼼に位置するように⻑⽅形を (300、300) だけ平⾏移動します。

このコード例をビルドしてみましょう。

$ pulp build -O --main Example.Refs --to dist/Main.js 

html/index.html ファイルを開いてみましょう。何度かキャンバスをクリックすると、キャ
ンバスの中⼼の周りを回転する緑の四⾓形が表⽰されるはずです。



演 習

1. (簡単) パスの線描と塗りつぶしを同時に⾏う⾼階関数を書いてください。その関
数を使⽤して Random.purs 例を書きなおしてください。

2. (やや難しい) Random 作⽤と DOM 作⽤を使⽤して、マウスがクリックされたとき
にキャンバスに無作為な位置、⾊、半径の円を描画するアプリケーションを作成し
てください。

3. (やや難しい) シーンを指定された座標を中⼼に回転する関数を書いてください。
ヒント：最初にシーンを原点まで平⾏移動しましょう。

9 .9  L-Systems

この章の最後の例として、 purescript-canvas パッケージを使⽤してL-
systems(Lindenmayer systems)を描画する関数を記述します。

L-Systemsはアルファベット、つまり初期状態となるアルファベットの⽂字列と、⽣成規則の
集合で定義されています。各⽣成規則は、アルファベットの⽂字をとり、それを置き換える⽂字
の配列を返します。この処理は⽂字の初期配列から始まり、複数回繰り返されます。

もしアルファベットの各⽂字がキャンバス上で実⾏される命令と対応付けられていれば、その
指⽰に順番に従うことでL-Systemsを描画することができます。

たとえば、アルファベットが⽂字 L (左回転)、 R (右回転)、 F (前進)で構成されていたとし
ます。また、次のような⽣成規則を定義します。

L -> L 

R -> R 

F -> FLFRRFLF 

配列 "FRRFRRFRR" から始めて処理を繰り返すと、次のような経過を辿ります。

FRRFRRFRR 

FLFRRFLFRRFLFRRFLFRRFLFRRFLFRR 

FLFRRFLFLFLFRRFLFRRFLFRRFLFLFLFRRFLFRRFLFRRFLF... 



この命令群に対応する線分パスをプロットすると、コッホ曲線と呼ばれる曲線に近似します。
反復回数を増やすと、曲線の解像度が増加していきます。

それでは型と関数の⾔語へとこれを翻訳してみましょう。

アルファベットの選択肢は型の選択肢によって表すことができます。今回の例では、以下のよ
うな型で定義することができます。

data Alphabet = L | R | F 

このデータ型では、アルファベットの⽂字ごとに１つづつデータ構築⼦が定義されています。

⽂字の初期配列はどのように表したらいいでしょうか。単なるアルファベットの配列でいいで
しょう。これを Sentence と呼ぶことにします。

type Sentence = Array Alphabet 

 

initial :: Sentence 

initial = [F, R, R, F, R, R, F, R, R] 

⽣成規則は Alphabet から Sentence への関数として表すことができます。

productions :: Alphabet -> Sentence 

productions L = [L] 

productions R = [R] 

productions F = [F, L, F, R, R, F, L, F] 

これはまさに上記の仕様をそのまま書き写したものです。

これで、この形式の仕様を受け取りキャンバスに描画する関数 lsystem を実装することがで
きます。 lsystem はどのような型を持っているべきでしょうか。この関数は初期状態 initial
と⽣成規則 productions のような値だけでなく、アルファベットの⽂字をキャンバスに描画す
る関数を引数に取る必要があります。

lsystem の型の最初の⼤まかな設計としては、次のようになるかもしれません。

forall eff. Sentence 

         -> (Alphabet -> Sentence) 

         -> (Alphabet -> Eff (canvas :: Canvas | eff) Unit) 

         -> Int 

         -> Eff (canvas :: Canvas | eff) Unit 



最初の2つの引数の型は、値 initial と productions に対応しています。

3番⽬の引数は、アルファベットの⽂字を取り、キャンバス上のいくつかのアクションを実⾏
することによって翻訳する関数を表します。この例では、⽂字 L は左回転、⽂字 R で右回転、
⽂字 F は前進を意味します。

最後の引数は、実⾏したい⽣成規則の繰り返し回数を表す数です。

最初に気づくことは、現在の lsystem 関数は Alphabet 型だけで機能しますが、どんなアル
ファベットについても機能すべきですから、この型はもっと⼀般化されるべきです。それでは、
量⼦化された型変数 a について、 Alphabet と Sentence を a と Array a で置き換えましょ
う。

forall a eff. Array a 

           -> (a -> Array a) 

           -> (a -> Eff (canvas :: Canvas | eff) Unit) 

           -> Int 

           -> Eff (canvas :: Canvas | eff) Unit 

次に気付くこととしては、「左回転」と「右回転」のような命令を実装するためには、いくつ
かの状態を管理する必要があります。具体的に⾔えば、その時点でパスが向いている⽅向を状態
として持たなければなりません。計算を通じて状態を関数に渡すように変更する必要がありま
す。ここでも lsystem 関数は状態がどんな型でも動作しなければなりませんから、型変数 s を
使⽤してそれを表しています。

型 s を追加する必要があるのは3箇所で、次のようになります。

forall a s eff. Array a 

             -> (a -> Array a) 

             -> (s -> a -> Eff (canvas :: Canvas | eff) s) 

             -> Int 

             -> s 

             -> Eff (canvas :: Canvas | eff) s 

まず追加の引数の型として lsystem に型 s が追加されています。この引数はL-Systemの初期
状態を表しています。

型 s は引数にも現れますが、翻訳関数( lsystem の第3引数)の返り値の型としても現れます。
翻訳関数は今のところ、引数としてL-Systemの現在の状態を受け取り、返り値として更新され
た新しい状態を返します。

この例の場合では、次のような型を使って状態を表す型を定義することができます。



type State = 

  { x :: Number 

  , y :: Number 

  , theta :: Number 

  } 

プロパティ x と y はパスの現在の位置を表しており、プロパティ theta は現在の向きを表
しており、ラジアンで表された⽔平線に対するパスの⾓度です。

システムの初期状態としては次のようなものが考えられます。

initialState :: State 

initialState = { x: 120.0, y: 200.0, theta: 0.0 } 

それでは、 lsystem 関数を実装してみます。定義はとても単純であることがわかるでしょ
う。

lsystem は第４引数の値(型 Number )に応じて再帰するのが良さそうです。再帰の各ステップ
では、⽣成規則に従って状態が更新され、現在の⽂が変化していきます。このことを念頭に置き
つつ、まずは関数の引数の名前を導⼊して、補助関数に処理を移譲することから始めましょう。

lsystem :: forall a s eff. Array a 

        -> (a -> Array a) 

        -> (s -> a -> Eff (canvas :: Canvas | eff) s) 

        -> Int 

        -> s 

        -> Eff (canvas :: Canvas | eff) s 

lsystem init prod interpret n state = go init n 

  where 

go 関数は第2引数に応じて再帰することで動きます。 n がゼロであるときと n がゼロでない
ときの2つの場合で分岐します。

n がゼロの場合では再帰は完了し、解釈関数に応じて現在の⽂を解釈します。ここでは引数
として与えられている、

型 Array a の⽂
型 s の状態
型 s -> a -> Eff (canvas :: Canvas | eff) s の関数

を参照することができます。これらの引数の型を考えると、以前定義した foldM の呼び出し
にちょうど対応していることがわかります。 foldM は purescript-control パッケージでも定



義されています。

  go s 0 = foldM interpret state s 

ゼロでない場合ではどうでしょうか。その場合は、単に⽣成規則を現在の⽂のそれぞれの⽂字
に適⽤して、その結果を連結し、そしてこの処理を再帰します。

  go s n = go (concatMap prod s) (n - 1) 

これだけです！ foldM や concatMap のような⾼階関数を使うと、このようにアイデアを簡潔
に表現することができるのです。

しかし、まだ完全に終わったわけではありません。ここで与えた型は、実際はまだ特殊化され
すぎています。この定義ではキャンバスの操作が実装のどこにも使われていないことに注⽬して
ください。それに、まったく Eff モナドの構造を利⽤していません。実際には、この関数はど
んなモナド m についても動作するのです！

この章に添付されたソースコードで定義されている、 lsystem のもっと⼀般的な型は次のよ
うになっています。

lsystem :: forall a m s . Monad m => 

           Array a 

        -> (a -> Array a) 

        -> (s -> a -> m s) 

        -> Int 

        -> s 

        -> m s 

この型が⾔っているのは、この翻訳関数はモナド m で追跡される任意の副作⽤をまったく⾃
由に持つことができる、ということだと理解することができます。キャンバスに描画したり、ま
たはコンソールに情報を出⼒するかもしれませんし、失敗や複数の戻り値に対応しているかもし
れません。こういった様々な型の副作⽤を使ったL-Systemを記述してみることを読者にお勧め
します。

この関数は実装からデータを分離することの威⼒を⽰す良い例となっています。この⼿法の利
点は、複数の異なる⽅法でデータを解釈する⾃由が得られることです。 lsystem は２つの⼩さ
な関数へと分解することができるかもしれません。ひとつめは concatMap の適⽤の繰り返しを
使って⽂を構築するもので、ふたつめは foldM を使って⽂を翻訳するものです。これは読者の
演習として残しておきます。



それでは翻訳関数を実装して、この章の例を完成させましょう  。 lsystem の型は型シグネチ
ャが⾔っているのは、翻訳関数の型は、何らかの型 a と s 、型構築⼦ m について、
s -> a -> m s でなければならないということです。今回は a を Alphabet 、 s を State 、

モナド m を Eff (canvas :: Canvas) というように選びたいということがわかっています。こ
れにより次のような型になります。

interpret :: State -> Alphabet -> Eff (canvas :: Canvas) State 

この関数を実装するには、 Alphabet 型の3つのデータ構築⼦それぞれについて処理する必要
があります。⽂字 L (左回転)と R (右回転)の解釈では、 theta を適切な⾓度へ変更するように
状態を更新するだけです。

interpret state L = pure $ state { theta = state.theta - Math.pi / 3 } 

interpret state R = pure $ state { theta = state.theta + Math.pi / 3 } 

⽂字 F (前進)を解釈するには、パスの新しい位置を計算し、線分を描画し、状態を次のよう
に更新します。

interpret state F = do 

  let x = state.x + Math.cos state.theta * 1.5 

      y = state.y + Math.sin state.theta * 1.5 

  moveTo ctx state.x state.y 

  lineTo ctx x y 

  pure { x, y, theta: state.theta } 

この章のソースコードでは、名前 ctx を参照できるようにするために、 interpret 関数は
main 関数内で let 束縛を使⽤して定義されていることに注意してください。 State 型がコン

テキストを持つように変更することは可能でしょうが、それはこのシステムの状態の変化部分で
はないので不適切でしょう。

このL-Systemsを描画するには、次のような strokePath アクションを使⽤するだけです。

strokePath ctx $ lsystem initial productions interpret 5 initialState 

L-Systemをコンパイルし、

$ pulp build -O --main Example.LSystem --to dist/Main.js 



html/index.html を開いてみましょう。キャンバスにコッホ曲線が描画されるのがわかると
思います。

演 習

1. (簡単) strokePath の代わりに fillPath を使⽤するように、上のL-Systemsの
例を変更してください。ヒント： closePath の呼び出しを含め、 moveTo の呼び
出しを interpret 関数の外側に移動する必要があります。

2. (簡単) 描画システムへの影響を理解するために、コード中の様々な数値の定数を
変更してみてください。

3. (やや難しい) lsystem 関数を2つの⼩さな関数に分割してください。ひとつめは
concatMap の適⽤の繰り返しを使⽤して最終的な結果を構築するもので、ふたつ

めは foldM を使⽤して結果を解釈するものでなくてはなりません。

4. (やや難しい) setShadowOffsetX アクション、 setShadowOffsetY アクション、
setShadowBlur アクション、 setShadowColor アクションを使い、塗りつぶされ

た図形にドロップシャドウを追加してください。ヒント： PSCi を使って、これら
の関数の型を調べてみましょう。

5. (やや難しい) 向きを変えるときの⾓度の⼤きさは今のところ⼀定( pi/3 )です。
その代わりに、 Alphabet データ型の中に⾓度の⼤きさを追加して、⽣成規則によ
って⾓度を変更できるようにしてください。

type Angle = Number 

 

data Alphabet = L Angle | R Angle | F Angle 

⽣成規則でこの新しい情報を使うと、どんな⾯⽩い図形を作ることができるでし
ょうか。

6. (難しい) L (60度左回転 )、 R (60度右回転)、 F (前進)、 M (これも前進)とい
う４つの⽂字からなるアルファベットでL-Systemが与えられたとします。

このシステムの⽂の初期状態は、単⼀の⽂字 M です。



このシステムの⽣成規則は次のように指定されています。

L -> L 

R -> R 

F -> FLMLFRMRFRMRFLMLF 

M -> MRFRMLFLMLFLMRFRM 

このL-Systemを描画してください。注意：最後の⽂のサイズは反復回数に従っ
て指数関数的に増⼤するので、⽣成規則の繰り返しの回数を削減することが必要に
なります。

ここで、⽣成規則における L と M の間の対称性に注⽬してください。ふたつの
「前進」命令は、次のようなアルファベット型を使⽤すると、 Boolean 値を使っ
て区別することができます。

data Alphabet = L | R | F Boolean 

このアルファベットの表現を使⽤して、もう⼀度このL-Systemを実装してくだ
さい。

7. (難しい) 翻訳関数で別のモナド m を使ってみましょう。 Trace 作⽤を利⽤して
コンソール上にL-Systemを出⼒したり、 Random 作⽤を利⽤して状態の型に無作
為の突然変異を適⽤したりしてみてください。

まとめ

この章では、 purescript-canvas ライブラリを使⽤することにより、PureScriptから
HTML5 Canvas APIを使う⽅法について学びました。マップや畳み込み、レコードと⾏多型、
副作⽤を扱うための Eff モナドなど、これまで学んできた⼿法を利⽤した実⽤的な例について
多く⾒ました。

この章の例では、⾼階関数の威⼒を⽰すとともに、実装からのデータの分離も実演してみせま
した。これは例えば、代数データ型を使⽤してこれらの概念を次のように拡張し、描画関数から
シーンの表現を完全に分離できるようになります。



data Scene = Rect Rectangle 

           | Arc Arc 

           | PiecewiseLinear (Array Point) 

           | Transformed Transform Scene 

           | Clipped Rectangle Scene 

           | ... 

この⼿法は purescript-drawing パッケージでも採⽤されており、描画前にさまざまな⽅法
でデータとしてシーンを操作することができるという柔軟性をもたらしています。

次の章では、PureScriptの外部関数インタフェース(foreign function interface)を使って、既存
のJavaScriptの関数をラップした purescript-canvas のようなライブラリを実装する⽅法につ
いて説明します。



第10章 外部関数インタフェース

10 .1  この章の⽬標

この章では、PureScriptコードからJavaScriptコードへの呼び出し、およびその逆を可能にす
る、PureScriptの外部関数インタフェース(foreign function interface, FFI)を紹介します。これか
ら扱うのは次のようなものです。

PureScriptから純粋なJavaScript関数を呼び出す⽅法
既存のJavaScriptコードに基づいて、作⽤型と Eff モナドと⼀緒に使⽤する新しいアクションを
作成する⽅法
JavaScriptからPureScriptコードを呼び出す⽅法
実⾏時のPureScriptの値の表現を知る⽅法
purescript-foreign パッケージを使⽤して型付けされていないデータを操作する⽅法

この章の終わりにかけて、再び住所録のコード例について検討します。この章の⽬的は、FFI
を使ってアプリケーションに次のような新しい機能を追加することです。

ポップアップ通知でユーザーに警告する
フォームのデータを直列化してブラウザのローカルストレージに保存し、アプリケーションが再
起動したときにそれを再読み込みする

10 .2  プロジェクトの準備

このモジュールのソースコードは、第3章、第7章及び第8章の続きになります。今回もそれぞ
れのディレクトリから適切なソースファイルがソースファイルに含められています。

この章では、2つの新しいBower依存関係を追加します。

. purescript-foreign - データ型と関数を提供しています。

. purescript-foreign-generic - データ型ジェネリックプログラミングを操作するためのデータ
型と関数を提供します。

注意：ウェブページがローカルファイルから配信されているときに起こる、ローカルストレー
ジとブラウザ特有の問題を避けるために、この章の例を実⾏するには、HTTPを経由してこの章
のプロジェクトを実⾏する必要があります。



10 .3  免責事項

JavaScriptとの共同作業をできる限り簡単にするため、PureScriptは単純な多⾔語関数インタ
フェースを提供します。しかしながら、FFIはPureScriptの⾼度な機能であることには留意して
いただきたいと思います。FFIを安全かつ効率的に使⽤するには、扱うつもりであるデータの実
⾏時の表現についてよく理解していなければなりません。この章では、PureScriptの標準ライブ
ラリのコードに関連する、そのような理解を与えることを⽬指しています。

PureScriptのFFIはとても柔軟に設計されています。実際には、外部関数に最低限の型だけを
与えるか、それとも型システムを利⽤して外部のコードの誤った使い⽅を防ぐようにするか、開
発者が選ぶことができるということを意味しています。標準ライブラリのコードは、後者の⼿法
を好む傾向にあります。

簡単な例としては、JavaScriptの関数で戻り値が null をされないことを保証することはでき
ません。実のところ、既存のJavaScriptコードはかなり頻繁に null を返します！しかし、
PureScriptの型は通常null値を持っていません。そのため、FFIを使ってJavaScriptコードのイン
ターフェイスを設計するときは、これらの特殊な場合を適切に処理するのは開発者の責任です。

10 .4  JavaScr iptからPureScr iptを呼び出す

少なくとも単純な型を持った関数については、JavaScriptからPureScript関数を呼び出すのは
とても簡単です。

例として以下のような簡単なモジュールを⾒てみましょう。

module Test where 

 

gcd :: Int -> Int -> Int 

gcd 0 m = m 

gcd n 0 = n 

gcd n m 

  | n > m     = gcd (n - m) m 

  | otherwise = gcd (m - n) n 

この関数は、減算を繰り返すことによって2つの数の最⼤公約数を⾒つけます。関数を定義す
るのにPureScriptを使いたくなるかもしれない良い例となっていますが、JavaScriptからそれを
呼び出すためには条件があります。 PureScriptでパターン照合と再帰を使⽤してこの関数を定義
するのは簡単で、実装する開発者は型検証器の恩恵を受けることができます。



この関数をJavaScriptから呼び出す⽅法を理解するには、PureScriptの関数は常に引数がひと
つのJavaScript関数へと変換され、引数へは次のようにひとつづつ適⽤していかなければならな
いことを理解するのが重要です。

var Test = require('Test'); 

Test.gcd(15)(20); 

ここでは、コードがPureScriptモジュールをCommonJSモジュールにコンパイルする
pulp build でコンパイルされていると仮定しています。 そのため、 require を使って Test

モジュールをインポートした後、 Test オブジェクトの gcd 関数を参照することができまし
た。

pulp build -O --to file.js を使⽤して、ブラウザ⽤のJavaScriptコードをバンドルするこ
ともできます。 その場合、グローバルなPureScript名前空間から Test モジュールにアクセスし
ます。デフォルトは PS です。

var Test = PS.Test; 

Test.gcd(15)(20); 

10 .5  名前の⽣成を理解する

PureScriptはコード⽣成時にできるだけ名前を保存することを⽬的としています。具体的に
は、トップレベルでの宣⾔では、JavaScriptのキーワードでなければ任意の識別⼦が保存されま
す。

識別⼦としてJavaScriptの予約語を使う場合は、名前はダブルダラー記号でエスケープされま
す。たとえば、次のPureScriptコードを考えてみます。

null = [] 

これは以下のようなJavaScriptへコンパイルされます。

var $$null = []; 

また、識別⼦に特殊⽂字を使⽤したい場合は、単⼀のドル記号を使⽤してエスケープされま
す。たとえば、このPureScriptコードを考えます。



example' = 100 

これは以下のJavaScriptにコンパイルされます。

var example$prime = 100; 

この⽅式は、ユーザー定義の中置演算⼦の名前を⽣成するためにも使⽤されます。

(%) a b = ... 

これは次のようにコンパイルされます。

var $percent = ... 

コンパイルされたPureScriptコードがJavaScriptから呼び出されることを意図している場合、
識別⼦は英数字のみを使⽤し、JavaScriptの予約語を避けることをお勧めします。ユーザ定義演
算⼦がPureScriptコードでの使⽤のために提供される場合でも、JavaScriptから使うための英数
字の名前を持った代替関数を提供しておくことをお勧めします。

10 .6  実⾏時のデータ表現

型はプログラムがある意味で「正しい」ことをコンパイル時に判断できるようにします。つま
り、実⾏時には中断されません。しかし、これは何を意味するのでしょうか？PureScriptでは式
の型は実⾏時の表現と互換性がなければならないことを意味します。

そのため、PureScriptとJavaScriptコードを⼀緒に効率的に使⽤できるように、実⾏時のデー
タ表現について理解することが重要です。これは、与えられた任意のPureScriptの式について、
その値が実⾏時にどのように評価されるかという挙動を理解できるべきであることを意味してい
ます。

PureScriptの式は、実⾏時に特に単純な表現を持っているということは朗報です。実際に標準
ライブラリのコードについて、その型を考慮すれば式の実⾏時のデータ表現を把握することが可
能です。

単純な型については、対応関係はほとんど⾃明です。たとえば、式が型 Boolean を持ってい
れば、実⾏時のその値 v は typeof v === 'boolean' を満たします。つまり、型 Boolean の式



は true もしくは false のどちらか⼀⽅の(JavaScriptの)値へと評価されます。実のところ、
null や undefined に評価される、型 Boolean のPureScriptの式はありません。

Number と String の型の式についても同様のことが成り⽴ちます。 Number 型の式は null
でないJavaScriptの数へと評価されますし、 String 型の式は null でないJavaScriptの⽂字列へ
と評価されます。

もっと複雑な型についてはどうでしょうか？

すでに⾒てきたように、PureScriptの関数は引数がひとつのJavaScriptの関数に対応していま
す。厳密に⾔えば、任意の型 a 、 b について、式 f の型が a -> b で、式 x が型 a につい
ての適切な実⾏時表現の値へと評価されるなら、 f はJavaScriptの関数へと評価され、 x を評
価した結果に f を適⽤すると、それは型 b の適切な実⾏時表現を持ちます。簡単な例として
は、 String -> String 型の式は、 null でないJavaScript⽂字列から null でないJavaScript⽂
字列への関数へと評価されます。

ご想像のとおり、PureScriptの配列はJavaScriptの配列に対応しています。しかし、PureScript
の配列は均質であり、つまりすべての要素が同じ型を持っていることは覚えておいてください。
具体的には、もしPureScriptの式 e が任意の型 a について型 Array a を持っているなら、 e
はすべての要素が型 a の適切な実⾏時表現を持った( null でない)JavaScript配列へと評価され
ます。

PureScriptのレコードがJavaScriptのオブジェクトへと評価されることはすでに⾒てきまし
た。ちょうど関数と配列の場合のように、そのラベルに関連付けられている型を考慮すれば、レ
コードのフィールドのデータの実⾏時の表現についても推論することができます。もちろん、レ
コードのそれぞれのフィールドは、同じ型である必要はありません。

10 .7  代数的データ型の実⾏時表現

PureScriptコンパイラは、代数的データ型のすべての構築⼦についてそれぞれ関数を定義し、
新たなJavaScriptオブジェクト型を作成します。これらの構築⼦はこれらのプロトタイプに基づ
いて新しいJavaScriptオブジェクトを作成する関数に対応しています。

たとえば、次のような単純な代数的データ型を考えてみましょう。

data ZeroOrOne a = Zero | One a 



PureScriptコンパイラは、次のようなコードを⽣成します。

function One(value0) { 

    this.value0 = value0; 

}; 

 

One.create = function (value0) { 

    return new One(value0); 

}; 

 

function Zero() { 

}; 

 

Zero.value = new Zero(); 

ここで2つのJavaScriptオブジェクト型 Zero と One を⾒てください。JavaScriptの予約語
new を使⽤すると、それぞれの型の値を作成することができます。引数を持つ構築⼦について

は、コンパイラは value0 、 value1 などと呼ばれるフィールドに対応するデータを格納しま
す。

PureScriptコンパイラは補助関数も⽣成します。引数のない構築⼦については、コンパイラは
構築⼦が使われるたびに new 演算⼦を使うのではなく、データを再利⽤できるように value プ
ロパティを⽣成します。ひとつ以上の引数を持つ構築⼦では、適切な表現を持つ引数を取り適切
な構築⼦を適⽤する create 関数をコンパイラは⽣成します。

２引数以上の構築⼦についてはどうでしょうか？その場合でも、PureScriptコンパイラは新し
いオブジェクト型と補助関数を作成します。しかし今回は、補助関数は2引数のカリー化された
関数です。たとえば、次のような代数的データ型を考えます。

data Two a b = Two a b 

このコードからは、次のようなJavaScriptコードを⽣成されます。

function Two(value0, value1) { 

    this.value0 = value0; 

    this.value1 = value1; 

}; 

 

Two.create = function (value0) { 

    return function (value1) { 

        return new Two(value0, value1); 

    }; 

}; 



ここで、オブジェクト型 Two の値は予約語 new または Two.create 関数を使⽤すると作成す
ることができます。

newtypeの場合はまた少し異なります。newtypeは単⼀の引数を取る単⼀の構築⼦を持つよう
制限された代数的データ型であることを思い出してください。この場合には、実際はnewtypeの
実⾏時表現は、その引数の型と同じになります。

例えば、電話番号を表す次のようなnewtypeを考えます。

newtype PhoneNumber = PhoneNumber String 

これは実⾏時にはJavaScriptの⽂字列として表されます。newtypeは型安全性の追加の層を提供
しますが、実⾏時の関数呼び出しのオーバーヘッドがないので、ライブラリを設計するのに役に
⽴ちます。

10 .8  量化された型の実⾏時表現

量化された型(多相型)の式は、制限された表現を実⾏時に持っています。実際には、量化され
た型の式が⽐較的少数与えられたとき、とても効率的に解決できることを意味しています。

例えば、次の多相型を考えてみます。

forall a. a -> a 

この型を持っている関数にはどんなものがあるでしょうか。少なくともひとつはこの型を持つ
関数が存在しています。すなわち、 Prelude で定義されている恒等関数 id です。

id :: forall a. a -> a 

id a = a 

実のところ、 id の関数はこの型の唯⼀の(全)関数です！これは確かに間違いなさそうに⾒え
ますが(この型を持った id とは明らかに異なる式を書こうとしてみてください)、これを確かめ
るにはどうしたらいいでしょうか。これは型の実⾏時表現を考えることによって確認することが
できます。

量化された型 forall a. t の実⾏時表現はどうなっているのでしょうか。さて、この型の実
⾏時表現を持つ任意の式は、型 a をどのように選んでも型 t の適切な実⾏時表現を持っていな



ければなりません。上の例では、型 forall a. a -> a の関数は、 String -> String 、
Number -> Number 、 Array Boolean -> Array Boolean などといった型について、適切な実

⾏時表現を持っていなければなりません。 これらは、数から数、⽂字列から⽂字列の関数でな
くてはなりません。

しかし、それだけでは⼗分ではありません。量化された型の実⾏時表現は、これよりも更に厳
しくなります。任意の式がパラメトリック多相的でなければなりません。つまり、その実装にお
いて、引数の型についてのどんな情報も使うことができないのです。この追加の条件は、考えら
れる多相型のうち、次のようなJavaScriptの関数として問題のある実装を禁⽌します。

function invalid(a) { 

    if (typeof a === 'string') { 

        return "Argument was a string."; 

    } else { 

        return a; 

    } 

} 

確かにこの関数は⽂字列から⽂字列、数から数へというような関数ではありますが、追加の条
件を満たしていません。引数の実⾏時の型を調べているからです。したがって、この関数は型
forall a. a -> a の正しい実装だとはいえないのです。

関数の引数の実⾏時の型を検査することができなければ、唯⼀の選択肢は引数をそのまま返す
ことだけであり、したがって id は、 forall a. a -> a のまったく唯⼀の実装なのです。

パラメトリック多相(parametric polymorphism)とパラメトリック性(parametricity)についての
詳しい議論は本書の範囲を超えています。しかしながら、PureScriptの型は、実⾏時に消去され
ているので、PureScriptの多相関数は(FFIを使わない限り)引数の実⾏時表現を検査することが
できないし、この多相的なデータの表現は適切であることに注意してください。

10 .9  制約された型の実⾏時表現

型クラス制約を持つ関数は、実⾏時に⾯⽩い表現を持っています。関数の振る舞いはコンパイ
ラによって選ばれた型クラスのインスタンスに依存する可能性があるため、関数には選択したイ
ンスタンスから提供された型クラスの関数の実装が含まれてた型クラス辞書(type class
dictionary)と呼ばれる追加の引数が与えられています。



例えば、 Show 型クラスを使⽤している制約された型を持つ、次のような単純なPureScript関
数について考えます。

shout :: forall a. Show a => a -> String 

shout a = show a <> "!!!" 

このコードから⽣成されるJavaScriptは次のようになります。

var shout = function (dict) { 

    return function (a) { 

        return show(dict)(a) + "!!!"; 

    }; 

}; 

shout は１引数ではなく、２引数の(カリー化された)関数にコンパイルされていることに注
意してください。最初の引数 dict は Show 制約の型クラス辞書です。 dict には型 a の
show 関数の実装が含まれています。

最初の引数として明⽰的にPreludeの型クラス辞書を渡すと、JavaScriptからこの関数を呼び出
すことができます。

shout(require('Prelude').showNumber)(42); 

演 習

1. (簡単) これらの型の実⾏時の表現は何でしょうか。

forall a. a 

forall a. a -> a -> a 

forall a. Ord a => Array a -> Boolean 

これらの型を持つ式についてわかることはなんでしょうか。

2. (やや難しい) pulp build を使ってコンパイルし、NodeJSの require 関数を使
ってモジュールをインポートすることで、JavaScriptから purescript-arrays ライ
ブラリの関数を使ってみてください。ヒント：⽣成されたCommonJSモジュールが



NodeJSモジュールのパスで使⽤できるように、出⼒パスを設定する必要がありま
す。

10 .10 PureScr iptからのJavaScr iptコードを使う

PureScriptからJavaScriptコードを使⽤する最も簡単な⽅法は、外部インポート宣⾔(foreign
import declaration)を使⽤し、既存のJavaScriptの値に型を与えることです。外部インポート宣⾔
では、対応するJavaScriptの宣⾔を外部JavaScriptモジュール(foreign JavaScript module)に持つ
必要があります。

たとえば、特殊⽂字をエスケープすることによりURIのコンポーネントを符号化する
JavaScriptの encodeURIComponent 関数について考えてみます。

$ node 

 

node> encodeURIComponent('Hello World') 

'Hello%20World' 

null でない⽂字列から null でない⽂字列への関数であり、副作⽤を持っていないので、こ
の関数はその型 String -> String について適切な実⾏時表現を持っています。

次のような外部インポート宣⾔を使うと、この関数に型を割り当てることができます。

module Data.URI where 

 

foreign import encodeURIComponent :: String -> String 

また、外部JavaScriptモジュールを書く必要があります。上記のモジュール
を src/Data/URI.purs として保存した場合、次のような外部JavaScriptモジュール
を src/Data/URI.js として保存します。

"use strict"; 

 

exports.encodeURIComponent = encodeURIComponent; 



Pulpは src ディレクトリにある .js ファイルを⾒つけ、それを外部JavaScriptモジュールとし
てコンパイラに提供します。

JavaScriptの関数と値は、通常のCommonJSモジュールと同じように exports オブジェクトに
代⼊することで、外部JavaScriptモジュールからエクスポートされます。 purs コンパイラは、こ
のモジュールを通常のCommonJSモジュールのように扱い、コンパイルされたPureScriptモジュ
ールへの依存関係として追加します。しかし、 psc-bundle や pulp build -O --to を使ってブ
ラウザ向けのコードをバンドルするときは、上記のパターンに従い、プロパティ代⼊を使っ
て exports オブジェクトにエクスポートする値を代⼊することがとても重要です。 これ
は、 psc-bundle がこの形式を認識し、未使⽤のJavaScriptのエクスポートをバンドルされたコ
ードから削除できるようにするためです。

これら2つの部品を使うことで、PureScriptで書かれた関数のように、PureScriptか
ら encodeURIComponent 関数を使うことができます。たとえば、この宣⾔をモジュールとして保
存してPSCiにロードすると、上記の計算を再現できます。

$ pulp repl 

 

> import Data.URI 

> encodeURIComponent "Hello World" 

"Hello%20World" 

この⼿法は簡単なJav  aScriptの値には適していますが、もっと複雑な値に使うには限界があり
ます。ほとんどの既存のJavaScriptコードは、基本的なPureScriptの型の実⾏時表現によって課せ
られた厳しい条件を満たしていないからです。このような場合のためには、適切な実⾏時表現に
従うことを強制するようにJavaScriptコードをラップするという別の⽅法があります。

10 .11 JavaScr iptの値のラッピング

これはPureScriptの型を与えるためにJavaScriptコードの既存の部分をラップする場合に特に
便利です。このようにしたくなる理由はいくつかあります。

関数が複数の引数を取るが、カリー化した関数と同じように呼び出したい。
任意のJavaScriptの副作⽤を追跡するために、 Eff モナドを使うことができます。
関数の適切な実⾏時表現を与えるために、 null や undefined のような特殊な場合を処理する
ために必要な場合があります。



外部インポート宣⾔を使⽤して、配列についての head 関数を作成したいとしましょう。
JavaScriptでは次のような関数になるでしょう。

function head(arr) { 

    return arr[0]; 

} 

しかし、この関数には問題があります。型 forall a. Array a -> a を与えようとしても、
空の配列に対してこの関数は undefined を返します。したがって、この特殊な場合を処理する
ために、ラッパー関数を使⽤する必要があります。

簡単な⽅法としては、空の配列の場合に例外を投げる⽅法があります。厳密に⾔えば、純粋な
関数は例外を投げるべきではありませんが、デモンストレーションの⽬的ではこれで⼗分です
し、安全性でないということを関数名で⽰しておけばいいでしょう。

foreign import unsafeHead :: forall a. Array a -> a 

JavaScriptモジュールでは、 unsafeHead を以下のように定義することができます。

exports.unsafeHead = function(arr) { 

  if (arr.length) { 

    return arr[0]; 

  } else { 

    throw new Error('unsafeHead: empty array'); 

  } 

}; 

10 .12 外部型の定義

失敗した場合に例外を投げるという⽅法は、あまり理想的とはいえません。PureScriptのコー
ドでは、⽋けた値のような副作⽤は型システムを使って扱うのが普通です。この⼿法としては
Maybe 型構築⼦を使う⽅法もありますが、この節ではFFIを使⽤した別の解決策を扱います。

実⾏時には型 a のように表現されますが undefined の値も許容するような新しい型
Undefined a を定義したいとしましょう。

外部インポート宣⾔とFFIを使うと、外部型(foreign type)を定義することができます。構⽂は
外部関数を定義するのと似ています。



foreign import data Undefined :: Type -> Type 

この予約語 data は値ではなく定義している型を表していることに注意してください。型シグ
ネチャの代わりに、新しい型の種を与えます。このとき、種 Undefined が Type -> Type であ
ると宣⾔しています。つまり Undefined は型構築⼦です。

これで head の定義を簡素化することができます。

exports.head = function(arr) { 

  return arr[0]; 

}; 

PureScriptモジュールには以下を追加します。

foreign import head :: forall a. Array a -> Undefined a 

2点変更がある注意してください。 head 関数の本体ははるかに簡単で、もしその値が未定義
であったとしても arr[0] を返し、型シグネチャはこの関数が未定義の値を返すことがあるとい
う事実を反映するよう変更されています。

この関数はその型の適切な実⾏時表現を持っていますが、型 Undefined a の値を使⽤する⽅
法がありませんので、まったく役に⽴ちません。しかし、FFIを使⽤して新しい関数を幾つか書
くことによって、それを修正することができます！

次の関数は、値が定義されているかどうかを教えてくれる最も基本的な関数です。

foreign import isUndefined :: forall a. Undefined a -> Boolean 

JavaScriptモジュールで次のように簡単に定義できます。

exports.isUndefined = function(value) { 

  return value === undefined; 

}; 

PureScriptから isUndefined と head を⼀緒に使⽤すると、便利な関数を定義することがで
きます。

isEmpty :: forall a. Array a -> Boolean 

isEmpty = isUndefined <<< head 



ここで、定義されたこの外部関数はとても簡単であり、PureScriptの型検査器を使うことによ
る利益をなるべく多く得るということを意味します。⼀般に外部関数は可能な限り⼩さく保ち、
アプリケーションの処理はPureScriptコードへ移動しておくことをおすすめします。

10 .13 多変数 関数

PureScriptのPreludeには、興味深い外部型がいくつかも含まれています。すでに扱ってきたよ
うに、PureScriptの関数型は単⼀の引数だけを取りますが、カリー化(Currying)を使うと複数の
引数の関数をシミュレートすることができます。これには明らかな利点があります。関数を部分
適⽤することができ、関数型の型クラスインスタンスを与えることができます。ただし、効率上
のペナルティが⽣じます。パフォーマンス重視するコードでは、複数の引数を受け⼊れる本物の
JavaScript関数を定義することが必要な場合があります。Preludeではそのような関数を安全に扱
うことができるようにする外部型が定義されています。

たとえば、Preludeの Data.Function.Uncurried モジュールには次の外部型宣⾔があります。

foreign import data Fn2 :: Type -> Type -> Type -> Type 

これは3つの型引数を取る型構築⼦ Fn2 を定義します。 Fn2 a b c は、型 a と b の２つの
引数、返り値の型 c をもつJavaScript関数の型を表現しています。

Preludeでは0引数から10引数までの関数について同様の型構築⼦が定義されています。

次のように mkFn2 関数を使うと、2引数の関数を作成することができます。

import Data.Function.Uncurried 

 

divides :: Fn2 Int Int Boolean 

divides = mkFn2 \n m -> m % n == 0 

そして、 runFn2 関数を使うと、2引数の関数を適⽤することができます。

> runFn2 divides 2 10 

true 

 

> runFn2 divides 3 10 

false 



ここで重要なのは、引数がすべて適⽤されるなら、コンパイラは mkFn2 関数や runFn2 関数
をインライン化するということです。そのため、⽣成されるコードはとてもコンパクトになりま
す。

exports.divides = function(n, m) { 

    return m % n === 0; 

}; 

10 .14 副作⽤の表現

Eff モナドもPreludeの外部型として定義されています。その実⾏時表現はとても簡単です。
型 Eff eff a の式は、任意の副作⽤を実⾏し型 a の適切な実⾏時表現で値を返す、引数なしの
JavaScript関数へと評価されます。

Eff 型の構築⼦の定義は、 Control.Monad.Eff モジュールで次のように与えられています。

foreign import data Eff :: # Effect -> Type -> Type 

Eff 型の構築⼦は作⽤の⾏と返り値の型によってパラメータ化されおり、それが種に反映さ
れることを思い出してください。

簡単な例として、 purescript-random パッケージで定義される random 関数を考えてみてく
ださい。その型は次のようなものでした。

foreign import random :: forall eff. Eff (random :: RANDOM | eff) Number 

random 関数の定義は次のように与えられます。

exports.random = function() { 

  return Math.random(); 

}; 

random 関数は実⾏時には引数なしの関数として表現されていることに注⽬してください。こ
れは乱数⽣成という副作⽤を実⾏しそれを返しますが、返り値は Number 型の実⾏時表現と⼀致
します。それは null でないJavaScriptの数です。



もう少し興味深い例として、 Preludeの Control.Monad.Eff.Console モジュールで定義され
た log 関数を考えてみましょう。 log 関数は次の型を持っています。

foreign import log :: forall eff. String -> Eff (console :: CONSOLE | eff) Unit 

この定義は次のようになっています。

exports.log = function (s) { 

  return function () { 

    console.log(s); 

  }; 

}; 

実⾏時の log の表現は、引数なしの関数を返す、単⼀の引数のJavaScript関数です。内側の関
数はコンソールにメッセージを書き込むという副作⽤を実⾏し、空のレコードを返します。
Unit は空のレコード型のnewtypeとしてPreludeで定義されているので、内側の関数の戻り値の

型は Unit 型の実⾏時表現と⼀致していることに注意してください。

作⽤ RANDOM と CONSOLE も外部型として定義されています。その種は ! 、つまり作⽤である
と定義されています。例えば次のようになります。

foreign import data RANDOM :: Effect 

詳しくはあとで⾒ていきますが、このように新たな作⽤を定義することが可能なのです。

Eff eff a 型の式は、通常のJavaScriptのメソッドのようにJavaScriptから呼び出すことがで
きます。例えば、この main 関数は作⽤の集合 eff と何らかの型 a について Eff eff a とい
う型でなければならないので、次のように実⾏することができます。

require('Main').main(); 

pulp build -O --to または pulp run を使⽤するときは、 main モジュールが定義されてい
ると、この main の呼び出しを⾃動的に⽣成することができます。

10 .15 新しい作⽤の定義



この章のソースコードでは、2つの新しい作⽤が定義されています。最も簡単なのは
Control.Monad.Eff.Alert モジュールで定義された ALERT 作⽤です。これはその計算がポップ

アップウィンドウを使⽤してユーザに警告しうることを⽰すために使われます。

この作⽤は最初に外部型宣⾔を使⽤して定義されています。

foreign import data ALERT :: Effect 

Alert は種 Effect が与えられており、 Alert が型ではなく作⽤であることを⽰していま
す。

次に、 alert アクションが定義されています。 alert アクションはポップアップを表⽰し、
作⽤の⾏に Alert 作⽤を追加します。

foreign import alert :: forall eff. String -> Eff (alert :: ALERT | eff) Unit 

JavaScriptモジュールは簡単で、 alert 関数を exports 変数に代⼊して定義します。

"use strict"; 

 

exports.alert = function(msg) { 

    return function() { 

        window.alert(msg); 

    }; 

}; 

 

このアクションは Control.Monad.Eff.Console モジュールの log アクションととてもよく
似ています。唯⼀の違いは、 log アクションが console.log メソッドを使⽤しているのに対
し、 alert アクションは window.alert メソッドを使⽤していることです。このように、
alert は window.alert が定義されているウェブブラウザのような環境で使⽤することができ

ます。

log の場合のように、 alert 関数は型 Eff (alert :: ALERT | eff) Unit の計算を表現す
るために引数なしの関数を使っていることに注意してください。

この章で定義される２つめの作⽤は、 Control.Monad.Eff.Storage モジュールで定義されて
いる STORAGE 作⽤です。これは計算がWeb Storage APIを使⽤して値を読み書きする可能性があ
ることを⽰すために使われます。

この作⽤も同じように定義されています。



foreign import data STORAGE :: Effect 

Control.Monad.Eff.Storage モジュールには、ローカルストレージから値を取得する
getItem と、ローカルストレージに値を挿⼊したり値を更新する setItem という、２つのアク

ションが定義されています。この⼆つの関数は、次のような型を持っています。

foreign import getItem :: forall eff . String 

  -> Eff (storage :: STORAGE | eff) Foreign 

foreign import setItem :: forall eff . String 

  -> String -> Eff (storage :: STORAGE | eff) Unit 

興味のある読者は、このモジュールのソースコードでこれらのアクションがどのように定義さ
れているか調べてみてください。

setItem はキーと値(両⽅とも⽂字列)を受け取り、指定されたキーでローカルストレージに
値を格納する計算を返します。

getItem の型はもっと興味深いものです。 getItem はキーを引数に取り、キーに関連付けら
れた値をローカルストレージから取得しようとします。 window.localStorage の getItem メソ
ッドは null を返すことがあるので、返り値は String ではなく、 purescript-foreign パッ
ケージの Data.Foreign モジュールで定義されている Foreign になっています。

Data.Foreign は、型付けされていないデータ、もっと⼀般的にいえば実⾏時表現が不明なデ
ータを扱う⽅法を提供しています。

演 習

1. (やや難しい) JavaScriptの Window オブジェクトの confirm メソッドのラッパを
書き、 Control.Monad.Eff.Alert モジュールにその関数を追加してください。

2. (やや難しい) localStorage オブジェクトの removeItem メソッドのラッパを書
き、 Control.Monad.Eff.Storage モジュールに追加してください



10 .16 型付けされていないデータの操作

この節では、型付けされていないデータを、その型の適切な実⾏時表現を持った型付けされた
データに変換する、 Data.Foreign ライブラリの使い⽅について⾒て⾏きます。

この章のコードは、第8章の住所録の上にフォームの⼀番下に保存ボタンを追加することで作
っていきます。保存ボタンがクリックされると、フォームの状態をJSONに直列化し、ローカル
ストレージに格納します。ページが再読み込みされると、JSON⽂書がローカルストレージから
取得され、構⽂解析されます。

Main モジュールではフォームデータの型を定義します。

newtype FormData = FormData 

  { firstName  :: String 

  , lastName   :: String 

  , street     :: String 

  , city       :: String 

  , state      :: String 

  , homePhone  :: String 

  , cellPhone  :: String 

  } 

問題は、このJSONが正しい形式を持っているという保証がないことです。別の⾔い⽅をすれ
ば、JSONが実⾏時にデータの正しい型を表しているかはわかりません。この問題は
purescript-foreign ライブラリによって解決することができます。他にも次のような使いかた

があります。

WebサービスからJSONレスポンス
JavaScriptコードから関数に渡された値

それでは、 PSCi で purescript-foreign 及び purescript-foreign-generic ライブラリを
試してみましょう。

⼆つのモジュールをインポートして起動します。

> import Data.Foreign 

> import Data.Foreign.Generic 

> import Data.Foreign.JSON 

Foreign な値を取得するためには、JSON⽂書を解析するのがいいでしょう。
purescript-foreign では次の2つの関数が定義されています。



parseJSON :: String -> F Foreign 

decodeJSON :: forall a. Decode a => String -> F a 

型構築⼦ F は、実際には Data.Foreign で定義されている型同義語です。

type F = Except (NonEmptyList ForeignError) 

ここで Except は、 Either のように、純粋なコードで例外を処理するためのモナドで
す。 runExcept 関数を使うと、 F モナドの値を Either モナドの値に変換することができます。

この Decode 型クラスは、それらの型が型付けされていないデータから得られることを表して
います。プリミティブ型や配列については型クラスインスタンスがすでに定義されていますが、
独⾃のインスタンスを定義することもできます。

それでは PSCi で readJSON を使⽤していくつかの簡単なJSON⽂書を解析してみましょう。

> import Control.Monad.Except 

 

> runExcept (decodeJSON "\"Testing\"" :: F String) 

Right "Testing" 

 

> runExcept (decodeJSON "true" :: F Boolean) 

Right true 

 

> runExcept (decodeJSON "[1, 2, 3]" :: F (Array Int)) 

Right [1, 2, 3] 

Either モナドでは Right データ構築⼦は成功を⽰していることを思い出してください。し
かし、その不正なJSONや誤った型はエラーを引き起こすことに注意してください。

> runExcept (decodeJSON "[1, 2, true]" :: F (Array Int)) 

(Left (NonEmptyList (NonEmpty (ErrorAtIndex 2 (TypeMismatch "Int" "Boolean")) Nil))) 

purescript-foreign-generic ライブラリはJSON⽂書で型エラーが発⽣した位置を教えてく
れます。

10 .17 nu l lとundefined値の取り扱い



実世界のJSON⽂書にはnullやundefined値が含まれているので、それらも扱えるようにしなけ
ればなりません。

purescript-foreign-generic では、この問題を解決する3種類の構築⼦、 Null 、
Undefined 、 NullOrUndefined が定義されています。先に定義した Undefined 型の構築⼦と

似た⽬的を持っていますが、省略可能な値を表すために Maybe 型の構築⼦を内部的に使ってい
ます。

それぞれの型の構築⼦について、ラップされた値から内側の値を取り出す関数、
runNullOrUndefined が提供されています。 null 値を許容するJSON⽂書を解析するには、
readJSON アクションまで対応する適切な関数を持ち上げます。

> import Prelude 

> import Data.Foreign.NullOrUndefined 

 

> runExcept (unNullOrUndefined <$> decodeJSON "42" :: F (NullOrUndefined Int)) 

(Right (Just 42)) 

 

> runExcept (unNullOrUndefined <$> decodeJSON "null" :: F (NullOrUndefined Int)) 

(Right Nothing) 

それぞれの場合で、型注釈が <$> 演算⼦の右辺に適⽤されています。たとえば、
readJSON "42" は型 F (NullOrUndefined Int) を持っています。 unNullOrUndefined 関数は

最終的な型 F (Maybe Number) 与えるために F まで持ち上げられます。

型 NullOrUndefined Int は数またはnullいずれかの値を表しています。各要素が null をか
もしれない数値の配列のように、より興味深いの値を解析したい場合はどうでしょうか。その場
合には、次のように readJSON アクションまで関数 map unNullOrUndefined を持ち上げます。

> runExcept (map unNullOrUndefined <$> decodeJSON "[1, 2, null]" 

    :: F (Array (NullOrUndefined Int))) (Right [(Just 1),(Just 2),Nothing]) 

⼀般的には、同じ型に異なる直列化戦略を提供するには、newtypesを使って既存の型をラッ
プするのがいいでしょう。 NullOrUndefined それぞれの型は、 Maybe 型構築⼦に包まれた
newtypeとして定義されています。

10 .18 住所録の項⽬の直列化



実のところ、 purescript-foreign-generic クラスはdatatype-generic programmingという
技術を使ってインスタンスの⾃動導出(derive)することが可能なので、 Decode クラスのインス
タンスを⾃分で書く必要はほとんどありません。このテクニックの完全な説明は本書の範囲を超
えていますが、関数を⼀度記述すれば、型⾃体の構造に基づいてさまざまなデータ型に再利⽤す
ることができます。

FormData 型の Decode インスタンスを派⽣させるためには、まず derive キーワードを使っ
て Generic 型クラスのインスタンスを派⽣させます。

derive instance genericFormData :: Generic FormData _ 

そして、 genericDecode 関数を使って、次のように decode 関数を定義します。

instance decodeFormData :: Decode FormData where 

  decode = genericDecode (defaultOptions { unwrapSingleConstructors = true }) 

実際、同じ⽅法で encoder を導出することもできます。

instance encodeFormData :: Encode FormData where 

  encode = genericEncode (defaultOptions { unwrapSingleConstructors = true }) 

デコーダとエンコーダで同じオプションを使⽤することが重要です。そうしないと、エンコー
ドされたJSONドキュメントが正しくデコードされないことがあります。

保存ボタンをクリックすると、JSON⽂書への直列化を⾏う encode 関数に FormData 型の値
が渡されます。 FormData 型はレコードのnewtypeで、 encode が渡された FormData 型の値は
JSONオブジェクトとして直列化されます。これは、JSONエンコーダを定義する際
に unwrapSingleConstructors オプションを指定したためです。

この Decode 型クラスのインスタンスは、 decodeJSON とともにローカル·ストレージから取得
したJSON⽂書を解析するために次のように使われています。

loadSavedData = do 

  item <- getItem "person" 

 

  let 

    savedData :: Either (NonEmptyList ForeignError) (Maybe FormData) 

    savedData = runExcept do 

      jsonOrNull <- traverse readString =<< readNullOrUndefined item 

      traverse decodeJSON jsonOrNull 



savedData アクションは２つの⼿順にわけて FormData 構造を読み取ります。まず、
getItem から得た Foreign 値を解析します。 jsonOrNull の型はコンパイラによって
Null String だと推論されます(読者への演習： この型はどのように推論されているのでしょ

うか？)。 traverse 関数は readJSON を Maybe.String 型の結果の(不⾜しているかもしれな
い)要素へと適⽤するのに使われます。 readJSON について推論される型クラスのインスタンス
はちょうどさっき書いたもので、型 F (Maybe FormData) の値で結果を返します。

traverse の引数には read が最初の⾏で得た結果 jsonOrNull を使っているので、 F のモ
ナド構造を使う必要があります。

結果の FormData には3つの可能性があります。

もし外側の構築⼦が Left なら、JSON⽂字列の解析中にエラーがあったか、それが間違った型
の値を表しています。この場合、アプリケーションは先ほど書いた alert アクションを使⽤し
てエラーを表⽰します。
もし外側の構築⼦が Right で内側の構築⼦が Nothing なら、 getItem が Nothing を返して
おり、キーがローカルストレージに存在していなかったことを意味しています。この場合、アプ
リケーションは静かに実⾏を継続します。
最後に、 Right (Just _) に適合した値はJSON⽂書としてただしく構⽂解析されたことを⽰し
ています。この場合、アプリケーションは適切な値でフォームフィールドを更新します。

pulp build -O --to dist/Main.js を実⾏してコードを試してから、ブラウザで
html/index.html を開いてください。 保存ボタンをクリックするとフォームフィールドの内容

をローカルストレージへ保存することができ、ページを再読込するとフィールドが再現されるは
ずです。

注意：ブラウザ特有の問題を避けるために、ローカルなHTTPサーバからHTMLファイルと
JavaScriptファイルを提供する必要があるかもしれません。

演 習

1. (簡単) decodeJSON を使って、 [[1, 2, 3], [4, 5], [6]] のようなJavaScript
の数の２次元配列を表現するJSON⽂書を解析してください。要素をnullにするこ
とが許容されている場合はどうでしょうか。配列⾃体をnullにすることが許容され
ている場合はどうなりますか。



2. (やや難しい) savedData の実装の型を検証し、計算のそれぞれの部分式の推論
された型を書き出してみましょう。

3. (難しい)次のデータ型は、葉で値を持つ⼆分⽊を表しています。

data Tree a = Leaf a | Branch (Tree a) (Tree a) 

purescript-foreign-generic を使ってこのタイプの Encode と Decode インス
タンスを導き、エンコードされた値がPSCiで正しくデコードできることを確認し
てください。

4. (難しい) 次の data 型は、整数か⽂字列のどちらかであるJSONを直接表現して
います。

data IntOrString 

  = IntOrString_Int Int 

  | IntOrString_String String 

この動作を実装する IntOrString データ型の Encode と Decode のインスタン
スを記述し、エンコードされた値が PSCi で正しくデコードできることを確認して
ください。

まとめ

この章では、PureScriptから外部のJavaScriptコードを扱う⽅法、およびその逆の⽅法を学び
ました。また、FFIを使⽤して信頼できるコードを書く時に⽣じる問題について⾒てきました。

データの実⾏時表現の重要性を⾒て、外部関数が正しい表現を持っていることを確かめました。
外部型、つまり Foreign データ型を使⽤することによって、null値のような特殊な場合や
JavaScriptの他の型のデータに対処する⽅法を学びました。
Preludeで定義されたいくつかの共通の外部型、既存のJavaScriptコードとどのように相互運⽤に
使⽤するかを⾒てきました。特に、 Eff モナドにおける副作⽤の表現を導⼊し、新たな副作⽤
を追跡するために Eff モナドを使⽤する⽅法を説明しました。
IsForeign 型クラスを使⽤して安全にJSONデータを復元する⽅法を説明しました。



その他の例については、Githubの purescript 組織、 purescript-contrib 組織および
purescript-node 組織が、FFIを使⽤するライブラリの例を多数提供しています。残りの章で

は、型安全な⽅法で現実世界の問題を解決するために使うライブラリを幾つか⾒ていきます。



第11章 モナドの探求

11 .1  この章の⽬標

この章の⽬標は、異なるモナドから提供された副作⽤を合成する⽅法を提供するモナド変換⼦
(monad transformers)について学ぶことです。NodeJSのコンソール上で遊ぶことができる、テキ
ストアドベンチャーゲームを題材として扱います。ゲームの様々な副作⽤(ロギング、状態、お
よび設定)がすべてモナド変換⼦スタックによって提供されます。

11 .2  プロジェクトの準備

このモジュールのプロジェクトでは以下のBower依存関係が新たに導⼊されます。

purescript-maps  - 不変のマップと集合のためのデータ型を提供します。
purescript-sets - 不変集合のデータ型を提供する標準的なモナド変換⼦の実装を提供する
purescript-transformers - 標準のモナド変換⼦の実装を提供します。
purescript-node-readline - NodeJSが提供する readline インターフェイスへのFFIバインデ

ィングを提供します。
purescript-yargs - yargs コマンドライン引数処理ライブラリにApplicativeなインターフェイ

スを提供します。

また、NPMを使って yargs モジュールをインストールする必要があります。

npm install 

11 .3  ゲームの遊びかた

プロジェクトを実⾏するには、 pulp run でソースコードをビルドしてから、NodeJSにコン
パイルされたJavaScriptを渡します。

デフォルトでは使い⽅が表⽰されます。

http://nodejs.org/api/readline.html
https://www.npmjs.org/package/yargs


node ./dist/Main.js -p  

 

Options: 

  -p, --player  Player name  [required] 

  -d, --debug   Use debug mode 

 

Missing required arguments: p 

The player name is required 

-p オプションを使ってプレイヤー名を指定してください。

pulp run -- -p Phil 

>  

プロンプトからは、 look 、 inventory 、 take 、 use 、 north 、 south 、 east 、
west などのコマンドを⼊⼒することができます。 --debug コマンドラインオプションが与え

られたときには、ゲームの状態を出⼒するための debug コマンドも使えます。

ゲームは2次元の碁盤の⽬の上でプレイし、コマンド north 、 south 、 east 、 west を発
⾏することによってプレイヤーが移動します。ゲームにはアイテムの配列があり、プレイヤーの
所持アイテム⼀覧を表したり、ゲーム盤上のその位置にあるアイテムの⼀覧を表すのに使われま
す。 take コマンドを使うと、プレイヤーの位置にあるアイテムを拾い上げることができます。

参考までに、このゲームのひと通りの流れは次のようになります。

$ pulp run -- -p Phil 

 

> look 

You are at (0, 0) 

You are in a dark forest. You see a path to the north. 

You can see the Matches. 

 

> take Matches 

You now have the Matches 

 

> north 

> look 

You are at (0, 1) 

You are in a clearing. 

You can see the Candle. 

 

> take Candle 

You now have the Candle 

 

> inventory 



You have the Candle. 

You have the Matches. 

 

> use Matches 

You light the candle. 

Congratulations, Phil! 

You win! 

このゲームはとても単純ですが、この章の⽬的は purescript-transformers パッケージを使
⽤してこのようなゲームを素早く開発できるようにするライブラリを構築することです。

11 .4  Stateモナド

purescript-transformers パッケージで提供されるモナドをいくつか⾒てみましょう。

最初の例は、純粋な変更可能状態を提供する State モナドです。すでに Eff モナド、すなわ
ち REF 作⽤と ST 作⽤によって提供された変更可能な状態という2つのアプローチについては⾒
てきました。 State は第3の選択肢を提供しますが、これは Eff モナドを使⽤して実装されて
いるわけではありません。

State 型構築⼦は、状態の型 s 、および返り値の型 a という2種類の引数を取ります。「
State モナド」というように説明はしていますが、実際には Monad 型クラスのインスタンスが

⽤意されているのは State に対してではなく、任意の型 s についての State s 型構築⼦に対
してです。

Control.Monad.State モジュールは以下のAPIを提供しています。

get    :: forall s.             State s s 

put    :: forall s. s        -> State s Unit 

modify :: forall s. (s -> s) -> State s Unit 

これは REF 作⽤や ST 作⽤が提供するAPIととてもよく似ています。しかし、これらのアク
ションに Ref や STRef に渡しているような、可変領域への参照を引数に渡さないことに注意し
てください。 State と Eff モナドが提供する解決策の違いは、 State モナドは暗黙的な単⼀
の状態だけを提供していることです。この状態は State モナドの型構築⼦によって隠された関
数の引数として実装されており、参照は明⽰的には渡されないのです。

例を⾒てみましょう。 State モナドの使いかたのひとつとしては、状態を数として、現在の
状態に配列の値を加算していくようなものかもしれません。状態の型 s として Number を選択



し、配列の⾛査に traverse_ を使って、配列の要素それぞれについて modify を呼び出すと、
これを実現することができます。

import Data.Foldable (traverse_) 

import Control.Monad.State 

import Control.Monad.State.Class 

 

sumArray :: Array Number -> State Number Unit 

sumArray = traverse_ \n -> modify \sum -> sum + n 

Control.Monad.State モジュールは State モナドでの計算を実⾏するための次の3つの関数
を提供します。

evalState :: forall s a. State s a -> s -> a 

execState :: forall s a. State s a -> s -> s 

runState  :: forall s a. State s a -> s -> Tuple a s 

３つの関数はそれぞれ初期値の型 s と計算の型 State s a を引数にとります。 evalState
は返り値だけを返し、 execState は最終的な状態だけを返し、 runState は Tuple a s 型の
値として表現された返り値と状態の両⽅を返します。

先ほどの sumArray 関数が与えられたとすると、 PSCi で次のように execState を使うと複
数の配列内の数字を合計することができます。

> :paste 

… execState (do 

…   sumArray [1, 2, 3] 

…   sumArray [4, 5] 

…   sumArray [6]) 0 

… ^D 

21 

演 習

1. (簡単) 上の例で、 execState を runState や evalState で 置き換えると結果
はどうなるでしょうか。



2. (やや難しい) State モナドと traverse_ 関数を使⽤して、次のような関数を書
いてください。

testParens :: String -> Boolean 

これは String が括弧の対応が正しく付けられているかどうかを調べる関数で
す。この関数は次のように動作しなくてはなりません。

> testParens "" 

true 

 

> testParens "(()(())())" 

true 

 

> testParens ")" 

false 

 

> testParens "(()()" 

false 

ヒント： ⼊⼒の⽂字列を⽂字の配列に変換するのに、 Data.String モジュール
の split 関数を使うと良いでしょう。

11 .5  Readerモナド

purescript-transformers パッケージでは Reader というモナドも提供されています。この
モナドは⼤域的な設定を読み取る機能を提供します。 State モナドがひとつの可変状態を読み
書きする機能を提供するのに対し、 Reader モナドはデータの読み取りの機能だけを提供しま
す。

Reader 型構築⼦は、構成の型を表す型 r 、および戻り値の型 a の2つの型引数を取りま
す。

Contro.Monad.Reader モジュールは以下のAPIを提供します。

ask   :: forall r. Reader r r 

local :: forall r a. (r -> r) -> Reader r a -> Reader r a 



ask アクションは現在の設定を読み取るために使い、 local アクションは局所的に設定を変
更して計算を実⾏するために使います。

たとえば、権限で制御されたアプリケーションを開発しており、現在の利⽤者の権限オブジェ
クトを保持するのに Reader モナドを使いたいとしましょう。型 r を次のようなAPIを備えた
型 Permission として選択します。

hasPermission :: String -> Permissions -> Boolean 

addPermission :: String -> Permissions -> Permissions 

利⽤者が特定の権限を持っているかどうかを確認したいときは、 ask を使って現在の権限オ
ブジェクトを取得すればいつでも調べることができます。たとえば、管理者だけが新しい利⽤者
の作成を許可されているとしましょう。

createUser :: Reader Permissions (Maybe User) 

createUser = do 

  permissions <- ask 

  if hasPermission "admin" permissions 

    then map Just newUser 

    else pure Nothing 

local アクションを使うと、計算の実⾏中に Permissions オブジェクトを局所的に変更し、
ユーザーの権限を昇格させることもできます。

runAsAdmin :: forall a. Reader Permissions a -> Reader Permissions a 

runAsAdmin = local (addPermission "admin") 

こうすると、利⽤者が admin 権限を持っていなかった場合であっても、新しい利⽤者を作成
する関数を書くことができます。

createUserAsAdmin :: Reader Permissions (Maybe User) 

createUserAsAdmin = runAsAdmin createUser 

Reader モナドの計算を実⾏するには、⼤域的な設定を与える runReader 関数を使います。

runReader :: forall r a. Reader r a -> r -> a 



演 習

以下の演習では、 Reader モナドを使って、字下げのついた⽂書を出⼒するための⼩
さなライブラリを作っていきます。「⼤域的な設定」は、現在の字下げの深さを⽰す数
になります。

type Level = Number 

 

type Doc = Reader Level String 

. (簡単) 現在の字下げの深さで⽂字列を出⼒する関数 line を書いてください。その関数は、
以下の型を持っている必要があります。

line :: String -> Doc 

ヒント：現在の字下げの深さを読み取るためには ask 関数を使⽤します。

. (やや難しい) local 関数を使⽤して、コードブロックの字下げの深さを⼤きくする次のよう
な関数を書いてください。

indent :: Doc -> Doc 

. (やや難しい) Data.Traversable で定義された sequence 関数を使⽤して、⽂書のリストを改
⾏で区切って連結する次のような関数を書いてください。

cat :: Array Doc -> Doc 

. (やや難しい) runReader 関数を使⽤して、⽂書を⽂字列として出⼒する次のような関数を書
いてください。

render :: Doc -> String 

これで、このライブラリを次のように使うと、簡単な⽂書を書くことができるはずです。



render $ cat 

  [ line "Here is some indented text:" 

  , indent $ cat 

      [ line "I am indented" 

      , line "So am I" 

      , indent $ line "I am even more indented" 

      ] 

  ] 

11 .6  Wri terモナド

Writer モナドは、計算の返り値に加えて、もうひとつの値を累積していく機能を提供しま
す。

よくある使い⽅としては型 String もしくは Array String でログを累積していくというも
のなどがありますが、 Writer モナドはこれよりもっと⼀般的なものです。これは累積するのに
任意のモノイドの値を使うことができ、 Sum モノイドを使って、合計を追跡し続けるのに使っ
たり、 Any モノイドを使って途中の Boolean 値がすべて真であるかどうかを追跡するのに使う
ことができます。

Writer 型の構築⼦は、 Monoid 型クラスのインスタンスである型 w 、および返り値の型 a
という2つの型引数を取ります。

Writer のAPIで重要なのは tell 関数です。

tell :: forall w a. Monoid w => w -> Writer w Unit 

tell アクションは、与えられた値を現在の累積結果に加算します。

例として、 Array String モノイドを使⽤して、既存の関数にログ機能を追加してみましょ
う。最⼤公約数関数の以前の実装を考えてみます。

gcd :: Int -> Int -> Int 

gcd n 0 = n 

gcd 0 m = m 

gcd n m = if n > m 

            then gcd (n - m) m 

            else gcd n (m - n) 



Writer (Array String) Int に返り値の型を変更することで、この関数にログ機能を追加す
ることができます。

import Control.Monad.Writer 

import Control.Monad.Writer.Class 

 

gcdLog :: Int -> Int -> Writer (Array String) Int 

各⼿順で⼆つの⼊⼒を記録するために、少し関数を変更する必要があります。

gcdLog n 0 = pure n 

gcdLog 0 m = pure m 

gcdLog n m = do 

  tell ["gcdLog " <> show n <> " " <> show m] 

  if n > m 

    then gcdLog (n - m) m 

    else gcdLog n (m - n) 

Writer モナドの計算を実⾏するには、 execWriter 関数と runWriter 関数のいずれかを使
います。

execWriter :: forall w a. Writer w a -> w 

runWriter  :: forall w a. Writer w a -> Tuple a w 

ちょうど State モナドの場合と同じように、 execWriter が累積されたログだけを返すのに
対して、 runWriter は累積されたログと結果の両⽅を返します。

PSCi で修正された関数を試してみましょう。

> import Control.Monad.Writer 

> import Control.Monad.Writer.Class 

 

> runWriter (gcdLog 21 15) 

Tuple 3 ["gcdLog 21 15","gcdLog 6 15","gcdLog 6 9","gcdLog 6 3","gcdLog 3 3"] 

演 習



1. (やや難しい) Writer モナドと purescript-monoid パッケージの
Additive Int のモノイドを使うように、上の sumArray 関数を書き換えてくださ

い。

2. (やや難しい)コラッツ関数は、⾃然数 n が偶数なら n / 2 、 n が奇数なら
3 * n + 1 であると定義されています。たとえば、 10 で始まるコラッツ数列は次

のようになります。

10, 5, 16, 8, 4, 2, 1, ... 

コラッツ関数の有限回の適⽤を繰り返すと、コラッツ数列は必ず最終的に 1 に
なるということとが予想できます。

数列が 1 に到達するまでに何回のコラッツ関数の適⽤が必要かを計算する再帰
的な関数を書いてください。

Writer モナドを使⽤してコラッツ関数のそれぞれの適⽤の経過を記録するよう
に、関数を変更してください。

11 .7  モナド変換⼦

上の3つのモナド、 State 、 Reader 、 Writer は、いずれもいわゆるモナド変換⼦(monad
transformers)の例となっています。対応するモナド変換⼦はそれぞれ StateT 、 ReaderT 、
WriterT と呼ばれています。

モナド変換⼦とは何でしょうか。さて、これまで⾒てきたように、モナドはPureScriptで適切
なハンドラ( runState 、 runReader 、 runWriter など)を使って解釈される、いろいろな種類
の副作⽤でPureScriptコードを拡張します。使⽤する必要がある副作⽤がひとつだけなら、これ
で問題ありません。しかし、同時に複数の副作⽤を使⽤できると便利なことがよくあります。例
えば、 Maybe と Reader を⼀緒に使⽤すると、ある⼤域的な設定の⽂脈で省略可能な結果を表
現することができます。もしくは、 Either モナドの純粋なエラー追跡機能と、 State モナド
が提供する変更可能な状態が同時に欲しくなるかもしれません。この問題を解決するのがモナド
変換⼦です。



拡張可能作⽤の⼿法を使うとネイティブな作⽤を混在させることができるので、 Eff モナド
はこの問題に対する部分的な解決策を提供していることをすでに⾒てきたことに注意してくださ
い。モナド変換⼦はまた異なった解決策を提供しますが、これらの⼿法にはそれぞれ利点と限界
があります。

モナド変換⼦は型だけでなく別の型構築⼦によってもパラメータ化される型構築⼦です。モナ
ド変換⼦はモナドをひとつ取り、独⾃のいろいろな副作⽤を追加した別のモナドへと変換しま
す。

例を⾒てみましょう。 Control.Monad.State.Trans で定義された StateT は State のモナ
ド変換⼦版です。 PSCi を使って StateT の種を⾒てみましょう。

> import Control.Monad.State.Trans 

> :kind StateT 

Type -> (Type -> Type) -> Type -> Type 

とても読みにくそうに思うかもしれませんが、使い⽅を理解するために、 StateT にひとつ引
数を与えてみましょう。

State の場合、最初の型引数は使いたい状態の型です。それでは型 String を与えてみまし
ょう。

> :kind StateT String 

(Type -> Type) -> Type -> Type 

次の引数は種 Type -> Type の型構築⼦です。これは StateT の機能を追加したい元のモナド
を表します。例として、 Either String モナドを選んでみます。

> :kind StateT String (Either String) 

Type -> Type 

型構築⼦が残りました。最後の引数は戻り値の型を表しており、たとえばそれを Number にす
ることができます。

> :kind StateT String (Either String) Number 

Type 

最後に、種 Type の何かが残りましたが、この型の値を探してみましょう。



構築したモナド StateT String (Either String) は、エラーで失敗する可能性があり、変更
可能な状態を使える計算を表しています。

外側の StateT String (Either String) モナドのアクション( get 、 put 、 modify )は直
接使うことができますが、ラップされている内側のモナド( Either String )の作⽤を使うため
には、これらの関数をモナド変換⼦まで「持ち上げ」なくてはいけません。
Control.MonadTrans モジュールでは、モナド変換⼦であるような型構築⼦を捕捉する
MonadTrans 型クラスを次のように定義しています。

class MonadTrans t where 

  lift :: forall m a. Monad m => m a -> t m a 

このクラスは、基礎となる任意のモナド m の計算をとり、それをラップされたモナド t m へ
と持ち上げる、 lift というひとつの関数だけを持っています。今回の場合、型構築⼦ t は
StateT String で、 m は Either String モナドとなり、 lift は型 Either String a の計算

を、型 State String (Either String) a の計算へと持ち上げる⽅法を提供することになりま
す。これは、型 Either String a の計算を使うときは、 lift を使えばいつでも作⽤
StateT String と Either String を隣り合わせに使うことができることを意味します。

たとえば、次の計算は StateT モナド変換⼦で導⼊されている状態を読み込み、状態が空の⽂
字列である場合はエラーを投げます。

import Data.String (drop, take) 

 

split :: StateT String (Either String) String 

split = do 

  s <- get 

  case s of 

    "" -> lift $ Left "Empty string" 

    _ -> do 

      put (drop 1 s) 

      pure (take 1 s) 

状態が空でなければ、この計算は put を使って状態を drop 1 s (最初の⽂字を取り除いた
s )へと更新し、 take 1 s ( s の最初の⽂字)を返します。

それでは PSCi でこれを試してみましょう。

> runStateT split "test" 

Right (Tuple "t" "est") 

 



> runStateT split "" 

Left "Empty string" 

これは StateT を使わなくても実装できるので、さほど驚くようなことではありません。しか
し、モナドとして扱っているので、do記法やApplicativeコンビネータを使って、⼩さな計算か
ら⼤きな計算を構築していくことができます。例えば、2回 split を適⽤すると、⽂字列から最
初の2⽂字を読むことができます。

> runStateT ((<>) <$> split <*> split) "test" 

(Right (Tuple "te" "st")) 

他にもアクションを幾つか⽤意すれば、 split 関数を使って、基本的な構⽂解析ライブラリ
を構築することができます。これは実際に purescript-parsing ライブラリで採⽤されている⼿
法です。これがモナド変換⼦の⼒なのです。必要な副作⽤を選択して、do記法とApplicativeコン
ビネータで表現⼒を維持しながら、様々な問題のための特注のモナドを作成することができるの
です。

11 .8  ExceptTモナド変換⼦

purescript-transformers パッケージでは、 Either e モナドに対応する変換⼦である
ExceptT e モナド変換⼦も定義されています。これは次のAPIを提供します。

class MonadError e m where 

  throwError :: forall a. e -> m a 

  catchError :: forall a. m a -> (e -> m a) -> m a 

 

instance monadErrorExceptT :: Monad m => MonadError e (ExceptT e m) 

 

runExceptT :: forall e m a. ExceptT e m a -> m (Either e a) 

MonadError クラスは e 型のエラーのスローとキャッチをサポートするモナドを取得し、
ExceptT e モナド変換⼦のインスタンスが提供されます。 Either e モナドの Left と同じよ

うに、 throwError アクションは失敗を⽰すために使われます。 catchError アクションを使う
と、 throwError でエラーが投げられたあとでも処理を継続することができるようになります。

runExceptT ハンドラを使うと、型 ExceptT e m a の計算を実⾏することができます。



このAPIは purescript-exceptions パッケージの Exception 作⽤によって提供されているも
のと似ています。しかし、いくつかの重要な違いがあります。

ExceptT モデルが代数的データ型を使っているのに対して、 Exception は実際のJavaScriptの
例外を使っています。
ExceptT が Error 型クラスのどんな型のエラーでも扱うのに対して、 Exception 作⽤は

JavaScriptの Error 型というひとつ例外の型だけを扱います。つまり、 ExceptT では新たなエ
ラー型を⾃由に定義できます。

試しに ExceptT を使って Writer モナドを包んでみましょう。ここでもモナド変換⼦
ExceptT e のアクションは⾃由に使えますが、 Writer モナドの計算は lift を使って持ちあ

げなければなりません。

import Control.Monad.Trans 

import Control.Monad.Writer 

import Control.Monad.Writer.Class 

import Control.Monad.Error.Class 

import Control.Monad.Except.Trans 

 

writerAndExceptT :: ExceptT String (Writer (Array String)) String 

writerAndExceptT = do 

  lift $ tell ["Before the error"] 

  throwError "Error!" 

  lift $ tell ["After the error"] 

  pure "Return value" 

PSCi でこの関数を試すと、ログの蓄積とエラーの送出という２つの作⽤がどのように相互作
⽤しているのかを⾒ることができます。まず、 runExceptT を使って外側の ExceptT 計算を実
⾏し、型 Write String (Either String String) の結果を残します。それから、 runWriter
で内側の Writer 計算を実⾏します。

> runWriter $ runExceptT writerAndExceptT 

Tuple (Left "Error!") ["Before the error"] 

実際に追加されるログは、エラーが投げられる前に書かれたログメッセージだけであることに
も注⽬してください。

11 .9  モナド変換⼦スタック



これまで⾒てきたように、モナド変換⼦を使うと既存のモナドの上に新しいモナドを構築する
ことができます。任意のモナド変換⼦ t1 と任意のモナド m について、その適⽤ t1 m もまた
モナドになります。これはふたつめのモナド変換⼦ t2 を先ほどの結果 t1 m に適⽤すると、第
３のモナド t2 (t1 m) を作れることを意味しています。このように、構成するモナドによって
提供された副作⽤を組み合わせる、モナド変換⼦のスタックを構築することができます。

実際には、基本となるモナド m は、ネイティブの副作⽤が必要なら Eff モナド、さもなくば
Control.Monad.Identity モジュールで定義されている Identity モナドになります。
Identity モナドは何の新しい副作⽤も追加しませんから、 Identity モナドの変換は、モナド

変換⼦の作⽤だけを提供します。実際に、 State モナド、 Reader モナド、 Writer モナド
は、 Identity モナドをそれぞれ StateT 、 ReaderT 、 WriterT で変換することによって実
装されています。

それでは3つの副作⽤が組み合わされている例を⾒てみましょう。 Identity モナドをスタッ
クの底にして、 StateT 作⽤、 WriterT 作⽤、 ExceptT 作⽤を使います。このモナド変換⼦ス
タックは、ログの蓄積し、純粋なエラー、可変状態の副作⽤を提供します。

このモナド変換⼦スタックを使うと、ロギングの機能が追加された split アクションを作る
ことができます。

type Errors = Array String 

 

type Log = Array String 

 

type Parser = StateT String (WriterT Log (ExceptT Errors Identity)) 

 

split :: Parser String 

split = do 

  s <- get 

  lift $ tell ["The state is " <> show s] 

  case s of 

    "" -> lift $ lift $ throwError ["Empty string"] 

    _ -> do 

      put (drop 1 s) 

      pure (take 1 s) 

この計算を PSCi で試してみると、 split が実⾏されるたびに状態がログに追加されること
がわかります。

モナド変換⼦スタックに現れる順序に従って、副作⽤を取り除いていかなければならないこと
に注意してください。最初に StateT 型構築⼦を取り除くために runStateT を使い、それから



runtWriteT を使い、その後 runExceptT を使います。最後に runIdentity を使⽤して
Identity モナドの演算を実⾏します。

> runParser p s = runIdentity $ runExceptT $ runWriterT $ runStateT p s 

 

> runParser split "test" 

(Right (Tuple (Tuple "t" "est") ["The state is test"])) 

 

> runParser ((<>) <$> split <*> split) "test" 

(Right (Tuple (Tuple "te" "st") ["The state is test", "The state is est"])) 

しかしながら解析が失敗した場合は、状態が空であるためログはまったく出⼒されません。

> runParser split "" 

(Left ["Empty string"]) 

これは、 ExceptT モナド変換⼦が提供する副作⽤が、 WriterT モナド変換⼦が提供する副
作⽤に影響を受けるためです。これはモナド変換⼦スタックが構成されている順序を変更するこ
とで解決することができます。スタックの最上部に ExceptT 変換⼦を移動すると、先ほど
Writer を ExceptT に変換したときと同じように、最初のエラーまでに書かれたすべてのメッ

セージが含まれるようになります。

このコードの問題のひとつは、複数のモナド変換⼦の上まで計算を持ち上げるために、 lift
関数を複数回使わなければならないということです。たとえば、 throwError の呼び出しは、1
回⽬は WriteT へ、2回⽬は StateT へと、2回持ちあげなければなりません。⼩さなモナド変
換⼦スタックならなんとかなりますが、そのうち不便だと感じるようになるでしょう。

幸いなことに、これから⾒るような型クラス推論によって提供されるコードの⾃動⽣成を使う
と、ほとんどの「多段持ち上げ」を⾏うことができます。

演 習

1. (簡単)  Identity 関⼿の上の ExceptT モナド変換⼦を使って、分⺟がゼロの
場合はエラーを投  げる、2つの数の商を求める関数 safeDivide を書いてくださ
い。



2. (やや難しい) 現在の状態が接頭辞に適合するか、エラーメッセージとともに失敗
する、次のような構⽂解析関数を書いてください。

string :: String -> Parser String 

この構⽂解析器は次のように動作しなくてはなりません。

> runParser (string "abc") "abcdef" 

(Right (Tuple (Tuple "abc" "def") ["The state is abcdef"])) 

ヒント：出発点として split の実装を使うといいでしょう。 stripPrefix 関数
も役に⽴ちます。

3. (難しい) 以前 Reader モナドを使⽤して書いた⽂書出⼒ライブラリを、
ReaderT と WriterT モナド変換⼦を使⽤して再実装してください。

⽂字列を出⼒する line や⽂字列を連結する cat を使うのではなく、 WriteT
モナド変換⼦と⼀緒に Array String モノイドを使い、結果へ⾏を追加するのに
tell を使ってください。

11 .10 救済のための型クラス

章の最初で扱った State モナドを⾒てみると、 State モナドのアクションには次のような型
が与えられていました。

get    :: forall s.             State s s 

put    :: forall s. s        -> State s Unit 

modify :: forall s. (s -> s) -> State s Unit 

Control.Monad.State.Class モジュールで与えられている型は、実際には次のようにもっと
⼀般的です。

get    :: forall m s. MonadState s m =>             m s 

put    :: forall m s. MonadState s m => s        -> m Unit 

modify :: forall m s. MonadState s m => (s -> s) -> m Unit 



Control.Monad.State.Class モジュールには「純粋な変更可能な状態を提供するモナド」へ
の抽象化を可能にする MonadState (多変数)型クラスが定義されています。予想できると思いま
すが、 State s 型構築⼦は MonadState s 型クラスのインスタンスになっており、このクラス
には他にも興味深いインスタンスが数多くあります。

特に、 purescript-transformers パッケージではモナド変換⼦ WriterT 、 ReaderT 、
ExceptT についての MonadState のインスタンスが提供されています。実際に、 StateT がモ

ナド変換⼦スタックのどこかに現れ、 StateT より上のすべてが MonadState のインスタンス
であれば、 get 、 put 、 modify を直接⾃由に使⽤することができます。

実は、これまで扱ってきた ReaderT 、 WriterT 、 ExceptT 変換⼦についても、同じことが
成り⽴っています。 purescript-transformers では、それらの操作をサポートするモナドの上
に抽象化することを可能にする、主な変換⼦それぞれについての型クラスが定義されています。

上の split 関数の場合、構築されたこのモナドスタックは型クラス MonadState 、
MonadWriter 、 MonadError それぞれのインスタンスです。これはつまり、 lift をまったく

呼び出す必要がないことを意味します！まるでモナドスタック⾃体に定義されていたかのよう
に、アクション get 、 put 、 tell 、 throwError をそのまま使⽤することができます。

split :: Parser String 

split = do 

  s <- get 

  tell ["The state is " <> show s] 

  case s of 

    "" -> throwError "Empty string" 

    _ -> do 

      put (drop 1 s) 

      pure (take 1 s) 

この計算はまるで、可変状態、ロギング、エラー処理という３つの副作⽤に対応した、独⾃の
プログラミング⾔語を拡張したかのようにみえます。しかしながら、内部的にはすべてはあくま
で純粋な関数と普通のデータを使って実装されているのです。

11 .11 Al ternat ive型クラス

purescript-control パッケージでは失敗しうる計算を操作するための抽象化がいくつか定義
されています。そのひとつは Alternative 型クラスです。



class Functor f <= Alt f where 

  alt :: forall a. f a -> f a -> f a 

 

class Alt f <= Plus f where 

  empty :: forall a. f a 

 

class (Applicative f, Plus f) <= Alternative f 

Alternative は、失敗しうる計算のプロトタイプを提供する empty 値、 エラーが起きたと
きに代替(Alternative)計算へ戻ってやり直す機能を提供する <|> 演算⼦ という、2つの新しい
コンビネータを提供しています。

Data.List モジュールでは Alternative 型クラスで型構築⼦を操作する2つの便利な関数を
提供します。

many :: forall f a. Alternative f => Lazy (f (List a)) => f a -> f (List a) 

some :: forall f a. Alternative f => Lazy (f (List a)) => f a -> f (List a) 

many コンビネータは計算をゼロ回以上繰り返し実⾏するために Alternative 型クラスを使
⽤しています。 some コンビネータも似ていますが、成功するために少なくとも１回の計算を必
要とします。

今回の Parser モナド変換⼦スタックの場合は、 ExceptT コンポーネントから導かれた、明
らかな⽅法で失敗をサポートする、 Alternative のインスタンスが存在します。これは、構⽂
解析器を複数回実⾏するために many 関数と some 関数を使うことができることを意味します。

> import Split 

> import Control.Alternative 

 

> runParser (many split) "test" 

(Right (Tuple (Tuple ["t", "e", "s", "t"] "") 

              [ "The state is \"test\"" 

              , "The state is \"est\"" 

              , "The state is \"st\"" 

              , "The state is \"t\"" 

              ])) 

ここで、⼊⼒⽂字列 "test" は、１⽂字の⽂字列４つの配列を返すように、繰り返し分割され
ています。残った状態は空⽂字列で、ログは split コンビネータが４回適⽤されたことを⽰し
ています。



Alternative 型構築⼦の他の例としては、 Maybe や、 Array つまり配列の型構築⼦があり
ます。

11 .12 モナド内包表記

Control.MonadPlus モジュールには MonadPlus と呼ばれる Alternative 型クラスの若⼲の
変形が定義されています。 MonadPlus はモナドと Alternative のインスタンスの両⽅である型
構築⼦を補⾜します。

class (Monad m, Alternative m) <= MonadZero m 

 

class MonadZero m <= MonadPlus m 

実際、 Parser モナドは MonadPlus のインスタンスです。

以前に本書中で配列内包表記を扱ったとき、不要な結果をフィルタリングするため使われる
guard 関数を導⼊しました。実際は guard 関数はもっと⼀般的で、 MonadPlus のインスタン

スであるすべてのモナドに対して使うことができます。

guard :: forall m. MonadZero m => Boolean -> m Unit 

<|> 演算⼦は失敗時のバックトラッキングをできるようにします。これがどのように役⽴つ
かを⾒るために、⼤⽂字だけに適合する split コンビネータの亜種を定義してみましょう。

upper :: Parser String 

upper = do 

  s <- split 

  guard $ toUpper s == s 

  pure s 

ここで、⽂字列が⼤⽂字でない場合に失敗するよう guard を使⽤しています。このコードは
前に⾒た配列内包表記とよく似ていることに注⽬してください。このように MonadPlus が使わ
れておりモナド内包表記(monad comprehensions)を構築するために参照することがあります。

11 .13 バックトラッキング



<|> 演算⼦を使うと、失敗したときに別の代替計算へとバックトラックすることができま
す。これを確かめるために、⼩⽂字に⼀致するもう⼀つの構⽂解析器を定義してみましょう。

lower :: Parser String 

lower = do 

  s <- split 

  guard $ toLower s == s 

  pure s 

これにより、まずもし最初の⽂字が⼤⽂字なら複数の⼤⽂字に適合し、さもなくばもし最初の
⽂字が⼩⽂字なら複数の⼩⽂字に適合する、という構⽂解析器を定義することができます。

> upperOrLower = some upper <|> some lower 

この構⽂解析器は、⼤⽂字と⼩⽂字が切り替わるまで、⽂字に適合し続けます。

> runParser upperOrLower "abcDEF" 

(Right (Tuple (Tuple ["a","b","c"] ("DEF")) 

              [ "The state is \"abcDEF\"" 

              , "The state is \"bcDEF\"" 

              , "The state is \"cDEF\"" 

              ])) 

many を使うと、⽂字列を⼩⽂字と⼤⽂字の要素に完全に分割することもできます。

> components = many upperOrLower 

 

> runParser components "abCDeFgh" 

(Right (Tuple (Tuple [["a","b"],["C","D"],["e"],["F"],["g","h"]] "") 

              [ "The state is \"abCDeFgh\"" 

              , "The state is \"bCDeFgh\"" 

              , "The state is \"CDeFgh\"" 

              , "The state is \"DeFgh\"" 

              , "The state is \"eFgh\"" 

              , "The state is \"Fgh\"" 

              , "The state is \"gh\"" 

              , "The state is \"h\"" 

              ])) 

繰り返しになりますが、これはモナド変換⼦がもたらす再利⽤性の威⼒を⽰しています。標準
的な抽象化を再利⽤することで、バックトラック構⽂解析器を宣⾔型のスタイルでわずか数⾏の
コードで書くことができました！



演 習

1. (簡単) string 構⽂解析器の実装から lift 関数の呼び出しを取り除いてくださ
い。新しい実装の型が整合していることを確認し、なぜそのようになるのかをよく
納得しておきましょう。

2. (やや難しい) string 構⽂解析器と many コンビネータを使って、⽂字列 "a" の
連続と、それに続く⽂字列 "b" の連続からなる⽂字列を認識する構⽂解析器を書
いてください。

3. (やや難しい) <|> 演算⼦を使って、⽂字 a と⽂字 b が任意の順序で現れるよう
な⽂字列を認識する構⽂解析器を書いてください。

4. (難しい) Parser モナドは次のように定義されるかもしれません。

type Parser = ExceptT Errors (StateT String (WriterT Log Identity)) 

このように変更すると、構⽂解析関数にどのような影響を与えるでしょうか。

11 .14 RWSモナド

モナド変換⼦のある特定の組み合わせは、 purescript-transformers パッケージ内の単⼀の
モナド変換⼦として提供されるのが⼀般的です。 Reader 、 Writer 、 State のモナドは、
Reader-Writer-Stateモナド( RWS モナド)へと結合されます。このモナドは RWST モナド変換⼦
と呼ばれる、対応するモナド変換⼦を持っています。

ここでは RWS モナドを使ってテキストアドベンチャーゲームの処理を設計していきます。

RWS モナドは(戻り値の型に加えて)3つの型変数で定義されています。

type RWS r w s = RWST r w s Identity 

副作⽤を提供しない Identity にベースモナドを設定することで、 RWS モナドが独⾃のモナ
ド変換⼦の観点から定義されていることに注意してください。



第1型引数 r は⼤域的な設定の型を表します。第2型引数 w はログを蓄積するために使⽤する
モノイド、第3型引数 s は可変状態の型を表しています。

このゲームの場合には、⼤域的な設定は Data.GameEnvironment モジュールの
GameEnvironment と呼ばれる型で定義されています。

type PlayerName = String 

 

newtype GameEnvironment = GameEnvironment 

  { playerName    :: PlayerName 

  , debugMode     :: Boolean 

  } 

GameEnvironment では、プレイヤー名と、ゲームがデバッグモードで動作しているか否かを
⽰すフラグが定義されています。これらのオプションは、モナド変換⼦を実⾏するときにコマン
ドラインから設定されます。

可変状態は Data.GameState モジュールの GameState と呼ばれる型で定義されています。

import qualified Data.Map as M 

import qualified Data.Set as S 

 

newtype GameState = GameState 

  { items       :: M.Map Coords (S.Set GameItem) 

  , player      :: Coords 

  , inventory   :: S.Set GameItem 

  } 

Coords データ型は2次元平⾯の点を表し、 GameItem データ型はゲーム内のアイテムです。

data GameItem = Candle | Matches 

GameState 型はソートされたマップを表す Map とソートされた集合を表す Set という2つの
新しいデータ構造を使っています。 items プロパティは、そのゲーム平⾯上の座標と、ゲーム
アイテムの集合へのマッピングになっています。 player プロパティはプレイヤーの現在の座標
を格納しており、 inventory プロパティは現在プレイヤーが保有するゲームアイテムの集合で
す。

Map と Set のデータ構造はキーによってソートされ、 Ord 型クラスの任意の型をキーとし
て使⽤することができます。これは今回のデータ構造のキーが完全に順序付けできることを意味
します。



ゲームのアクションを書くために、 Map と Set 構造がどのように使っていくのかを⾒ていき
ましょう。

ログとしては List String モノイドを使います。 RWS を使って Game モナドのための型同義
語を定義しておきます。

type Log = L.List String 

 

type Game = RWS GameEnvironment Log GameState 

11 .15 ゲームロジックの実装

今回は、 Reader モナド、 Writer モナド、 State モナドのアクションを再利⽤し、 Game
モナドで定義されている単純なアクションを組み合わせてゲームを構築していきます。このアプ
リケーションの最上位では、 Game モナドで純粋な計算を実⾏しており、 Eff モナドはコンソ
ールにテキストを出⼒するような追跡可能な副作⽤へと結果を変換するために使っています。

このゲームで最も簡単なアクションのひとつは has アクションです。このアクションはプレ
イヤーの持ち物に特定のゲームアイテムが含まれているかどうかを調べます。これは次のように
定義されます。

has :: GameItem -> Game Boolean 

has item = do 

  GameState state <- get 

  pure $ item `S.member` state.inventory 

この関数は、現在のゲームの状態を読み取るために Monad.State 型クラスで定義されている
get アクションを使っており、指定した GameItem が持ち物の Set のなかに出現するかどうか

を調べるために Data.Set で定義されている member 関数を使っています。

他にも pickUp アクションがあります。現在の位置にゲームアイテムがある場合、プレイヤー
の持ち物にそのアイテムを追加します。これには MonadWriter と MonadState 型クラスのアク
ションを使っています。まず、現在のゲームの状態を読み取ります。

pickUp :: GameItem -> Game Unit 

pickUp item = do 

  GameState state <- get 



次に pickUp は現在の位置にあるアイテムの集合を検索します。これは Data.Map で定義され
た lookup 関数を使って⾏います。

  case state.player `M.lookup` state.items of 

lookup 関数は Maybe 型構築⼦で⽰されたオプショナルな結果を返します。 lookup 関数
は、キーがマップにない場合は Nothing を返し、それ以外の場合は Just 構築⼦で対応する値
を返します。

関⼼があるのは、指定されたゲームアイテムが対応するアイテムの集合に含まれている場合で
す。 member 関数を使うとこれを調べることができます。

    Just items | item `S.member` items -> do 

この場合、 put を使ってゲームの状態を更新し、 tell を使ってログにメッセージを追加し
ます。

      let newItems = M.update (Just <<< S.delete item) state.player state.items 

          newInventory = S.insert item state.inventory 

      put $ GameState state { items     = newItems 

                            , inventory = newInventory 

                            } 

      tell (L.singleton ("You now have the " <> show item)) 

ここで、 MonadState と MonadWriter の両⽅について Game モナド変換⼦スタックについて
の適切なインスタンスが存在するので、２つの計算はどちらも lift は必要ないことに注意して
ください。

put の引数では、レコード更新を使ってゲームの状態の items と inventory フィールドを
変更しています。特定のキーの値を変更するには Data.Map の update 関数を使います。このと
き、 delete 関数を使い指定したアイテムを集合から取り除くことで、プレイヤーの現在の位置
にあるアイテムの集合を変更します。

最後に、 pickUp 関数は tell を使ってユーザに次のように通知することにより、残りの場合
を処理します。

    _ -> tell (L.singleton "I don't see that item here.") 



Reader モナドを使う例として、 debug コマンドのコードを⾒てみましょう。ゲームがデバ
ッグモードで実⾏されている場合、このコマンドを使うとユーザは実⾏時にゲームの状態を調べ
ることができます。

  GameEnvironment env <- ask 

  if env.debugMode 

    then do 

      state <- get 

      tell (L.singleton (show state)) 

    else tell (L.singleton "Not running in debug mode.") 

ここでは、ゲームの設定を読み込むために ask アクションを使⽤しています。繰り返します
が、どんな計算の lift も必要なく、同じdo記法ブロック内で MonadState 、 MonadReader 、
MonadWriter 型クラスで定義されているアクションを使うことができることに注意してくださ

い。

debugMode フラグが設定されている場合、 tell アクションを使ってログに状態が追加され
ます。そうでなければ、エラーメッセージが追加されます。

Game.purs モジュールでは、 MonadState 型クラス、 MonadReader 型クラス、
MonadWriter 型クラスでそれぞれ定義されたアクションだけを使って、同様のアクションが定

義されています。

11 .16 計算の実⾏

このゲームロジックは RWS モナドで動くため、ユーザのコマンドに応答するためには計算を
実⾏する必要があります。

このゲームのフロントエンドは、 yargs コマンドライン構⽂解析ライブラリへのApplicative
なインターフェイスを提供する purescript-yargs パッケージと、対話的なコンソールベース
のアプリケーションを書くことを可能にするNodeJSの readline モジュールをラップする
purescript-node-readline パッケージという２つのパッケージで構成されています。

このゲームロジックへのインタフェースは Game モジュール内の関数 game によって提供され
ます。

game :: Array String -> Game Unit 



この計算を実⾏するには、ユーザが⼊⼒した単語のリストを⽂字列の配列として渡してから、
runRWS を使って RWS の計算を実⾏します。

data RWSResult state result writer = RWSResult state result writer 

 

runRWS :: forall r w s a. RWS r w s a -> r -> s -> RWSResult s a w 

runRWS は runReader 、 runWriter 、 runState を組み合わせたように⾒えます。これは、
引数として⼤域的な設定および初期状態をとり、ログ、結果、最的な終状態を含むレコードを返
します。

このアプリケーションのフロントエンドは、次の型シグネチャを持つ関数 runGame によって
定義されます。

runGame :: forall eff . GameEnvironment 

  -> Eff ( exception :: EXCEPTION 

         , readline :: RL.READLINE 

         , console :: CONSOLE 

         | eff 

         ) Unit 

Console 作⽤は、この関数が purescript-node-readline パッケージを使ってコンソールを
介してユーザと対話することを⽰しています。 runGame は関数の引数としてのゲームの設定と
ります。

purescript-node-readline パッケージでは、端末からのユーザ⼊⼒を扱う Eff モナドのア
クションを表す LineHandler 型が提供されています。対応するAPIは次のとおりです。

type LineHandler eff a = String -> Eff eff a 

 

setLineHandler :: forall eff a. Interface 

                -> LineHandler (readline :: READLINE | eff) a 

                -> Eff (readline :: READLINE | eff) Unit 

Interface 型はコンソールのハンドルを表しており、コンソールと対話する関数への引数と
して渡されます。 createInterface 関数を使⽤すると Interface を作成することができます。

runGame env = do 

  interface <- createConsoleInterface noCompletion 



最初の⼿順はコンソールにプロンプトを設定することです。 interface ハンドルを渡し、プ
ロンプト⽂字列とインデントレベルを提供します。

setPrompt "> " 2 interface 

今回は lineHandler 関数を実装してみましょう。 lineHandler は let 宣⾔内の補助関数を
使って次のように定義されています。

lineHandler :: GameState -> String 

             -> Eff ( exception :: EXCEPTION 

                    , console :: CONSOLE 

                    , readline :: RL.READLINE 

                    | eff 

                    ) Unit 

lineHandler currentState input = do 

  case runRWS (game (split " " input)) env currentState of 

    RWSResult state _ written -> do 

      for_ written log 

      setLineHandler interface $ lineHandler state 

  prompt interface 

  pure unit 

lineHandler では env という名前のゲーム構成や、 interface という名前のコンソールハ
ンドルを参照しています。

このハンドラは追加の最初の引数としてゲームの状態を取ります。ゲームのロジックを実⾏す
るために runRWS にゲームの状態を渡さなければならないので、これは必要となっています。

このアクションが最初に⾏うことは、 Data.String モジュールの split 関数を使⽤して、
ユーザーの⼊⼒を単語に分割することです。それから、ゲーム環境と現在のゲームの状態を渡
し、 runRWS を使⽤して( RWS モナドで) game アクションを実⾏しています。

純粋な計算であるゲームロジックを実⾏し、画⾯にすべてのログメッセージを出⼒して、ユー
ザに次のコマンドのプロンプトを表⽰する必要があります。 for_ アクションは( List String

型の)ログを⾛査し、コンソールにその内容を出⼒するために使われています。そして
setLineHandler を使って lineHandler 関数を更新することで、ゲームの状態を更新します。

最後に prompt アクションを使ってプロンプトが再び表⽰しています。

runGame 関数ではコンソールインターフェイスに最初のlineHandlerを設定して、最初のプロ
ンプトを表⽰します。



  setLineHandler interface $ lineHandler initialGameState 

  prompt interface 

演 習

1. (やや難しい) ゲームフィールド上にあるすべてのゲームアイテムをユーザの持ち
ものに移動する新しいコマンド cheat を実装してください。

2. (難しい) RWS モナドの Writer コンポーネントは、エラーメッセージと情報メ
ッセージの２つの種類のメッセージのために使われています。このため、コードの
いくつかの箇所では、エラーの場合を扱うためにcase式を使⽤しています。

エラーメッセージを扱うのに ExceptT モナド変換⼦を使うようにし、情報メッ
セージを扱うのに RWS を使うようにするよう、コードをリファクタリングしてく
ださい。

11 .17 コマンドラインオプションの扱い

このアプリケーションの最後の部品は、コマンドラインオプションの解析と
GameEnvironment レコードを作成する役⽬にあります。このためには purescript-yargs パッ

ケージを使⽤します。

purescript-yargs はApplicativeなコマンドラインオプション構⽂解析器の例です。
Applicative関⼿を使うと、いろいろな副作⽤の型を表す型構築⼦まで任意個数の引数の関数をを
持ち上げられることを思い出してください。 purescript-yargs パッケージの場合には、コマ
ンドラインオプションからの読み取りの副作⽤を追加する Y 関⼿が興味深い関⼿になっていま
す。これは次のようなハンドラを提供しています。

runY :: forall a eff. YargsSetup -> 

                      Y (Eff (exception :: EXCEPTION, console :: CONSOLE | eff) a) -> 

                      Eff (exception :: EXCEPTION, console :: CONSOLE | eff) a 



この関数の使いかたは、例で⽰すのが最も適しているでしょう。このアプリケーションの
main 関数は runY を使って次のように定義されています。

main = runY (usage "$0 -p <player name>") $ map runGame env 

最初の引数は yargs ライブラリを設定するために使⽤されます。今回の場合、使⽤⽅法のメ
ッセージだけを提供していますが、 Node.Yargs.Setup モジュールには他にもいくつかのオプ
ションを提供しています。

2番⽬の引数では、 Y 型構築⼦まで runGame 関数を持ち上げるために <$> コンビネータを使
⽤しています。引数 env は where 節でApplicative演算⼦ <$> 、 <*> を使って構築されていま
す。

  where 

  env :: Y GameEnvironment 

  env = gameEnvironment 

          <$> yarg "p" ["player"] 

                   (Just "Player name") 

                   (Right "The player name is required") 

                   false 

          <*> flag "d" ["debug"] 

                   (Just "Use debug mode") 

PlayerName -> Boolean -> GameEnvironment という型を持つこの gameEnvironment 関数
は、 Y まで持ち上げられています。このふたつの引数は、コマンドラインオプションからプレ
イヤー名とデバッグフラグを読み取る⽅法を指定しています。最初の引数は -p もしくは
--player オプションで指定されるプレイヤー名オプションについて記述しており、２つ⽬の引

数は -d もしくは --debug オプションで指定されるデバッグモードフラグについて記述してい
ます。

これは Node.Yargs.Applicative モジュールで定義されているふたつの基本的な関数につい
て⽰しています。 yarg は(型 String 、 Number 、 Boolean の)オプショナルな引数を取りコ
マンドラインオプションを定義し、 flag は型 Boolean のコマンドラインフラグを定義してい
ます。

Applicative演算⼦によるこの記法を使うことで、コマンドラインインターフェイスに対してコ
ンパクトで宣⾔的な仕様を与えることが可能になったことに注意してください。また、 env の
定義で runGame 関数に新しい引数を追加し、 <*> を使って追加の引数まで runGame を持ち上
げるだけで、簡単に新しいコマンドライン引数を追加することができます。



演 習

1. (やや難しい) GameEnvironment レコードに新しい真偽値のプロパティ cheatMode
を追加してください。 また、 yargs 設定に、チートモードを有効にする新しいコ
マンドラインフラグ -c を追加してください。チートモードが有効になっていない
場合、 cheat コマンドは禁⽌されなければなりません。

まとめ

モナド変換⼦を使⽤したゲームの純粋な定義、コンソールを使⽤したフロントエンドを構築す
るための Eff モナドなと、この章ではこれまで学んできた⼿法を実⽤的に使いました。

ユーザインターフェイスからの実装を分離したので、ゲームの別のフロントエンドを作成する
ことも可能でしょう。例えば、 Eff モナドでCanvas APIやDOMを使⽤して、ブラウザでゲー
ムを描画するようなことができるでしょう。

モナド変換⼦によって、型システムによって作⽤が追跡される命令型のスタイルで、安全なコ
ードを書くことができることを⾒てきました。また、型クラスは、コードの再利⽤を可能にする
モナドによって提供される、アクション上の抽象化の強⼒な⽅法を提供します。標準的なモナド
変換⼦を組み合わせることにより、 Alternative や MonadPlus のような標準的な抽象化を使
⽤して、役に⽴つモナドを構築することができました。

モナド変換⼦は、⾼階多相や多変数型クラスなどの⾼度な型システムの機能を利⽤することに
よって記述することができ、表現⼒の⾼いコードの優れた実演となっています。

次の章では、⾮同期なJavaScriptのコードにありがちな不満、コールバック地獄の問題に対し
て、モナド変換⼦がどのような洗練された解決策を与えるのかを⾒ていきます。



第12章 コールバック地獄

12 .1  この章の⽬標

この章では、これまでに⾒てきたモナド変換⼦やApplicative関⼿といった道具が、現実世界の
問題解決にどのように役⽴つかを⾒ていきましょう。ここでは特に、コールバック地獄(callback
hell)の問題を解決について⾒ていきます。

12 .2  プロジェクトの準備

この章のソースコードは、 pulp run を使ってコンパイルして実⾏することができます。 ま
た、 request モジュールをNPMを使ってインストールする必要があります。

npm install 

12 .3  問題

通常、JavaScriptの⾮同期処理コードでは、プログラムの流れを構造化するためにコールバッ
ク(callbacks)を使⽤します。たとえば、ファイルからテキストを読み取るのに好ましいアプロー
チとしては、 readFile 関数を使⽤し、コールバック、つまりテキストが利⽤可能になったとき
に呼び出される関数を渡すことです。

function readText(onSuccess, onFailure) { 

  var fs = require('fs'); 

  fs.readFile('file1.txt', { encoding: 'utf-8' }, function (error, data) { 

    if (error) { 

      onFailure(error.code); 

    } else { 

      onSuccess(data); 

    }    

  }); 

} 



しかしながら、複数の⾮同期操作が関与している場合には⼊れ⼦になったコールバックを⽣じ
ることになり、すぐに読めないコードになってしまいます。

function copyFile(onSuccess, onFailure) { 

  var fs = require('fs'); 

  fs.readFile('file1.txt', { encoding: 'utf-8' }, function (error, data1) { 

    if (error) { 

      onFailure(error.code); 

    } else { 

      fs.writeFile('file2.txt', data, { encoding: 'utf-8' }, function (error) { 

        if (error) { 

          onFailure(error.code); 

        } else { 

          onSuccess(); 

        } 

      }); 

    }    

  }); 

}  

この問題に対する解決策のひとつとしては、独⾃の関数に個々の⾮同期呼び出しを分割するこ
とです。

function writeCopy(data, onSuccess, onFailure) { 

  var fs = require('fs'); 

  fs.writeFile('file2.txt', data, { encoding: 'utf-8' }, function (error) { 

    if (error) { 

      onFailure(error.code); 

    } else { 

      onSuccess(); 

    } 

  }); 

} 

 

function copyFile(onSuccess, onFailure) { 

  var fs = require('fs'); 

  fs.readFile('file1.txt', { encoding: 'utf-8' }, function (error, data) { 

    if (error) { 

      onFailure(error.code); 

    } else { 

      writeCopy(data, onSuccess, onFailure); 

    }    

  }); 

}  

この解決策は⼀応は機能しますが、いくつか問題があります。



上で writeCopy へ data を渡したのと同じ⽅法で、⾮同期関数に関数の引数として途中の結果
を渡さなければなりません。これは⼩さな関数についてはうまくいきますが、多くのコールバッ
ク関係する場合はデータの依存関係は複雑になることがあり、関数の引数が⼤量に追加される結
果になります。
どんな⾮同期関数でもコールバック onSuccess と onFailure が引数として定義されるという
共通のパターンがありますが、このパターンはソースコードに付随したモジュールのドキュメン
トに記述することで実施しなければなりません。このパターンを管理するには型システムのほう
がよいですし、型システムで使い⽅を強制しておくほうがいいでしょう。

次に、これらの問題を解決するために、これまでに学んだ⼿法を使⽤する⽅法について説明し
ていきます。

12 .4  継続モナド

copyFile の例をFFIを使ってPureScriptへと翻訳していきましょう。PureScriptで書いていく
につれ、計算の構造はわかりやすくなり、 purescript-transformers パッケージで定義されて
いる継続モナド変換⼦ ContT が⾃然に導⼊されることになるでしょう。

まず、FFIを使って readFile と writeFile に型を与えなくてはなりません。型同義語をいく
つかと、ファイル⼊出⼒のための作⽤を定義することから始めましょう。

foreign import data FS :: Effect 

 

type ErrorCode = String 

type FilePath = String 

readFile はファイル名と2引数のコールバックを引数に取ります。ファイルが正常に読み込
まれた場合は、2番⽬の引数にはファイルの内容が含まれますが、そうでない場合は、最初の引
数がエラーを⽰すために使われます。

今回は readFile を2つのコールバックを引数としてとる関数としてラップすることにしま
す。先ほどの copyFile や writeCopy とまったく同じように、エラーコールバック
( onFailure )と結果コールバック( onSuccess )の2つです。簡単のために Data.Function の多
引数関数の機能を使うと、このラップされた関数 readFileImpl は次のようになるでしょう。

foreign import readFileImpl 

  :: forall eff 

   . Fn3 FilePath 

         (String -> Eff (fs :: FS | eff) Unit) 



         (ErrorCode -> Eff (fs :: FS | eff) Unit) 

         (Eff (fs :: FS | eff) Unit) 

外部JavaScriptモジュールでは、 readFileImpl は次のように定義されます。

exports.readFileImpl = function(path, onSuccess, onFailure) { 

  return function() { 

    require('fs').readFile(path, { 

      encoding: 'utf-8' 

    }, function(error, data) { 

      if (error) { 

        onFailure(error.code)(); 

      } else { 

        onSuccess(data)(); 

      } 

    }); 

  }; 

}; 

readFileImpl はファイルパス、成功時のコールバック、失敗時のコールバックという３つの
引数を取り、空( Unit )の結果を返す副作⽤のある計算を返す、ということをこの型は⾔ってい
ます。コー  ルバック⾃⾝にも、その作⽤を追跡するために Eff モナドを使うような型が与えら
れていることに注意してください。

この readFileImpl の実装がその型の正しい実⾏時表現を持っている理由を、よく理解して
おくようにしてください。

writeFileImpl もよく似ています。違いはファイルがコールバックではなく関数⾃⾝に渡さ
れるということだけです。実装は次のようになります。

foreign import writeFileImpl 

  :: forall eff 

   . Fn4 FilePath 

         String 

         (Eff (fs :: FS | eff) Unit) 

         (ErrorCode -> Eff (fs :: FS | eff) Unit) 

         (Eff (fs :: FS | eff) Unit) 

exports.writeFileImpl = function(path, data, onSuccess, onFailure) { 

  return function() { 

    require('fs').writeFile(path, data, { 

      encoding: 'utf-8' 

    }, function(error) { 

      if (error) { 



        onFailure(error.code)(); 

      } else { 

        onSuccess(); 

      } 

    }); 

  }; 

}; 

これらのFFIの宣⾔が与えられれば、 readFile と writeFile の実装を書くことができます。
Data.Function ライブラリを使って、多引数のFFIバインディングを通常の(カリー化され

た)PureScript関数へと変換するので、もう少し読みやすい型になるでしょう。

さらに、成功時と失敗時の２つの必須のコールバックに代わって、成功か失敗のどちらか
(Either) に対応した単⼀のコールバックを要求するようにします。つまり、新しいコールバック
は引数として Either ErrorCode モナドの値をとります。

readFile :: forall eff . FilePath 

  -> (Either ErrorCode String -> Eff (fs :: FS | eff) Unit) 

  -> Eff (fs :: FS | eff) Unit 

readFile path k = 

  runFn3 readFileImpl 

         path 

         (k <<< Right) 

         (k <<< Left) 

 

writeFile :: forall eff . FilePath 

  -> String 

  -> (Either ErrorCode Unit -> Eff (fs :: FS | eff) Unit) 

  -> Eff (fs :: FS | eff) Unit 

writeFile path text k = 

  runFn4 writeFileImpl 

         path 

         text 

         (k $ Right unit) 

         (k <<< Left) 

ここで、重要なパターンを⾒つけることができます。これらの関数は何らかのモナド(この場
合は Eff (fs :: FS | eff) )で値を返すコールバックをとり、同⼀のモナドで値を返します。
これは、最初のコールバックが結果を返したときに、そのモナドは次の⾮同期関数の⼊⼒に結合
するためにその結果を使⽤することができることを意味しています。実際、 copyFile の例で⼿
作業でやったことがまさにそれです。

これは purescript-transformers の Control.Monad.Cont.Trans モジュールで定義されてい
る継続モナド変換⼦(continuation monad transformer)の基礎となっています。



ContT は次のようなnewtypeとして定義されます。

newtype ContT r m a = ContT ((a -> m r) -> m r) 

継続(continuation)はコールバックの別名です。継続は計算の残余(remainder)を捕捉します。
ここで「残余」とは、⾮同期呼び出しが⾏われ、結果が提供された後に起こることを指していま
す。

ContT データ構築⼦の引数は readFile と writeFile の型ととてもよく似ています。実際、
もし型 a を ErrorCode String 型、 r を Unit 、 m をモナド Eff(fs :: FS | eff) というよう
に選ぶと、 readFile の型の右辺を復元することができます。

readFile や writeFile のような⾮同期のアクションを組み⽴てるために使う Async モナド
を定義するため、次のような型同義語を導⼊します。

type Async eff = ContT Unit (Eff eff) 

今回の⽬的では Eff モナドを変換するために常に ContT を使い、型 r は常に Unit になり
ますが、必ずそうしなければならないというわけではありません。

ContT データ構築⼦を適⽤するだけで、 readFile と writeFile を Async モナドの計算と
して扱うことができます。

readFileCont 

  :: forall eff 

   . FilePath 

  -> Async (fs :: FS | eff) (Either ErrorCode String) 

readFileCont path = ContT $ readFile path 

 

writeFileCont 

  :: forall eff 

   . FilePath 

  -> String 

  -> Async (fs :: FS | eff) (Either ErrorCode Unit) 

writeFileCont path text = ContT $ writeFile path text 

ここで ContT モナド変換⼦に対してdo記法を使うだけで、ファイル複製処理を書くことがで
きます。

copyFileCont 

  :: forall eff 

   . FilePath 



  -> FilePath 

  -> Async (fs :: FS | eff) (Either ErrorCode Unit) 

copyFileCont src dest = do 

  e <- readFileCont src 

  case e of 

    Left err -> pure $ Left err 

    Right content -> writeFileCont dest content 

readFileCont の⾮同期性がdo記法によってモナドの束縛に隠されていることに注⽬してくだ
さい。これはまさに同期的なコードのように⾒えますが、 ContT モナド変換⼦は⾮同期関数を
書くのを⼿助けしているのです。

継続を与えて runContT ハンドラを使うと、この計算を実⾏することができます。この継続は
次に何をするか、例えば⾮同期なファイル複製処理が完了した時に何をするか、を表していま
す。この簡単な例では、型 Either ErrorCode Unit の結果をコンソールに出⼒する logShow
関数を単に継続として選んでいます。

import Prelude 

 

import Control.Monad.Eff.Console (logShow) 

import Control.Monad.Cont.Trans (runContT) 

 

main = 

  runContT 

    (copyFileCont "/tmp/1.txt" "/tmp/2.txt") 

    logShow 

演 習

1. (簡単) readFileCont と writeFileCont を使って、2つのテキストフ  ァイルを連
結する関数を書いてください。

2. (やや難しい) FFIを使って、 setTimeout 関数に適切な型を与えてください。ま
た、 Async モナドを使った次のようなラッパー関数を書いてください。

type Milliseconds = Int 

 

foreign import data TIMEOUT :: Effect 

 



setTimeoutCont 

  :: forall eff 

   . Milliseconds 

  -> Async (timeout :: TIMEOUT | eff) Unit 

12 .5  ExceptTを機能させる

この⽅法はうまく動きますが、まだ改良の余地があります。

copyFileCont の実装において、次に何をするかを決定するためには、パターン照合を使って
(型 Either ErrorCode String の) readFileCont 計算の結果を解析しなければなりません。し
かしながら、 Either モナドは対応するモナド変換⼦ ExceptT を持っていることがわかってい
るので、 ContT を持つ ExceptT を使って⾮同期計算とエラー処理の２つの作⽤を結合できると
期待するのは理にかなっています。

実際にそれは可能で、 ExceptT の定義を⾒ればそれがなぜかがわかります。

newtype ExceptT e m a = ExceptT (m (Either e a)) 

ExceptT は基礎のモナドの結果を単純に a から Either e a に変更します。現在のモナドス
タックを ExceptT ErrorCode 変換⼦で変換するように、 copyFileCont を書き換えることがで
きることを意味します。それは現在の⽅法に ExceptT データ構築⼦を適⽤するだけなので簡単
です。型同義語を与えると、ここでも型シグネチャを整理することができます。

readFileContEx 

  :: forall eff 

   . FilePath 

  -> ExceptT ErrorCode (Async (fs :: FS | eff)) String 

readFileContEx path = ExceptT $ readFileCont path 

 

writeFileContEx 

  :: forall eff 

   . FilePath 

  -> String 

  -> ExceptT ErrorCode (Async (fs :: FS | eff)) Unit 

writeFileContEx path text = ExceptT $ writeFileCont path text 



⾮同期エラー処理が ExceptT モナド変換⼦の内部に隠されているので、このファイル複製処
理ははるかに単純になります。

copyFileContEx 

  :: forall eff 

   . FilePath 

  -> FilePath 

  -> ExceptT ErrorCode (Async (fs :: FS | eff)) Unit 

copyFileContEx src dest = do 

  content <- readFileContEx src 

  writeFileContEx dest content 

演 習

1. (やや難しい) 任意のエラーを処理するために、 ExceptT を使⽤して2つのファ
イルを連結しする先ほどの解決策を書きなおしてください。

2. (やや難しい) ⼊⼒ファイル名の配列を与えて複数のテキストファイルを連結する
関数 concatenateMany を書く。 ヒント： traverse を使⽤します。

12 .6  HTTPクライアント

ContT を使って⾮同期機能を処理する例として、この章のソースコードの
Network.HTTP.Client モジュールについても⾒ていきましょう。このモジュールでは Async モ

ナドを使⽤して、NodeJSの⾮同期を request モジュールを使っています。

request モジュールは、URLとコールバックを受け取り、応答が利⽤可能なとき、またはエ
ラーが発⽣したときにHTTP（S）リクエストを⽣成してコールバックを呼び出す関数を提供し
ます。 リクエストの例を次に⽰します。

require('request')('http://purescript.org'), function(err, _, body) { 

  if (err) { 

    console.error(err); 

  } else { 



    console.log(body); 

  } 

}); 

Async モナドを使うと、この簡単な例をPureScriptで書きなおすことができます。

Network.HTTP.Client モジュールでは、 request メソッドは以下のようなAPIを持つ関数
getImpl としてラップされています。

foreign import data HTTP :: Effect 

 

type URI = String 

 

foreign import getImpl 

  :: forall eff 

   . Fn3 URI 

         (String -> Eff (http :: HTTP | eff) Unit) 

         (String -> Eff (http :: HTTP | eff) Unit) 

         (Eff (http :: HTTP | eff) Unit) 

exports.getImpl = function(uri, done, fail) { 

  return function() { 

    require('request')(uri, function(err, _, body) { 

      if (err) { 

        fail(err)(); 

      } else { 

        done(body)(); 

      } 

    }); 

  }; 

}; 

再び Data.Function.Uncurried モジュールを使って、これを通常のカリー化された
PureScript関数に変換します。先ほどと同じように、２つのコールバックを Maybe Chunk 型の値
を受け⼊れるひとつのコールバックに変換しています。 Either String String 型の値を受け取
り、 ContT データ構築⼦を適⽤して Async モナドのアクションを構築します。

get :: forall eff. 

  URI -> 

  Async (http :: HTTP | eff) (Either String String) 

get req = ContT \k -> 

  runFn3 getImpl req (k <<< Right) (k <<< Left) 



演 習

1. (やや難しい) runContT を使ってHTTP応答の各チャンクをコンソールへ出⼒す
ることで、 get を試してみてください。

2. (やや難しい) readFileCont と writeFileCont に対して以前に⾏ったように、
ExceptT を使い get をラップする関数 getEx を書いてください。

1.（難しい） getEx と writeFileContEx を使って、ディスク上のファイルからの内容をを保
存する関数を書いてください。

12 .7  並列計算

ContT モナドとdo記法を使って、⾮同期計算を順番に実⾏されるように合成する⽅法を⾒て
きました。⾮同期計算を並列に合成することもできたら便利でしょう。

もし ContT を使って Eff モナドを変換しているなら、単に２つの計算のうち⼀⽅を開始した
後に他⽅の計算を開始すれば、並列に計算することができます。

purescript-parallel パッケージは型クラス Parallel を定義します。この型クラスはモナド
のために並列計算を提供する Async のようなものです。以前に本書でApplicative関⼿を導⼊し
たとき、並列計算を合成するときにApplicative関⼿がどのように便利なのかを観察しました。実
は Parallel のインスタンスは、( Async のような)モナド m と、並列に計算を合成するために
使われるApplicative関⼿ f との対応関係を定義しているのです。

class (Monad m, Applicative f) <= Parallel f m | m -> f, f -> m where 

  sequential :: forall a. f a -> m a 

  parallel :: forall a. m a -> f a 

このクラスは2つの関数を定義しています。

parallel ：モナド m を計算し、それを応⽤ファンクタ f の計算に変換します。
sequential ：反対⽅向の変換を⾏います。



purescript-parallel ライブラリは Async モナドの Parallel インスタンスを提供します。
これは、2つの継続(continuation)のどちらが呼び出されたかを追跡することによって、変更可能
な参照を使⽤して並列に Async アクションを組み合わせます。 両⽅の結果が返されたら、最終
結果を計算してメインの継続に渡すことができます。

parallel 関数を使うと readFileCont アクションの別のバージョンを作成することもできま
す。これは並列に組み合わせることができます。２つのテキストファイルを並列に読み取り、連
結してその結果を出⼒する簡単な例は次のようになります。

import Prelude 

import Control.Apply (lift2) 

import Control.Monad.Cont.Trans (runContT) 

import Control.Monad.Eff.Console (logShow) 

import Control.Monad.Parallel (parallel, sequential) 

 

main = flip runContT logShow do 

  sequential $ 

   lift2 append 

     <$> parallel (readFileCont "/tmp/1.txt") 

     <*> parallel (readFileCont "/tmp/2.txt") 

readFileCont は Either ErrorCode String 型の値を返すので、 lift2 を使って Either 型
構築⼦より append 関数を持ち上げて結合関数を形成する必要があることに注意してください。

Applicative関⼿では任意個引数の関数の持ち上げができるので、このApplicativeコンビネータ
を使ってより多くの計算を並列に実⾏することができます。 traverse と sequence のような
Applicative関⼿を扱うすべての標準ライブラリ関数から恩恵を受けることもできます。

必要に応じて Parralel と runParallel を使って型構築⼦を変更することで、do記法ブロッ
クのApplicativeコンビネータを使って、直列的なコードの⼀部で並列計算を結合したり、または
その逆を⾏ったりすることができます。

演 習

1. (簡単) parallel と sequential を使って2つのHTTPリクエストを作成し、そ
れらのレスポンス内容を並⾏して収集します。あなたの結合関数は2つのレスポン
ス内容を連結しなければならず、続けて print を使って結果をコンソールに出⼒
してください。



2. (やや難しい) Async に対応するapplicative関⼿は Alternative のインスタンス
です。このインスタンスによって定義される <|> 演算⼦は2つの計算を並列に実⾏
し、最初に完了する計算結果を返します。

この Alternative インスタンスを setTimeoutCont 関数と共に使⽤して関数を
定義してください。

timeout :: forall a eff 

         . Milliseconds 

        -> Async (timeout :: TIMEOUT | eff) a 

        -> Async (timeout :: TIMEOUT | eff) (Maybe a) 

指定された計算が指定されたミリ秒数以内に結果を提供しない場合、 Nothing
を返します。

3. (やや難しい) purescript-parallel は ExceptT を含むいくつかのモナド変換⼦
のための Parallel クラスのインスタンスも提供します。

lift2 で append を持ち上げる代わりに、 ExceptT を使ってエラー処理を⾏う
ように、並列ファイル⼊出⼒の例を書きなおしてください。解決策は Async モナ
ドを変換するために ExceptT 変換⼦を使うとよいでしょう。

同様の⼿法で複数の⼊⼒ファイルを並列に読み込むために concatenateMany 関
数を書き換えてください。

4. (難しい、拡張) ディスク上のJSON⽂書の配列が与えられ、それぞれの⽂書はデ
ィスク上の他のファイルへの参照の配列を含んでいるとします。

{ references: ['/tmp/1.json', '/tmp/2.json'] } 

⼊⼒として単⼀のファイル名をとり、そのファイルから参照されているディスク
上のすべてのJSONファイルをたどって、参照されたすべてのファイルの⼀覧を収
集するユーティリティを書いてください。

そのユーティリティは、JSON⽂書を解析するために purescript-foreign ライ
ブラリを使⽤する必要があり、単⼀のファイルが参照するファイルは並列に取得し
なければなりません！



まとめ

この章ではモナド変換⼦の実⽤的なデモンストレーションを⾒てきました。

コールバック渡しの⼀般的なJavaScriptのイディオムを ContT モナド変換⼦によって捉えること
ができる⽅法を説明しました。
どのようにコールバック地獄の問題を解決するかを説明しました。 直列の⾮同期計算を表現す
るdo記法を使⽤して、かつ並列性を表現するためにApplicative関⼿によって解決することがで
きる⽅法を説明しました。
⾮同期エラーを表現するために ExceptT を使いました。



第13章 テストの⾃動⽣成

13 .1  この章の⽬標

この章では、テスティングの問題に対する、型クラスの特に洗練された応⽤について⽰しま
す。どのようにテストするのかをコンパイラに教えるのではなく、コードがどのような性質を持
っているべきかを教えることでテストします。型クラスを使って無作為データ⽣成のための定型
コードを隠し、テストケースを仕様から無作為に⽣成することができます。これは⽣成的テステ
ィング(generative testing、またはproperty-based testing）と呼ばれ、HaskellのQuickCheckライ
ブラリによって知られるようになった⼿法です。

purescript-quickcheck パッケージはHaskellのQuickCheckライブラリをPureScriptにポーテ
ィングしたもので、型や構⽂はもとのライブラリとほとんど同じようになっています。
purescript-quickcheck を使って簡単なライブラリをテストし、Pulpでテストスイートを⾃動

化されたビルドに統合する⽅法を⾒ていきます。

13 .2  プロジェクトの準備

この章のプロジェクトにはBower依存関係として purescript-quickcheck が追加されます。

Pulpプロジェクトでは、テストソースは test ディレクトリに置かれ、テストスイートのメ
インモジュールは Test.Main と名づけられます。 テストスイートは、 pulp test コマンドを
使⽤して実⾏できます。

13 .3  プロパティの書き込み

Merge モジュールでは purescript-quickcheck ライブラリの機能を実演するために使う簡単
な関数 merge が実装されています。

merge :: Array Int -> Array Int -> Array Int 

http://www.haskell.org/haskellwiki/Introduction_to_QuickCheck1


merge は2つのソートされた数の配列をとって、その要素を統合し、ソートされた結果を返し
ます。例えば次のようになります。

> import Merge 

> merge [1, 3, 5] [2, 4, 6] 

 

[1, 2, 3, 4, 5, 6] 

典型的なテストスイートでは、⼿作業でこのような⼩さなテストケースをいくつも作成し、結
果が正しい値と等しいことを確認することでテスト実施します。しかし、 merge 関数について
知る必要があるものはすべて、2つの性質に要約することができます。

(既ソート性) xs と ys がソート済みなら、 merge xs ys もソート済みになります。
（部分配列） xs と ys ははどちらも merge xs ys の部分配列で、要素は元の配列と同じ順序
で現れます。

purescript-quickcheck では、無作為なテストケースを⽣成することで、直接これらの性質
をテストすることができます。コードが持つべき性質を、次のような関数として述べるだけで
す。

main = do 

  quickCheck \xs ys -> 

    isSorted $ merge (sort xs) (sort ys) 

  quickCheck \xs ys -> 

    xs `isSubarrayOf` merge xs ys 

ここで、 isSorted と isSubarrayOf は次のような型を持つ補助関数として実装されていま
す。

isSorted :: forall a. Ord a => Array a -> Boolean 

isSubarrayOf :: forall a. Eq a => Array a -> Array a -> Boolean 

このコードを実⾏すると、 purescript-quickcheck は無作為な⼊⼒ xs と ys を⽣成してこ
の関数に渡すことで、主張しようとしている性質を反証しようとします。何らかの⼊⼒に対して
関数が false を返した場合、性質は正しくないことが⽰され、ライブラリはエラーを発⽣させ
ます。幸いなことに、次のように100個の無作為なテストケースを⽣成しても、ライブラリはこ
の性質を反証することができません。

$ pulp test 

 

* Build successful. Running tests... 



 

100/100 test(s) passed. 

100/100 test(s) passed. 

 

* Tests OK. 

もし merge 関数に意図的にバグを混⼊した場合（例えば、⼤なりのチェックを⼩なりのチェ
ックへと変更するなど）、最初に失敗したテストケースの後で例外が実⾏時に投げられます。

Error: Test 1 failed: 

Test returned false 

このエラーメッセージではあまり役に⽴ちませんが、これから⾒ていくように、少しの作業で
改良することができます。

13 .4  エラーメッセージの改善

テストケースが失敗した時に同時にエラーメッセージを提供するには、
purescript-quickcheck の 演算⼦を使います。次のように性質の定義に続けて で区切って

エラーメッセージを書くだけです。

quickCheck \xs ys -> 

  let 

    result = merge (sort xs) (sort ys) 

  in 

    xs `isSubarrayOf` result <?> show xs <> " not a subarray of " <> show result 

このとき、もしバグを混⼊するようにコードを変更すると、最初のテストケースが失敗したと
きに改良されたエラーメッセージが表⽰されます。

Error: Test 6 failed: 

[79168] not a subarray of [-752832,686016] 

⼊⼒ xs が無作為に選ばれた数の配列として⽣成されていることに注⽬してください。



演 習

1. （簡単） 空の配列を持つ配列を統合しても元の配列は変更されない、と主張す
る性質を書いてください。

2. （簡単） merge の残りの性質に対して、適切なエラーメッセージを追加してく
ださい。

13 .5  多相的なコードのテスト

Merge モジュールでは、数の配列だけでなく、 Ord 型クラスに属するどんな型の配列に対し
ても動作する、 merge 関数を⼀般化した mergePoly という関数が定義されています。

mergePoly :: forall a. Ord a => Array a -> Array a -> Array a 

merge の代わりに mergePoly を使うように元のテストを変更すると、次のようなエラーメッ
セージが表⽰されます。

No type class instance was found for 

 

  Test.QuickCheck.Arbitrary.Arbitrary t0 

 

The instance head contains unknown type variables. 

Consider adding a type annotation. 

このエラーメッセージは、配列に持たせたい要素の型が何なのかわからないので、コンパイラ
が無作為なテストケースを⽣成できなかったということを⽰しています。このような場合、補助
関数を使と、コンパイラが特定の型を推論すること強制できます。例えば、恒等関数の同義語と
して ints という関数を定義します。

ints :: Array Int -> Array Int 

ints = id 

それから、コンパイラが引数の２つの配列の型 Array Int を推論するように、テストを変更
します。



quickCheck \xs ys -> 

  isSorted $ ints $ mergePoly (sort xs) (sort ys) 

quickCheck \xs ys -> 

  ints xs `isSubarrayOf` mergePoly xs ys 

ここで、 numbers 関数が不明な型を解消するために使われるので、 xs と ys はどちらも型
Array Int を持っています。

演 習

1. (簡単) xs と ys の型を Array Boolean に強制する関数 bools を書き、
mergePoly をその型でテストする性質を追加してください。

2. (やや難しい) 標準関数から(例えば purescript-arrays パッケージから)ひとつ
関数を選び、適切なエラーメッセージを含めてQuickCheckの性質を書いてくださ
い。その性質は、補助関数を使って多相型引数を Int か Boolean のどちらかに固
定しなければいけません。

13 .6  任意のデータの⽣成

purescript-quickcheck ライブラリを使って性質についてのテストケースを無作為に⽣成す
る⽅法について説明します。

無作為に値を⽣成することができるような型は、次のような型クラス Arbitary のインスタ
ンスを持っています。

class Arbitrary t where 

  arbitrary :: Gen t 

Gen 型構築⼦は決定的無作為データ⽣成の副作⽤を表しています。 決定的無作為データ⽣成
は、擬似乱数⽣成器を使って、シード値から決定的無作為関数の引数を⽣成します。



Test.QuickCheck.Gen モジュールは、ジェネレータを構築するためのいくつかの有⽤なコンビ
ネータを定義します。

Gen はモナドでもApplicative関⼿でもあるので、 Arbitary 型クラスの新しいインスタンス
を作成するのに、いつも使っているようなコンビネータを⾃由に使うことができます。

例えば、 purescript-quickcheck ライブラリで提供されている Int 型の Arbitrary インス
タンスは、関数を整数から任意の整数値のバイトまでマップするための Functor インスタンス
を Gen に使⽤することで、バイト値の分布した値を⽣成します。

newtype Byte = Byte Int 

 

instance arbitraryByte :: Arbitrary Byte where 

  arbitrary = map intToByte arbitrary 

    where 

    intToByte n | n >= 0 = Byte (n `mod` 256) 

                | otherwise = intToByte (-n) 

ここでは、0から255までの間の整数値であるような型 Byte を定義しています。 Arbitrary
インスタンスの <$> 演算⼦を使って、 uniformToByte 関数を arbitrary アクションまで持ち
上げています。この型の arbitrary アクションの型は Gen Number だと推論されますが、これ
は0から1の間に均⼀に分布する数を⽣成することを意味しています。

この考え⽅を merge に対しての既ソート性テストを改良するのに使うこともできます。

quickCheck \xs ys -> 

  isSorted $ numbers $ mergePoly (sort xs) (sort ys) 

このテストでは、任意の配列 xs と ys を⽣成しますが、 merge はソート済みの⼊⼒を期待
しているので、 xs と ys をソートしておかなければなりません。⼀⽅で、ソートされた配列を
表すnewtypeを作成し、ソートされたデータを⽣成する Arbitrary インスタンスを書くことも
できます。

newtype Sorted a = Sorted (Array a) 

 

sorted :: forall a. Sorted a -> Array a 

sorted (Sorted xs) = xs 

 

instance arbSorted :: (Arbitrary a, Ord a) => Arbitrary (Sorted a) where 

  arbitrary = map (Sorted <<< sort) arbitrary 



この型構築⼦を使うと、テストを次のように変更することができます。

quickCheck \xs ys -> 

  isSorted $ ints $ mergePoly (sorted xs) (sorted ys) 

これは些細な変更に⾒えるかもしれませんが、 xs と ys の型はただの Array Int から
Sorted Int へと変更されています。これにより、 mergePoly 関数はソート済みの⼊⼒を取

る、という意図を、わかりやすく⽰すことができます。理想的には、 mergePoly 関数⾃体の型
が Sorted 型構築⼦を使うようにするといいでしょう。

より興味深い例として、 Tree モジュールでは枝の値でソートされた⼆分⽊の型が定義されて
います。

data Tree a 

  = Leaf 

  | Branch (Tree a) a (Tree a) 

Tree モジュールでは次のAPIが定義されています。

insert    :: forall a. Ord a => a -> Tree a -> Tree a 

member    :: forall a. Ord a => a -> Tree a -> Boolean 

fromArray :: forall a. Ord a => Array a -> Tree a 

toArray   :: forall a. Tree a -> Array a 

insert 関数は新しい要素をソート済みの⼆分⽊に挿⼊するのに使われ、 member 関数は特定
の値の有無を⽊に問い合わせるのに使われます。例えば次のようになります。

> import Tree 

 

> member 2 $ insert 1 $ insert 2 Leaf 

true 

 

> member 1 Leaf 

false 

toArray 関数と fromArray 関数は、ソートされた⽊とソートされた配列を相互に変換するた
めに使われます。 fromArray を使うと、⽊についての Arbitrary インスタンスを書くことがで
きます。

instance arbTree :: (Arbitrary a, Ord a) => Arbitrary (Tree a) where 

  arbitrary = map fromArray arbitrary 



型 a についての有効な Arbitary インスタンスが存在していれば、テストする性質の引数の
型として Tree a を使うことができます。例えば、 member テストは値を挿⼊した後は常に
true を返すことをテストできます。

quickCheck \t a -> 

  member a $ insert a $ treeOfInt t 

ここでは、引数 t は Tree Number 型の無作為に⽣成された⽊です。型引数は、識別関数
treeOfInt によって明確化されています。

演 習

1. （やや難しい） a-z の範囲から無作為に選ばれた⽂字の集まりを⽣成する
Arbitrary インスタンスを持った、 String のnewtypeを作ってください。ヒン

ト： Test.QuickCheck.Gen モジュールから elements と arrayOf 関数を使いま
す。

2. （難しい） ⽊に挿⼊された値は、任意に多くの挿⼊があった後も、その⽊の構
成要素であることを主張する性質を書いてください。

13 .7  ⾼階関数のテスト

Merge モジュールは merge 関数についての他の⽣成も定義します。 mergeAith 関数は、統
合される要素の順序を決定するのに使われる、追加の関数を引数としてとります。つまり
mergeWith は⾼階関数です。

例えば、すでに⻑さの昇順になっている２つの配列を統合するのに、 length 関数を最初の引
数として渡します。このとき、結果も⻑さの昇順になっていなければなりません。

> import Data.String 

 



> mergeWith length 

    ["", "ab", "abcd"] 

    ["x", "xyz"] 

 

["","x","ab","xyz","abcd"] 

このような関数をテストするにはどうしたらいいでしょうか。理想的には、関数であるような
最初の引数を含めた、３つの引数すべてについて、値を⽣成したいと思うでしょう。

関数を無作為に⽣成せきるようにする、もうひとつの型クラスがあります。この型クラスは
Coarbitrary と呼ばれており、次のように定義されています。

class Coarbitrary t where 

  coarbitrary :: forall r. t -> Gen r -> Gen r 

coarbitrary 関数は、型 t と、関数の結果の型 r についての無作為な⽣成器を関数の引数
としてとり、無作為な⽣成器をかき乱すのにこの引数を使います。つまり、この引数を使って、
乱数⽣成器の無作為な出⼒を変更しているのです。

また、もし関数の定義域が Coarbitrary で、値域が Arbitrary なら、 Arbitrary の関数を
与える型クラスインスタンスが存在しています。

instance arbFunction :: (Coarbitrary a, Arbitrary b) => Arbitrary (a -> b) 

実は、これが意味しているのは、引数として関数を取るような性質を記述できるということで
す。 mergeWith 関数の場合では、新しい引数を考慮するようにテストを修正すると、最初の引
数を無作為に⽣成することができます。

既ソート性の性質については、必ずしも Ord インスタンスを持っているとは限らないので、
結果がソートされているということを保証することができませんが、引数として渡す関数 f に
したがって結果がソートされている期待することはできます。さらに、２つの⼊⼒配列が f に
従ってソートされている必要がありますので、 sortBy 関数を使って関数 f が適⽤されたあと
の⽐較に基づいて xs と ys をソートします。

quickCheck \xs ys f -> 

  isSorted $ 

    map f $ 

      mergeWith (intToBool f) 

                (sortBy (compare `on` f) xs) 

                (sortBy (compare `on` f) ys) 



ここでは、関数 f の型を明確にするために、関数 intToBool を使⽤しています。

intToBool :: (Int -> Boolean) -> Int -> Boolean 

intToBool = id 

部分配列性については、単に関数の名前を mergeWith に変えるだけです。引き続き⼊⼒配列
は結果の部分配列になっていると期待できます。

quickCheck \xs ys f -> 

  xs `isSubarrayOf` mergeWith (numberToBool f) xs ys 

関数は Arbitrary であるだけでなく Coarbitrary でもあります。

instance coarbFunction :: (Arbitrary a, Coarbitrary b) => Coarbitrary (a -> b) 

これは値の⽣成が単純な関数だけに限定されるものではないことを意味しています。つまり、
⾼階関数や、引数が⾼階関数であるような関数すら無作為に⽣成することができるのです。

13 .8  Coarbi t raryのインスタンスを書く

Gen の Monad や Applicative インスタンスを使って独⾃のデータ型に対して Arbitrary イ
ンスタンスを書くことができるのとちょうど同じように、独⾃の Coarbitrary インスタンスを
書くこともできます。これにより、無作為に⽣成される関数の定義域として、独⾃のデータ型を
使うことができるようになります。

Tree 型の Coarbitrary インスタンスを書いてみましょう。枝に格納されている要素の型に
Coarbitrary インスタンスが必要になります。

instance coarbTree :: Coarbitrary a => Coarbitrary (Tree a) where 

型 Tree a の値を与えられた乱数発⽣器をかき乱す関数を記述する必要があります。⼊⼒値が
Leaf であれば、そのままの⽣成器を返します。

coarbitrary Leaf = id 



もし⽊が Branch なら、 関数合成で独⾃のかき乱し関数を作ることにより、 左の部分⽊、
値、右の部分⽊を使って⽣成器をかき乱します。

coarbitrary (Branch l a r) = 

  coarbitrary l <<< 

  coarbitrary a <<< 

  coarbitrary r 

これで、⽊を引数にとるような関数を含む性質を⾃由に書くことができるようになりました。
たとえば、 Tree モジュールでは述語が引数のどんな部分⽊についても成り⽴っているかを調べ
る関数 anywhere が定義されています。

anywhere :: forall a. (Tree a -> Boolean) -> Tree a -> Boolean 

これで、無作為にこの述語関数 anywhere を⽣成することができるようになりました。例え
ば、 anywhere 関数が次のようなある命題のもとで不変であることを期待します。

quickCheck \f g t -> 

  anywhere (\s -> f s || g s) t == 

    anywhere f (treeOfInt t) || anywhere g t 

ここで、 treeOfInt 関数は⽊に含まれる値の型を型 Int に固定するために使われています。

treeOfInt :: Tree Int -> Tree Int 

treeOfInt = id 

13 .9  副作⽤のないテスト

テストの⽬的では通常、テストスイートの main アクションには quickCheck 関数の呼び出
しが含まれています。しかし、副作⽤を使わない quickCheckPure と呼ばれる quickCheck 関数
の亜種もあります。 quickCheckPure は、⼊⼒として乱数の種をとり、テスト結果の配列を返す
純粋な関数です。

PSCi を使⽤して quickCheckPure を使ってみましょう。ここでは merge 操作が結合法則を
満たすことをテストしてみます。



> import Prelude 

> import Merge 

> import Test.QuickCheck 

> import Test.QuickCheck.LCG (mkSeed) 

 

> :paste 

… quickCheckPure (mkSeed 12345) 10 \xs ys zs -> 

…   ((xs `merge` ys) `merge` zs) == 

…     (xs `merge` (ys `merge` zs)) 

… ^D 

 

Success : Success : ... 

quickCheckPure は乱数の種、⽣成するテストケースの数、テストする性質の３つの引数をと
ります。もしすべてのテストケースに成功したら、 Success データ構築⼦の配列がコンソール
に出⼒されます。

quickCheckPure は、性能ベンチマークの⼊⼒データ⽣成や、ウェブアプリケーションのフォ
ームデータ例を無作為に⽣成するというような状況で便利かもしれません。

演 習

1. （簡単） Byte と Sorted 型構築⼦についての Coarbitrary インスタンスを書
いてください。

2. （やや難しい）任意の関数 f について、 mergeWith f 関数の結合性を主張する
(⾼階)性質を書いてください。 quickCheckPure を使って PSCi でその性質をテス
トしてください。

3. （やや難しい）次のデータ型の Coarbitrary インスタンスを書いてください。

data OneTwoThree a = One a | Two a a | Three a a a 

ヒント： Test.QuickCheck.Gen で定義された oneOf 関数を使って Arbitrary
インスタンスを定義します。

4. (やや難しい) all 関数を使って quickCheckPure 関数の結果を単純化してくだ
さい。その関数はもしどんなテストもパスするなら true を返し、そうでなければ



false を返さなくてはいけません。 purescript-monoids で定義されている
First モノイドを、失敗時の最初のエラーを保存するために foldMap 関数と⼀緒

に使ってみてください。

まとめ

この章では、⽣成的テスティングのパラダイムを使って宣⾔的な⽅法でテストを書くための、
purescript-quickcheck パッケージを導⼊しました。

pulp test 使ってQuickCheckをテストを⾃動化する⽅法を説明しました。
エラーメッセージを改良する 演算⼦の使い⽅と、性質を関数として書く⽅法を説明しました。
Arbitrary と Coarbitrary 型クラスは、定型的なテストコードの⾃動⽣成を可能にし、⾼階性

質関数を可能にすることも説明しました。
独⾃のデータ型に対して Arbitrary と Coarbitrary インスタンスを実装する⽅法を説明しま
した。



第14章 領域特化⾔語

14 .1  この章の⽬標

この章では、多数の標準的な⼿法を使ったPureScriptにおける領域特化⾔語(domain-specific
language, DSL) の実装について探求していきます。

領域特化⾔語とは、特定の問題領域での開発に適した⾔語のことです。領域特化⾔語の構⽂お
よび機能は、その領域内の考え⽅を表現するコードの読みやすさを最⼤限に発揮すべく選択され
ます。本書の中では、すでに領域特化⾔語の例を幾つか⾒てきています。

第11章で開発された Game モナドと関連するアクションは、テキストアドベンチャーゲーム開
発という領域に対しての領域特化⾔語を構成しています。
第12章で ContT と Parallel 関⼿のために書いたコンビネータのライブラリは、⾮同期プログ
ラミングの領域に対する領域特化⾔語の例と考えることができます。
第13章で扱った purescript-quickcheck パッケージは、⽣成的テスティングの領域の領域特化
⾔語です。このコンビネータはテストの性質対して特に表現⼒の⾼い記法を可能にします。

この章では、領域特化⾔語の実装において、いくつかの標準的な⼿法による構造的なアプロー
チを取ります。これがこの話題の完全な説明だということでは決してありませんが、独⾃の⽬的
に対する具体的なDSLを構築するには⼗分な知識を与えてくれるでしょう。

この章で実⾏している例は、HTML⽂書を作成するための領域特化⾔語になります。正しい
HTML⽂書を記述するための型安全な⾔語を開発することが⽬的で、少しづつ実装を改善する
ことによって作業していきます。

14 .2  プロジェクトの準備

この章で使うプロジェクトには新しいBower依存性が追加されます。これから使う道具のひと
つであるFreeモナドが定義されている purescript-free ライブラリです。

このプロジェクトのソースコードは、PSCiを使ってビルドすることができます。

14 .3  HTMLデータ型



このHTMLライブラリの最も基本的なバージョンは Data.DOM.Simple モジュールで定義され
ています。このモジュールには次の型定義が含まれています。

newtype Element = Element 

  { name         :: String 

  , attribs      :: Array Attribute 

  , content      :: Maybe (Array Content) 

  } 

 

data Content 

  = TextContent String 

  | ElementContent Element 

 

newtype Attribute = Attribute 

  { key          :: String 

  , value        :: String 

  } 

Element 型はHTMLの要素を表しており、各要素は要素名、属性のペア  の配列と、要素の内
容でで構成されています。 content プロパティでは、 Maybe タイプを使って要素が開いている
(他の要素やテキストを含む)か閉じているかを⽰しています。

このライブラリの鍵となる機能は次の関数です。

render :: Element -> String 

この関数はHTML要素をHTML⽂字列として出⼒します。 PSCi で明⽰的に適当な型の値を
構築し、ライブラリのこのバージョンを試してみましょう。

$ pulp repl 

 

> import Prelude 

> import Data.DOM.Simple 

> import Data.Maybe 

> import Control.Monad.Eff.Console 

 

> :paste 

… log $ render $ Element 

…   { name: "p" 

…   , attribs: [ 

…       Attribute 

…         { key: "class" 

…         , value: "main" 

…         } 



…     ] 

…   , content: Just [ 

…       TextContent "Hello World!" 

…     ] 

…   } 

… ^D 

 

Hello World!

 

unit 

現状のライブラリにはいくつかの問題があります。

HTML⽂書の作成に⼿がかかります。すべての新しい要素が少なくとも1つのレコードと1つの
データ構築⼦が必要です。
無効な⽂書を表現できてしまいます。
要素名の⼊⼒を間違えるかもしれません
要素に間違った型の属性を関連付けることができてしまいます
開いた要素が正しい場合でも、閉じた要素を使⽤することができてしまいます

この章では、さまざまな⼿法を⽤いてこれらの問題を解決し、このライブラリーをHTML⽂
書を作成するために使える領域特化⾔語にしていきます。

14 .4  スマート構築⼦

最初に導⼊する⼿法は⽅法は単純なものですが、とても効果的です。モジュールの使⽤者にデ
ータの表現を露出する代わりに、モジュールエクスポートリスト(module exports list)を使って
データ構築⼦ Element 、 Content 、 Attribute を隠蔽し、正しいことが明らかなデータだけ
構築する、いわゆるスマート構築⼦(smart constructors)だけをエクスポートします。

例を⽰しましょう。まず、HTML要素を作成するための便利な関数を提供します。

element :: String -> Array Attribute -> Maybe (Array Content) -> Element 

element name attribs content = Element 

  { name:      name 

  , attribs:   attribs 

  , content:   content 

  } 



次に、 element 関数を適⽤することによってHTML要素を作成する、スマート構築⼦を作成
します。

a :: Array Attribute -> Array Content -> Element 

a attribs content = element "a" attribs (Just content) 

 

p :: Array Attribute -> Array Content -> Element 

p attribs content = element "p" attribs (Just content) 

 

img :: Array Attribute -> Element 

img attribs = element "img" attribs Nothing 

最後に、正しいデータ構造だけを構築することがわかっているこれらの関数をエクスポートす
るように、モジュールエクスポートリストを更新します。

module Data.DOM.Smart 

  ( Element 

  , Attribute(..) 

  , Content(..) 

 

  , a 

  , p 

  , img 

 

  , render 

  ) where 

モジュールエクスポートリストはモジュール名の直後の括弧内に書きます。各モジュールのエ
クスポートは次の3種類のいずれかです。

値の名前で⽰された、値(または関数)
クラスの名で⽰された、型クラス
型の名前で⽰された型構築⼦、およびそれに続けて括弧で囲まれた関連するデータ構築⼦のリス
ト

ここでは、 Element の型をエクスポートしていますが、データ構築⼦はエクスポートしてい
ません。もしデータ構築⼦をエクスポートすると、モジュールの使⽤者が不正なHTML要素を
構築できてしまいます。

Attribute と Content 型についてはデータ構築⼦をすべてエクスポートしています(エクス
ポートリストの記号 .. で⽰されています)。これから、これらの型にスマート構築⼦の⼿法を
適⽤していきます。



すでにライブラリにいくつかの⼤きな改良を加わっていることに注意してください。

不正な名前を持つHTML要素を表現することは不可能です(もちろん、ライブラリが提供する要
素名に制限されています)。
閉じた要素は、構築するときに内容を含めることはできません。

Content 型にもとても簡単にこの⼿法を適⽤することができます。単にエクスポートリスト
から Content 型のデータ構築⼦を取り除き、次のスマート構築⼦を提供します。

text :: String -> Content 

text = TextContent 

 

elem :: Element -> Content 

elem = ElementContent 

Attribute 型にも同じ⼿法を適⽤してみましょう。まず、属性のための汎⽤のスマート構築
⼦を⽤意します。最初の試みとしては、次のようなものになるかもしれません。

attribute :: String -> String -> Attribute 

attribute key value = Attribute 

  { key: key 

  , value: value 

  } 

 

infix 4 attribute as := 

この定義では元の Element 型と同じ問題に悩まされています。存在しなかったり、名前が間
違っているような属性を表現することが可能です。この問題を解決するために、属性名を表す
newtypeを作成します。

newtype AttributeKey = AttributeKey String 

それから、この演算⼦を次のように変更します。

attribute :: AttributeKey -> String -> Attribute 

attribute (AttributeKey key) value = Attribute 

  { key: key 

  , value: value 

  } 

AttributeKey データ構築⼦をエクスポートしなければ、明⽰的にエクスポートされた次のよ
うな関数を使う以外に、使⽤者が型 AttributeKey の値を構築する⽅法はありません。いくつ



かの例を⽰します。

href :: AttributeKey 

href = AttributeKey "href" 

 

_class :: AttributeKey 

_class = AttributeKey "class" 

 

src :: AttributeKey 

src = AttributeKey "src" 

 

width :: AttributeKey 

width = AttributeKey "width" 

 

height :: AttributeKey 

height = AttributeKey "height" 

新しいモジュールの最終的なエクスポートリストは次のようになります。もうどんなデータ構
築⼦も直接エクスポートしていないことに注意してください。

module Data.DOM.Smart 

  ( Element 

  , Attribute 

  , Content 

  , AttributeKey 

 

  , a 

  , p 

  , img 

 

  , href 

  , _class 

  , src 

  , width 

  , height 

 

  , attribute, (:=) 

  , text 

  , elem 

 

  , render 

  ) where 

PSCi でこの新しいモジュールを試してみると、コードが⼤幅に簡潔になり、改良されている
ことがわかります。



$ pulp repl 

 

> import Prelude 

> import Data.DOM.Smart 

> import Control.Monad.Eff.Console 

> log $ render $ p [ _class := "main" ] [ text "Hello World!" ] 

 

Hello World!

 

unit 

しかし、基礎のデータ表現が変更されていないので、 render 関数を変更する必要はなかった
ことにも注⽬してください。これはスマート構築⼦による⼿法の利点のひとつです。外部APIの
使⽤者によって認識される表現から、モジュールの内部データ表現を分離することができるので
す。

演 習

1. (簡単) Data.DOM.Smart モジュールで render を使った新しいHTML⽂書の作成
を試してみましょう。

2. (やや難しい)  checked と disabled など、値を要求しないHTML属性があり
ますが、これらは次のような空の属性として表⽰されるかもしれません。

&lt;input disabled&gt; 

空の属性を扱えるように Attribute の表現を変更してください。要素に空の属
性を追加するために、 attribute または := の代わりに使える関数を記述してく
ださい。

14 .5  幻影型



次に適⽤する⼿法についての動機を与えるために、次のコードを考えてみます。

> log $ render $ img 

    [ src    := "cat.jpg" 

    , width  := "foo" 

    , height := "bar" 

    ] 

 

<img src="cat.jpg" width="foo" height="bar" /> 

unit 

ここでの問題は、 width と height についての⽂字列値を提供しているということで、ここ
で与えることができるのはピクセルやパーセントの単位の数値だけであるべきです。

AttributeKey 型にいわゆる幻影型(phantom type)引数を導⼊すると、この問題を解決できま
す。

newtype AttributeKey a = AttributeKey String 

定義の右辺に対応する型 a の値が存在しないので、この型変数 a は幻影型と呼ばれていま
す。この型 a はコンパイル時により多くの情報を提供するためだけに存在しています。任意の
型 AttributeKey a の値は実⾏時には単なる⽂字列ですが、そのキーに関連付けられた値に期
待されている型を教えてくれます。

AttributeKey の新しい形式で受け取るように、 attribute 関数の型を次のように変更しま
す。

attribute :: forall a. IsValue a => AttributeKey a -> a -> Attribute 

attribute (AttributeKey key) value = Attribute 

  { key: key 

  , value: toValue value 

  } 

ここで、幻影型の引数 a は、属性キーと属性値が互換性のある型を持っていることを確認す
るために使われます。使⽤者は AttributeKey a を型の値を直接作成できないので(ライブラリ
で提供されている定数を介してのみ得ることができます)、すべての属性が正しくなります。

IsValue 制約は、キーに関連付けられた値がなんであれ、その値を⽂字列に変換し、⽣成し
たHTML内に出⼒できることを保証します。 IsValue 型クラスは次のように定義されていま
す。



class IsValue a where 

  toValue :: a -> String 

String と Int 型についての型クラスインスタンスも提供しておきます。

instance stringIsValue :: IsValue String where 

  toValue = id 

 

instance intIsValue :: IsValue Int where 

  toValue = show 

また、これらの型が新しい型変数を反映するように、 AttributeKey 定数を更新しなければ
いけません。

href :: AttributeKey String 

href = AttributeKey "href" 

 

_class :: AttributeKey String 

_class = AttributeKey "class" 

 

src :: AttributeKey String 

src = AttributeKey "src" 

 

width :: AttributeKey Int 

width = AttributeKey "width" 

 

height :: AttributeKey Int 

height = AttributeKey "height" 

これで、不正なHTML⽂書を表現することが不可能で、 width と height 属性を表現するの
に数を使うことが強制されていることがわかります。

> import Prelude 

> import Data.DOM.Phantom 

> import Control.Monad.Eff.Console 

 

> :paste 

… log $ render $ img 

…   [ src    := "cat.jpg" 

…   , width  := 100 

…   , height := 200 

…   ] 

… ^D 

 



<img src="cat.jpg" width="100" height="200" /> 

unit 

演 習

1. (簡単) ピクセルまたはパーセントの⻑さのいずれかを表すデータ型を作成してく
ださい。その型について IsValue のインスタンスを書いてください。この型を使
うように width と height 属性を変更してください。

2. (難しい) 幻影型を使って真偽値 true 、 false についての表現を最上位で定義
することで、 AttributeKey が disabled や chacked のような空の属性を表現し
ているかどうかを符号化することができます。

data True 

data False 

幻影型を使って、使⽤者が attribute 演算⼦を空の属性に対して使うことを防
ぐように、前の演習の解答を変更してください。

14 .6  Freeモナド

APIに施す最後の変更は、 Content 型をモナドにしてdo記法を使えるようにするために、
Freeモナドと呼ばれる構造を使うことです。Freeモナドは、⼊れ⼦になった要素をわかりやすく
なるよう、HTML⽂書の構造化を可能にします。次のようなコードを考えます。

p [ _class := "main" ] 

  [ elem $ img 

      [ src    := "cat.jpg" 

      , width  := 100 

      , height := 200 

      ] 

  , text "A cat" 

  ] 



これを次のように書くことができるようになります。

p [ _class := "main" ] $ do 

  elem $ img 

    [ src    := "cat.jpg" 

    , width  := 100 

    , height := 200 

    ] 

  text "A cat" 

しかし、do記法だけがFreeモナドの恩恵だというわけではありません。モナドのアクション
の表現をその解釈から分離し、同じアクションに複数の解釈を持たせることをFreeモナドは可能
にします。

Free モナドは purescript-free ライブラリの Control.Monad.Free モジュールで定義され
ています。 PSCi を使うと、次のようにFreeモナドについての基本的な情報を⾒ることができま
す。

> import Control.Monad.Free 

 

> :kind Free 

(Type -> Type) -> Type -> Type 

Free の種は、引数として型構築⼦を取り、別の型構築⼦を返すことを⽰しています。実は、
Free モナドは任意の Functor を Monad にするために使うことができます！

モナドのアクションの表現を定義することから始めます。これを⾏うには、サポートする各モ
ナドアクションそれぞれについて、ひとつのデータ構築⼦を持つ Functor を作成する必要があ
ります。今回の場合、2つのモナドのアクションは elem と text になります。実際には、
Content 型を次のように変更するだけです。

data ContentF a 

  = TextContent String a 

  | ElementContent Element a 

 

instance functorContentF :: Functor ContentF where 

  map f (TextContent s x) = TextContent s (f x) 

  map f (ElementContent e x) = ElementContent e (f x) 

ここで、この ContentF 型構築⼦は以前の Content データ型とよく似ています。 Functor イ
ンスタンスでは、単に各データ構築⼦で型 a の構成要素に関数 f を適⽤します。



これにより、最初の型引数として ContentF 型構築⼦を使うことで構築された、新しい
Content 型構築⼦を Free モナドを包むnewtypeとして定義することができます。

type Content = Free ContentF 

型のシノニムの代わりにnewtypeを使⽤して、使⽤者に対してライブラリの内部表現を露出す
ることを避ける事ができます。 Content データ構築⼦を隠すことで、提供しているモナドのア
クションだけを使うことを仕様者に制限しています。

ContentF は Functor なので、 Free ContentF に対する Monad インスタンスが⾃動的に⼿
に⼊り、このインスタンスを Content 上の Monad インスタンスへと持ち上げることができま
す。

Content の新しい型引数を考慮するように、少し Element データ型を変更する必要がありま
す。モナドの計算の戻り値の型が Unit であることだけが要求されます。

newtype Element = Element 

  { name         :: String 

  , attribs      :: Array Attribute 

  , content      :: Maybe (Content Unit) 

  } 

また、 Content モナドについての新しいモナドのアクションになる elem と text 関数を変
更する必要があります。これを⾏うには、 Control.Monad.Free モジュールで提供されている
liftF 関数を使います。この関数の(簡略化された)型は次のようになっています。

liftF :: forall f a. (Functor f) => f a -> Free f a 

liftF は、何らかの型 a について、型 f a の値からFreeモナドのアクションを構築できる
ようにします。今回の場合、 ContentF 型構築⼦のデータ構築⼦を次のようにそのまま使うだけ
です。

text :: String -> Content Unit 

text s = liftF $ TextContent s unit 

 

elem :: Element -> Content Unit 

elem e = liftF $ ElementContent e unit 

他にもコードの変更はありますが、興味深い変更は render 関数に対してのものです。ここで
は、このFreeモナドを解釈しなければいけません。



14 .7  モナドの解釈

Control.Monad.Free モジュールでは、Freeモナドで計算を解釈するための多数の関数が提供
されています。

runFree 

  :: forall f a 

   . Functor f 

  => (f (Free f a) -> Free f a) 

  -> Free f a 

  -> a 

 

runFreeM 

  :: forall f m a 

   . (Functor f, MonadRec m) 

  => (f (Free f a) -> m (Free f a)) 

  -> Free f a 

  -> m a 

runFree 関数は、純粋な結果を計算するために使⽤されます。 runFreeM 関数は、フリーモ
ナドの動作を解釈するためにモナドを使⽤することを可能にします

厳密には、 MonadRec のより強い制約を満たすモナド m を使⽤する制限がされています。こ
れはスタックオーバーフローを⼼配する必要がないことを意味します。なぜなら m は安全な末
尾再帰モナド(monadic tail recursion)をサポートするからです。

まず、アクションを解釈することができるモナドを選ばなければなりません。
Writer String モナドを使って、結果のHTML⽂字列を累積することにします。

新しい render メソッドは補助関数 renderElement に移譲して開始し、 Writer モナドで計
算を実⾏するため execWriter を使⽤します。

render :: Element -> String 

render = execWriter <<< renderElement 

renderElement はwhereブロックで定義されています。

  where 

    renderElement :: Element -> Writer String Unit 

    renderElement (Element e) = do 



renderElement の定義は簡単で、いくつかの⼩さな⽂字列を累積するために Writer モナド
の tell アクションを使っています。

      tell "<" 

      tell e.name 

      for_ e.attribs $ \x -> do 

        tell " " 

        renderAttribute x 

      renderContent e.content 

次に、同じように簡単な renderAttribute 関数を定義します。

    where 

      renderAttribute :: Attribute -> Writer String Unit 

      renderAttribute (Attribute x) = do 

        tell x.key 

        tell "=\"" 

        tell x.value 

        tell "\"" 

renderContent 関数は、もっと興味深いものです。ここでは、 runFreeM 関数を使って、
Freeモナドの内部で補助関数 renderContentItem に移譲する計算を解釈しています。

      renderContent :: Maybe (Content Unit) -> Writer String Unit 

      renderContent Nothing = tell " />" 

      renderContent (Just content) = do 

        tell ">" 

        runFreeM renderContentItem content 

        tell "</" 

        tell e.name 

        tell ">" 

renderContentItem の型は runFreeM の型シグネチャから推測することができます。関⼿ f
は型構築⼦ ContentF で、モナド m は解釈している計算のモナド、つまり Writer String で
す。これにより renderContentItem について次の型シグネチャがわかります。

      renderContentItem :: ContentF (Content Unit) -> Writer String (Content Unit) 

ContentF の⼆つのデータ構築⼦でパターン照合するだけで、この関数を実装することができ
ます。



      renderContentItem (TextContent s rest) = do 

        tell s 

        pure rest 

      renderContentItem (ElementContent e rest) = do 

        renderElement e 

        pure rest 

それぞれの場合において、式 rest は型 Writer String を持っており、解釈計算の残りを表
しています。 rest アクションを呼び出すことによって、それぞれの場合を完了することができ
ます。

これで完了です！ PSCi で、次のように新しいモナドのAPIを試してみましょう。

> import Prelude 

> import Data.DOM.Free 

> import Control.Monad.Eff.Console 

 

> :paste 

… log $ render $ p [] $ do 

…   elem $ img [ src := "cat.jpg" ] 

…   text "A cat" 

… ^D 

 

<p><img src="cat.jpg" />A cat</p> 

unit 

演 習

1. (やや難しい) ContentF 型に新しいデータ構築⼦を追加して、⽣成されたHTMLに
コメントを出⼒する新しいアクション comment に対応してください。 liftF を使
ってこの新しいアクションを実装してください。新しい構築⼦を適切に解釈するよ
うに、解釈 renderContentItem を更新してください。

14 .8  ⾔語の拡張



すべてのアクションが型 Unit の何かを返すようなモナドは、さほど興味深いものではありま
せん。実際のところ、概ね良くなったと思われる構⽂は別として、このモナドは Monoid 以上の
機能は何の追加していません。

意味のある結果を返す新しいモナドアクションでこの⾔語を拡張することで、Freeモナド構造
の威⼒を説明しましょう  。

アンカーを使⽤して⽂書のさまざまな節へのハイパーリンクが含まれているHTML⽂書を⽣
成するとします。⼿作業でアンカーの名前を⽣成すればいいので、これは既に実現できていま
す。⽂書中で少なくとも２回、ひとつはアンカーの定義⾃⾝に、もうひとつはハイパーリンク
に、アンカーが含まれています。しかし、この⽅法には根本的な問題がいくつかあります。

開発者は⼀意なアンカー名を⽣成するために失敗することがあります。
開発者は、アンカー名のひとつまたは複数のインスタンスを誤って⼊⼒するかもしれません。

⾃分の間違いから開発者を保護するために、アンカー名を表す新しい型を導⼊し、新しい⼀意
な名前を⽣成するためのモナドアクションを提供することができます。

最初の⼿順は、名前の型を新しく追加することです。

newtype Name = Name String 

 

runName :: Name -> String 

runName (Name n) = n 

繰り返しになりますが、 Name は String のnewtypeとして定義しており、モジュールのエク
スポートリスト内でデータ構築⼦をエクスポートしないように注意する必要があります。

次に、属性値として Name を使うことができるように、新しい型 IsValue 型クラスのインス
タンスを定義します。

instance nameIsValue :: IsValue Name where 

  toValue (Name n) = n 

また、次のように a 要素に現れるハイパーリンクの新しいデータ型を定義します。

data Href 

  = URLHref String 

  | AnchorHref Name 

 

instance hrefIsValue :: IsValue Href where 



  toValue (URLHref url) = url 

  toValue (AnchorHref (Name nm)) = "#" <> nm 

href 属性の型の値を変更して、この新しい Href 型の使⽤を強制します。また、要素をアン
カーに変換するのに使う新しい name 属性を作成します。

href :: AttributeKey Href 

href = AttributeKey "href" 

 

name :: AttributeKey Name 

name = AttributeKey "name" 

残りの問題は、現在モジュールの使⽤者が新しい名前を⽣成する⽅法がないということです。
Content モナドでこの機能を提供することができます。まず、 ContentF 型構築⼦に新しいデー

タ構築⼦を追加する必要があります。

data ContentF a 

  = TextContent String a 

  | ElementContent Element a 

  | NewName (Name -> a) 

NewName データ構築⼦は型 Name の値を返すアクションに対応しています。データ構築⼦の
引数として Name を要求するのではなく、型 Name -> a の関数を提供するように使⽤者に要求
していることに注意してください。型 a は計算の残りを表していることを思い出すと、この関
数は、型 Name の値が返されたあとで、計算を継続する⽅法を提供するというように直感的に理
解することができます。

新しいデータ構築⼦を考慮するように、 ContentF についての Functor インスタンスを更新
する必要があります。

instance functorContentF :: Functor ContentF where 

  map f (TextContent s x) = TextContent s (f x) 

  map f (ElementContent e x) = ElementContent e (f x) 

  map f (NewName k) = NewName (f <<< k) 

そして、先ほど述べたように、 liftF 関数を使うと新しいアクションを構築することができ
ます。

newName :: Content Name 

newName = liftF $ NewName id 



id 関数を継続として提供していることに注意してください。型 Name の結果を変更せずに返
すということを意味しています。

最後に、新しいアクションを解釈するために、解釈関数を更新する必要があります。以前は計
算を解釈するために Writer String モナドを使っていましたが、このモナドは新しい名前を⽣
成する能⼒を持っていないので、何か他のものに切り替えなければなりません。 WriterT モナ
ド変換⼦を State モナドと⼀緒に使うと、必要な作⽤を組み合わせることができます。型注釈
を短く保てるように、この解釈モナドを型同義語として定義しておきます。

type Interp = WriterT String (State Int) 

Int型の引数は状態の型で、この場合は増加していくカウンタとして振る舞う数であり、⼀意
な名前を⽣成するのに使われます。

Writer と WriterT モナドはそれらのアクションを抽象化するのに同じ型クラスメンバを使
うので、どのアクションも変更する必要がありません。必要なのは、 Writer String への参照
すべてを Interp で置き換えることだけです。しかし、この計算を実⾏するために使われるハン
ドラを変更しなければいけません。 execWriter の代わりに、 evalState を使います。

render :: Element -> String 

render e = evalState (execWriterT (renderElement e)) 0 

新しい NewName データ構築⼦を解釈するために、 renderContentItem に新しい場合分けを
追加しなければいけません。

renderContentItem (NewName k) = do 

  n <- get 

  let fresh = Name $ "name" <> show n 

  put $ n + 1 

  pure (k fresh) 

ここで、型 Name -> Interp a の継続 k が与えられているので、型 Interp a の解釈を構築
しなければいけません。この解釈は単純です。 get を使って状態を読み、その状態を使って⼀
意な名前を⽣成し、それから put で状態をインクリメントしています。最後に、継続にこの新
しい名前を渡して、計算を完了します。

これにより、 PSCi で、 Content モナドの内部で⼀意な名前を⽣成し、要素の名前とハイパ
ーリンクのリンク先の両⽅を使って、この新しい機能を試してみましょう。



> import Prelude 

> import Data.DOM.Name 

> import Control.Monad.Eff.Console 

 

> :paste 

… render $ p [ ] $ do 

…   top <- newName 

…   elem $ a [ name := top ] $ 

…     text "Top" 

…   elem $ a [ href := AnchorHref top ] $ 

…     text "Back to top" 

… ^D 

 

TopBack to top

 

unit 

複数回の newName 呼び出しの結果が、実際に⼀意な名前になっていることを確かめてみてく
ださい。

演 習

1. (やや難しい) 使⽤者から Element 型を隠蔽すると、さらにAPIを簡素化するこ
とができます。次の⼿順に従って、これらの変更を⾏ってください。

p や img のような(返り値が Element の)関数を elem アクションと結合し
て、型 Content Unit を返す新しいアクションを作ってください。
型 Content a の引数を許容し、結果の型 Tuple String を返すように、
render 関数を変更してください。

2. (やや難しい) 型同義語の代わりに newtype を使って Content モナドの実装を
隠し、 newtype のためにデータ構築⼦をエクスポートしないでください。

3. (難しい) ContentF 型を変更して、次の新しいアクションをサポートするように
してください。

isMobile :: Content Boolean 



このアクションは、この⽂書がモバイルデバイス上での表⽰のためにレンダリングされている
かどうかを⽰す真偽値を返します。   
  ヒント： ask アクションと ReaderT 型変換⼦を使って、このアクションを解釈してみてくだ
さい。あるいは、 RWS モナドを使うほうが好みの⼈もいるかもしれません。

まとめ

この章では、いくつかの標準的な技術を使って、単純な実装を段階的に改善することにより、
HTML⽂書を作成するための領域特化⾔語を開発しました。

データ表現の詳細を隠蔽し、構築⽅法により正しい⽂書を作ることだけを許可するために、スマ
ート構築⼦を使いました。
⾔語の構⽂を改善するために、ユーザ定義の中置２項演算⼦を使⽤しました。
使⽤者が間違った型の属性値を提供するのを防ぐために、データの型に追加の情報を符号化する
幻影型を使⽤しました。
Freeモナドを使って、内容の集まりの配列的な表現を、do表記を提供するモナド的な表現に変
換しました。この表現を拡張してモナドの新しいアクションを提供し、標準のモナド型変換⼦で
モナドの計算を解釈しました。

使⽤者が間違いを犯すのを防ぎ、領域特化⾔語の構⽂を改良するために、これらの⼿法はすべ
てPureScriptのモジュールと型システムを活⽤しています。

関数型プログラミング⾔語による領域特化⾔語の実装は活発に研究されている分野ですが、い
くつかの簡単なテクニックに対して役に⽴つ導⼊を提供し、表現⼒豊かな型を持つ⾔語で作業す
ること威⼒を⽰すことができていれば幸いです。




